
符號、演算、與抽象化

我們先從一個故事開始。
一個島上住著兩種人，騎士 (knight)與惡棍 (knave)。騎士總說實話，惡棍

總說謊話 —所謂「謊話」的意思是實話的相反。從外表看不出一個人是騎士或
惡棍。你站在一個山洞前。傳說中，山洞盡頭藏著價值連城的黃金。但也有可
能並非如此，洞裡其實是一頭龍，一進去就會被吃了。洞口站著一位老先生，
看不出是騎士或是惡棍。你該如何設計一個問題問他，得知山洞裡到底有沒有
金子呢？
我們不妨先做個熱身題。

例 0.1. 假設在你面前站著兩位居民，A與 B.居民 A說：「如果你問 B他自己是
不是騎士，他會說是。」由此你可知道 A是騎士或是惡棍嗎？B呢？

在讀下去之前，請先嘗試解解看。

0.1 騎士與惡棍之島

剛剛你是怎麼解這問題的？許多人解這類問題用的是窮舉法：假設 A是騎士、
B也是騎士，看看會不會有矛盾，接著假設 A是騎士、B是惡棍，看看是否有
矛盾… 依此類推。
我們想介紹另一種做法。我們將「A是騎士」簡寫成 A. 如果 A說了某個陳

述 S,我們無法知道 S是否成立，但我們知道

A≡ S .

其中的 (≡)可以理解為邏輯中「若且唯若」的關係，也可以簡單理解為（真假
值的）「等於」。若用口語說，當我們寫 P≡ Q,表示 P和 Q的真假值相同。確

5

實，假定 A說了陳述 S，如果 A是騎士，S便是真的；如果 A不是騎士，S便是
假的。因此「A說了陳述 S」可寫成 A≡ S。
「若且唯若」這符號有幾個性質。首先，對任何陳述 P,可知 P ≡ P必定為
真：

(P≡ P)≡ true .

我們將這條規則命名為「自反恆真」。1 其次，既然 (≡)可以理解為「等於」，
理所當然該有交換律：2

P≡ Q

= Q≡ P .

最後，很少人注意到，真假值的「等於」是有結合律的：對任何 P, Q,和 R,我
們有這樣的性質：3

(P≡ Q)≡ R

= P≡ (Q≡ R) .

為了練習，我們多考慮一些情況。假設我們問 A：「你是騎士嗎？」而 A回
答「是的！」這可以記成 A ≡ A. 然而，根據自反恆真，A ≡ A可以化簡為 true.
這意味著：在這島上不管你問誰「你是騎士嗎」，對方都會回答「是」。
如果 A說：「B是騎士」呢？這可以記成 A≡ B. 由此我們無法知道 B到底是

騎士還是惡棍，但我們可以知道 A和 B是同一種人。
現在想想問題 0.1：居民 A說「如果你問 B他自己是不是騎士，他會說是。」

這可以記成 A≡ (B≡ B). 我們來演算看看：

A≡ (B≡ B)

= {自反恆真 }
A≡ true

= {結合律與自反恆真 }
A .

也就是說，我們不知道 B到底是騎士還是惡棍，卻可知道 A一定是騎士！這
段演算中，把 A ≡ true 換成 A 的最後一步，直覺上似乎很直觀。如果要探
究理由，其實用了結合律與自反恆真：(A ≡ A) ≡ true 根據結合律可代換為
A≡ (A≡ true),因此 A≡ true可代換成 A.
再考慮一個熱身題：如果 A說「我和 B是同一種人」呢？這可記成 A≡ (A≡

B)。我們算算看:

1給定二元關係 ⊴，如果 x ⊴ x對任何 x均成立，我們說 ⊴有自反性 (reflexivity). 「自反恆真」
即「自反性恆為真」的意思。

2(≡)和 (=)都是「等於」。我們在此處使用不同符號，僅是為了區分出先後次序：同一行內部
的相等用 (≡),行與行之間的相等用 (=). 本書其他地方的用法可能不同。

3(P≡Q)≡ R只在 P, Q,和 R都是真假值（或邏輯陳述）時有意義，否則 P, Q,和 R無法用 (≡)
串起。反例：(3 = 3) = 3可化簡成 true = 3,但後式中等號兩邊的值型別不同，不是合法的句子。
也許因為我們較常考慮數字的相等，而較少用符號表達真假值的相等，(≡)的結合律並不廣為人
知。

A≡ (A≡ B)

= {結合律 }
(A≡ A)≡ B

= {自反恆真 }
true≡ B

= {交換律與自反恆真 }
B .

這次，我們不知 A是騎士還是惡棍，卻可知 B一定是騎士。
最後，我們該回到本章開頭的問題了。你站在山洞口，面對洞口的老人。

姑且稱呼他為 A. 要怎麼想出一個問題問他，用來判斷島上到底有沒有金子
呢？之前介紹邏輯符號的目的是讓我們可避免瞎猜，而用代數解未知數的方式
把問題推演出來。把「島上有金子」這個命題記為 G. 我們希望問某個問題 Q。
如果 A對問題 Q回答「是」，就表示島上有金子，反之則否。

•「A對問題 Q回答『是』」記為 A≡ Q；
•「A對問題 Q回答『是』，島上便有金子」其實應該是一個若且唯若的命
題，寫成 G≡ (A≡ Q)。

而根據交換律和結合律，我們可以推演：

G≡ (A≡ Q)

= {結合律 }
(G≡ A)≡ Q

= {交換律 }
Q≡ (G≡ A) .

我們可得知 Q ≡ (G ≡ A)，也就是說我們要問 A的問題 Q就是 G ≡ A：「請
問，『島上有金子』和『你是騎士』是不是等價的呢？」4

希望這個島上不論騎士或惡棍，邏輯都蠻不錯，才聽得懂這種問題囉！

0.2 讓符號為你工作

回顧看看，我們方才解各種「騎士與惡棍」問題的方法都分為兩個步驟：第一
步先把情況以一組符號描述出來，第二步則是依照這些符號的規則下去推演。
理想上，第二步比第一步容易，因為我們在進行推演時已不用回頭去思考這些
符號的意思，只需照著符號本身的規則不加思索地推演。這麼的好處是能減輕
我們思考的負擔。
探究怎樣的符號組成是合法的、一組合法的符號能否經由給定的規則轉換

成另一組符號，是語法 (syntax)層次上的問題。問這些符號的「意思」是什麼，
則是語意 (semantics)的層次。以騎士與惡棍問題為例，如果我們回到第一原則
去窮舉「如果 A是騎士，B也是騎士.. 如果 A是惡棍，B也是惡棍.. 」等等種種
可能，這是在語意上思考。至少對這個問題來說，語意上的解法不僅較繁瑣，
也只能用來回答「某人是騎士還是惡棍」類型的問題，而難以用來推演出「該
如何問老人，才能得知山洞中是否有黃金？」。上一節的解法則是利用語法幫助

4不妨試試看能否把這句子講得不那麼繞口？

我們：將問題寫下，利用交換律、結合律等等規則時，我們其實沒有回頭去思
考「等等，這個式子是什麼意思？」但在符號的幫助之下，我們以更簡潔的方
式解了許多個問題。
再舉一個「以符號思考」的例子：你會如何算 28× 18? 如果你的思路是

「28乘以 19是 10個 28加上 8個 28，而 10個 28是 280，8個 28是… 10個 28減掉
2個 28!」那麼這種思考方式比較接近語意上的。你不斷在思考這個式子「代表
什麼」，並試圖尋找較簡單的捷徑。如果你拿起紙筆，甚至在心中畫出了這樣
的符號：

28
× 17
--

然後開始照背誦的九九乘法表去推演，此時你用的是語法式的解法。你也許不
會思考許多，只以熟悉的規則推著演算的進行。以乘法而言，這兩種作法各有
用處。但不可否認地，這套語法式解法的存在使我們可安心地把乘法當作已經
解決的問題：我們可以不用思考地作乘法。
許多人見到數學符號就覺得難。事實上，數學符號是發明來簡化問題的。

符號與符號之間的規則讓我們可僅用語法、不需在語意上思考，我們因而可把
腦力用在更難的問題上。大家看到數學符號就覺得難，其實正是因為它們常被
用在不靠數學符號就難以解決的難題上。難的是這些問題，而符號給了我們解
決它們的能力。
這一切和程式語言有什麼關係呢？因為，一套好的程式語言就是一套設計

良好的符號。它能幫助我們描述問題，並找出解決問題的方法。一些程式語言
學者提出的口號「讓符號為你工作 (let the symbols do the work!)」恰好地描述
了這門學問的精神。

0.3 抽象化

如前所述，我們解決問題的第一步是以符號將它表達出來。這一步稱作「抽象
化 (abstraction)」，通常是較難的一步。

例 0.2. Mary有的蘋果數目是 John的兩倍。Mary發現她的蘋果中有一半已經
壞掉了，於是丟了它們。John則吃了一顆蘋果。現在 Mary有的蘋果數目仍是
John的兩倍。請問他們最初各有多少顆蘋果？

你會如何解這問題呢？如果回到第一原則，我們也許可以一步步嘗試：試
試看 John最初有一顆蘋果，Mary有兩顆的情況是不是合理解答，然後試試看
John有兩顆，Mary有四顆的情況… 這是在語意上解問題。就如同我們解騎士
與惡棍問題時用窮舉法一樣。
但許多人也許會用代數：把 Mary最初的蘋果數目用 m表示，John的蘋果

數目用 j表示，寫成這樣的式子：

m = 2 j ,

m/2 = 2( j−1) .

接下來，用高中程度的代數方法，就可以找出 m和 j的值了。
代數方法是純粹基於語法的技術：我們觀察式子的結構，決定該怎麼作

（例如把第二個式子乘以二，兩式相減；或著把 m代換成 2 j），而不需記得 m和
j分別是什麼意思。
不需依賴「意義」在此是個重要的優點：這表示這套代數方法和特定問題

的意義無關，可應用在許許多多場合。可用來解這個問題，也可用來解雞兔問
題… 只要能把問題表示成代數式子，就有一個機械化、不需費神思考的解法。
從「Mary有的蘋果數目是 John的兩倍… 」到 m = 2 j;m/2 = 2( j−1)的轉變

是一個「抽象化」。抽象化在此的意義是將不重要的資訊拋棄，只『抽取出』此
問題中最關鍵的元素。由於此處我們的目的是找出「Mary與 John最初各有多
少顆蘋果」，我們可猜想關於數字的資訊 (例如 m = 2 j)是重要的，並猜想其他
的一些資訊（蘋果是壞掉還是被吃掉了？兩人丟/吃蘋果的先後順序？）可能是
不重要的。因此我們決定只留下關於數字的資訊，並幸運地確實靠此解出了問
題。
電腦是個只會處理符號的機器。實體世界的問題之所以能用電腦來解決，

仰賴的就是「抽象化」這一步。有人說，整個計算科學就是關於抽象化的學
問。
日常用語中，當我們說某事物「很抽象」，通常意味該事物模糊而難以理

解。在計算科學中，「抽象」的事物才是最關鍵、最確實、最該把握的。我們
前面才說過「數學符號讓事情變容易」，現在又說「抽象過的事物是最確實的」。
看來學程式語言久了，讓我們越來越難與常人溝通。最後只好離群索居，躲到
騎士與惡棍之島上捉弄來訪的遊客囉！

0.4 抽象化與表達力

在開發較具規模的軟體前，許多軟體方法建議我們分析問題、需求，並寫成形
式化的規格。5 這是一種抽象化。面對真實的問題時，這一步不容易。
我們再回到蘋果的例子，回想起在該例中，我們選擇抽取出和值有關的資

訊，而拋棄了許多其他：因果、先後順序… 等等。
在某些問題或應用中，因果或時間順序是否有可能是重要的？確實有。解

這些問題時，我們可能在中途發現選錯了抽象的方式，使得留下的資訊不足以
解決該問題。於是我們只好重新開始。有時，我們甚至會發現現有的符號不足
以表達這些問題，得設計另一套符號。
這讓我們討論到解決問題重要的第零步：在我們能用符號表達問題之前，

得先有人設計出一套完善的、足以表達許多問題的符號及其運算規則。為不同
目的，我們可能設計不同的符號。一套符號也伴隨著該符號可如何轉變、操作
（例如我們見過的交換律、結合律，某些符號碰在一起可化簡、等等）的規則。
符號及其規則合起來成為一個「形式 (formal)系統」。使用設計過的符號與其規
則解決問題的研究被稱作「形式方法 (formal method)」。此處的”formal”一詞，
意指我們利用符號的「形式 (form)」來解決問題。6

5例如，Abrial [1996]的 B method是以一套基於集合論與一階邏輯的形式語言描述軟體規格的
方法。

6大學資訊系所的一門必修課”formal language”中的”formal”也為「形式」之意。早期台灣將該
門課稱為「正規語言」，可能是將 formal誤解為相對於 casual的「正式/正規」。

如果我們希望電腦幫我們解問題，需設計的符號就是一套程式語言。這不
是一件容易的事：一個程式語言的表達力得強到足以描述所有該語言被設計來
解決的問題。一套程式語言是設計者看待世界的抽象觀點。有些程式語言主張
世界的狀態可以被一個個指令改變（「把 x的值更新為 x+ y;在螢幕上畫一個方
塊… 」）；有些語言認為世界應該看待為一個個物件；有些語言認為世界是許
多彼此傳送訊息的共時程序；有些語言認為只要描述出每個個體之前的邏輯關
係，交給電腦推論結果，就是很好的計算模型；本書中將介紹的抽象觀點則主
張把一切都看成函數：寫程式是定義函數，圖形是座標到顏色的函數，動畫則
是時間到圖形的函數…。
晚近的語言設計，尤其是型別系統的設計，採用了許多邏輯學界的結果。

本章談「騎士與惡棍之島」時，使用的是命題邏輯 (propositional logic)的一種
分支。一套「邏輯」也可視作一群符號以及其規則。許多人可能不知道，邏輯
有許多種。命題邏輯是一套簡單的邏輯，其優點是給定任一合法的句子，都有
一套機械性作法可得知其成立與否 —這個性質稱作「可判定性 (decidability)」。
但命題邏輯的表達力並不強：許多事情無法在命題邏輯中表達出來。
想表達更細緻的陳述，可使用表達力更強的邏輯。有些邏輯能表達「對所

有」、「存在」；有些邏輯能表達先後順序以及「將一直成立」、「將在未來某
時間成立」等觀念；有些邏輯可用來表達概念與概念之間的關係。但有得必有
失，一套邏輯的表達力只要強到某個程度，就不再有可判定性了，沒有一套固
定的方法可知道任意一個邏輯式子的真假。這是理性的限制。在知道形式系統
的侷限之下盡力探索並發揮其極限，便是這套學問迷人之處。
因此，設計程式語言時總得掌握這樣的平衡：我們希望語言的表達力強到

足以描述我們希望表達的運算程序，但又不希望強到無法掌握其性質。

0.5 正確性

程式語言是設計者選擇用來看待世界的抽象方式。這樣的選擇必然是基於一些
考量：如果設計者希望程式能很容易分割、重用，他可能選擇易於將大問題分
解成小問題的觀點。如果設計者希望該語言能有效率很高的實作，這語言看待
世界的方式可能就和機器的世界觀很接近。
本書選擇的方向則是「程式語言應能幫助我們確保程式的正確性。」
但什麼是「正確」？直觀說來，一個程式「正確」的意思是該程式「做到了

我們要它做的事。」但，電腦本來不就只能一個指令一個動作地做我們要它做
的事嗎？那麼「正確」的意思到底是什麼？
我們考慮一下這個問題：

例 0.3 (最大區段和). 給定一串（有正有負）的數字。請找出一段連續的數字，
使其總和越大越好，並傳回這個總和。

這是經典的「最大區段和」問題。如前面的章節所述，面對一個問題，我們
先把它描述成符號。如果 a是給定的那串數字，N是其長度，a [i ]是數字中的
地 i個，而 m是我們想求出的、最大的那個和，最大區段和問題可以描述成這

樣：

m = max{sum(i, j) | 0 ⩽ i ⩽ j ⩽ N} ,

sum(i, j) = Σ j−1
k=i a[k] ,

(1)

其中 sum (i, j)代表 a [i ] . .a [j−1 ]的和，而 (↑) S找出集合 S中最大的那個。
但（許多讀者可能也知道）這個問題其實有個線性時間內可執行完畢的解：

將這串數字 a從頭到尾看過一遍，維持兩個變數 s和 m。每看到一個新數字
a [j ],將 s更新為 0與 s+a [k ]中較大的那個，並把 m更新為 s與 m中較大的那
個。寫成程式的話，可能是這樣：

s, m = 0, 0
for k in range(0,n-1)
s = max (0, s + a[k])
m = max (s, m)

那麼，明顯的問題來了。上面的程式和問題描述(1)一點也不像。我們怎麼知道
程式真正實作了(1)的要求？
由此談「正確性」，相信清楚多了。數學式(1)給的是一個規格 (specification)，

談的是我們要的結果 (what),而程式描述的則是怎麼做 (how). 「正確性」總是
相對於一個規格而言的：描述「怎麼做」的程式是否真做到了規格的要求？
「最大區段和」的例子可以給我們幾個啟示。首先，把「要什麼」和「怎麼
做」牽上關聯，有時是很困難的。當我們說一個程式「很難懂」，其中一個意
思是很難看出「為什麼這個程式實作出了這個規格？」（也就是說「為什麼這個
程式是正確的？」）看懂一個程式，也就是了解他為何是正確的。
「最大區段和」的例子也告訴我們，「難懂」的程式不一定要很長。上面的
程式短短四行，但若沒有輔助解釋，一般人可能很難「看懂」它。7

如果規格已經不見了呢？這樣的情形更糟。當我們在沒有規格的情況下問
一個程式「是做什麼的」，我們真正問的是「這個程式符合的規格是什麼？」這
可能是個難題。回頭看看上面的程式，如果沒有給(1)，你能說得出這個程式在
做什麼嗎？

0.6 可演算的程式語言

正確性是如此重要又難以掌握的性質，我們因此希望程式語言能幫助我們確保
程式的正確性。我們採用的方法是：給定一個規格，我們希望正確的程式能夠
由其規格演算、推導出來。就如同在騎士與惡棍問題中，我們把希望 Q滿足
的性質寫下，然後利用符號的規則算出 Q. 我們希望程式也能如此從其規格被
算出來。
以最大區段和問題為例，用我們之後將介紹的符號，我們可以把(1)改寫成：

mss = (↑) ·map sum · segments ,

7理解這個程式的關鍵是：m永遠是 a [0 ] . .a [k ]中所有連續區段最大的和，而 s永遠是
a [0 ] . .a [k ]中最右端為 a [k ]的連續區段中最大的和。這兩個條件是該迴圈的恆式 (loop invariant)。
恆式是了解一個迴圈最重要的資訊。

α β γ δ ε ϛ ζ η θ ι κ λ μ ν ξ ο π ϙ
1 2 3 4 5 6 7 8 9 10 20 30 40 50 60 70 80 90

ρ σ τ υ φ χ ψ ω ϡ ͵α ͵β ͵γ
100 200 300 400 500 600 700 800 900 1000 2000 3000

Figure 0.1: 希臘數字。

其意思大約是「給定一串數字，找出它的所有連續區段，算每一個區段的和，
然後傳回其中最大的。」接下來的問題是，mss有沒有比較快的實作？我們希
望可以用代數方法，由 mss的規格出發：

(↑) ·map sum · segments
= { segments的定義 }
(↑) ·map sum · concat ·map inits · tails

= {某定理 }
...

= {另一些定理 }
(↑) · scanr (⊕) 0 .

最後的 (↑) · scanr (⊕) 0相當於前一節的程式，只是以不同的符號表示。
我們可以由規格出發，在符號上操作，將程式如同求代數的解一樣地算出

來。只要每個小步驟都正確，最後的程式就是正確的。這套技術稱作「程式推
導 (program derivation)」或「程式演算 (program calculation)」，將是本書重要
的主題。而在本書的觀點中，函數編程相較於其他典範的特色與優勢是函數語
言是一套適合演算的符號系統。

適合演算的符號 我們舉個例子，談談符號「適合演算」是怎麼回事。圖 0.1中
顯示的是西元前四世紀左右希臘人使用的數字表示法。以希臘字母表為順序，
1, 2,…到 9分別寫成 α, β,…θ. 每遇到 10的乘冪便換一組符號，如 10, 20, …寫成
ι, κ…。11, 12, 13分別寫成 ια, ιβ, ιγ, 21, 22, 23則寫成 κα, κβ, κγ. 653可寫成 χνγ.
這套表示法的缺點包括不易表示大數，以及不易做演算：我們需要知道 δ加上
β等於 ϛ,以及 μ加上 κ等於 ξ。8

與之對比的是瑪雅數字。這套數字系統的年代不詳，據估計可能在西元四
至世紀採用。一個點代表 1,一條橫杠代表 5。例如 19寫成 4555 . 瑪雅數字採用 20
進位，較大的位數寫在上方。因此

25155

代表 7(25 )× 20+ 11(

155 ),也就是 151. 另有個例外：第三個數字為 18（而非 20）
的乘冪 —因為 18×20 = 360接近一年的日數。圖 0.2中左邊由上至下是以瑪雅
數字寫成的 3212199. 值得注意的是，此類以符號的位置表示其量級的數字表

8西元前八世紀，希臘人曾使用以 I, Π, Δ, H分別表示 1, 5, 10, 100的表示法。後來被圖 0.1中的
系統取代。無論如何，希臘人仍以此發展出了進步的數學。

1 1×18×20×20×20×20 = 2880000

2 2×18×20×20×20 = 288000

15 6×18×20×20 = 43200

2 2×18×20 = 720

355 13×20 = 260

4555 19
總和: 3212199

Figure 0.2: 以瑪雅數字表示 3212199.

示法大都需要一個相當於「零」的佔位符號。在瑪雅數字中，該符號寫成0。
因此 20可寫成 0̇,與只有一個點的 1區分。9

如何用瑪雅數字作加法？以下顯示的是 151+55:

25 2 45 55

155 +

555 =

155555 = 15

151 55 206 206

我們可先把兩個個位數（155 與 555 ）和兩個 20位數（25 與 2 ）分別相加。其中，個
位數相加的結果是

155555 —超過了 20！於是我們把其中四個橫槓（即一個 20）往
上進位成一個點，和 20位數的四個點結合成另一個橫槓。
有趣的是，我相信大部分的讀者即使可能是第一次看到這種數系，也從未

使用過 20進位，還是可迅速理解以上的加法，甚至可以舉一反三地自己試試
看。這就是一個適合演算的符號系統給我們的便利。
對本書來說，函數語言的價值便是 —它是個適於演算的語言。

0.7 相關資料

本章中的騎士與惡棍問題所使用的邏輯形式稱作演算邏輯 (calculational logic)，
其主要精神是將命題邏輯與述語邏輯表達為適合用於等式推導與演算的形式。
Gries and Schneider [2003]認為大部分形式邏輯系統是為了研究邏輯本身的特
性而設計，而演算邏輯則是為了以邏輯解決問題而開發的。演算邏輯由程式
語言學者們於 80年代初期漸漸發展出來。10 Dijkstra and Scholten [1990]將演
算邏輯用於指令式程式語言的語意中。本章中的騎士與惡棍問題由 Backhouse
[2003] 中節錄而來。這是一本介紹以形式方式構思演算法以解決問題的教
科書。本章只用到了 (≡)一個運算子，該書中則有更完整的介紹。Gries and
Schneider [October 22, 1993]則是一本以演算邏輯為基礎的離散數學教科書。
「讓符號為你工作」這句口號可在 Dijkstra的許多著作中見到（例如 Dijkstra

and van Gasteren [1986] , Dijkstra [2000]）。Dijkstra [2004]寫道：「設計程式的
9一些早期數字表示法中有此種佔位符號，但並不見得把「零」視為可獨立存在的一個數。「將

零視為一個數字」是數學史上的重要發明，許多數學性質都需有它的存在才得以描述。程式語言
中也有類似的情況：有一個「不做任何事的指令」或一個「將輸入照樣傳回的函數 (即 id)」對於描
述程式的性質是很重要的。

10Gries and Schneider [2003] 將此歸功於 Roland Backhouse, Edsger W. Dijkstra, Wim H.J. Feijen,
David Gries, Carel S. Scholten,以及 Netty van Gasteren等人。他們的共同研究主題是以形式方法開
發程式。

人最好把程式當作精巧的算式。而我們知道只有一種設計精巧算式的可靠方
法：用符號操作來推導它。我們得讓符號為我們工作，因為這是已知唯一在
大尺度下仍可行的方法。」Misra則在 1988年 Marktoberdorf暑期課程的演講
[Misra, 1989]中虛構了一個故事：使用羅馬數字的羅馬人被推銷比較易於演算
的印度-阿拉伯數字系統，卻不以為然地說「我們已經有奴隸為我們工作了！」
希臘與瑪雅數字的例子取自 Mazur [2014]。該書對於種種符號的演進有更

詳盡的介紹。

值、函數、與定義

語言是概念的載體。如第 0章所述，程式語言不僅用來表達概念，也用於演
算，分擔我們思考的負擔。在本書中，為便於精確說明，我們也不得不選一個
語言。
本書中使用的語言大致上基於 Haskell,但為適合本書使用而經過簡化、修

改。我們將在本章初步介紹 Haskell語言的一小部分，包括在 Haskell中何謂
「計算」、值與函數的定義、常見的資料結構，以及串列上的常用函數。目的是
讓讀者具備足夠的基本概念，以便進入之後的章節。也因此，本書中所介紹的
語言並非完整的 Haskell,本書也不應視作 Haskell語言的教材。對於有此需求
的讀者，我將在本章結尾推薦一些適合的教科書。

1.1 值與求值

Haskell是個可被編譯 (compile)、也可被直譯 (interpret)的語言。Haskell直譯
器延續了 LISP系列語言的傳統，是個「讀、算、印 (read, evaluate, pring)」的
迴圈 —電腦從使用者端讀取一個算式，算出結果，把結果印出，然後再等使
用者輸入下一個算式。一段與 Haskell直譯器的對話可能是這樣：

Main>3+4
7
Main> sum [1 . .100 ]
5050
Main>

在此例中，Main>是 Haskell直譯器的提示符號。1 使用者輸入 3+ 4, Haskell
算出其值 7並印出. 接著，使用者想得知 1到 100的和，Haskell算出 5050.

1此處的人機互動紀錄擷取自 GHCi (Glasgow Haskell Compiler Interactive). GHC為目前最被廣
泛使用的 Haskell實作。

15

演算格式

本書中將使用如下的格式表達（不）等式演算或推論：

expr0
= {reason0}

expr1
⩾ {reason1}

expr2
:

= exprn .

這是一個 expr0 ⩾ exprn 的證明。式子 expr0 . .exprn 用具有遞移律的運算子 (如
(=), (⩾)等等)串起。放在大括號中的是註解，例如 reason0 是 expr0 = expr1 的原
因，reason1 是 expr1 ⩾ expr2 的原因。
根據 van de Snepscheut [1993, p19],此格式最早由W.H.J. Feijen所建議。

上例中的算式只用到 Haskell已知的函數（如 (+), sum等）。使用者若要自
己定義新函數，通常得寫在另一個檔案中，命令 Haskell直譯器去讀取。例如，
我們可把如下的定義寫在一個檔案中：

double :: Int→ Int

double x = x+ x .

上述定義的第一行是個型別宣告。當我們寫 e :: t, 代表 e具有型別 t. Int是整
數的型別，而箭號 (→)為函數型別的建構元。第一行 double :: Int→ Int便是
告知電腦我們將定義一個新識別字 double,其型別為「從整數 (Int)到整數的函
數」。2 至於該函數的定義本體則在第二行 double x = x+ x,告知電腦「凡看到
double x,均可代換成 x+ x.」

求值 第 0章曾提及：一套程式語言是設計者看待世界的抽象觀點。程式通常
用來表達計算，因此程式語言也得告訴我們：在其假想的世界中，「計算」是
怎麼一回事。指令式語言的世界由許多容器般的變數組成，計算是將值在變數
之間搬來搬去。邏輯編程中，描述一個問題便是寫下事物間的邏輯關係，計
算是邏輯規則「歸結 (resolution)」的附帶效果。共時 (concurrent)程式語言著
眼於描述多個同時執行的程式如何透過通道傳遞訊息，計算也靠此達成。函
數語言中，一個程式便是一個數學式，而「計算」便是依照該式子中各個符
號的定義，反覆地展開、歸約，直到算出一個「值」為止。這個過程又稱作
「求值 (evaluation)」。在 Haskell直譯器中，若 double的定義已被讀取，輸入

double (9×3),電腦會算出 54:

Main>double (9×3)
54

但 54這個值是怎麼被算出來的？以下是其中一種可能：
2Haskell標準中有多種整數，其中 Int為有限大小（通常為該電腦中一個字組 (word)）的整數，

Integer則是所謂的大整數或任意精度整數，無大小限制。本書中只使用 Int.

double (9×3)
= { (×)的定義 }

double 27
= { double的定義 }

27+27
= { (+)的定義 }

54 .

上述演算的第一步將 9×3歸約成 27 —我們尚未定義 (×)與 (+),但目前只需知
道它們與大家所熟悉的整數乘法、加法相同。第二步將 double 27變成 27+27,
根據的是 double的定義：double x = x+ x. 最後，27+27理所當然地歸約成 54.
「歸約」一詞由 β -reduction而來，在此處指將函數本體中的形式參數代換為實
際參數。3 在上述例子中，我們遇到如 double (9× 3) 的函數呼叫，先將參數
(9×3)化簡，再展開函數定義，可說是由內到外的歸約方式。大部分程式語言
都依這樣的順序求值，讀者可能也對這種順序較熟悉。
但這並不是唯一的求值順序。我們能否由外到內，先把 double展開呢？

double (9×3)
= { double的定義 }
(9×3)+(9×3)

= { (×)的定義 }
27+(9×3)

= { (×)的定義 }
27+27

= { (+)的定義 }
54 .

以這個順序求值，同樣得到 54.
一般說來，一個式子有許多種可能的求值順序：可能是由內往外、由外往

內、或其他更複雜的順序。我們自然想到一個問題：這些不同的順序都會把該
式子化簡成同一個值嗎？有沒有可能做出一個式子，使用一個順序會被化簡成
54,另一個順序化簡成 53?
我們看看如下的例子。函數 three顧名思義，不論得到什麼參數，都傳回

3；inf 則是一個整數：

three :: Int→ Int

three x = 3 ,

inf :: Int
inf = 1+ inf .

在指令式語言中，inf 的定義可能會被讀解為「將變數 inf 的值加一」。但函數
語言中「變數」的值是不能更改的。此處的意義應是：inf 是一個和 1+ inf 相
等的數值。我們來看看 three inf 會被為甚麼？如果我們由內往外求值：

3Reduction的另一個常見譯名是「化簡」，然而，許多情況下，一個式子被 reduce後變得更長
而不「簡」，因此本書譯為「歸約」。

three inf
= { inf 的定義 }

three (1+ inf )
= { inf 的定義 }

three (1+(1+ inf ))
= { inf 的定義 }
...

看來永遠停不下來！但如果我們由外往內，three inf 第一步就可變成 3：

three inf
= { three的定義 }

3 .

我們該如何理解、討論這樣的現象呢？

範式與求值順序 為描述、討論相關的現象，我們得非正式地介紹一些術語。
用較直觀、不形式化的方式理解，一個式子中「接下來可歸約之處」稱作其歸
約點 (redex)。例如 (9×3)+(4×6)中，9×3與 4×6都是歸約點。如果一個式
子已沒有歸約點、無法再歸約了，我們說該式已是個範式 (normal form)。
回頭看，經由之前的例子我們已得知：
• 有些式子有範式 (如 double (9×3)有個範式 54)，有些沒有（如 inf）。
• 同一個式子可用許多順序求值。有些求值順序會碰到範式，有些不會。

給一個式子，我們很自然地希望知道它有沒有值，並算出其值。如果一個式子
有很多個範式，我們便難說哪一個才是該式子的「值」。如此一來，立刻衍生
出幾個問題。給定一個式子，我們是否能得知它有沒有範式呢？如果有，用哪
個求值順序才能得到那個範式？以及，有沒有可能用一個求值順序會得到某範
式，換另一個求值順序卻得到另一個範式？
很不幸地，第一個問題的答案是否定的：沒有一套固定的演算法可判定

任意一個式子是否有範式。這相當於計算理論中常說到的停機問題 (halting
problem) —沒有一個演算法能準確判斷任意一個演算法（對某個輸入）是否能
正常終止。
但對於另兩個問題，計算科學中有個重要的 Church-Rosser定理。非常粗略

地說，該定理告訴我們：在我們目前討論的這類語言中4

• 一個式子最多只有一個範式。
• 如果一個式子有範式，使用由外到內的求值順序可得到該範式。

如此一來，給定任一個式子，我們都可用由外到內的方式算算看。假設一算之
下得到（例如）54。用其他的求值順序可能得不到範式，但若有了範式，必定
也是 54. 如果由外到內的順序得不到範式，用其他任何順序也得不到。
由於「由外到內」的求值順序有「最能保證得到範式」的性質，又被稱

作「範式順序 (normal order evaluation)」。「由內到外」的則被稱作「應用順序
(applicative order evaluation)」。以本書的目的而言，我們可大致把 Haskell使用

4此處討論的可粗略說是以 λ -calculus為基礎的函數語言。基於其他概念設計的程式語言當然
可能不遵守 Church-Rosser定理。

的求值方式視為範式順序。但請讀者記得這是個粗略、不盡然正確的說法 —
Haskell實作上使用的求值方式經過了更多最佳化。正式的 λ -calculus教科書中
對於歸約點、求值順序、Church-Rosser定理等概念會有更精確的定義。

被迫求值 型別 Bool表示真假，有兩個值 False和 True。常用的函數 not可定
義如下：

not ::Bool→ Bool

not False= True

not True = False .

此處 not的定義依輸入值不同而寫成兩個條款。這種定義方式在 Haskell中稱
作樣式配對 (pattern matching)：False與 True在此處都是樣式 (patterns)。遇到
這樣的定義時，Haskell 將輸入依照順序與樣式們一個個比對。如果對得上，
便傳回相對應的右手邊。本例中，若輸入為 False，傳回值為 True，否則傳回
值為 False。
我們來看看 not (5 ⩽ 3)該怎麼求值？若依照範式順序，照理來說應先將 not

的定義展開。但若不知 5 ⩽ 3的值究竟是 False還是 True,我們不知該展開哪行
定義！因此，要計算 not (5 ⩽ 3)，也只好先算出 5 ⩽ 3了：

not (5 ⩽ 3)
= { (⩽)之定義 }

not False
= { not之定義 }
True .

求值過程中若必須得知某子算式的值才能決定如何進行，只好先算那個子
算式。在 Haskell中有不少（有清楚定義的）場合得如此，包括遇上 (⩽)、(⩾)

等運算子、樣式配對、case算式（將於第 1.6.1節中介紹）…等等。

習題 1.1 — 定義一個函數 myeven :: Int→ Bool，判斷其輸入是否為偶數。
你可能用得到以下函數：a

mod :: Int→ Int→ Int ,

(= =) :: Int→ Int→ Bool .

其中 mod x y為 x除以 y之餘數，(= =)則用於判斷兩數是否相等。
習題 1.2 — 定義一個函數 area ::Float→ Float,給定一個圓的半徑，計算其
面積。（可粗略地將 22/7當作圓周率。）

a此處所給的並非這些函數最一般的型別。

惰性求值 實作上，Haskell求值的方式還經過了更多的最佳化：例如將歸約
過的算式改寫為它的值，避免重複計算。這套求值方式稱為惰性求值 (lazy
evaluation).

技術上說來，惰性求值和範式順序求值並不一樣。但前者可視為後者的最
佳化實作 —惰性求值的結果必須和範式順序求值相同。因此，在本書之中大
部分地方可忽略他們的差異。和惰性求值對偶的是及早求值 (eager evaluation)，
可視為應用順序求值的實作 —在呼叫一個函數之前，總是把其參數先算成範
式。這也是一般程式語言較常見的計算方法。
本書中談到偏向實作面的議題時會用「惰性求值/及早求值」，在談不牽涉

到特定實作的理論時則使用「範式順序求值/應用順序求值」。

1.2 函數定義

考慮如下的函數定義：

smaller :: Int→ Int→ Int

smaller x y = if x ⩽ y then x else y .

我們可粗略地理解為：smaller 是一個函數，拿兩個參數 x與 y，傳回其中
較小的那個。如 smaller (double 6) (3+4)的值為 7.

習題 1.3 — 用前一節介紹的求值順序，將 smaller (double 6) (3+4)化簡為
範式。

守衛 如果函數本體做的第一件事就是條件判斷，Haskell提供另一種語法：

smaller :: Int→ Int→ Int

smaller x y | x ⩽ y = x
| x> y = y .

這麼寫出的 smaller的行為仍相同：如果 x ⩽ y成立，傳回 x；如果 x> y,傳回
y。但這種語法較接近一些數學教科書中定義函數的寫法。其中，放在 x ⩽ y和
x> y等位置的必須是型別為 Bool的算式。它們擋在那兒，只在值為 True的時
候才讓執行「通過」，因此被稱為守衛 (guard)。如果有三個以上的選項，如下
述例子，使用守衛比 if . . then . .else更清晰可讀：

sign :: Int→ Int

sign x | x>0 = 1
| x = = 0 = 0
| x<0 =−1 .

遇到數個守衛時，Haskell將依照順序嘗試每個選項，直到碰到第一個為真
的守衛，然後只執行該項定義。也就是說，即使我們把 sign定義中的 x = = 0改
為 x ⩾ 0, sign 10的值仍會是 1。若每個守衛都是 False,程式則將中斷執行（並
傳回一個錯誤訊息）。在許多程式中，我們會希望最後一個守衛能捕捉所有的
漏網之魚：如果其他條款都不適用，就執行最後一個。一種做法是讓一個守衛
是 True。或著，在 Haskell中有個 otherwise關鍵字可讓定義讀來更口語化些：

sign :: Int→ Int

sign x | x>0 = 1
| x = = 0 = 0
| otherwise =−1 .

事實上，otherwise的定義就是 otherwise = True.

區域識別字 Haskell提供兩種宣告區域識別字的語法：let與 where. 也許最容
易理解的方式是用例子說明。

例 1.1. 工讀生每小時的時薪為新台幣 130元。假設一週有五個工作天，每天有
八小時上班時間。定義一個函數 payment :: Int→ Int，輸入某學生工作的週數，

計算其薪資。

答. 我們當然可直接用一個式子算出薪資。但為清楚起見，我們可用兩個區域
識別字 days和 hours, 分別計算該學生工作的日數和時數。如果用 let, 可這麼
做。

payment :: Int→ Int

payment weeks = let days = 5×weeks
hours = 8×days

in 130×hours .

let算式的語法為

let x1 = e1

x2 = e2 ....

in e .

其中 e為整個算式的值，而 x1, x2 等等為區域識別字。兩個區域識別字的有效
範圍包括 e,以及 e1 與 e2。
另一種語法是 where子句。若用它定義 payment,可寫成這樣：

payment :: Int→ Int

payment weeks = 130×hours ,

where hours = 8×days
days = 5×weeks .

□

該用 let 或是 where? 大部份時候這可依個人習慣，看寫程式的人覺得怎
麼說一件事情比較順。使用 let 時，敘事的順序是由小到大，先給「工作日
數」、「工作時數」等小元件的定義，再用他們組出最後的式子 130×hours。使
用 where時則是由大到小，先說我們要算「工作時數乘以 130」，然後補充「其
中，工作時數的定義是…」。
但，Haskell之所以保留了兩種語法，是為了因應不同的用途。語法上，let

是一個算式，可出現在算式中。如下的算式是合法的，其值為 24:

(1+1)× (let y = double 5 in y+2) .

where子句的一般語法則如下例所示：

f x0 = d0

where y0 = e0

f x1 = d1

where y1 = e1 .

由語法設計上即可看出，子句 where y0 = e0 只能放在 f x0 = ...的一旁當作補
述，不能出現在 d0之中。這個例子中，y0的有效範圍含括 d0與 e0。另，e0可
以使用 x0.
算式中只能用 let. 相對地，也有些只能使用 where的場合。我們來看我們

來看一個只能使用 where的例子：

例 1.2. 延續例 1.1。今年起，新勞動法規規定工作超過 19週的工讀生必須視為
正式雇員，學校除了薪資外，也必須付給勞保、健保費用。學校需負擔的勞健

保金額為雇員薪資的百分之六。請更新函數 payment,輸入某工讀生工作週數，
計算在新規定之下，學校需為工讀生付出的總額。

答. 一種可能寫法是先使用守衛，判斷工作週數是否大於 19：

payment :: Int→ Int

payment weeks | weeks>19 = round (fromIntegral baseSalary×1.06)
| otherwise = baseSalary ,

where baseSalary = 130×hours
hours = 8×days
days = 5×weeks .

在 where子句中，我們先算出不含勞健保費用的薪資，用識別字 baseSalary表
達。如果 weeks大於 19,我們得將 baseSalary乘以 1.06；否則即傳回 baseSalary.
函數 fromIntegral把整數轉為浮點數，round 則把浮點數四捨五入為整數。請
注意：兩個被守衛的算式都用到了 baseSalary — where 子句中定義的識別
字是可以跨越守衛的。相較之下，let 算式只能出現在等號的右邊，而在
守衛 weeks > 19 = ... 之後出現的 let 所定義出的區域識別字，顯然無法在
otherwise = ...之中被使用，反之亦然。 □

巢狀定義 let算式之中還可有 let算式，where子句中定義的識別字也可有自
己的 where子句。我們看看兩個關於 let的例子：

例 1.3. 猜猜看 nested和 recursive的值分別是什麼。將他們載入 Haskell直譯器，
看看和你的猜測是否相同。

nested :: Int
nested = let x = 3

in (let x = 5
in x+ x)+ x ,

recursive :: Int
recursive = let x = 3

in let x = x+1
in x .

一級公民

在程式語言中，若說某物/某概念是一級公民 (first-class citizen)或「一級的」，通
常指它和其他事物被同等對待：如果其他事物可被當作參數、可被當作傳回值、
可被覆寫… 那麼它也可以。這是一個沒有嚴格形式定義的說法，由 Christopher
Strachey在 1960年代提出，可用在型別、值、物件、模組… 等等之上。
例如：OCaml是個有「一級模組」的語言，因為 OCaml模組也可當作參數，可
定義從模組到模組的函數（OCaml中稱之為 functor）。在 C語言之中函數是次級
的，因為函數不能當參數傳（能傳的是函數的指標，而非函數本身）。Strachey指
出，在 Algol中實數是一級的，而程序是次級的。

答. nested的值是 13，因為 x+ x之中的 x在 let x = 5的範圍中,而 . .+ x中的 x
則在 let x = 3的範圍中。5 至於 recursive的值，關鍵在於 x = x+1中右手邊的
x指的是哪個。若是 x = 3的那個 x,整個算式的值將是 4. 若 x = x+1中，等號
右手邊的 x也是左手邊的 x, recursive就是 ((. .)+1)+1 ,沒有範式。這兩種設計
都有其道理。Haskell選了後者：在 let x = e in...之中，x的有效範圍包括 e. 因
此 recursive在 Haskell中會無窮地求值下去。但也有些函數語言中的 let採用
前者的設計。通常這類語言中會另有一個 letrec結構，和 Haskell的 let功能相
同。 □

1.3 高階函數

目前為止，我們看過由整數到整數的函數、由整數到真假值的函數… 那麼，可
以有由函數到函數的函數嗎？函數語言將函數視為重要的構成元件，因此函數
也被視為一級公民。如果整數、真假值… 可以當作參數、可以被函數傳回，函
數當然也可以。一個「輸入或輸出也是函數」的函數被稱為高階函數 (higher
order function)。Haskell甚至設計了許多鼓勵我們使用高階函數的機制。本書
中我們將見到許多高階函數。其實，我們已經看過一個例子了。

Currying 回顧 smaller的定義：

smaller :: Int→ Int→ Int

smaller x y = if x ⩽ y then x else y .

1.2節中說「smaller是一個函數，拿兩個參數 x與 y」。但這僅是口語上方便
的說法。事實上，在 Haskell中（如同在 λ -calculus中），所有函數都只有一個
參數。函數 smaller其實是一個傳回函數的函數：

• smaller的型別 Int→ Int→ Int其實應看成 Int→ (Int→ Int)：這個函數拿
到一個 Int後，會傳回一個型別為 Int→ Int的函數。

• smaller 3的型別是 Int→ Int。這個函數還可拿一個 Int參數，將之和 3比
大小，傳回較小的那個。

5在各種語言中，範圍的設計都是為了給程式員方便：在寫子算式時，可不用擔心是否與外層
的識別字撞名。在教學時，我們難免舉各種撞名的例子作為說明。若把這些刁鑽例子當作考題，
就是違反設計者本意的發展了。

• smaller 3 4 是一個 Int。它其實是將函數 smaller 3 作用在 4 之上。也
就是說，smaller 3 4 其實應看成 (smaller 3) 4. 根據定義，它可展開為
if 3 ⩽ 4 then 3 else 4，然後化簡為 3.

習題 1.4 — 將 smaller的定義鍵入一個檔案，載入 Haskell直譯器中。
1.smaller 3 4的型別是什麼？在 GHCi中可用:t e指令得到算式 e的型
別。

2.smaller 3的型別是什麼？
3.在檔案中定義 st3 = smaller 3. 函數 st3的型別是什麼？
4.給 st3一些參數，觀察其行為。

「用『傳回函數的函數』模擬『接收多個參數的函數』」這種做法稱作 currying.
6 Haskell鼓勵大家使用 currying —它的內建函數大多依此配合設計，語法設
計上也很給 currying方便。當型別中出現連續的 (→)時，預設為往右邊結合，
例如 Int→ Int→ Int 應看成 Int→ (Int→ Int). 這使得「傳回函數的函數」容
易寫。而在值的層次，連續的函數應用往左結合。例如，(smaller 3) 4可寫成
smaller 3 4。這讓我們能很方便地將參數一個個餵給 curried函數。
另一方面，如果我們想使用 double兩次，計算 x的四倍，應寫成 double (double x).

若寫 double double x，會被視為 (double double) x — double作用在自身之上，而
這顯然是個型別錯誤。
我們再看一個使用 currying的練習：

例 1.4. 給定 a, b, c, x,下述函數 poly計算 ax2 +bx+ c:

poly :: Int→ Int→ Int→ Int→ Int

poly a b c x = a× x× x+b× x+ c .

請定義一個函數 poly1,使得 poly1 x = x2 +2x+1. 函數 poly1 需使用 poly.

答. 一種作法是：

poly1 :: Int→ Int

poly1 x = poly 1 2 1 x .

但這相當於 poly1 x = (poly 1 2 1) x — poly拿到 x後，立刻把 x傳給 poly 1 2 1
這個函數。因此 poly1 可更精簡地寫成：

poly1 :: Int→ Int

poly1 = poly 1 2 1 .

兩種寫法都有人使用。有提及 x的寫法著重於描述拿到參數 x之後要對它進行
什麼操作。而省略 x的寫法則是在函數的層次上思考：我們要定義一個函數，
稱作 poly1。這個函數是什麼呢？就是 poly拿到 1, 2, 1之後傳回的那個函數。

6Currying為動名詞，形容詞則為 curried。此詞來自於邏輯學家 Haskell B. Curry的姓氏。詳見
第 1.11節。

如果我們想用 poly定義出另一個函數 poly2 a b c = a×22 +b×2+2呢？最
好理解的可能是不怎麼意外的寫法：

poly :: Int→ Int→ Int→ Int

poly2 a b c = poly a b c 2 .

我們可以用一些技巧使 a, b,和 c不出現在定義中，但如此得到的程式並不
會更好懂。 □

二元運算子 在進入其他主題前，我們交代一些語法細節。Haskell 鼓勵
currying 的使用，也把二元運算子都設計成 curried 的。例如加法的型別是
Int→ Int→ Int. Haskell也配套設計了種種關於二元運算子的特殊語法，希望
讓它們更好用。但這些語法規則的存在都僅是為了方便我們寫出（主觀上）語
法漂亮的程式，而不是非有不可、非學不可的規定。假設某二元運算子 (⊕)的
型別是 a→ b→ c, (x⊕)是給定了 (⊕)的第一個參數後得到的函數；(⊕y)則是
給定了 (⊕)的第二個參數後得到的函數：7

(x⊕) y = x⊕ y { (x⊕)的型別為 b→ c; }
(⊕y) x = x⊕ y { (⊕y)的型別為 a→ c. }

例如：
• (2×)和 (×2)都是把一個數字乘以二的函數；
• (/2)則把輸入數字除以二；
• (1/)計算輸入數字的倒數。
名字以英文字母開頭的函數預設為前序的。例如，計算餘數的函數 mod使

用時得寫成 mod 6 4。若把它放在「倒引號 (backquote)」中，表示將其轉為中
序 —如果我們比較喜歡把 mod放到中間，可以寫成 6 ‘mod‘ 4. 首字元非英文字
母的函數（如 (+), (/)等）則會被預設為中序的二元運算子。若把一個中序二
元運算子放在括號中，表示將其轉為前序運算子。例如，(+) 1 2和 1+2的意
思相同。
在Haskell的設計中，函數應用的優先順序比中序運算子高。因此 double 3+

4會被視作 (double 3)+4,而不是 double (3+4). 將中序運算子放在括號中也有
「讓它不再是個中序運算子，只是個一般識別字」的意思。例如算式 f +x中，f
和 x是中序運算子 (+)的參數。但在 f (+) x中，(+)和 x都是 f 的參數（這個
式子可以讀解為 (f (+)) x）。

以函數為參數 下述函數 square計算輸入的平方：

square :: Int→ Int

square x = x× x .

我們可另定義一個函數 quad :: Int→ Int，把 square用兩次，使得 quad x算出
x4.

7根據 Hudak et al. [2007]，此種「切片」(sectioning)語法最早見於 David Wile的博士論文。後
來被包括 Richard Bird在內的 IFIP WG 2.1成員使用，並由 David A. Turner實作在他的語言Miranda
中。

quad :: Int→ Int

quad x = square (square x) .

但，「把某函數用兩次」是個常見的編程模式。我們能不能把 quad與 square
抽象掉，單獨談「用兩次」這件事呢？下面的函數 twice把參數 f 在 x之上用
兩次：

twice :: (a→ a)→ (a→ a)
twice f x = f (f x) .

有了 twice,我們可以這麼定義 quad:

quad :: Int→ Int

quad = twice square .

函數 twice是本書中第一個「以函數為參數」的函數。我們可看到「讓函數
可作為參數」對於抽象化是有益的：我們可以把「做兩次」這件事單獨拿出來
說，把「做什麼」抽象掉。
「函數可以當作參數」意味著我們可以定義作用在函數上的運算子。twice就
是這麼一個運算子：它拿一個函數 f，把它加工一下，做出另一個函數（後者
的定義是把 f 用兩次）。

參數式多型 函數 twice也是本書中第一個多型函數。在 Haskell的型別中，小
寫開頭的識別字（如其中的 a）是型別層次的參數。讀者可想像成在 twice的型別
最外層有一個省略掉的 ∀a. 也就是說, twice的完整型別是 ∀a ·(a→ a)→ (a→ a)
—對所有的型別 a, twice都可拿一個型別為 a→ a的函數，然後傳回一個型別
為 a→ a的函數。
在 twice的型別 (a→ a)→ (a→ a)中，
• 第一個 a→ a是參數 f 的型別，
• 在第二個 a→ a中，第一個 a是參數 x的型別，
• 第二個 a則是整個計算結果的型別。

參數 f 的型別必須是 a→ a: 輸出入型別必須一樣，因為 f 的結果必須可當作 f
自己的輸入。
在 twice被使用時，型別參數 a會依照上下文被特化 (instantiate)成別的型

別。例如 twice square中，因為 square的型別是 Int→ Int,這一個 twice的型別
變成了 (Int→ Int)→ (Int→ Int) — a被特化成 Int. 若某函數 k的型別是 Float→
Float,在 twice k中，twice的型別是 (Float→ Float)→ (Float→ Float)。同一個
函數 twice可能依其上下文而有許多不同的型別，但都是 (a→ a)→ (a→ a)的
特例。「一段程式可能有許多不同型別」的現象稱作多型 (polymorphism)。多型
又有許多種類，此處為其中一種。詳情見… [todo: ]

習題 1.5 — 為何 twice的型別不可以是 (a→ b)→ (a→ b)?

例 1.5. forktimes f g x = f x×g x .

算式 forktimes f g x把 f x和 g x的結果乘起來。

1. 請想想 forktimes的型別該是什麼？
2. 試定義函數 compute :: Int→ Int, 使用 forktimes計算 x2 + 3× x+ 2。提示：

x2 +3× x+2 = (x+1)× (x+2).

答. 如同 twice, forktimes可以有很多型別，但都應該是 (a→ Int)→ (a→ Int)→
a→ Int的特例：在 forktimes f g x中，f 和 g的型別可以是 a→ Int,其中 a可以
是任何型別 a，而 x的型別必須也是同一個 a. 函數 compute可定義如下：

compute :: Int→ Int

compute = forktimes (+1) (+2) .

其中 forktimes型別中的 a被特化為 Int. □

如前所述，forktimes f g x 把 f x 和 g x 的結果乘起來。但，一定得是乘法
嗎？我們當然可以再多做一點點抽象化。

例 1.6. 考慮函數 lift2 h f g x = h (f x) (g x).
1. lift2 的型別是什麼？

2. 用 lift2 定義 forktimes.
3. 用 lift2 計算 x2 +3× x+2.

答. 我們把 lift2 最泛用的型別和其定義重複如下：

lift2 :: (a→ b→ c)→ (d→ a)→ (d→ b)→ d→ c .

lift2 h f g x = h (f x) (g x) .

有了 lift2, forktimes可定義為：

forktimes :: (a→ Int)→ (a→ Int)→ a→ Int

forktimes = lift2 (×) ,

請讀者觀察：lift型別中的 a, b, c都特化成 Int, d則改名為 a.
我們也可用 lift2 定義 compute:

compute :: Int→ Int

compute = lift2 (×) (+1) (+2) .

函數 lift2 可以看作一個作用在二元運算子上的運算子，功用是把二元運算
子「提升」到函數層次。例如，原本 (×)只能拿兩個 Int當作參數，（例：1×2
是「把 1和 2乘起來」），但現在 lift2 (×)可將函數 (+1)和 (+2)當參數了，意
思為「把 (+1)和 (+2)的結果乘起來」。 □

1.4 函數合成

拿到一個函數 f，我們能做的基本操作包括把 f 作用在某個參數上、把 f 傳給
別的函數… 此外，另一個常用的基本操作是將 f 和別的函數合成 (compose) 為

一個新函數。8 「合成」運算子在 Haskell中寫成 (·). 這個運算子的形式定義如
下（我們先看定義本體，待會兒再看型別）：

(f ·g) x = f (g x) .

若用口語說，f ·g是將 f 和 g兩個函數「串起來」得到的新函數：輸入 x先
丟給 g,後者算出的結果再傳給 f .

例 1.7. square ·double與 double · square都是由 Int到 Int的函數. 直覺上，前者
把輸入先給 double,其結果再給 square。後者則反過來。如何了解它們的行為？
既然它們是函數，我們便餵給它們一個參數，看看會展開成什麼！兩者分別展

開如下：

(square ·double) x
= { (·)的定義 }

square (double x)
= (x+ x)× (x+ x) ,

(double · square) x
= { (·)的定義 }

double (square x)
= (x× x)+(x× x) .

所以，如果輸入為 x, (square ·double) x計算 (2x)2; (double · square) x則是 2x2.

但，並不是所有函數都可以串在一起：f ·g之中，g的輸出型別和 f 的輸入
型別必須一致才行。運算子 (·)包括型別的完整定義為：

(·) :: (b→ c)→ (a→ b)→ a→ c
(f ·g) x = f (g x) .

如果 g的型別是 a→ b, f 的型別是 b→ c,將他們串接起來後，便得到一個型別
為 a→ c的函數。
有了 (·),函數 twice可以定義如下：

twice :: (a→ a)→ (a→ a)
twice f = f · f .

確實，根據 (·)的定義，twice f x = (f · f ) x = f (f x)，和 twice原來的定義相同。
為了討論函數合成的性質，我們先介紹一個函數 id:

id :: a→ a
id x = x .

它又稱作單位函數或恆等函數。這是一個似乎沒有在做什麼的函數：給任何輸
入，id都原封不動地把它送到輸出 —這也反映在他的型別 a→ a上。這個函數
有什麼重要性呢？原來，(·)滿足結合律，並且以 id為單位元素（這也是「單
位函數」這名字的由來）：

id · f = f = f · id ,

(f ·g) ·h = f · (g ·h) .

8Robert Glück認為函數上應有三個基本操作：函數應用、函數合成、以及求一個函數的反函
數。前兩者已經提到，第三者則是大部分語言欠缺的。

用數學術語來說的話，id 與 (·)形成一個幺半群 (monoid)。函數 id 的重要
性就如同 0在代數中的重要性（0與 (+)也是一個幺半群）。我們在許多計算、
證明中都會見到它。
以下我們試著證明 (·)的結合律。我們想論證 (f · g) · h = f · (g · h)，但該如

何下手？該等式的等號左右兩邊都是函數。當我們說兩個整數相等，意思很清
楚：如果等號左邊是 0,右邊也是 0;如果左邊是 1,右邊也是 1… 但說兩個函數
「相等」，是什麼意思呢？

定義 1.8 (外延相等 (extensional equality)). 給定兩個型別相同的函數 f 和 g,當
我們說它們外延相等 (extensionally equal)，意思是給任何一個輸入，f 和 g都算
出相等的輸出。也就是：(∀x · f x = g x).
本書中，當我們寫兩個函數相等 (f = g)時，指的便是外延相等，除非例外

註明。

在外延相等的假設下，證明 (f · g) · h = f · (g · h) 也就是證明對任何一個 x,
((f ·g) ·h) x = (f · (g ·h)) x均成立. 我們推論如下：

((f ·g) ·h) x
= { (·)的定義 }
(f ·g) (h x)

= { (·)的定義 }
f (g (h x))

= { (·)的定義 }
f ((g ·h) x) .

= { (·)的定義 }
(f · (g ·h)) x .

既然 (f ·g) ·h = f · (g ·h)，我們便可統一寫成 f ·g ·h,不用加括號了。

習題 1.6 — 證明 id · f = f = f · id.

合成 (·)也是一個中序運算子。和其他中序運算子一樣，其優先性低於函數應
用。因此，當我們寫 f ·g x,指的是 f · (g x) — g x為一個函數，和 f 合成，而不
是 (f ·g) x（後者根據 (·)的定義，是 f (g x)）。

例 1.9. 下列程式中，有些是合法的 Haskell式子、有些則有型別錯誤。對每個
程式，如果它是合法的，請找出它的型別，並說說看該程式做什麼。如果有型

別錯誤，請簡述為什麼。

1. square · smaller 3;
2. smaller 3 · square;
3. smaller (square 3);
4. smaller · square 3.

答. 前三者都是 Int→ Int。
1. 根據 (·)的定義，(square · smaller 3) x = square (smaller 3 x). 因此 square ·

smaller 3是一個函數，將其輸入和 3比較，取較小者的平方。

2. (smaller 3 · square) x = smaller 3 (square x). 因此它讀入 x,並在 3或 x↑2之
中選較小的那個。

3. smaller (square 3)是一個函數，讀入 x之後，選擇 x與 3 ↑ 2之中較小的
那個。

4. smaller · square 3有型別錯誤：square 3不是一個函數（而是一個整數），
無法和 smaller合成。

□

函數應用運算子 本書中有些時候會將許多函數組合成一串，例如 square ·
double · (+1) · smaller 3。由於函數應用的優先順序比一般二元運算元高，把上
述式子應用在參數 5之上時得寫成

(square ·double · (+1) · smaller 7) 5 ,

（這個式子的值為 (2× (5+1))2。）每次都得加一對括號似乎有些累贅。Haskell
另有一個運算子 ( $),唸作 “apply”,代表函數應用：

( $) :: (a→ b)→ a→ b
f $ x = f x .

f $ x和 f x意思一樣。那麼我們為何需要這個運算子呢？原因之一是 ( $)的優
先度比 (·)低，因此上式可省去括號改寫如下：

square ·double · (+1) · smaller 7 $ 5 .

運算子 ( $)的另一個重要意義是：「函數應用」這個動作有了符號，成為可
以獨立討論的事物。例如，( $)可以當作參數。一個這麼做的例子是習題 1.11。

常量函數 既然介紹了 id,本節也順便介紹一個以後將使用到的基本組件。給
定 x之後，函數 const x是一個不論拿到什麼函數，都傳回 x的函數。函數 const
的定義如下：

const :: a→ b→ a
const x y = x .

第 1.1節開頭的範例 three可定義為 three = const 3.
「無論如何都傳回 x」聽來好像是個沒用的函數，但和 id一樣，我們日後會
看到它在演算、證明中時常用上。事實上，組件邏輯理論告訴我們：所有函數
都可以由 id, const,和下述的 subst三個函數組合出來。

subst :: (a→ b→ c)→ (a→ b)→ (a→ c)
subst f g x = f x (g x) .

1.5 λ 算式

雖說函數是一級市民，在本書之中，目前為止，仍有一項功能是其他型別擁
有、函數卻還沒有的：寫出一個未命名的值的能力。整數、真假值都能不經命
名、直接寫在算式中，例如，我們可寫 smaller (square 3) 4,而不需要先定義好

num1,num2 :: Int
num1 = 3
num2 = 4 ,

才能說 smaller (square num1) num2. 但使用函數時，似乎非得先給個名字，才
能使用它：

square :: Int→ Int

square x = ... ,

quad = twice square .

如果為了某些原因（例如，如果在我們的程式中 square只會被用到一次），我
們不想給 square一個名字，我們能不能直接把它寫出來呢？

λ 算式便是允許我們這麼做的語法。以直觀的方式解釋，λx→ e便是一個
函數，其中 x是參數，e是函數本體。例如，(λx→ x×x)是一個函數，計算其
輸入 (x)的平方。如果我們不想給 square一個名字,我們可將 quad定義為：

quad = twice (λx→ x× x) .

寫成 λ 算式的函數也可直接作用在參數上，例如 (λx→ e1) e2。這個式子歸
約的結果通常表示為 e1 [e2 / x ],意思是「將 e1 之中的 x代換為 e2」。例如，算
式 (λx→ x× x) (3+4)可歸約為 (3+4)× (3+4).

例 1.10. 以下是一些 λ 算式的例子：
• 函數 (λx→ 1+ x)把輸入遞增 —和 (1+)相同。其實，把 (1+)的語法糖

去掉後，得到的就是這個 λ 算式。
• (λx→ λy→ x+2× x× y)是一個傳回 λ 算式的函數。

– (λx→ λy→ x+ 2× x× y) (3+ 4)可歸約為 (λy→ (3+ 4)+ 2× (3+
4)× y)。注意 λx不見了，函數本體中的 x被代換成 3+ 4，λy→ . .

則仍留著。

– (λx→ λy→ x+2× x× y) (3+4) 5可歸約為 (3+4)+2× (3+4)×5.
• 由於傳回函數的函數是常見的，Haskell (如同 λ -calculus)提供較短的語法。
上述例子中的函數也可簡寫成：(λx y→ x+2× x× y).

• 函數也可以當參數。例如，(λx→ x 3 3) (+)可歸約為 (+) 3 3,或 3+3.
• 以下是 (λ f x→ f x x) (λy→ 2× y) 3的求值過程：

(λ f x→ f x x) (λy z→ 2× y+ z) 3
= (λx→ (λy→ 2× y+ z) x x) 3
= (λy→ 2× y+ z) 3 3
= 2×3+3
= 9 .

• 在 λx→ e之中，x是範圍限於 e的區域識別字。因此：

(λ f x→ x+ f x) (λx→ x+ x) 3
= (λx→ x+(λx→ x+ x) x) 3
= 3+(λx→ x+ x) 3
= 3+3+3
= 9 .

有了 λ 算式後，函數 smaller又有另一種寫法：

smaller :: Int→ Int→ Int

smaller = λx y→ if x ⩽ y then x else y .

事實上，λ 算式可視為更基礎的機制 —目前為止我們所介紹的種種語法結
構都僅是 λ 算式的語法糖，都可展開、轉譯為 λ 算式。Haskell的 λ 算式源於
一套稱為 λ 演算 (λ calculus)的形式語言 —這是一個為了研究計算本質而發展
出的理論，也是函數語言的理論核心。我們將在爾後的章節中做更詳盡的介
紹。[todo: which?]

1.6 簡單資料型態

藉由一些例子，我們已經看過 Haskell的一些數值型別：Int, Float等等。在本
節中我們將簡短介紹我們將用到的一些其他型別。

1.6.1 布林值

布林值 (Boolean)常用於程式中表達真和假。在 Haskell中，我們可假想有這樣
的一個型別定義：

data Bool= False | True .

其中，data 是 Haskell 宣告新資料型別的保留字。上述定義可用口語描述成
「定義一個稱作 Bool的新資料型別，有兩個可能的值，分別為 False和 True.」
False和 True是型別 Bool的唯二兩個建構元 —任何型別為 Bool的值，如果有
正規式，必定是它們兩者之一。在 Haskell之中，建構元必須以大寫英文字母
或冒號 (:)開頭。

樣式配對 有了資料，我們來看看怎麼定義該型別上的函數。以布林值為輸入
的函數中，最簡單又常用的可能是 not:

not ::Bool→ Bool

not False= True

not True = False .

這和我們的的直覺理解一致：not False是 True, not True是 False. 我們看到
這個定義寫成兩行（正式說來是兩個「子句」），每一個子句分別對應到 Bool的

一個可能的值。以下則是邏輯上的「且」和「或」（分別寫作 (∧)與 (∨)）的定
義：9

(∧),(∨) ::Bool→ Bool→ Bool

False ∧ y = False

True ∧ y = y ,

False ∨ y = y
True ∨ y = True .

運算子 (∧)與 (∨)的定義同樣是各兩個子句，每個子句分別考慮其第一個參
數的值。以 x ∧ y為例：如果 x是 False,不論 y的值為何，x ∧ y都是 False；如
果 x是 True, x & y的值和 y相同。(∨)的情況類似。

例 1.11. 以下函數判斷給定年份 y是否為閏年。

leapyear :: Int→ Bool

leapyear y = (y ‘mod‘ 4 = = 0) ∧
(y ‘mod‘ 100 ̸= = 0 ∨ y ‘mod‘ 400 = = 0) .

我們來算算看 leapyear 2016。依照定義展開為

(2016 ‘mod‘ 4 = = 0) ∧ (2016 ‘mod‘ 100 ̸= = 0 ∨ 2016 ‘mod‘ 400 = = 0) .

接下來該怎麼做呢？函數 (∧)的定義有兩個子句，我們得知道 2016 ‘mod‘
4 = = 0的值才能得知該歸約成哪個。因此只好先算 2016 ‘mod‘ 4 = = 0，得到 True:

True ∧ (2016 ‘mod‘ 100 ≠ = 0 ∨ 2016 ‘mod‘ 400 = = 0) ,

然後依照 (∧)的定義歸約為 2016 ‘mod‘ 100 ≠ = 0 ∨ 2016 ‘mod‘ 400 = = 0. 接下來
也依此類推。
我們發現這是第 1.1節中所提及的被迫求值的例子：我們得先把參數算出，

才知道接下來如何走。函數 not, (∧), (∨) 定義成許多個子句，每個都分析其
參數的可能外觀，據此決定該怎麼走。這種定義方式稱作樣式配對 (pattern
matching)：等號左手邊的 False, True等等在此是樣式 (pattern)。使用這些函數
時，例如 x ∧ y中，x得先被算到可以和這些樣式配對上的程度，才能決定接下
來的計算如何進行。
樣式配對也可用在不止一個參數上。例如，以下的運算元 (= =)判斷兩個布

林值是否相等。

(= =) ::Bool→ Bool→ Bool

False = = False= True

False = = True = False

True = = False= False

True = = True = True .

9邏輯「且」又稱作合取 (conjunction)；邏輯「或」又稱作析取 (disjunction). 在 Haskell中，「且」
與「或」需分別寫成 (&&)和 (||)。本書中採用數學與邏輯領域較常使用的 (∧)與 (∨).

讀者可能注意到我們用了同一個符號 (= =)來表示整數與布林值的相等測試。
請讀者暫且接受，相信 Haskell有某些方式可得知任一個算式中的 (= =)到底是
什麼型別的相等。詳情 [todo: where?]

Haskell中另有一個專用來做樣式配對的 case算式。例如，(∧)也可寫成如
下的形式：

(∧) ::Bool→ Bool

x ∧ y = case x of
False→ False

True → y .

由於 case是算式，如同 let一樣可出現在其他算式中，也可巢狀出現。

習題 1.7 — 以 case算式定義 not, (∨),和 (= =).

習題 1.8 — 另一個定義 (= =) ::Bool→ Bool→ Bool的方式是

x = = y = (x ∧ y) ∨ (not x ∧ not y) .

請將 (x,y) :=(False,False), (x,y) :=(False,True)等四種可能分別代入化簡，
看看是否和本節之前的 (= =)定義相同。

1.6.2 字元

我們可把「字元」這個型別想成一個很長的 data宣告：

data Char = 'a' | 'b' | ... | 'z' | 'A' | 'B' ....

其中包括所有字母、符號、空白… 目前的 Haskell甚至有處理 Unicode字元的
能力。但無論如何，Char 之中的字元數目是有限的。我們可用樣式配對定義
字元上的函數。注意：字元以單引號括起來。
我們也可假設字元是有順序的，每個字元對應到一個內碼。關於 Char的常

用函數中，ord將字元的內碼找出，chr則將內碼轉為字元：

ord ::Char→ Int ,

chr :: Int→ Char .

例 1.12. 下列函數 isUpper判斷一個字元是否為大寫英文字母；toLower則將大
寫字母轉成小寫字母，若輸入並非大寫字母則不予以變動。

isUpper ::Char→ Bool

isUpper c = let x = ord c in ord 'A' ⩽ x ∧ x ⩽ ord 'Z' ,

toLower ::Char→ Char

toLower c | isUpper c = chr (ord c−ord 'A'+ord 'a')
| otherwise = c .

1.6.3 序對

數學上，將兩個值（如 3和 'a')放在一起，就成了一個有序對 (ordered pair)，
可寫成 (3,'a')。之所以稱作「有序」對，因為其中兩個元素的順序是不可忽
略的 — (3,'a')與 ('a',3)是不同的有序對。另一個常見譯名是「數對」。由於
我們處理的不只是數字，本書將之簡稱為「序對」。
給兩個集合 A和 B,從 A之中任取一元素 x，從 B之中也任取一元素 y，兩

者的序對 (x,y) 形成的集合稱作 A 和 B 的笛卡兒積 (Cartesian product)，寫成
A×B:

A×B= {(x,y) | x ∈ A,y ∈ B} .

Haskell之中也有類似的構造。給定型別 a與 b,它們的序對的型別是 (a×b).
10 我們可以想像 Haskell有這麼一個型別定義：

data (a×b) = (a,b) .

以口語說的話，(a× b)是一個新型別，而具有此型別的值若有範式，必定是
(x,y)的形式，其中 x的型別是 a, y的型別是 b. 11 序對的建構元寫成 (,)，型別
為 a→ b→ (a×b). 例如 (,) 4 'b'= (4,'b').
兩個常用的函數 fst與 snd分別取出序對中的第一個和第二個元素：

fst :: (a×b)→ a
fst (x,y) = x ,

snd :: (a×b)→ b
snd (x,y) = y .

函數 fst與 snd的定義方式也是樣式配對：輸入值必須先計算成 (x,y)的形式。

例 1.13. 以下是一些序對與其相關函數的例子。
• (3,'a')是一個型別為 (Int×Char)的序對。

• fst (3,'a') = 3, snd (3,'a') = 'a'
• 函數 swap將序對中的元素調換：

swap :: (a×b)→ (b×a)
swap (x,y) = (y,x) .

另一個定義方式是 swap p = (snd p, fst p). 但這兩個定義並不盡然相同。詳見
第 1.7節。

序對也可以巢狀構成。例如 ((True,3),'c') 是一個型別為 ((Bool× Int)×
Char)的序對，而 snd (fst ((True,3),'c')) = 3. 在 Haskell之中，((a×b)× c)與
(a× (b× c))被視為不同的型別，但他們是同構的 —我們可定義一對函數在這
兩個型別之間作轉換：

assocr :: ((a×b)× c)→ (a× (b× c))
assocr ((x,y),z) = (x,(y,z)) ,

10然而，由於「程式可能不終止」這個因素作怪，a×b的元素並不僅是 a與 b（如果視做集合）
的笛卡兒積。詳見 [todo: where?]

11其實這個定義並不符合 Haskell 的語法，因此只是方便理解的想像。另，型別 (a× b) 在
Haskell中寫成 (a,b). 我的經驗中，讓型別與值的語法太接近，反易造成困惑。

同構

兩個集合 A與 B同構 (isomorphic)，意思是 A之中的每個元素都唯一地對應到
B之中的一個元素，反之亦然。
一個形式定義是：A與 B同構意謂我們能找到兩個全 (total)函數 to ::A→ B和

from ::B→ A,滿足

from · to = id ,
to · from = id .

此處的兩個 id型別依序分別為 A→ A和 B→ B。將定義展開，也就是說，對所
有 x ::A，from (to x) = x;對所有 y ::B, to (from y) = y. 這個定義迫使對每個 x都存在
一個唯一的 to x,反之亦然。
我們已有兩個例子：((a× b)× c)與 (a× (b× c))同構，此外，(a× b)與 (b× a)

也同構，因為 swap · swap = id.
如果集合 A與 B同構，不僅 A之中的每個元素都有個在 B之中相對的元素，給

任一個定義在 A之上的函數 f ,我們必可構造出一個 B之上的函數，具有和 f 相同
的性質。即使 A與 B並不真正相等，我們也可把它們視為基本上沒有差別的。在
許多無法談「相等」的領域中，同構是和「相等」地位一樣的觀念。

assocl :: (a× (b× c))→ ((a×b)× c)
assocl (x,(y,z)) = ((x,y),z) ,

並且滿足 assocr ·assocl = id,和 assocl ·assocr = id.

習題 1.9 — 試試看不用樣式配對，而以 fst和 snd定義 assocl和 assocr:

assocl p = ...

assocr p = ...

另外可一提的是，Haskell允許我們在 λ 算式中做樣式配對。例如 fst的另一種
寫法是：

fst = λ (x,y)→ x .

Haskell另有提供更多個元素形成的有序組，例如 (True,3,'c')是一個型別
為 (Bool× Int×Char)的值。但本書暫時不使用他們。

分裂與積 在我們將介紹的程式設計風格中，以下兩個產生序對的運算子相當
好用。第一個運算子利用兩個函數產生一個序對：

⟨·, ·⟩ :: (a→ b)→ (a→ c)→ a→ (b× c)
⟨f ,g⟩ x = (f x,g x) .

給定兩個函數 f :: a→ b和 g :: a→ c, ⟨f ,g⟩ :: a→ (b× c)是一個新函數，將 f 和
g的結果收集在一個序對中。我們借用範疇論的詞彙，將此稱作分裂 (split) —
⟨f ,g⟩可讀成「f 與 g的分裂」。

如果我們已經有了一個序對，我們可用 (f ×g)算出一個新序對：

(×) :: (a→ b)→ (c→ d)→ (a× c)→ (b×d)
(f ×g) (x,y) = (f x,g x) .

函數 (f ×g)將 f 和 g分別作用在序對 (x,y)的兩個元素上。這個操作稱作「f 和
g的乘績 (product)」，同樣是借用範疇論的詞彙。
目前 Haskell 的標準階層函式庫將分裂與乘積收錄在 Control.Arrow 中，

⟨f ,g⟩寫作：f &&& g, (f ×g)則寫作 f *** g.

Currying 與 Uncurrying 如前所述，Haskell 的每個函數都只拿一個參數。
拿多個參數的函數可以傳回函數的函數來模擬，稱作 currying. 有了序對之
後，另一種模擬多參數的方式是把參數都包到一個序對中。例如，型別為
(a×b)→ c)的函數可視為拿了兩個型別為 a與 b的參數。
函數 curry與 uncurry幫助我們在這兩種表示法之間轉換 — curry將拿序對

的函數轉換成 curried函數，uncurry則讓 curried函數改拿序對當作參數：

curry :: ((a×b)→ c)→ (a→ b→ c)
curry f x y = f (x,y) ,

uncurry :: (a→ b→ c)→ ((a×b)→ c)
uncurry f (x,y) = f x y .

例：如果 (= =)的型別為 Int→ Int→Bool, uncurry (= =)的型別為 (Int× Int)→Bool。
後者檢查一個序對中的兩個值是否相等（例：uncurry (= =) (3,3) = True）。

習題 1.10 — 事實上，curry 與 uncurry 的存在證明了 (a× b)→ c 與 a→
b→ c是同構的。試證明 curry ·uncurry = id,以及 uncurry · curry = id.

習題 1.11 — 請說明 map (uncurry ( $))的型別與功能。關於 ( $)請參考第
30頁。

1.7 弱首範式

我們現在可把第 1.1節中提及的求值與範式談得更仔細些。第一次閱讀的讀者
可把本節跳過。回顧 fst使用樣式配對的定義 fst (x,y) = x. 假設我們把 swap定
義如下：

swap p = (snd p, fst p) .

考慮 fst (swap (3,'a'))該怎麼求值：

fst (swap (3,'a'))
= { swap的定義 }

fst (snd (3,'a'), fst (3,'a'))
= { fst的定義 }

snd (3,'a')

= { snd的定義 }
'a' .

在第一步中，由於 fst使用樣式配對，我們得先把 swap (3,'a')算出來。若把
swap (3,'a')算到底，得到的範式是 ('a',3). 但如第一步中所顯示，如果目的
只是為了配對 (x,y)這個樣式，我們並不需要把 swap (3,'a')算完，只需算到
(snd (3,'a'), fst (3,'a'))即可 — x可對應到 snd (3,'a'), y可對應到 fst (3,'a'),
fst的計算便可以進行。在下一步中，子算式 fst (3,'a')便被丟棄了，並沒有必
要算出來。
做樣式配對時，Haskell只會把算式歸約到剛好足以與樣式配對上的程度。

當樣式的深度只有一層（如 (x,y)）時，與之配對的式子會被歸約成一種稱作弱
首範式 (weak head normal form)的形式。弱首範式有其嚴格定義，但本書讀者
只需知道：算式會被歸約到露出最外面的建構元，（在此例中是被歸約成 ( , )

的形式），然後便停下來。

例 1.14. 回顧 swap的兩種定義方式，分別命名為

swap1 (x,y) = (y,x) ,

swap2 p = (snd p, fst p) ,

若考慮不終止的參數，兩者的行為並不盡然相同。定義 three (x,y) = 3,並
假設 ⊥是一個沒有範式的式子 —一旦開始對 ⊥求值，便停不下來。試計算
three (swap1 ⊥)和 three (swap2 ⊥).

答. 由於 three使用樣式配對，計算 three (swap1 ⊥)時得把 swap1 ⊥歸約成弱首
範式。同理，計算 swap1 ⊥時，第一步便是先試圖把 ⊥歸約成弱首範式，然
後便停不下來了。
至於 three (swap2 ⊥)則可如下地求值：

three (swap2 ⊥)
= { swap2 之定義 }

three (snd ⊥, fst ⊥)
= { three之定義 }

3 .

第一步中，swap2 ⊥則依照範式順序求值的原則展開為 (snd ⊥, fst ⊥) —這是一
個序對，只是該序對含有兩個沒有範式的元素。該序對可對應到樣式 (x,y),因
此整個式子歸約為 3.
附帶一提，three⊥是一個不停止的計算。 □

1.8 串列

一個串列 (list)抽象說來便是將零個或多個值放在一起變成一串。串列是函數
語言傳統上的重要資料結構：早期的函數語言 LISP便是 LISt Processing的縮
寫。Haskell 中的串列多了一個限制：串列中的每個元素必須有同樣的型別。

(:)與 (::)

大部分有型別的函數語言（如 ML, Agda等）之中，(:)表示型別關係，(::)則
是串列的建構元。Haskell 的前身之一是 David A. Turner 的語言 Miranda. 在其
Hindley-Milner型別系統中，Miranda使用者幾乎不需寫出程式的型別 —型別可由
電腦自動推導。而串列是重要資料結構。把兩個符號調換過來，使常用的符號較
短一些，似乎是合理的設計。

Haskell繼承了Miranda的語法。然而，後來 Haskell的型別發展得越來越複雜，
使用者偶爾需要寫出型別來幫助編譯器。即使型別簡單，程式語言界也漸漸覺得
將函數的型別寫出是好習慣。而串列建構元的使用量並不見得比型別關係多。但
此時想改符號也為時已晚了。

本書中將「元素型別均為 a的串列」的型別寫成 List a. 12 Hasekell以中括號表
示串列，其中的元素以逗號分隔。例如，[1,2,3,4 ]是一個型別為 List Int的串
列，其中有四個元素；[True,False,True ]是一個型別為 List Bool的串列，有三
個元素。至於 [ ]則是沒有元素的空串列（通常唸做 “nil”），其最通用的型別為
List a，其中 a可以是任何型別。
串列的元素也可以是串列。例如 [[1,2,3 ], [ ], [4,5 ]] 的型別是 List (List Int),

含有三個元素，分別為 [1,2,3 ], [ ],和 [4,5 ].
事實上，上述的寫法只是語法糖。我們可想像 Haskell有這樣的型別定義：

data List a = [ ] | a :List a .

意謂一個元素型別為 a的串列只有兩種可能構成方式：可能是空串列 [ ]，也可
能是一個元素 (a)接上另一個串列 (List a)。後者的情況中，元素和串列之間用
符號 (:)銜接。
符號 (:)唸作 “cons”,為「建構 (construct)」的字首。其型別為 a→ List a→

List a —它總是將一個型別為 a的元素接到一個 List a之上，造出另一個 List a.
上述的 [1,2,3,4 ]其實是 1 : (2 : (3 : (4 : [ ])))的簡寫：由空串列 [ ]開始，將元素
一個個接上去。為了方便，Haskell 將 (:) 運算元視做右相依的，因此我們可
將括號省去，寫成 1 : 2 : 3 : 4 : [ ]。無論如何，這樣的串列表示法是偏一邊的 —
元素總是從左邊放入，最左邊的元素也最容易取出。如果一個串列不是空的，
其最左邊的元素稱作該串列的頭 (head)，剩下的元素稱作其尾 (tail)。例如，
[1,2,3,4 ]的頭是 1,尾是 [2,3,4 ].

Haskell中將字串當作字元形成的串列。標準函式庫中這麼定義著：

type String = List Char .

意謂 String 就是 List Char。在 Haskell 中，data 用於定義新型別，而 type 並
不產生一個新的型別，只是給現有的型別一個較方便或更顯出當下意圖的名
字。此外，Haskell另提供一個語法糖，用雙引號表達字串。因此，"fun"是
['f','u','n' ]的簡寫，後者又是 'f' :'u' :'n' : [ ]的簡寫。
本節接下來將介紹許多與串列相關的內建函數。

12Haskell中的寫法是 [a ]. 同樣地，在我的教學經驗中，將中括號同時使用在值與型別上造成不
少誤解。例如學生可能認為 [1,2 ]的型別是 [Int, Int ] —其實應該是 [Int ].

1.8.1 串列解構

我們先從拆解串列的函數開始。函數 head和 tail分別取出一個串列的頭和尾：

head ::List a→ a
head (x : xs) = x ,

tail ::List a→ List a
tail (x : xs) = xs .

注意其型別：head 傳回一個元素，tail 則傳回一個串列。例：head "fun" 和
tail "fun"分別是字元 'f'和字串 "un". 函數 head和 tail都可用樣式配對定義，
但此處的樣式並不完整，尚缺 [ ]的情況。如果將空串列送給 head 或 tail，則
會出現執行時錯誤。因此，head和 tail都是部分函數 (partial functions) —它們
只將某些值（非空的串列）對應到輸出，某些值（空串列）則沒有。
函數 null判斷一個串列是否為空串列。它也可用樣式配對定義如下：

null ::List a→ Bool

null [ ] = True

null (x : xs) = False .

本書依循 Bird [1998]中的變數命名習慣，將型別為串列的變數以 s做結尾，例
如 xs, ys等等。至於「元素為串列的串列」則命名為 xss, yss等等。但這只是為
方便理解而設計的習慣。Haskell本身並無此規定。
除了 head 與 tail，也有另一組函數 last與 init分別取出一個串列最右邊的

元素，以及剩下的串列：

last ::List a→ a ,

init ::List a→ List a .

例：last "fun"與 init "fun"分別為字元 'n'與字串 "fu". 但 last與 init的定義
比起 head 與 tail來得複雜：記得我們的串列表示法是偏向一邊的，從左邊存
取元素容易，從右邊存取元素則較麻煩。我們會在 [todo: where]之中談到 last
與 init的定義。

1.8.2 串列生成

第 1.8.1節中的函數均將串列拆開。本節之中我們來看一些生成串列的方法。如
果元素的型別是有順序的（例如 Int, Char等型別），Haskell提供了一個方便我
們依序生成串列的語法。以例子說明：

例 1.15. 以下為 Haskell的列舉語法的一些例子：
• [0 . .10 ]可展開為 [0,1,2,3,4,5,6,7,8,9,10 ].
• 可用頭兩個元素來指定間隔量。例如 [0,3 . .10 ] = [0,3,6,9 ]. 注意該串列
的元素不超過右界 10.

• 在 [10 . .0 ]之中，10一開始就超過了右界 0,因此得到 [ ]. 如果想要產生由
10倒數到 0的串列，可這樣指定間隔：[10,9 . .0 ].

• 字元也是有順序的，因此 ['a' . .'z' ]可展開為含所有英文小寫字母的串
列。

• 至於沒有右界的 [0 . . ]則會展開為含 [0,1,2,3... ]的無限長串列。

函數 iterate :: (a→ a)→ a→ List a用於產生無限長的串列：iterate f x可展
開為 [x, f x, f (f x), f (f (f x))... ].

例 1.16. 一些 iterate的例子：
• iterate (1+) 0展開為 [0,1,2,3... ]. 其實 [n . . ]可視為 iterate (1+) n的簡寫。
• 在例 1.24中我們會看到 [m . .n ]也可用 iterate與其他函數做出。
• iterate not False可得到無窮串列 [False,True,False. . . ].

數學中描述集合時常使用一種稱作集合建構式 (set comprehension)的語法。
例如，{x× x | x ∈ S,odd x}表示收集所有 x× x形成的集合，其中 x由集合 S中
取出，並且必須為奇數。Haskell將類似的語法用在串列上。同樣以例子說明：

例 1.17. 串列建構式 (list comprehension)的例子：
• [x | x← [0 . .9 ]]表示「從 [0 . .9 ]之中取出 x,並收集 x」，可展開為 [0,1,2,3,4,5,6,7,8,9 ].
• [x× x | x ← [0 . .10 ]] 的 x 來源和之前相同，但收集的是 x× x，得到
[0,1,4,9,25,36,49,64,81 ].

• [(x,y) | x← [0 . .2 ],y← "abc" ]展開得到 [(0,'a'), (0,'b'), (0,'c'), (1,'a'),
(1,'b'), (1,'c'), (2,'a'), (2,'b'), (2,'c')]. 注意序對出現的順序：先固
定 x，將 y跑過一遍，再換成下一個 x.

• [x×x | x← [0 . .10 ],odd x ]從 [0 . .10 ]之中取出 x，但只挑出滿足 odd x的那
些，得到 [1,9,25,49,81 ].

例 1.18. 以下算式的值分別為何？
1. [(a,b) | a← [1 . .3 ],b← [1 . .2 ]].
2. [(a,b) | b← [1 . .2 ],a← [1 . .3 ]].
3. [(i, j) | i← [1 . .4 ], j← [(i+1) . .4 ]]. 這是一個有了 i← . .之後，i即可在右
方被使用的例子。

4. [(i, j) | i← [1 . .4 ],even i, j← [(i+1) . .4 ],odd j ].
5. ['a' | i← [0 . .10 ]]. 這個例子顯示 i並不一定非得出現在被收集項目中。

答. 分別展開如下：
1. [(1,1),(1,2),(2,1),(2,2),(3,1),(3,2)].
2. [(1,1),(2,1),(3,1),(1,2),(2,2),(3,2)].
3. [(1,2),(1,3),(1,4),(2,3),(2,4),(3,4)].
4. [(2,3)].
5. "aaaaaaaaaaa".

□

串列建構式在寫程式時相當好用，但它也僅是個語法糖 —所有的串列建構
式都可轉換為後面的章節將介紹的 map, concat, filter等函數的組合。

1.8.3 串列上的種種組件函數

我們將在本節介紹大量與串列有關的函數。它們常被稱做組件 (combinators)函
數。每一個組件都負責一項單一、但具通用性而容易重用的功能。它們常用來
彼此結合以組出更大的程式。

介紹這些函數有兩個原因。首先，它們可用來組出許多有趣的程式。我們
將一邊逐一介紹這些函數，一邊以例子示範，同時也逐漸帶出本章鼓勵的一種
特殊編程風格。另一個原因是在日後的章節中我們也都將以這些函數作為例
子，討論並證明關於它們的性質。
第一次閱讀的讀者可能訝異：這麼多函數，怎麼記得住？事實上，這些組

件大都僅是把常見的編程模式具體地以形式方式表達出來。辨識出這些模式
後，不僅會發現它們其實很熟悉，對於我們日後了解其他程式也有助益。

長度 函數 length :: List a→ Int 計算串列的長度。空串列 [ ] 的長度為 0. 例：
length "function"= 8.

索引 函數 (!!) 的型別為 List a→ Int→ a. 給定串列 xs 和整數 i, 如果 0 ⩽
i< length xs, xs !! i為 xs中的第 i個元素，但由 0起算。例 "function" !!0 = 'f',
"function" !! 3 = 'c'. 如果 i> length xs, 則會成為執行期錯誤。注意：如果
length xs = n，其中的元素編號分別為 0, 1 .. n−1.
在指令式語言中，索引是處理陣列常用的基本操作。處理陣列的常見模式

是用一個變數 i指向目前正被處理的元素，將 a [i ]的值讀出或覆寫，然後更新
i的值。但由接下來的許多範例中，讀者會發現本章盡量避免這種做法。也因
此 (!!)在本章中使用的機會不多。

連接 函數 (++) :: List a → List a → List a 將兩個串列相接。例：[1,2,3 ] ++
[4,5 ] = [1,2,3,4,5 ].
函數 (++) 和 (:) 似乎都是把串列接上東西。兩者有什麼不同呢？答案是：

(:) :: a→ List a→ List a永遠把一個元素接到串列的左邊，而 (++)把兩個串列
接在一起，兩個串列都有可能含有零個或多個元素。例：[ ]++[4,5 ] = [4,5 ]. 事
實上，(:)是比 (++)更基礎的操作。在第 2.4.1節中，我們會看到 (++)是用 (:)
定義而成的。
另一個關於連接的函數是 concat ::List (List a)→ List a：它以一個元素都是

串列的串列作為輸入，將其中的串列接在一起。例：concat [[1,2,3 ], [ ], [4 ], [5,6 ]]
= [1,2,3,4,5,6 ]. 它和 (++)的不同之處在哪呢？顯然，(++)總把兩個串列接在
一起，而 concat 的參數中可含有零個或多個串列。在第 2.4.1節中，我們會看
到 concat是用 (++)定義而成的。

取與丟 take的型別為 Int→ List a→ List a. take n xs取 xs的前 n個元素。若
xs的長度不到 n, take n xs能拿幾個就拿幾個. 例：take 3 "function" = "fun",
take 5 "ox"= "ox".
相對地，drop n xs丟掉 xs的前 n個元素。若 xs的長度不到 n, drop n xs能丟

幾個就拿幾個. 例：drop 3 "function" = "ction", take 5 "ox" = "". 函數 drop
的型別也是 Int→ List a→ List a.
函數 take和 drop顯然有些關聯，但它們的關聯該怎麼具體地寫下來呢？一

個可能是：對所有的 n和 xs,

take n xs++drop n xs = xs .

乍看之下似乎言之成理。但這個性質真的成立嗎？我們將在第 2章中討論到。

映射 map ::(a→ b)→ List a→ List b是串列上一個很重要的高階函數：map f xs
將 f 作用在 xs的每一個元素上。例：

map square [1,2,3,4 ] = [1,4,9,16 ] ,

map (1+) [2,3,4 ] = [3,4,5 ] .

回憶我們之前關於高階函數的討論，另一個理解方式是：map是一個處理函數
的操作。給一個「將 a變成 b」的函數 f :: a→ b，map將這個函數提升到串列
的層次，得到一個「將 List a變成 List b」的函數 map f ::List a→ List b.

例 1.19. 如果一個串列 xs可分解為 ys++ zs,我們說 ys是 xs的一個前段 (prefix),
zs則是 xs的一個後段 (suffix). 例如，串列 [1,2,3 ]的前段包括 [ ], [1 ], [1,2 ],與
[1,2,3 ]（注意：[ ]是一個前段，[1,2,3 ]本身也是）,後段則包括 [1,2,3 ], [2,3 ],
[3 ],與 [ ]。

試定義函數 inits ::List a→ List (List a),計算輸入串列的所有前段。13 提示：

目前我們可以用 map, take和其他函數組出 inits. 在第 2.6節中將會介紹另一個
做法。

答. 一種使用 map和 take的可能作法如下：

inits ::List a→ List (List a)
inits xs = map (λn→ take n xs) [0 . . length xs ] .

或著也可用串列建構式寫成 [take n xs | n← [0 . . length xs ]]. 讀者可能已發現：
[f x | x← xs ]就是 map f xs. □

習題 1.12 — 定義函數 tails ::List a→ List (List a), 計算輸入串列的所有後
段。

過濾 一個型別為 a→ Bool 的函數稱作一個「述語」(predicate). 給定述語
p, filter p xs 將 xs 之中滿足 p 的元素挑出。函數 filter 的型別為 (a→ Bool)→
List a→ List a。例：filter even [2,5,1,7,6 ] = [2,6 ].

例 1.20. 該怎麼得知一個字串中大寫字母的個數？將大寫字母過濾出來，計算
所得串列的長度即可。如下所示：

numUpper ::String→ Int

numUpper = length ·filter isUpper .

例 1.21. 下列算式求出 02 到 502 的平方數（能寫成 n2 的數字）中，結尾為 25
的數字。

filter ((= = 25) · (‘mod‘100)) (map square [0 . .50 ]) .

13請注意該函數的名字是 inits,和之前介紹過的 init不同。這是 Haskell函式庫中使用的命名。

歸約後得到 [25,225,625,1225,2025 ]. 其中 (= = 25) ·(‘mod‘100)部分使用了第 25頁
中提到的語法。如果覺得不習慣，也可用 λ 算式寫成：

filter (λn→ n ‘mod‘ 100 = = 25) (map square [0 . .50 ]) .

例 1.22. 接續上例。另一個可能寫法是先過濾出「平方之後結尾為 25」的數字，
再算這些數字的平方：

map square (filter ((= = 25) · (‘mod‘100) · square) [0 . .50 ]) .

這個算式也歸約出一樣的結果：[25,225,625,1225,2025 ]。
稍微推廣一些，這個例子暗示我們 filter p ·map f 和 map f ·filter (p · f )似乎是

等價的。但確實如此嗎？我們也將在第 2章中討論。

例 1.23. 接續上例。如果我們不僅希望找到結尾為 25的平方數，也希望知道它
們是什麼數字的平方，一種寫法如下：

filter ((= = 25) · (‘mod‘100) · snd) (map ⟨id,square⟩ [0 . .50 ]) .

我們用 map ⟨id,square⟩ 將每個數字與他們的平方放在一個序對中，得到
[(0,0),(1,1),(2,4),(3,9)... ]. 而 filter 的述語多了一個 snd, 表示我們只要那些
「第二個元素符合條件」的序對。上式化簡後可得到 [(5,25), (15,225), (25,625),
(35,1225), (45,2025)]. 運算元 ⟨·, ·⟩的定義詳見第 36頁。
述語 (= = 25) · (‘mod‘100) · snd可以展開為 (λ (i,n)→ n ‘mod‘ 100 = = 25).

取、丟、與過濾 函數 takeWhile, dropWhile和 filter有一樣的型別。

takeWhile :: (a→ Bool)→ List a→ List a ,

dropWhile :: (a→ Bool)→ List a→ List a .

它們之間的差異也許用例子解釋得最清楚：

filter even [6,2,4,1,7,8,2 ] = [6,2,4,8,2 ] ,

takeWhile even [6,2,4,1,7,8,2 ] = [6,2,4 ] ,

dropWhile even [6,2,4,1,7,8,2 ] = [1,7,8,2 ] .

filter p挑出所有滿足 p的元素；takeWhile p由左往右逐一取出元素，直到遇
上第一個不滿足 p的元素，並將剩下的串列丟棄；dropWhile p則與 takeWhile p
相對，將元素丟棄，直到遇上第一個不滿足 p的元素。直覺上，後兩者似乎也
應該滿足 takeWhile p xs++dropWhile p xs = xs,但這仍尚待驗證。

例 1.24. 給定整數 m與 n, [m . .n ]可視為 takeWhile (⩽ n) (iterate (1+) m)的簡寫。

例 1.25. 讀者也許覺得 takeWhile或 dropWhile似乎和迴圈有密切關係。確實，
利用 iterate與 dropWhile，我們可定義出類似 while迴圈的操作：

until :: (a→ Bool)→ (a→ a)→ a→ a
until p f = head ·dropWhile (not ·p) · iterate f

until p f x 由 x 算出 f x, 由 f x 算出 f (f x) … 直到 p (f (f . .x)) 成立為止。例：
until ((>50) · square) (1+) 0得到 8,因為 82 = 64,是第一個平方大於 50的非負
整數。由於惰性求值，iterate f 在意義上雖然是個無窮串列，但只會被執行到
dropWhile (not ·p)擷取的長度為止。
下述函數則實作了用輾轉相減法求最大公因數的古典演算法。函數 minus

不斷將大數減去小數，直到兩數相等為止：

gcd :: (Int× Int)→ Int

gcd = fst ·until (uncurry (= =)) minus ,

where minus (x,y) | x> y = (y,x− y)
| x< y = (y− x,x) .

關於 uncurry詳見第 37頁。

習題 1.13 — 試 定 義 一 個 函 數 squaresUpTo :: Int → List Int, 使 得
squaresUpTo n傳回所有不大於 n的平方數。例：squaresUpTo 10 = [1,4,9 ],
squaresUpTo (−1) = [ ].

拉鍊 函數 zip ::List a→ List b→ List (a×b)的作用可由下述例子示範：

zip [1,2,3 ] "abc"= [(1,'a'),(2,'b'),(3,'c')] ,

zip [1,2 ] "abc"= [(1,'a'),(2,'b')] ,

zip [1,2,3 ] "ab" = [(1,'a'),(2,'b')] ,

zip [1 . . ] "abc"= [(1,'a'),(2,'b'),(3,'c')] ,

zip [1 . . ] [2 . . ] = [(1,2),(2,3),(3,4) . . ] ,

zip xs ys 將串列 xs 與 ys 相對應的元素放在序對中。如果兩個串列長度不同，
zip將其中一個用完後即停止。zip也能處理無限長的串列。由於這個動作看來
像是把 xs與 ys當作拉鍊的兩側「拉起來」，因此用拉拉鍊的狀聲詞 “zip”命名。
相對地，也有一個函數 unzip ::List (a×b)→ (List a×List b)，將「拉鍊」拉

開。例：unzip [(1,'a'),(2,'b'),(3,'c')]可得到 ([1,2,3 ],"abc").
許多情況下，我們不想要把兩兩對應的元素放到序對中，而是分別餵給一

個二元運算子。這時可用另一個相關函數 zipWith,例：zipWith (+) [1,2,3 ] [4,5,6 ] =
[5,7,9 ]函數 zipWith可以這樣定義：

zipWith :: (a→ b→ c)→ List a→ List b→ List c
zipWith f = map (uncurry f ) · zip .

習題 1.14 — 用 zipWith定義 zip.

例 1.26. 試定義函數 positions ::Char→ String→ List Int,使得 positions z xs傳回
z在 xs中出現的所有位置。例：positions 'o' "hoola hooligans" = [1,2,7,8 ].

答. 一種可能寫法如下：

positions z = map fst ·filter ((= = z) · snd) · zip [0 . . ] .

我們用 zip [0 . . ]為輸入串列標上位置，用 filter ((= = z) · snd)取出第二個元素等
於 z的序對，最後用 map fst取出所有位置。注意函數合成與 currying的使用。

□

例 1.27. 接續上例。如果我們僅想要 z出現的第一個位置呢？我們可以定義：

pos ::Char→ String→ Int

pos z = head ·positions z .

這是一個部分函數，pos z xs傳回 positions z xs的第一個結果。如果 z沒有出現，
positions z xs傳回 [ ], pos z xs會得到執行期錯誤。如果 z出現在 xs中，由於惰性
求值，pos得到第一個位置後 positions便會停下，不會把串列整個產生。
如果我們希望 pos在 z沒有出現時傳回 −1,可以這麼做：

pos ::Char→ String→ Int

pos z xs = case positions z xs of
[ ] →−1
(i : is)→ i .

1.9 全麥編程

讀者至此應已注意到本章採用的特殊編程風格。一般說到串列，大家會先想到
資料結構課程中常提到的連結串列 (linked list)。介紹連結串列的範例程式大多
用迴圈或遞迴追蹤著指標，一個一個地處理串列中的元素。在指令式語言中做
關於陣列的操作時，也常用變數或指標指著「目前的」元素，並在一個迴圈中
將該變數逐次遞增或減。總之，我們處理聚合型資料結構時，總是將其中元素
一個個取出來處理。但本章的做法不同：我們將整個串列視為一個整體，對整
個串列做 map, filter, dropWhile等動作，或將它和另一個串列整個 zip起來…。
這種編程方式被稱作全麥編程 (wholemeal programming)，第 1.11節中將解釋

此詞的由來。全麥編程的提倡者們認為：一個個地處理元素太瑣碎，而鼓勵我
們拉遠些，使用組件，以更抽象的方式組織程式的結構。
諸如 map, filter, iterate, zipWith等等組件其實都是常見的編程模式。它們可

被視為為了特定目的已先寫好的迴圈。拜高階函數與惰性求值之賜，這些組件
能容易地被重用在許多不同脈絡中。這麼做的好處之一是：諸如 map, filter, zip
等組件的意義清楚，整個程式的意義也因此會比起自行在迴圈中一個個處理
元素來得容易理解。事實上，這麼做可以養成我們思考演算法的新習慣。一些
常見的編程模式現在是有名字的，我們把編程模式抽象出來了。而如同第 0章
所述，抽象化是我們理解、掌握、操作事物的重要方法。我們現在有了更多
詞彙去理解、討論程式與演算法：「這個演算法其實就是先做個 map，把結果
concat起來，然後做 filter…」

在本書其他章節中我們也將看到：這些抽象化方便我們去操作、轉換程
式。具體說來，如果程式用這些組件拼湊成，我們對這些組件知道的性質都可
用在我們的程式上。例如，如果我們知道 map f ·map g = map (f ·g)，當我們看
到程式中有兩個相鄰的 map,我們可用已知的性質把他們合併成一個 —這相當
於合併兩個迴圈。或著我們可以把一個 map拆成兩個，以方便後續的其他處
理。程式的建構方法使得程式含有更多資訊，使我們有更多可操作的空間。
全麥編程之所以成為可能，有賴程式語言的支援。例如，高階函數使得我

們能將與特定問題相關的部分（如 map f 與 filter p中的 f 與 p）抽象出來；惰
性求值使我們勇於使用大串列或無限串列作為中間值，不用擔心它們被不必要
地真正算出。
此外，全麥編程也需要豐富的組件函式庫。設計良好的組件捕捉了常見的

編程模式，有了它們的幫忙，我們的程式可寫得簡潔明暸 —本章之中大部分
的程式都是都是一行搞定的 “one-liner”. 但，這些組件不可能窮舉所有的編程
模式。我們仍會需要自行從頭寫些函數。受到全麥編程影響，在自行寫函數
時，我們也常會希望將它們寫得更通用些，藉此發現常見的編程模式，設計出
可重用的組件。
全麥編程能寫出多實用的程式？第 1.11節中會提及其他學者嘗試過的，包

含解密碼、解數獨在內的有趣例子。在本節，我們則想示範一個小練習：由下
至上的合併排序 (merge sort)。

合併排序 假設我們已有一個函數 merge′ :: (List Int×List Int)→ List Int, 如果
xs與 ys已經排序好，merge′ (xs,ys)將它們合併為一個排序好的串列。14 函數
merge′ 可用第 2.13節的方式歸納寫成，也可使用將在第[todo: where]章提及的
組件 unfoldr做出。我們如何用 merge′將整個串列排序好呢？
一般書中較常提及由上至下的合併排序：將輸入串列（或陣列）切成

長度大致相等的兩半，分別排序，然後合併。本節則以由下至上的方式試
試看。如果輸入串列為 [4,2,3,5,8,0,1,7 ]，我們先把每個元素都單獨變成串
列，也就是變成 [[4 ], [2 ], [3 ], [5 ], [8 ], [0 ], [1 ], [7 ]]。然後把相鄰的串列兩兩合併：
[[2,4 ], [3,5 ], [8,0 ], [1,7 ]]，再兩兩合併成為 [[2,3,4,5 ], [0,1,8,7 ]]，直到只剩下
一個大串列為止。
如果我們定義兩個輔助函數：wrap將一個元素包成一個串列，isSingle判

斷一個串列是否只剩下一個元素，

wrap :: a→ List a
wrap x = [x ] ,

isSingle ::List a→ Bool

isSingle [x ] = True

isSingle xs = False .

那麼上述的合併排序可以寫成：

msort = head ·until isSingle mergeAdj ·map wrap .

這幾乎只是把口語描述逐句翻譯：先把每個元素都包成串列，反覆做 mergeAdj
直到只剩下一個大串列，然後將那個大串列取出來。

14之所以取名為merge′，因為在第 2.12節中我們將使用一個類似且相關的函數merge ::List Int→
List Int→ List Int.

下一項工作是定義 mergeAdj ::List (List Int)→ List (List Int),其功能是將相鄰
的串列兩兩合併。如果我們能訂出一個函數 adjs ::List a→ List (a× a),將相鄰
的元素放在序對中，mergeAdj就可以寫成：

mergeAdj = map merge′ ·adjs .

但 adjs該怎麼定義呢？對大部分讀者來說，最自然的方式也許是用第 2章
將討論的歸納法。但作為練習，我們姑且用現有的組件試試看。先弄清楚
我們對 adjs的期待。當 xs = [x0,x1,x2,x3 ],我們希望 adjs xs = [(x0,x1),(x2,x3)].
但當 xs 有奇數個元素時，例如 xs = [x0,x1,x2,x3,x4 ], 最後一個元素 x4 便落
單了。如果是為了合併排序，我們也許可以把 x4 和 [ ] 放在一起，adjs xs =
[(x0,x1),(x2,x3),(x4, [ ])]. 但為使 adjs 適用於更多的情況，也許我們應該讓
它多拿一個參數，當作落單的元素的配對。因此我們把 adjs 的型別改為
a→ List a→ List (a×a),希望 adjs z xs = [(x0,x1),(x2,x3),(x4,z)].
我們試著看看這可如何辦到。
• 首先，zip xs (tail xs) 可把 xs 的每個元素和其下一個放在序對中。例：
當 xs = [x0,x1,x2,x3,x4 ]時，zip xs (tail xs)的值是 [(x0,x1), (x1,x2), (x2,x3),

(x3,x4)].
• 如果我們為 zip的第二個參數補上一個 z,成為 zip xs (tail (xs++[z ]))，這
可歸約為 [(x0,x1),(x1,x2),(x2,x3),(x3,x4),(x4,z)]。

• 再將位置（由 0 算起）為奇數的元素丟棄，我們便得到原先希望的
[(x0,x1),(x2,x3),(x4,z)]了！

讀者可試試看當 xs有偶數個元素時的情況。總之，adjs可定義成：

adjs :: a→ List a→ List (a×a)
adjs z xs = everyother (zip xs (tail xs++[z ])) ,

其中 everyother ys把 ys中位置為奇數的元素丟棄。
最後，考慮如何把串列中位置為奇數的元素丟棄。一種做法是：一直從串

列中丟掉頭兩個元素，直到串列用完：

everyother ::List a→ List a
everyother = map head · takeWhile (not ·null) · iterate (drop 2) .

總而言之，由下至上的合併排序可寫成：

msort ::List Int→ List Int

msort = head ·until isSingle mergeAdj ·map wrap ,

其中 mergeAdj的定義是：

mergeAdj ::List (List Int)→ List (List Int)

mergeAdj = map merge′ ·adjs [ ] .

如果我們想看到合併排序完成前的每一步驟，可將 msort中（以 iterate與
dropWhile定義出）的 until改為 iterate與 takeWhile:

msortSteps ::List Int→ List (List (List Int))

msortSteps = takeWhile (not · isSingle) · iterate mergeAdj ·map wrap .

例如，msortSteps [9,2,5,3,6,4,7,0,5,1,8,2,3,1 ]可得到

[ [[9 ], [2 ], [5 ], [3 ], [6 ], [4 ], [7 ], [0 ], [5 ], [1 ], [8 ], [2 ], [3 ], [1 ]],
[[2,9 ], [3,5 ], [4,6 ], [0,7 ], [1,5 ], [2,8 ], [1,3 ]],
[[2,3,5,9 ], [0,4,6,7 ], [1,2,5,8 ], [1,3 ]],
[[0,2,3,4,5,6,7,9 ], [1,1,2,3,5,8 ]]] .

最後兩個串列合併為 [0,1,1,2,2,3,3,4,5,5,6,7,8,9 ]，即為 msort的結果。

1.10 自訂資料型別

本章目前為止給讀者看到的 data定義其實都是 Haskell已內建的型別。使用者
也可自己定義新資料型別。例如，我們可能定義一個新型別表達四個方向：

data Direction= North | East | South |West ,

或著定義一個表示顏色的型別，用三個浮點數表達紅、綠、藍的比例：

data RGBColor = RGB Float Float Float ,

例：土耳其藍 (turquoise)可寫成 RGB 0.25 0.875 0.8125. 下列函數則降低一
個顏色的彩度：15

desaturate ::Float→ RGBColor→ RGBColor

desaturate p (RGB r g b) =
RGB (r+p× (gr− r)) (g+p× (gr−g)) (b+p× (gr−b)) ,

where gr = r×0.299+g×0.587+b×0.144 .

我們也可定義如 List一樣的遞迴資料型別。例如，資料結構中可能談到兩
種二元樹狀結構，一種僅在內部節點有標示（稱作 internally labelled），另一種
僅在葉節點有表示（稱作 externally labelled）。這兩種二元樹可分別表示如下：

data ITree a = Null | Node a (ITree a) (ITree a) ,

data ETree a = Tip a | Bin (ETree a) (ETree a) .

怎麼編寫這種資料結構上的程式呢？我們將在下一章中說到。

1.11 參考資料

本章中的許多想法取自 Bird [1998]，該書是我相當推薦的 Haskell教材。
15這是一個簡便的做法：算出該顏色的灰度 gr,然後計算每個原色與該灰度的線性內插。更準

確的作法應將 RGB轉成 HSV，以後者調整飽和度。

Haskell為何叫 Haskell?

1980年代中期，程式語言學者們已各自開發出了許多個語法、語意類似但卻稍
有不同、大都只在出生機構被使用的惰性純函數語言。沒有一個語言取得壓倒性
的優勢。為溝通方便、以及為了讓整個領域能走向下一步，大家有了該設計個統
合、共通的惰性純函數語言的共識。1988年一月，新語言設計小組在耶魯大學開
會，眾多討論項目中包括幫語言取個名字。以下軼事節錄自 Hudak et al. [2007] .
當天被提出的選項包括 Haskell, Vivaldi, Mozart, CFL (Common Functional Lan-

guage), Curry, Frege, Peano, Nice, Fun, Light… 等等。最後經程序選出的名字是
“Curry”, 紀念邏輯學家 Haskell B. Curry — 他在組件邏輯 (combinatorial logic)、
Curry-Howard同構等領域的研究一直深遠影響函數語言學界。
但當天晚上就有人覺得這名字會招惹太多雙關語笑話。除了咖哩之外，小組成

員覺得實在不行的是：TIM (three instruction machine)是函數語言用的一種抽象機
器，但 Tim Curry則成了電影洛基恐怖秀（Rocky Horror Picture Show, 1975）的男
主角。
於是新語言的名字就改成 Haskell了。
小組成員 Paul Hudak和 David Wise寫信給 Curry的遺孀 Virginia Curry，徵求

她的同意。Hudak後來親自登門拜訪，Virginia Curry和他聊了之前的訪客（包含
Church與 Kleene）的故事；後來她也去聽了 Hudak關於 Haskell（語言）的演講，
表現得十分友善。臨別前，她說：「其實呀，Haskell一直都不喜歡他的名字。」

Currying Moses [Schönfinkel, 1924] 提出多參數函數可用單參數函數表達。
Haskell Curry在許多著作中（例：Curry [1980]）使用 currying，但當時並沒有
currying一詞。為何此概念最後會以 Curry命名呢？David A. Turner（Haskell
語言的前身之一 Miranda 的設計人）在一次網路討論 [Sankar et al., 1997] 中
表示 currying一詞由 Christopher Strachey取名，於 1967年前後使用在其上課
資料中。這種說法目前廣被大家接受，但我目前尚未找可佐證的上課資料。
相反地，Strachey [1967]之中明確表示他認為 currying的概念是由 Schönfinkel
發明的，並稱之為「Schönfinkel 的裝置」。16 但 currying 的想法可追溯得
比 Schönfinkel 或 Curry 都早。F. L. Gottlob Frege 1891 年的 Über Funktion und
Begriff (英譯 Function and Concept) [Frege, 1960]結尾幾頁的概念即是 currying.

全麥編程 「全麥編程」一詞由牛津大學 Geraint Jones取名，由來可能是模仿
健康食物的說詞。如 Bird [2010,第 19章]便寫道，「全麥編程好處多多，可預
防『索引症』(indexitis)，鼓勵合法的程式建構。」在該章之中，Richard Bird以
大量使用串列組件函數的全麥編程為起點，推導出能相當迅速地解數獨的程
式。Hinze [2009]以全麥編程為工具，示範了河內塔問題 (Tower of Hanoi)的許
多性質，以及其與謝爾賓斯基 (Sierpiński)三角形的關係。其中寫道「函數語言
擅長全麥編程。這個詞彙由 Geraint Jones命名。全麥編程意謂由大處去想：處
理整個串列，而不是一連串的元素；發展出整個解答的空間，而不是個別的解
答；想像整個圖，而不是單一的路徑。對於已知的問題，全麥編程常給我們新
洞察與新觀點。」Hutton [2016,第五章]則以編、解密碼為例。凱撒加密 (Caesar
cipher)為一種簡單的加密方式：將明文中的每個字母都往前或後偏移固定的

16原文：“There is a device originated by Schönfinkel, for reducing operators with several operands
to the successive application of single operand operators.”

量，例如當偏移量為 2時，'a'變成 'c'，'b'變成 'd' … 一種解凱撒密碼的
有效方式是計算密文中每個字母的分佈，和一般英文文章中的平均字母分佈做
比較，藉以猜出偏移量。Graham Hutton在書中示範如何用組件函數、完全不
用遞迴地寫出編碼與解碼程式。以上都是相當值得一看的例子。

LISP的串列 誕生於 1958年的 LISP是目前仍被使用的高階程式語言中歷史第
二悠久的 —最早的是 FORTRAN.但 LISP與 FORTRAN是風格截然不同的語言。
雖然具有含副作用的指令，LISP仍被認為是函數語言的先驅。

LISP為「串列處理 (list processing)」的縮寫。但事實上，LISP中的聚合資料
結構「S算式 (S-expression)」不只可用來表達串列。CONS函數做出的是一個
序對，其中第一個元素被稱作 CAR (contents of the address part of register),第
二個稱作 CDR (contents of the decrement part of register). 如果 CDR的部分仍是
一個 CONS做出的序對，或是特殊值 NIL,整個結構表達的就是一個串列。S算
式也可用來做出二元樹、語法樹… 等等。Haskell串列的建構元 [ ]可唸成 ‘nil”,
(:)唸成 ‘cons”,這兩個詞彙都從 LISP而來。
在前幾波人工智慧熱潮時，大家認為符號與邏輯的處理是人工智慧的基

礎。但早期的程式語言大多針對數值運算而設計，會處理串列的 LISP便被視
為最適合做符號處理的語言 —「人工智慧用的語言」。另一個被視為「人工智
慧專用語言」的是奠基於述語邏輯與歸結 (resolution)的 PROLOG.今日的人工
智慧技術以神經網路為基礎，「人工智慧專用語言」的頭銜則給了 Python。

歸納定義與證明

「全麥編程」的觀念鼓勵我們以小組件組織出大程式。但這些個別組件該如何
實作呢？或著，沒有合用的組件時該怎麼辦？我們可以回到更基礎的層次，用
遞迴 (recursion)定義它們。「遞迴」意指一個值的定義又用到它本身，是數學中
常見的定義方式。在早期的編程教材中，遞迴常被視為艱澀、難懂、進階的主
題。但在函數程設中，遞迴是唯一可使程式不定次數地重複一項計算的方法。
一旦跨過了門檻，遞迴其實是個很清晰、簡潔地描述事情的方式。
對於遞迴，許多初學者一方面不習慣、覺得如此構思程式很違反「直覺」，

另一方面也納悶：以自己定義自己，到底是什麼意思？這兩個難處其實都談到
了好問題。對於前者，我們希望發展一些引導我們構思遞迴程式的思路；希望
有了這些依據，能使寫遞迴程式變得直覺而自然。關於後者，其實並非所有遞
迴定義都有「意思」—有些「不好」的遞迴並沒有定義出東西。我們討論遞迴
的意義時必須限定在「好」的、有意義的程式上。最好有些方式確保我們寫出
的遞迴定義是好的。
在本章我們將討論一種型式較單純的遞迴：歸納 (induction)。對上述兩個

問題，本章的回應是：先有歸納定義出的資料結構，再依附著該資料結構撰寫
歸納定義的程式，是一種思考、解決程式問題的好方法，也是一種理解遞迴程
式的方式。此外，依循這種方法也能確保該定義是「好」的。我們將從數學歸
納法出發，發現歸納程式與數學歸納法的相似性 —寫程式和證明其實是很相
似的活動。
本書以 Haskell為學習工具，但在之後的幾章，我們僅使用 Haskell的一小

部分。Haskell支援無限大的資料結構，也允許我們寫出不會終止的程式。但
我們將假設所有資料結構都是有限的（除非特別指明），所有函數皆會好好終
止（也就是說函數都是「全函數 (total function)」—對每一個值，都會好好地算
出一個結果，不會永遠算下去，也不會丟回一個錯誤）。這麼做的理由將在本
章解釋。

53

2.1 數學歸納法

在討論怎麼寫程式之前，我們得先複習一下數學歸納法 —晚點我們就會明白
理由。回顧：自然數在此指的是 0,1,2...等所有非負整數。1 自然數有無限多
個，但每個自然數都是有限大的。自然數的型別記為 N. 若 a是一個型別, a之
上的述語 (predicate)可想成型別為 a→ Bool的函數，常用來表示某性質對某特
定的 a是否成立。自然數上的述語便是 N→ Bool。數學歸納法可用來證明某性
質對所有自然數都成立：

給定述語 P ::N→ Bool. 若

1. P對 0成立，並且
2. 若 P對 n成立，P對 1+n亦成立，

我們可得知 P對所有自然數皆成立。

為何上述的論證是對的？我們在 2.9節將提供一個解釋。但此處我們可以提
供一個和程式設計較接近的理解方式。自然數可被想成如下的一個資料結構：

data N= 0 | 1+ N .

這行定義有幾種讀解法，目前我們考慮其中一種 —該定義告訴我們：
1. 0的型別是 N;
2. 如果 n的型別是 N, 1+ n的型別也是 N;
3. 此外，沒有其他型別是 N的東西。

這種定義方式稱作歸納定義 (inductive definition)。其中「沒有其他型別是 N的
東西」一句話很重要 —這意味著任一個自然數只可能是 0，或是另一個自然數
加一，沒有別的可能。任一個自然數都是這麼做出來的：由 0開始，套上有限
個 1+。反過來說，給任意一個自然數，我們將包覆其外的 1+ 一層層拆掉，在
有限時間內一定會碰到 0. 有人主張將 inductive definition翻譯為迭構定義，著
重在從基底（此處為 0）開始，一層層堆積上去（此處為套上 1+）的概念。
本書中，我們把自然數的 0寫成粗體，表明它是資料建構元；把 1+的加號

寫得小些並和 1放得很近，以強調「加一」是資料建構元、是一個不可分割的
動作（和我們之後將介紹的一般自然數加法 (+)不同）。數字 2其實是 1+ (1+ 0)
的簡寫, 3其實是 1+ (1+ (1+ 0))的簡寫。
歸納定義出的資料型別允許我們做歸納證明。由於 P 是 N 到 Bool 的函

數，「P對 0成立」可記為 P 0，「若 P對 n成立，P對 1+ n亦成立」可記為
P (1+ n)⇐ P n。2 我們假設兩者都已被證明，用它們證明 P 3：

P (1+ (1+ (1+ 0)))
⇐ {因 P (1+ n)⇐ P n }
P (1+ (1+ 0))
⇐ {因 P (1+ n)⇐ P n }
P (1+ 0)

1有些數學派別的「自然數」是從 1算起的。計算科學中則通常習慣以 0起算。
2P⇐ Q意思是「若 Q則 P」。依此順序寫，有「為證明 P，我們想辦法讓 Q成立」的感覺。

許多人不習慣由右到左的箭頭，但不論數學上或日常生活中，這都是常使用的論證思考方向。

⇐ {因 P (1+ n)⇐ P n }
P 0 .

第一步中，我們希望 P (1+ (1+ (1+ 0))) 成立，根據 P (1+ n) ⇐ P n, 只要
P (1+ (1+ 0))即可。第二步中，我們希望P (1+ (1+ 0))成立，同樣根據P (1+ n)⇐
P n，只要 P (1+ 0)成立即可… 最後，只要 P 0成立，P (1+ 0)即成立，但 P 0
是已知的。因此我們已論證出 P 3成立！
由上述推演中，我們發現：數學歸納法的兩個前提 P 0與 P (1+ n)⇐ P n給

了我們一個對任一個自然數 m,生成一個 P m之證明的方法。這是由於自然數
本就是一個歸納定義出的資料型別：任一個自然數 m都是有限個 1+ 套在 0之
上的結果，因此，只要反覆用 P (1+ n)⇐ P n拆，總有碰到 P 0的一天。既然
對任何 m,都做得出一個 P m的證明，我們就可安心相信 P對任何自然數都成
立了。
為了之後討論方便，我們將前述的數學歸納法寫得更形式化些：

自然數上之歸納法： (∀n ·P n) ⇐ P 0 ∧ (∀n ·P (1+ n)⇐ P n) .

這只是把之前的文字描述改寫成二階邏輯，但可清楚看出：給定 P,我們希望
證明它對所有自然數都成立，只需要提供 P 0和 P (1+ n)⇐ P n兩個證明。其
中 P 0是確定 P對 0成立的基底 (base case)，P (1+ n)⇐ P n則被稱作歸納步驟
(inductive step)：在假設 P n成立的前提下，想辦法「多做一步」，論證 P (1+ n)
也成立。餘下的就可交給數學歸納法這個架構了。

2.2 自然數上之歸納定義

數學歸納法和編程有什麼關係呢？考慮一個例子：給定 b,n ::N,我們希望寫個
函數 exp計算乘冪，使得 exp b n = bn. 我們先把型別寫下：

exp ::N→ N→ N
exp b n = ?

問號部分該怎麼寫？沒有其他線索很難進行，因此我們回想：n是自然數，而
任何自然數只可能是 0或 1+做出的。我們便分成這兩個狀況考慮吧：

exp ::N→ N→ N
exp b 0 = ?
exp b (1+ n) = ?

其中，exp b 0較簡單：顯然應該是 b0 = 1. 至於 exp b (1+ n)的右手邊該怎麼寫？
似乎很難一步定出來。但假設 exp b n已經順利算出了 bn, 由於 b1+n = b× bn,
exp b (1+ n)與之的關係可寫成：

exp b (1+ n) = b× exp b n .

如此一來我們便完成了一個計算乘冪的程式：

Haskell v.s Math

很不幸地，Haskell並不接受 2.2節中 exp的定義。
首先，Haskell並沒有獨立的自然數型別。我們可自己定（並將其宣告為 Num

類別的一員），或著直接使用 Haskell內建的 Int型別。其次，Haskell原本允許我
們在定義的左手邊寫 exp b (n+1)，但這套稱作 “n+ k pattern”的語法已在 Haskell
2010中被移除。目前我們得將 exp寫成：

exp :: Int→ Int→ Int
exp b 0 = 1
exp b n = b× exp b (n−1) .

n+ k pattern曾引起激烈討論。支持者主要著眼於它在教學上的方便：這方便我
們討論數學歸納法、做證明、並讓我們更明顯地看出自然數與串列的相似性。反
對者則批評它與 type class的衝突。後來由反方勝出。
有些 Haskell教科書堅持書中出現的程式碼須是能在一個字一個字地鍵入電腦後

即可執行的。本書的定位並非 Haskell教材，而是函數編程概念的入門書。為此目
的，我們希望選擇適合清晰表達概念、易於操作、演算、證明的符號。而一個實
用目的的語言得在許多設計上妥協尋求平衡，基於種種考量，往往得犧牲符號的
簡潔與便利性（這點我們完全能理解）。因此本書中的程式語法偶爾會和 Haskell
語法有所不同。我們會盡量指明這些不同處，使讀者知道如何將本書中的程式轉
換成現下的 Haskell語法。

exp ::N→ N→ N
exp b 0 = 1
exp b (1+ n) = b× exp b n .

回顧一下剛剛的思路：我們難以一步登天地對任何 n寫出 exp b n,但我們提
供 exp b 0 該有的值，並在假設 exp b n 已算出該有的值的前提下，試著做一
點加工、多算那一步，想法做出 exp b (1+ n) 該有的值。這和前述的數學歸
納法是一樣的！寫歸納程式和做歸納證明是很類似的行為。使用數學歸納法
證明 P需要提供一個基底 P 0和歸納步驟 P (1+ n)⇐ P n. 歸納定義程式也一
樣。在 exp b n 的定義中，基底是 exp b 0，歸納步驟則是由 exp b n 想法變出
exp b (1+ n). 有這兩個元件，我們便有了一個對任何自然數 n,保證算出 exp b n
的方法。作為例子，我們看看 exp 2 3是怎麼被算出來的：

exp 2 (1+ (1+ (1+ 0)))
= { exp之歸納步驟 }

2× exp (1+ (1+ 0))
= { exp之歸納步驟 }

2×2× exp (1+ 0)
= { exp之歸納步驟 }

2×2×2× exp 0
= { exp之基底 }

2×2×2×1 .

第一步中，要算出 exp 2 (1+ (1+ (1+ 0))),我們得先算出 exp (1+ (1+ 0)). 要算出
後者，在第二步中我們得先算出 exp (1+ 0)… 直到我們碰到 exp b 0.

自然數上的歸納定義 我們將 b固定，稍微抽象一點地看 exp b ::N→ N這個
函數。該定義符合這樣的模式：

f ::N→ a
f 0 = e
f (1+ n) = ...f n ... .

這類函數的輸入是 N，其定義中 f (1+ n)的狀況以 f n定出，此外沒有其他對 f
的呼叫。若一個函數符合這樣的模式，我們說它是在自然數上歸納定義出的，
其中 f 0那條稱作其基底，f (1+ n)那條稱作其歸納步驟。我們日後將看到的許
多程式都符合這個模式。
數學上，若一個函數能為其值域內的每個值都找到一個輸出，我們說它是

個全函數 (total function)，否則是部分函數 (partial function). 計算上，當我們
說 f 是一個全函數，意謂只要 x型別正確並可算出值，f x便能終止並算出一個
值，不會永久跑下去，也不會丟出錯誤。
如前所述的、在自然數上歸納定義的 f 會是全函數嗎？首先，任何自然數

都可拆成 0或是 1+ n，而這兩個情況已被 f 的兩行定義涵括，不會出現漏失的
錯誤。其次，f 每次呼叫自己，其參數都少了一層 1+. 長此以往，不論輸入多
大，總有一天會遇到基底 f 0 —因為任何自然數都是從 0開始，套上有限個 1+.
只要基底狀況的 e以及在歸納步驟中 f n前後的計算都正常終止，對任何輸入，
f 都會正常終止。因此 f 是個全函數。
「程式會終止」是很重要的性質，我們之後會常談到。在本書目前為止示範
的編程方法中，「一個函數若呼叫自己，只能給它更小的參數」是個單純但重
要的規範（例如 f (1+ n)的右手邊可以有 f n,不能有 f (1+ n)或 f (1+ (1+ n))）。
操作上這確保程式會終止，而在第 [todo: where?] 章之中，我們將提到這也確
保該遞迴定義是「好」的、有意義的。
順便一提：在 f (1+ n)的右手邊中，f n可以出現不只一次 —因為 ...f n ...

f n...可看成 (λx→ ...x ... x...) (f n). 在 f n的前後 ...的部分可以出現 n —將在第
60頁中介紹的階層函數就是一個這樣的例子。有些情況下我們不允許 n出現在
...中，此時會額外說明。

乘法、加法 我們多看一些歸納定義的例子。在 exp中我們用到乘法，但假若
我們的程式語言中只有加法、沒有乘法呢？我們可自己定定看：

(×) ::N→ N→ N
m×n = ?

若不用組件，我們目前會的寫程式方法只有歸納法，也只有這招可試試看了。
但，(×)有兩個參數，我們該把 (m×) ::N→ N視為一個函數，分別考慮 n是 0
或 1+...的情況，還是把 (×n) ::N→ N視為一個函數，考慮 m是 0或 1+...的
情況？答案是兩者皆可，並無根本性的差異。只是現在我們做的選擇會影響到
之後與 (×)相關的證明怎麼寫 (見第 2.3節)。本書中的習慣是拆左手邊的參數，
因此我們考慮以下兩種情況。

(×) ::N→ N→ N
0 ×n = ?
(1+ m)×n = ...m×n ....

基底狀況中，0×n的合理結果應是 0. 歸納步驟中，我們得想法算出 (1+ m)×n,
但我們可假設 m×n已經算出了。稍作思考後，讀者應可同意以下的做法：

(×) ::N→ N→ N
0 ×n = 0
(1+ m)×n = n+(m×n) ,

如果已有 m×n，多做一個 (n+),就可得到 (1+ m)×n了。
如果我們的程式語言中連加法都沒有呢？加法可看成連續地做 1+:

(+) ::N→ N→ N
0 +n = n
(1+ m)+n = 1+ (m+n) .

此處 (+) 是我們定義的、可將任意兩個自然數相加的加法，而 1+ 只做「加
一」，是基本的資料建構元。為求一致，我們同樣在左邊的參數上做歸納。基
底狀況中，0+n只應是 n. 想計算 (1+ m)+ n,先假設 m+n已經算出，再多套
一個 1+. 不難看出 m+n是把 n當做基底，在外面套上 m個 1+的結果。3

2.3 自然數上之歸納證明

上一節中我們定出了函數 exp。如果定義正確，exp b n算的應是 bn. 例如，我
們知道 bm+n = bm×bn. 我們定出的函數 exp是否真有此性質呢？

定理 2.1. 對任何 b,m,n ::N, exp b (m+n) = exp b m× exp b n.

我們試著證明定理 2.1。數學歸納法是我們目前唯一的工具，而要使用它，第
一個問題是：該用 b, m,或 n的哪一個來做歸納呢（意即把哪一個拆解）？
觀察定理 2.1中待證明式的等號兩邊，並參照 exp, (+),與 (×)的定義。等號

左手邊的 exp b (m+n)之中，化簡 exp b前得知道 m+n究竟是 0還是 1+ k。而
根據 (+)的定義，化簡 m+n前需知道 m究竟是 0還是 1+ k. 再看右手邊，根
據 (×)的定義，要化簡 exp b m×exp b n得先化簡 exp b m,而後者也得知道 m是
什麼。對兩邊的分析都指向：我們應針對 m做歸納！
策略擬定後，我們便試試看吧！

證明定理 thm:exp-plus-times. 欲證明 exp b (m+ n) = exp b m× exp b n,我們在 m
之上做歸納。m要不就是 0,要不就是 1+ k.
情況 m :=0. 此時需證明 exp b (0+n) = exp b 0× exp b n. 推論如下：

exp b (0+n)
= { (+)之定義 }

3「這樣做不是很慢嗎？」是的。本章的自然數表示法，以及其引申出的運算元都不應看作有效
率的實作，而是理論工具。了解加法與乘法可這樣看待後，許多其相關性質都可依此推導出來。

exp b n
= {因 1× k = k }

1× exp b n
= { exp之定義 }

exp b 0× exp b n .

情況 m :=1+ m. 此時需證明 exp b ((1+ m)+n) = exp b (1+ m)×exp b n,但可假設
exp b (m+n) = exp b m× exp b n已成立。推論如下：

exp b ((1+ m)+n)
= { (+)之定義 }

exp b (1+ (m+n))
= { exp之定義 }

b× exp b (m+n)
= {歸納假設 }

b× (exp b m× exp b n)
= { (×)之結合律 }
(b× exp b m)× exp b n

= { exp之定義 }
exp b (1+ m)× exp b n .

對這個證明，讀者是否有所懷疑？最大的疑問可能在「假設 exp b (m+n) =
exp b m× exp b n成立」這句上。這不就是我們要證明的性質嗎？在證明中假設
它成立，似乎是用該性質自己在證明自己。這是可以的嗎？
為清楚說明，我們回顧一下第 2.1節中的數學歸納法（並把區域識別字改為

k以避免混淆）：

自然數上之歸納法： (∀k ·P k) ⇐ P 0 ∧ (∀k ·P (1+ k)⇐ P k) .

證明 2.1欲證的是 exp b (m+n) = exp b m×exp b n，並在 m上做歸納。更精確地
說，就是選用了下述的 P:4

P m≡ (exp b (m+n) = exp b m× exp b n) ,

在證明中改變的是 m，而 b與 n是固定的。數學歸納法可證明 (∀m ·P m),展開
後正是 (∀m · exp b (m+n) = exp b m× exp b n). 而根據數學歸納法，我們需提供
P 0與 (∀m ·P (1+ m)⇐ P m)的證明。
證明 2.1中「情況 m :=0」的部分，就是 P 0的證明。而「情況 m :=1+ m」則

是 (∀m ·P (1+ m)⇐ P m)的證明。「假設 exp b (m+n) = exp b m× exp b n成立」
指的是假設 P m成立，我們在此前提之下試圖證明 P (1+ m). 因此，證明 2.1並
沒有「用該性質自己證明自己」。我們是以一個比較小的結果（P m）證明稍大

4在程式推導圈子中，(≡)常用來代表「只用在真假值上、且滿足結合律的等號」。本書中使用
(≡)以和 (=)做區分。

一點的結果（P (1+ m)）。就如同我們寫歸納程式時，假設 f n已經算出，試著
用它定出 f (1+ n).
在證明之中，如 P m 這種在歸納步驟假設成立的性質被稱作歸納假設

(induction hypothesis)。

程式與證明 證明 2.1還有一些能啟發我們之處。方才，我們看到 exp b (m+

n) = exp b m× exp b n，決定以數學歸納法證明，但接下來怎麼著手？怎麼選定
在哪個變數上做歸納？
答案是：分析該式中程式的行為。程式怎麼拆其參數，我們在證明中就怎

麼拆。我們試圖證明某些程式的性質，但程式本身便提供了證明可怎麼進行的
提示。「使證明的結構符合程式的結構」是許多證明的秘訣。並非所有證明都
可以如此完成，但本原則在許多情況下適用。
再看看 exp, (+), (×)等函數的定義。為何他們都分出了兩個情況：0與 1+ n,

並且 1+ n的情況使用到該函數對於 n的值？因為自然數的資料型別是這麼定
的！自然數只可能是 0或 1+ n，而後者是由 n做出來的。因此程式也如此寫。
程式的結構依循與其處理的資料型別之結構。
資料、程式、與證明原來有著這樣的關係：證明的結構依循著程式的結構，

而程式的結構又依循著資料型別的結構。歸納定義出了一個型別後，自然知道
怎麼在上面寫歸納程式；歸納程式有了，自然知道如何做關於這些程式的歸納
證明。一切由定義資料結構開始。掌握這個原則，大部分的證明就不是難事。

讓符號為你工作 再考慮證明 2.1中的狀況 m := 1+ m. 假想由你做這個證明，
由第一行 exp b ((1+ m)+n)開始。接下來該怎麼進行？
既然我們已經打定主意用數學歸納法，在證明的某處必定會使用 exp b (m+

n) = exp b m× exp b n這個歸納假設。因此，證明前幾行的目的便是想辦法
將 exp b ((1+ m) + n) 中的 1+ 往外提，將 exp b 往內側推，使得式子中出現
exp b (m+ n)。一旦成功，就可運用歸納假設，將其改寫成 exp b m× exp b n！
接下的就是機械化地收尾、將式子整理成 exp b (1+ m)× exp b n了。
這呼應到第 0.2節所說的「讓符號為你工作」。光從語意上想，我們不易理解

為何 exp b ((1+ m)+n)能夠等於 exp b (1+ m)× exp b n. 但符號給我們線索。我
們可觀察式子的結構，暫時不去想語意；我們的目標是操作、移動這些符號，
將它們轉換成可使用歸納假設的形式。因此，接下來的演算推導便有所依據而
非盲目進行：已知目標是把某符號往外提或往內推，我們就可尋找、使用可達
到此目的的規則。這些規則包括已知函數的定義、或諸如分配律、遞移律、結
合律等等數學性質。若很明顯地缺了一個想要的性質，也許可以把它當作引理
另外證證看。符號幫助我們，使我們的思考清晰而有方向。

習題 2.1 — 證明 1× k = k. 這個證明並不需要歸納。
習題 2.2 — 證明 (+)之結合律: m+(n+ k) = (m+n)+ k. 此證明中你使用
的述語是什麼？
習題 2.3 — 證明 k+1 = k. 你需要使用歸納法嗎？用什麼述語？

最後，說到自然數上的歸納定義，似乎不得不提階層 (factorial)。用非正式的
寫法，fact n = n× (n−1)× (n−2)× ...×1. 形式化的定義如下：

fact ::N→ N
fact 0 = 1
fact (1+ n) = (1+ n)× fact n .

我們在定理 2.5中將會談到階層與排列的關係。

2.4 串列與其歸納定義

如同第 1.8所述，「元素型別為 a的串列」可定義成如下的資料型別：5

data List a = [ ] | a :List a .

這個定義可以理解為
1. [ ]是一個串列，
2. 若 xs是一個元素型別為 a的串列，x型別為 a,則 x : xs也是一個元素型別
為 a的串列，

3. 此外沒有其他元素型別為 a的串列。
我們不難發現 List a和 N是相當類似的資料結構：[ ]相當於 0, (:)則類似

1+,只是此處我們不只「加一」，添加的那個東西多了一些資訊，是一個型別為
a的元素。或著我們可反過來說，串列「只是」在每個 1+上都添了一些資訊的
自然數！既然自然數與串列有類似的結構，不難想像許多自然上的函數、自然
數的性質，都有串列上的類似版本，

2.4.1 串列上之歸納定義

和自然數類似，許多串列上的函數可歸納地定義出來。由於串列只可能由 [ ]或
(:)做出，定義串列上的函數時也分別處理這兩個情況。基底情況為 [ ],而欲定
義 f (x : xs)的值時，可假設 f xs已算出來了：

f ::List a→ b
f [ ] = e
f (x : xs) = ...f xs ...

來看些例子吧！「算一個陣列的和」可能是許多人學到陣列後得寫的頭幾個練
習程式。串列版的和可以這麼寫：

sum ::List Int→ Int

sum [ ] = 0
sum (x : xs) = x+ sum xs .

基底狀況中，空串列的和應是 0。歸納步驟中，我們要算 x : xs的和，可假設我
們已算出 xs的和，再加上 x即可。計算串列長度的 length有很類似的定義：

length ::List a→ N
length [ ] = 0
length (x : xs) = 1+ (length xs) .

5Haskell中「元素型別為 a的串列」寫成 [a ]. 由於這樣的符號在教學中遇到許多困難，本書中
寫成 List a.

在歸納步驟中，我們想計算 x : xs的長度，只需假設我們已知 xs的長度，然後
加一。事實上，length剛好體現了前述「List a只是在每個 1+上添了資訊的自
然數」一事：length把串列走過一遍，將 [ ]代換成 0，並將每個 ( :)中附加的
資訊拋棄，代換成 1+。
函數 map f ::List a→ List b，也就是 map給定函數 f 的結果，也可在串列上

歸納定義：

map :: (a→ b)→ List a→ List b
map f [ ] = [ ]

map f (x : xs) = f x : map f xs .

基底狀況的合理結果是 [ ]. 歸納步驟中，要對 x : xs中的每個元素都做 f ,我們可
假設已經知道如何對 xs中的每個元素都做 f ,把其結果接上 f x即可。
函數 (++)把兩個串列接起來。如果我們在其左邊的參數上做歸納定義，可

得到：6

(++) ::List a→ List a→ List a
[ ] ++ ys = ys
(x : xs)++ ys = x : (xs++ ys) .

空串列接上 ys仍是 ys. 歸納步驟中，要把 x : xs接上 ys,我們可假設已有辦法把
xs接上 ys,然後只需添上 x即可。
請讀者比較一下 (++)與自然數加法 (+)的定義，會發現兩者的結構一模一

樣！如果串列是在每個 1+中加上資料的自然數，(++)就是串列上的加法了。
若要形式化地把 List a, N, (++),與 (+)牽上關係，連接他們的橋樑就是 length
— xs++ ys的長度，應是 xs與 ys的長度之和！意即：

length (xs++ ys) = length xs+ length ys . (2.1)

習題 2.10中將證明此性質。
最後，(++)是反覆使用 (:),函數 concat則是反覆使用 (++):

concat ::List (List a)→ List a
concat [ ] = [ ]

concat (xs : xss) = xs++ concat xss .

2.4.2 串列上之歸納證明

如果 List a是一個歸納定義出的資料結構，我們應可以在 List a之上做歸納證
明。確實，串列上的歸納法可寫成：

串列上之歸納法： (∀xs ·P xs) ⇐ P [ ] ∧ (∀x xs ·P (x : xs)⇐ P xs) .

以文字敘述的話：給定一個述語 P ::List a→ Bool,若要證明 P xs對所有 xs都
成立，只需證明 P [ ]和「對所有 x和 xs,若 P xs則 P (x : xs)」。

6依照 Haskell的運算元優先順序，x : (xs++ ys)其實可寫成 x : xs++ ys,一般也常如此寫。此處為
了清楚而加上括號。

下述的 map融合定理 (map-fusion theorem)是關於 map極常用的定理之一。
所謂「融合」在此處是把兩個 map融合為一。我們日後會見到更多的融合定
理。

定理 2.2 (map融合定理). 對任何 f 與 g, map f ·map g = map (f ·g).

作為一個例子，我們試著證明定理 2.2。我們目前只會用歸納證明，但是
map f ·map g = map (f · g)的左右邊都是函數，沒有出現串列也沒有出現自然
數。該拿什麼東西來歸納呢？
回顧外延相等（定義 1.8）：當 h, k均是函數，h = k的意思是對任何參數 x,

h x = k x. 因此，將待證式左右邊各補上參數，並將 (·)展開，可得知其意義為
對任何 xs,

map f (map g xs) = map (f ·g) xs .

我們便可以在 xs上做歸納了！

Proof. 當 xs :=[ ]，等式兩邊皆歸約為 [ ]. 考慮 xs := x : xs的情況：

map f (map g (x : xs))
= { map之定義 }

map f (g x : map g xs)
= { map之定義 }

f (g x) : map f (map g xs)
= { (·)之定義 }
(f ·g) x : map f (map g xs)

= {歸納假設 }
(f ·g) x : map (f ·g) xs

= { map之定義 }
map (f ·g) (x : xs) .

習題 2.4 — 證明對所有 xs, xs++[ ] = xs. 比較本題與習題 2.3的證明。
習題 2.5 — 定義函數 reverse :: List a→ List a, 將輸入的串列反轉。例如
reverse [1,2,3,4,5 ] = [5,4,3,2,1 ].

習題 2.6 — 證明對所有 f , length ·map f = length.

習題 2.7 — 證明對所有 x, sum ·map (x×) = (x×) · sum.

習題 2.8 — 證明對所有 xs, sum (map (1+) xs) = length xs+ sum xs.

習題 2.9 — 證明對所有 xs與 y, sum (map (const y) xs) = y× length xs.

討論自然數時，習題 2.2曾請讀者證明加法都滿足結合律。此處示範證明類似
定理的串列版：

定理 2.3. (++) 滿足結合律。意即，對任何 xs, ys, 和 zs, (xs++ ys)++ zs = xs++
(ys++ zs).

等式證明的步驟該多詳細？

本書中目前為止的等式證明相當細：每一個定義展開都成為獨立的步驟。這是
為了教學目的，實務上不一定得如此。以我而言，自己的研究手稿中可能會將步
驟寫得極詳細，為確保每個細節正確，並讓他人（或幾年後已經忘記細節的自己）
在不需知道上下文的情況下也能機械化地檢查每個步驟。但在論文中，因篇幅有
限，及考量讀者一次能處理的資訊量有限，發表出的證明可能會省略許多步驟。
實務上，被認為簡單、不寫出也不妨礙理解的步驟或說明都可被省略。但何謂

簡單則很依靠作者的判斷與習慣。一般說來，僅展開定義的步驟用電腦便可自動
做到，通常是可精簡掉的。最好寫出的步驟則可能是決定整個證明之結構的、不
易以電腦決定而得靠人類智慧與經驗的，等等。這可能包括使用歸納假設的那步，
或使用較特別的引理時。例如，性質 2.1的歸納步驟證明可能被精簡如下：

length ((x : xs)++ ys)
= 1+ (length (xs++ ys))
= {歸納假設 }

1+ (length xs+ length ys)
= length (x : xs)+ length ys .

Proof. 上述式子中有三個變數，我們怎麼得知該在哪一個變數上做歸納呢？此
時絕對別急著把三個變數都拆開，變成多達八種狀況。觀察：如果要歸約等號
左邊的 (xs++ ys)++ zs，根據 (++)的定義，得對 xs++ ys做狀況分析；要歸約
xs++ ys，又得對 xs做狀況分析。同樣地，根據 (++)的定義，要歸約等號右邊
的 xs++(ys++ zs)得對 xs做狀況分析。不論左右邊，最關鍵的值都是 xs. 因此
我們在 xs之上做歸納。
狀況 xs :=[ ]:

([ ]++ ys)++ zs
= { (++)之定義 }

ys++ zs
= { (++)之定義 }
[ ]++(ys++ zs) .

狀況 xs := x : xs:

((x : xs)++ yz)++ zs
= { (++)之定義 }
(x : (xs++ ys))++ zs

= { (++)之定義 }
x : ((xs++ ys)++ zs)

= {歸納假設 }
x : (xs++(ys++ zs))

= { (++)之定義 }
(x : xs)++(ys++ zs) .

基底狀況的證明很簡單。至於歸納步驟，同樣地，前兩步都是為了湊
出 (xs++ ys) ++ zs, 以便使用歸納假設。既然 (++) 滿足結合律，日後我們寫
xs++ ys++ zs就可不加括號了。

習題 2.10 — 證明性質(2.1)：對所有 xs與 ys, length (xs++ ys) = length xs+
length ys.

習題 2.11 — 證明對所有 f , xs,與 ys, map f (xs++ ys) = map f xs++map f ys.

2.5 從資料、程式、到證明

本節再用一個例子談談「讓符號為我們工作」。考慮證明下述性質：

sum · concat = sum ·map sum . (2.2)

根據外延相等，這相當於證明對所有 xss, sum (concat xss)= sum (map sum xss).
當 xss :=[ ],等號兩邊都歸約成 0. 考慮 xss := xs : xss的情況：

sum (concat (xs : xss))
= { concat之定義 }

sum (xs++ concat xss)
= {因 sum (xs++ ys) = sum xs+ sum ys }

sum xs+ sum (concat xss)
= {歸納假設 }

sum xs+ sum (map sum xss)
= { sum之定義 }

sum (sum xs : map sum xss)
= { map之定義 }

sum (map sum (xs : xss)) .

讀者對這個證明最大的疑問可能是：我們怎麼知道該用 sum (xs ++ ys) =
sum xs+ sum ys 呢？為什麼當我們遇到 sum (xs++ concat xss), 我們不是把 xs
再拆成首和尾，甚至把 xss拆開？這是許多綜合考量的結果。首先，我們預期
會使用歸納假設，因此證明前幾步的目的均為把 sum推到 concat xss左側，試
圖做出 sum (concat xss). 此處我們再強調一個觀念：證明中的步驟不是瞎猜的，
而是有目的的。符號給我們提示：我們需要把 sum往右推，因此我們試圖尋找
能完成這個目標的性質。
其次，證明的結構跟隨著程式的結構。由於 concat是由 (++)定義的，每一

個關於 concat的定理，均很有可能奠基在一個關於 (++)的相對定理上。為了
證明一個描述 sum與 concat的關係的定理，我們可能會需要一個關於 sum與
(++)的性質。
有了符號給我們的這兩個線索，我們便可以運用我們對語意的了解：sum

與 (++) 應該會滿足什麼性質？我們可猜測應該是 sum (xs++ ys) = sum xs+
sum ys —根據我們對 sum與 (++)的語意上的理解，這性質成立的機會很大。
而且，這個性質允許我們把 sum推到右邊。

符號仍舊幫助了我們。我們不可能期待所有定理都只靠符號推導出，在關
鍵時刻我們仍會需要對於特定領域的、語意上的知識。但符號給了我們許多引
導，縮小我們需要猜測的範圍。
至於 sum (xs++ ys) = sum xs+ sum ys該怎麼證明呢？不要急著把 xs與 ys都

拆開。觀察等號左右邊的式子，根據 (++), (+)與 sum的定義，兩個式子的歸
約都是先對 xs做情況分析。因此我們可試著在 xs上做歸納。
證明的結構跟隨著程式的結構 —這是做證明時相當好用的指引。程式對哪

個參數做歸納，我們做證明時就在那個參數上做歸納。函數 f 用函數 g定義，
我們證明 f 的性質時就可以預期會需要一個 g的相關性質。並非所有證明都可
以如此做出，但這個模式在許多情況下都適用。
而如我們所知，程式的結構又跟隨著資料的結構：例如，串列有 [ ]和 x : xs

兩個情況，我們定義串列上的程式時也分出這兩個情況，定義 f (x : xs)時可以
使用 f xs的值。資料結構、程式、與證明於是有著這樣的連續關係：歸納定義
出一個資料結構，我們就有了一個在上面歸納定義程式的方法；有了歸納定義
的程式，我們便知道怎麼做關於它的歸納證明。一切由資料結構開始。

習題 2.12 — 證明 length · concat = sum ·map length. 我們會需要什麼關於
(++)的相關性質？
習題 2.13 — 證明對所有 f , map f · concat = concat ·map (map f ). 我們會需
要什麼關於 (++)的相關性質？

2.6 更多歸納定義與證明

為讓讀者熟悉，本節中我們多看一些自然數或串列上的歸納定義。

2.6.1 filter, takeWhile,與 dropWhile

filter 我們曾見過的函數 filter可寫成如下的歸納定義：

filter :: (a→ Bool)→ List a→ List a
filter p [ ] = [ ]

filter p (x : xs) = if p x then x : filter p xs else filter p xs .

在 filter的許多性質中，我們試著證明下述性質作為例子：

定理 2.4. filter p ·map f = map f ·filter (p · f ).

Proof. 和定理 2.2一樣，待證式的左右邊都是函數。根據外延相等，我們將左
右邊各補上參數 xs，並在 xs上做歸納：

filter p (map f xs) = map f (filter (p · f ) xs) .

情況 x :=[ ]中左右邊都可化簡成 [ ]. 我們看看 xs := x : xs的情況：

filter p (map f (x : xs))
= { map之定義 }

filter p (f x : map f xs)
= { filter之定義 }

if p (f x) then f x : filter p (map f xs) else filter p (map f xs)
= {歸納假設 }

if p (f x) then f x : map f (filter (p · f ) xs) else map f (filter (p · f ) xs)
= { map之定義 }

if p (f x) then map f (x : filter (p · f ) xs) else map f (filter (p · f ) xs)
= {因 f (if p then e1 else e2) = if p then f e1 else f e2,如後述 }

map f (if p (f x) then x : filter (p · f ) xs else filter (p · f ) xs)
= { filter之定義 }

map f (filter (p · f ) (x : xs)) .

終止與證明 上述證明的倒數第二步為將 map f 提到外面，用了一個關於 if的
性質：

f (if p then e1 else e2) = if p then f e1 else f e2 . (2.3)

這性質對嗎？若 p成立，左右手邊都化簡為 f e1,若 p不成立，左右手邊都
化簡為 f e2. 因此 (2.3)應該成立，是嗎？
答案是：如果我們假設的世界中有不終止的程式，(2.3)便不正確了。例如，

當 f 是 three x = 3，而 p是個永遠執行、不終止的算式（例：let b = not b in b)：

three (if p then e1 else e2)
?
= if p then three e1 else three e2−” .”−

上述式子的左手邊直接化簡成 3,但右手邊卻不會終止，因為 if得知道 p的值。
我們找到了 (2.3)的反例！
在允許可能不終止的程式存在的世界中，(2.3)得多些附加條件。通常的做

法是限定 f 須是個嚴格函數,意即 f 的輸入若不終止，f也不會終止。但(2.3)並
不是唯一帶著附加條件的性質 —許多常用性質都得加上類似的附加條件。所
有狀況分析也都得將不終止考慮進去，例如，自然數除了 0與 1+ n之外，還
多了第三種情況「不終止」。7推論與證明變得更複雜。有些人因此較喜歡另一
條路：藉由種種方法確保我們只寫出會終止的程式，便可假設我們確實活在所
有程式都正常終止的世界中。

保護式 v.s. 條件分支 有些人喜歡用保護式語法定義 filter：

filter p [ ] = [ ]

filter p (x : xs) | p x = x : filter p xs
| otherwise = filter p xs .

若在此定義下證明定理 2.4，依「證明的結構與程式的結構相同」的原則，順
理成章地，我們可在 xs := x : xs中再分出 p (f x)成立與不成立的兩個子狀況：

7Bird [1987]就採用這種作法。

狀況 xs :=[ ] : ...
狀況 xs := x : xs : ...
狀況 p (f x) :

filter p (map f (x : xs))
= { p (f x)成立 }

f x : filter p (map f xs)
= ... .

狀況 not (p (f x)) :
filter p (map f (x : xs))

= { (not p (f x)) }
filter p (map f xs)

= ... .

這個定義中不用 if,因此證明中也用不上 (2.3),但該證明要成立仍須假設所有程
式都正常終止 —我們少證了一個「p (f x)不終止」的情況（而確實，在此情況
下(2.3)並不成立）。喜歡用哪個方式純屬個人偏好。
前幾章提過的 takeWhile與 dropWhile兩函數型別與 filter相同。他們可寫成

如下的歸納定義：

takeWhile :: (a→ Bool)→ List a→ List a
takeWhile p [ ] = [ ]

takeWhile p (x : xs) = if p x then x : takeWhile p xs else [ ] ,

dropWhile :: (a→ Bool)→ List a→ List a
dropWhile p [ ] = [ ]

dropWhile p (x : xs) = if p x then dropWhile p xs else x : xs .

兩者都是在輸入串列上做歸納。兩者也都可用保護式語法定義。

習題 2.14 — 證明 takeWhile p xs++dropWhile p xs = xs.

習題 2.15 — 以保護式語法定義 takeWhile與 dropWhile,以此定義做做看習
題 2.14.

2.6.2 elem與不等式證明

不等式證明 給定如下的定義，elem x xs判斷 x是否出現在串列 xs中：

elem x [ ] = False

elem x (y : xs) = (x = = y) ∨ elem x xs .

目前為止，我們所練習的都是以 (=)將式子串起的等式證明。以下以
elem為例，我們嘗試證明一個「不等式」：

elem z xs ⇒ elem z (xs++ ys) .

以口語說出的話：「若 z出現在 xs中，z也出現在 xs++ ys中」。欲證明上式，
該從哪一側推到哪一側呢？一般認為從式子較長、或結構較複雜的那側開始，

化簡成較短、較簡單的那側，是較容易的。因此我們嘗試由右側推到左側：由
elem z (xs++ ys)開始，尋找使之成立的條件，並希望 elem z xs是足夠的。

Proof. 在 xs上做歸納。基底 xs :=[ ]的狀況在此省略，看 xs := x : xs的狀況：

elem z ((x : xs)++ ys)
≡ { (++)之定義 }

elem z (x : (xs++ ys))
≡ { elem之定義 }
(z = = x) ∨ elem z (xs++ ys)
⇐ {歸納假設 }
(z = = x) ∨ elem z xs
≡ { elem之定義 }

elem z (x : xs) .

讀者可注意：第 1, 2, 4步使用的邏輯關係都是 (≡)，第 3步卻是 (⇐)，因此整
個證明建立了「若 elem z (x : xs)，則 elem z ((x : xs)++ ys)」。

習題 2.16 — 證明 not (elem z (xs++ ys))⇒ not (elem z xs).

習題 2.17 — 證明 elem z xs⇒ elem z (ys++ xs).

習題 2.18 — 證明 (∀x ·p x⇒ q x)⇒ all p xs⇒ all q xs. 其中 all的定義為：

all :: (a→ Bool)→ List a→ Bool

all p [ ] = True

all p (x : xs) = p x ∧ all p xs .

習題 2.19 — 證明 all (∈ xs) (filter p xs). 其中 x ∈ xs是 elem x xs的中序寫法。
我們可能需要習題 2.17和 2.18的結果，以及下述性質：

if p then x else x = x . (2.4)

2.6.3 串列區段

前段與後段 本章目前為止討論的歸納定義都依循著這樣的模式：欲定義
f ::List a→ b,只要為 f [ ]與 f (x : xs)找到定義。在定義後者時，只需定義出由
f xs做出 f (x : xs)的關鍵一步。目前為止，這關鍵一步都是加一、加上一個元
素等簡單的動作。現在我們來看些更複雜的例子。
例 1.19中曾提及：如果一個串列 xs可分解為 ys++ zs,我們說 ys是 xs的一個

前段 (prefix), zs是 xs的一個後段 (suffix). 例如，串列 [1,2,3 ]的前段包括 [ ], [1 ],
[1,2 ],與 [1,2,3 ]（注意：[ ]是一個前段，串列 [1,2,3 ]本身也是）,後段則包括
[1,2,3 ], [2,3 ], [3 ], 與 [ ]。我們是否能定義一個函數 inits ::List a→ List (List a),
計算給定串列的所有前段呢？例 1.19給的答案是：

inits xs = map (λn→ take n xs) [0 . . length xs ] .

如果不用組件，改用歸納定義呢？我們試試看：

inits ::List a→ List (List a)
inits [ ] = ?
inits (x : xs) = ?

基底狀況 inits [ ]的可能選擇是 [[ ]]（見後述）。至於歸納步驟該怎麼寫？我們
用例子來思考。比較 inits [2,3 ]與 inits [1,2,3 ]:

inits [2,3 ] = [[ ], [2 ], [2,3 ]] ,

inits [1,2,3 ] = [[ ], [1 ], [1,2 ], [1,2,3 ]] .

假設我們已算出 inits [2,3 ],如何把它加工變成 inits [1,2,3 ]? 請讀者暫停一下，
思考看看！
一個思路是：如果在 [[ ], [2 ], [2,3 ]]中的每個串列前面都補一個 1,我們就有

了 [[1 ], [1,2 ], [1,2,3 ]]. 再和 inits [1,2,3 ]比較，就只差一個空串列了！因此 inits
的一種定義方式是：

inits ::List a→ List (List a)
inits [ ] = [[ ]]

inits (x : xs) = [ ] : map (x:) (inits xs) .

在此得提醒：有些讀者認為基底狀況 inits [ ]的值選為 [[ ]]，是因為結果的
型別是 List (List a)（直覺地把每個 List都對應到一組中括號，或認為 [[ ]]是型
別為 List (List a)的最簡單的值）。但事實並非如此：畢竟，[ ]的型別也可以是
List (List a)！我們讓 inits [ ] = [[ ]]的原因是空串列 [ ]的「所有前段」只有一個，
恰巧也是 [ ]。就如同在自然數上的歸納函數定義中，有些基底狀況是 0,有些
是 1,有些是別的值，此處我們也依我們的意圖，選定最合適的基底值。

習題 2.20 — 試定義 inits+ ::List a→ List (List a),計算一個串列的所有非空
前段。例如 inits+ [1,2,3 ]是 [[1 ], [1,2 ], [1,2,3 ]]。當然，其中一個定義方式
是 inits+ = tail · inits. 你能以歸納方式定義出 inits+嗎？

inits+ [ ] = ?
inits+ (x : xs) = ?

習題 2.21 — 我們驗證一下 inits在例 1.19中的組件定義與本章的歸納定義
是相等的。定義 upto ::N→ List N:

upto 0 = [0 ]
upto (1+ n) = 0 : map (1+) (upto n) .

使得 upto n = [0 . .n ]. 假設 inits已如本節一般地歸納定義，證明對所有 xs,
inits xs = map (λn→ take n xs) (upto (length xs)). 您可能會需要 map融合定
理（定理 2.2），

定義傳回後段的函數 tails時可依循類似的想法：如何把 tails [2,3 ] = [[2,3 ], [3 ], [ ]]加
工，得到 tails [1,2,3 ] = [[1,2,3 ], [2,3 ], [3 ], [ ]]? 這次較簡單：加上 [1,2,3 ]即可。
的確，串列 x : xs的後段包括 x : xs自己，以及 xs的後段：

tails ::List a→ List (List a)
tails [ ] = [[ ]]

tails (x : xs) = (x : xs) : tails xs .

在習題 2.44中我們將證明一個將 inits與 tails牽上關係的定理：將 inits傳回的
前段與 tails傳回的後段照其順序對應，每對接起來都是原來的串列。

連續區段 給定一個串列，許多傳統最佳化問題的目標是計算符合某條件的連
續區段（簡稱「區段」）。例如，[1,2,3 ]的區段包括 [ ],[1 ],[2 ],[3 ],[1,2 ],[2,3 ],以
及 [1,2,3 ]本身。我們可用 inits與 tails得到一個串列的所有區段。

segments ::List a→ List (List a)
segments = concat ·map inits · tails .

但 segments無法寫成本章目前這種形式的歸納定義。我們將在以後的章節再討
論到 segments.

習題 2.22 — 試著把 segments寫成如下的歸納定義：

segments ::List a→ List (List a)
segments [ ] = ?
segments (x : xs) = ...segments xs... ,

在歸納步驟中希望由 segments xs湊出 segments (x : xs)。這是能在不對輸入
串列（型別為 List a）做任何限制之下做得到的嗎？如果做不到，為什麼？

2.6.4 插入、排列、子串列、與劃分

排列 函數 fan x xs把 x插入 xs的每一個可能空隙。例如，fan 1 [2,3,4 ]可得到
[[1,2,3,4 ], [2,1,3,4 ], [2,3,1,4 ], [2,3,4,1 ]]. 讀者不妨想想它該怎麼定義？一種可
能方式如下：

fan :: a→ List a→ List (List a)
fan x [ ] = [[x ]]
fan x (y : xs) = (x : y : xs) : map (y:) (fan x xs) .

有了 fan,我們不難定義 perms ::List a→ List (List a),計算一個串列所有可能的
排列。例如，perms [1,2,3 ] = [[1,2,3 ], [2,1,3 ], [2,3,1 ], [1,3,2 ], [3,1,2 ], [3,2,1 ]]：

perms ::List a→ List (List a)
perms [ ] = [[ ]]

perms (x : xs) = concat (map (fan x) (perms xs)) .

讀者可思考為何我們需要 concat? 如果沒有，會出現什麼錯誤？
基於 perm的這個定義，我們證明一個定理：長度為 n的串列有 fact n種排

列。這個證明將使用到許多輔助性質與引理，有些已經是我們之前證明過的習
題，有些則可作為接下來的習題。在本證明之中我們也練習將連續的函數應用
f (g (h x))寫成函數合成 f ·g ·h $ x以方便計算。

定理 2.5. 對任何 xs, length (perms xs) = fact (length xs).

Proof. 在 xs上做歸納。
基底狀況 xs :=[ ]:

length (perms [ ])
= length [[ ]]

= 1
= fact (length [ ]) .

歸納步驟 xs := x : xs:

length (perms (x : xs))
= { perms, (·),與 ( $)之定義 }

length · concat ·map (fan x) ·perms $ xs
= {因 length · concat = sum ·map length (習題 2.12), map融合 }

sum ·map (length · fan x) ·perms $ xs
= {因 length · fan x = (1+) · length (習題 2.25) }

sum ·map ((1+) · length) ·perms $ xs
= {因 map length (perms xs) = map (const (length xs)) (perms xs) (習題 2.26) }

sum ·map ((1+) · const (length xs)) ·perms $ xs
= {因 sum (map (1+) xs) = length xs+ sum xs (習題 2.8) }

length (perms xs)+ sum (map (const (length xs)) (perms xs))
= {因 sum (map (const y) xs) = y× length xs (習題 2.9) }

length (perms xs)+ length xs× length (perms xs)
= {四則運算: x+ y× x = (1+ y)× x }
(1+ (length xs))× length (perms xs)

= {歸納假設 }
(1+ (length xs))× fact (length xs)

= { fact之定義 }
fact (1+ (length xs))

= { length之定義 }
fact (length (x : xs)) .

習題 2.23 — 證明 map f · fan x = fan (f x) ·map f .

習題 2.24 — 證明 perm ·map f = map (map f ) ·perm.

習題 2.25 — 證明 length (fan x xs) = 1+ (length xs).

習題 2.26 — 證明 perms xs 傳回的每個串列都和 xs 一樣長，也就是
map length (perms xs) = map (const (length xs)) (perms xs). 其中 const定義於
第 30頁 — const y是一個永遠傳回 y的函數。

子串列 函數 sublists :: List a→ List (List a) 計算一個串列的所有子串列。後
者是類似子集合的概念，只是把順序也考慮進去：ys是 xs的子串列，如果將
xs 中的零個或數個元素移除後可得到 ys: 例如 sublists [1,2,3 ] 的結果可能是：
[[ ], [3 ], [2 ], [2,3 ], [1 ], [1,3 ], [1,2 ], [1,2,3 ]]。怎麼定義 sublists呢？在基底狀況中，
空串列仍有一個子串列 [ ]. 歸納步驟中，x : xs的子串列可分為兩種：不含 x的，
以及含 x的。不含 x的子串列就是 xs的所有子串列（以下稱作 yss），而含 x的
子串列就是 yss中的每個串列接上 x. 因此我們可定義：

sublists ::List a→ List (List a)
sublists [ ] = [[ ]]

sublists (x : xs) = yss++map (x:) yss ,

where yss = sublists xs .

習題 2.27 — 定義 splits ::List a→ List (List a×List a)，使 splits xs傳回所有
滿足 ys++ zs的 (ys,zs). 例：

splits [1,2,3 ] = [([ ], [1,2,3 ]),([1 ], [2,3 ]),([1,2 ], [3 ]),([1,2,3, [ ]])] .

另一種說法是 splits xs = zip (inits xs) (tails xs).

習題 2.28 — 證明 length · sublists = exp 2 · length. 也就是說，長度為 n的串
列的子串列數目為 2n. 你會需要的性質可能包括 (2.1) (length (xs++ ys) =
length xs+ length ys),以及 length ·map f = length.

劃分 給定串列 xs ::List a. 如果 yss ::List (List a)滿足
1. concat yss = xs,
2. yss中的每個串列都不是空的，我們便說 yss是 xs的一個劃分 (partition)。
例如，[[1 ], [2,3 ], [4 ]]是 [1,2,3,4 ]的一個劃分。

我們是否能定義一個函數 parts ::List a→ List (List (List a)),計算一個串列所
有的劃分？可能的方式有很多，其中一種思路如下。首先，空串列 [ ]只有一
個可能的劃分，即是 [ ]. 考慮 x : xs,如果 xs已經被劃分完畢，對於 x我們有兩
個選擇：讓它成為單獨的串列，或著將 xs加入最左邊的串列。寫成歸納定義
如下：

parts ::List a→ List (List (List a))
parts [ ] = [[ ]]

parts (x : xs) = concat (map (extend x) (parts xs)) ,

where extend x [ ] = [[[x ]]]
extend x (ys : yss) = [[x ] : ys : yss,(x : ys) : yss ] .

遞迴呼叫 parts xs 找出 xs 的所有劃分，輔助函數 extend x :: List (List a) →
List (List (List a))則作用在其中一個劃分上。我們得分出兩個狀況：

1. 如果該劃分含第一個串列 ys和剩下的 yss,我們可選擇讓 x自成一個串列，
或著加入 ys中。我們把這兩種選擇收集到一個串列中，因此 extend的結
果有三層 List。

2. 如果該劃分是空的，加入 x的劃分必定是 [[x ]]，但仍需放入一個串列中，
表示「只有這一個選擇」。

呼叫 map (extend x)的結果需再用一個 concat聯集在一起。

2.7 再談「讓符號為你工作」

看了這麼多歸納定義，我們做個較複雜的練習，並以此為例再談談「讓符號為
你工作」。本節在初次閱讀時可略過。
以下的函數 ins將一個元素插到一個串列的空隙中。它和第 2.6.4節中的 fan

類似但稍有不同 — ins x xs 不會把 x 放到 xs 的結尾。8 例如：ins 1 [2,3,4 ] =
[[1,2,3,4 ], [2,1,3,4 ], [2,3,1,4 ]]。

8本節不使用 fan而另外定義了 ins的原因僅是因為如此使得性質 (2.5)較單純而方便說明。

反函數

假設我們有一個取反函數的運算元：給定 f :: a→ b, 其反函數 f−1 的型別是
b→ List a。f−1 y將所有使得 f x = y的 x收集在串列中。則 parts可以有一個更接
近本節中文字描述的定義：

parts = filter (all (not ·null)) · concat−1 .

計算 parts xs時，我們用 concat−1 找出所有滿足 concat yss = xs的 List (List a),
並且只挑出不含空串列的。
上述定義可以轉換成本節的歸納定義。相關研究可參考Mu and Bird [2003].

ins :: a→ List a→ List (List a)
ins x [ ] = [ ]

ins x (y : ys) = (x : y : ys) : map (y:) (ins x ys) .

請證明對所有 x, ys,與 zs,

map (++zs) (ins x ys)++map (ys++) (ins x zs) = ins x (ys++ zs) . (2.5)

意即：在 ys++ zs之中插入 x的所有方法，是 1. 把 x插入 ys之中，並把每個結
果的右邊接上 zs,以及 2. 把 x插入 zs之中，並把每個結果的左邊接上 ys.
面對一個如此冗長的式子，我們該如何著手？我們可猜想如此複雜的

性質大概會需要歸納證明。但在哪個變數上做歸納呢？觀察(2.5)，等號左
手邊最外側的運算子是 (++)。計算 (++) 時，根據定義，先被樣式配對的
是 (++) 左邊的參數，也就是 map (++zs) (ins x ys)。再來，根據 map 的定義，
map (++zs) (ins x ys) 的運算要有進展，得先把 ins x ys 算出來。最後，根據
ins 的定義，計算 ins x ys 會先拆開 ys. 由此可知，等號左手邊的運算是從拆
解 ys 開始的。等號右手邊的情況也一樣：根據 ins 與 (++) 的定義，要計算
ins x (ys++ zs)，第一個被拆開的是 ys. 因此，欲證明(2.5)，合理的猜測是在 ys
上做歸納！
我們省略 ys :=[ ]的情況，僅考慮歸納情況 ys := y : ys。經代換後的式子是

map (++zs) (ins x (y : ys))++map ((y : ys)++) (ins x zs). 我們採取的策略和證明定
理 2.1時一樣：想辦法湊出 map (++zs) (ins x ys)++map (ys++) (ins x zs),以便使
用歸納假設。由此論證如下（步驟加上編號以利說明）：

map (++zs) (ins x (y : ys))++map ((y : ys)++) (ins x zs)
= { 1. ins與 map之定義 }
(x : y : ys++ zs) : map (++zs) (map (y:) (ins x ys))++map ((y : ys)++) (ins x zs)

= { 2. map融合 }
(x : y : ys++ zs) : map ((++zs) · (y:)) (ins x ys)++map ((y : ys)++) (ins x zs)

= { 3. 由於 (++zs) · (y:) = (y:) · (++zs) (習題 2.30) }
(x : y : ys++ zs) : map ((y:) · (++zs)) (ins x ys)++map ((y : ys)++) (ins x zs)

= { 4. map融合 }
(x : y : ys++ zs) : map (y:) (map (++zs) (ins x ys))++map ((y : ys)++) (ins x zs)

= { 5. 由於 ((y : ys)++) = (y:) · (ys++) (習題 2.29) }
(x : y : ys++ zs) : map (y:) (map (++zs) (ins x ys))++map ((y:) · (ys++)) (ins x zs)

= { 6. map融合 }
(x : y : ys++ zs) : map (y:) (map (++zs) (ins x ys))++map (y:) (map (ys++) (ins x zs))

= { 7. 由於 map f (xs++ ys) = map f xs++map f ys }
(x : y : ys++ zs) : map (y:) (map (++zs) (ins x ys)++map (ys++) (ins x zs))

= { 8. 歸納假設 }
(x : y : ys++ zs) : map (y:) (ins x (ys++ zs))

= { 9. ins與 (++)之定義 }
ins x ((y : ys)++ zs) .

雖然步驟多、式子長，但策略確定後，整個證明的架構便很清楚。第 1 到 7
步都是為了可在第 8步使用歸納而做的準備。為湊出 map (++zs) (ins x ys)++
map (ys++) (ins x zs), 第 2 至第 4 步把 map (++zs) 往裡推，把第一個 map (y:)
往外提；第 5, 6 步則提出另一個 map (y:)。式子變成 map (y:) ...++map (y:)...
的形式後，共同的 map (y:)可在第 7步提出來。如此，終於可使用歸納假設
了。整個證明中，要配對哪個變數、把式子中的哪些項往哪兒移動，都被符
號引導著而有線索可循。證明中用到的兩個性質 (++zs) · (y:) = (y:) · (++zs)與
((y : ys)++) = (y:) · (ys++)並非天外飛來，而是確立了目標後，因為有需求（兩
者分別讓我們把 (++zs)與 (ys++)往右推，把 (y:)往左拉）而設計出並另外證明
的性質。
初學者常犯的一個錯誤是：以為所有情況下、所有的變數都需要樣式配

對。事實上，過度的樣式配對是形式證明中希望避免的。藉由觀察 (2.5)，我們
決定對 ys做歸納，證明過程中發現如此便已經足夠。如果我們對 ys和 zs都
做配對，將必須處理四種情況：ys,zs :=[ ], [ ], [ ], ys,zs :=[ ],z : zs, ys,zs :=[ ],z : zs
ys,zs := y : ys,z : zs。一個待證明的性質若有 n個串列，似乎得分出 2n 種情況 —
只有萬不得已我們才會這麼做。
最嚴重的問題並不是情況數目太多，而是過度的樣式配對破壞了證明的結

構。如果對 zs做配對，許多人會很自然地想把 ins x (z : zs)展開。證明可能如
此進行：

map (++(z : zs)) (ins x (y : ys))++map ((y : ys)++) (ins x (z : zs))
= {將 (++)的左手邊展開 }
(x : y : ys++(z : zs)) : map (++(z : zs)) (map (y:) (ins x ys))++

map ((y : ys)++) (ins x (z : zs))
= {展開 ins x (z : zs) }
(x : y : ys++(z : zs)) : map (++(z : zs)) (map (y:) (ins x ys))++

map ((y : ys)++) ((x : z : zs) : map (z:) (ins x zs))
= { map的定義 }
(x : y : ys++(z : zs)) : map (++(z : zs)) (map (y:) (ins x ys))++
((y : ys++(x : z : zs)) : map ((y : ys)++) (map (z:) (ins x zs))) .

要從這樣的式子中再整理出map (++(z:zs)) (ins x (y:ys))++map ((y:ys)++) (ins x (z:
zs))以便做歸納假設，我們得看出如何把（本來不需展開的）ins x (z : zs)給收

回來。式子的結構已經被破壞，操作起來的難度便大大提高了。9

習題 2.29 — 證明：對所有 y與 ys, (y:) · (ys++) = ((y : ys)++).

習題 2.30 — 證明：對所有 y與 zs, (++zs) · (y:) = (y:) · (++zs). 以上兩者都是
不需歸納、也不須狀況分析的單純證明。

2.8 其他歸納資料結構

不僅自然數與串列，歸納式的函數定義與證明可用在所有歸納定義的資料結構
上。以第 1.10節提及的兩種二元樹為例：

data ITree a = Null | Node a (ITree a) (ITree a) ,

data ETree a = Tip a | Bin (ETree a) (ETree a) .

型別 ETree可讀解為一個歸納資料結構，我們可在它之上以歸納法定義函數。
例如，下述的函數 minE計算一個 ETree中最小的元素，mapE f 則將函數 f 作
用在樹中的每個元素上。

minE ::ETree Int→ Int

minE (Tip x) = x
minE (Bin t u) = minE t ↓minE u ,

mapE :: (a→ b)→ ETree a→ ETree b
mapE f (Tip x) = Tip (f x)
mapE f (Bin t u) = Bin (mapE f t) (mapE f u) .

我們也有如下的歸納原則：

ETree上之歸納法： (∀t ::ETree a ·P t) ⇐
(∀x ·P (Tip x)) ∧ (∀t u ·P (Bin t u)⇐ P t ∧ P u) .

作為例子，我們證明下列性質：

minE (mapE (x+) t) = x+minE t .

左手邊將樹中的每個元素都加上 x,然後取最小的。右手邊則告訴我們不必這
麼麻煩，先取最小的，再加上 x即可！

Proof. 基底狀況 t :=Tip y很容易成立。至於歸納狀況 t :=Bin t u,使其成立的關
鍵性質是加法可分配至 (↓)中：x+(y↓ z) = (x+ y)↓ (x+ z):

minE (mapE (x+) (Bin t u))
= { mapE與 minE之定義 }

minE (mapE (x+) t)↓minE (mapE (x+) u)
= {歸納假設 }

9這其實是我於 2019年開設的程式語言課程的期末考題。在十數位將 ins x (z : zs)展開的同學
中，只有一位成功地將它收了回來。

(x+minE t)↓ (x+minE u)
= {算術：加法分配至 (↓)中 }

x+(minE t ↓minE u)
= { minE之定義 }

x+minE (Bin t u) .

習題 2.31 — 請寫出 ITree的歸納原則？
習題 2.32 — 試定義函數 tags :: ITree a→ List a,由左至右傳回給定之 ITree

的所有標籤。例如，若

t = Node 3 (Node 2 (Node 1 Null Null) Null)

(Node 5 (Node 4 Null Null)

(Node 6 Null Null)) ,

則 tags t = [1,2,3,4,5,6 ].

習題 2.33 — 試定義函數 size :: ITree a→ Int,傳回給定的 ITree之標籤數目
—例如 size Null= 0, size (Node 3 Null (Node 4 Null Null)) = 2.

習題 2.34 — 接續前兩題，證明 length (tags t) = size t.

2.9 由集合論看歸納法

歸納證明為何成立？本節試圖以集合論的角度理解歸納法，試圖建立幾個觀
念：當我們說某型別是「歸納定義」的，常常代表它是具有某個結構的最小集
合。而歸納證明可以理解為證明某型別是所有滿足某性質的事物的子集。本節
於初次閱讀時可跳過。

序理論回顧 為了本章的完整性，我們在這兒回顧一些重要定義。對學過序理
論或抽象代數的讀者來說，以下概念應不陌生。如果您還不熟悉這些定義，由
於它們在程式語言語意中常常使用，值得花些時間弄懂。

定義 2.6 (前序、偏序). 給定集合 S,令 (⩽)為 S上的一個二元關係。如果 (⩽)滿

足：

• 自反律：對所有 x ∈ S, x ⩽ x.
• 遞移律：對所有 x,y,z ∈ S, x ⩽ y ∧ y ⩽ z⇒ x ⩽ z,
則 (⩽)被稱為 S上的一個前序 (preorder). 如果 (⩽)除上述兩性質外還滿足：

• 反對稱律：對所有 x,y ∈ S, x ⩽ y ∧ y ⩽ x⇒ x = y.
則 (⩽)被稱為 S上的一個偏序 (partial order). 如果 S上有偏序 (⩽),我們常

把它們放在一起，稱 (S,(⩽))為一個偏序集合 (partially ordered set,或 poset).

定義 2.7 (最小上界、最大下界). 給定偏序集合 (S,(⩽))，考慮其子集 T ⊆ S:
• 如果 x ∈ S滿足 (∀y ∈ T : y ⩽ x),則 x是 T 的一個上界 (upper bound).

• 如果 T 的所有上界中存在「最小」的，該值稱作 T 的最小上界 (supremum,
或 least upper bound),記為 sup T . 依據定義，sup T 滿足 (∀y∈ T : y ⩽ x)⇒
sup T ⩽ x.

• 如果 x ∈ S滿足 (∀y ∈ T : y ⩾ x),則 x是 T 的一個下界 (lower bound).
• 如果 T 的所有下界存在「最大」的，該值稱作 T 的最大下界 (infimum,或

greatest lower bound),記為 inf T . 依據定義，inf T 滿足 (∀y ∈ T : y ⩾ x)⇒
inf T ⩾ x.

定義 2.8 (格、完全格). 考慮偏序集合 (S,(⩽)):
• 如果對任何 x,y∈ S, sup {x,y}和 inf {x,y}均存在且都在 S之中，則 (S,(⩽))

是一個格 (lattice)。
• 如果對任何 T ⊆ S，sup T 和 inf T 均存在且都在 S之中，則 (S,(⩽))是一

個完全格 (complete lattice)。

在本節之中我們只會考慮一種格。令 D 代表所有範式（如 0, 1+ 0, True,
1 : 2 : [ ], (λx→ x)…）的集合，我們考慮的格是PD，即 D的所有子集形成的集
合。其上的偏序就是子集關係 (⊆).

定點 再回顧一些與定點相關的理論。

定義 2.9 (定點). 給定完全格 (A,(⩽))和函數 f ::A→ A,
1. 如果 f x ⩽ x,我們說 x是 f 的一個前定點 (prefixed point).
2. 如果 f x = x,我們說 x是 f 的一個定點 (fixed point).
3. 如果 f x ⩾ x,我們說 x是 f 的一個後定點 (postfixed point).

定義 2.10 (最小前定點、最大後定點). 給定完全格 (A,(⩽))和函數 f ::A→ A,
• 如果 f 的前定點之中存在最小值，我們將它記為 µ f . 根據此定義，µ f 滿
足 f x ⩽ x⇒ µ f ⩽ x.

• 如果 f 的後定點之中存在最大值，我們將它記為 ν f . 根據此定義，ν f 滿
足 x ⩽ f x⇒ x ⩽ ν f .

定理 2.11. 給定完全格 (A,(⩽))和函數 f ::A→ A,
• f 的最小前定點 µ f 也是最小的定點，
• f 的最小前定點 ν f 也是最大的定點。

歸納定義 回顧自然數的定義：

data N= 0 | 1+ N .

第 2.1節對這行定義的解釋是：
1. 0的型別是 N;
2. 如果 n的型別是 N, 1+ n的型別也是 N;
3. 此外，沒有其他型別是 N的東西。

如果我們把一個型別視作一個集合，上述條件定出了怎麼樣的集合呢？10 用
N表示我們定出的這個新型別。上述第 1.點告訴我們 0是 N的成員，也就是

10請注意：「把型別視為集合」僅在簡單的語意之中成立。本書後來將會討論更複雜的語意，屆
時型別並不只是集合。

{0} ⊆ N. 第 2. 點則表示，從 N 這個集合中取出任一個元素 n, 加上 1+, 得到
的結果仍會在 N之中。也就是說 {1+ n | n ∈ N} ⊆ N. 集合基本定理告訴我們
X⊆ Z ∧ Y ⊆ Z和 X∪Y ⊆ Z是等價的，所以：11

{0}∪{1+ n | n ∈ N} ⊆ N . (2.6)

意思是說，如果我們定義一個集合到集合的函數 NatF:

NatF X= {0}∪{1+ n | n ∈ X} ,

那麼 (2.6)可以改寫為

NatF N ⊆ N ,

也就是說，N是 NatF的一個前定點！
至於 3.呢？它告訴我們 N僅含恰巧能滿足 1.和 2.的元素，沒有多餘。意

即，N是滿足 1.和 2.的集合之中最小的。若有另一個集合 Z也滿足 1.和 2.,我
們必定有 N⊆ Z. 也就是說 N是 NatF的最小前定點：N= µ NatF!
給定述語 P,我們把所有「滿足 P的值形成的集合」也記為 P. 12 數學歸納

法的目的是證明所有自然數都滿足 P。但，「所有自然數都滿足 P」其實就是
N⊆ P.
如何證明 N ⊆ P? 如前所述，N 是 NatF 的最小前定點。如果 P 恰巧也是

NatF的一個前定點，N⊆ P一定得成立。寫成推論如下：

N⊆ P

⇐ { N是 NatF的最小前定點，定義 2.10 }
NatF P⊆ P

≡ { NatF的定義 }
{0}∪{1+ n | n ∈ P} ⊆ P

≡ { (∪)的泛性質：X∪Y ⊆ Z ≡ X⊆ Z ∧ Y ⊆ Z }
{0} ⊆ P ∧ {1+ n | n ∈ P} ⊆ P .

也就是說，如果證出 {0} ⊆ P和 {1+ n | n ∈ P} ⊆ P，我們就有 N⊆ P。其中，
1. {0} ⊆ P翻成口語便是「P對 0成立」，
2. {1+ n | n ∈ P} ⊆ P則是「若 P對 n成立，P對 1+ n亦成立」。

正是數學歸納法的兩個前提！
原來，數學歸納法之所以成立，是因為自然數被定義為某函數的最小前定

點。事實上，當我們說某型別是「歸納定義」的，意思便是它是某個函數的最
小前定點。
以串列為例。為單純起見，先考慮元素皆為 N的串列。如下的定義

data ListNat = [ ] | N :ListNat ,

可理解為 ListNat= µ ListNatF，而 ListNatF定義為：
11X⊆ Z ∧ Y ⊆ Z≡ X∪Y ⊆ Z被稱為「(∪)的泛性質」。
12對任何 A,函數 P ::A→ Bool和「滿足 P的 A形成的集合」是同構的。我們可把它們等同視

之。

ListNatF X= { [ ]}∪{n : xs | xs ∈ X,n ∈ N} .

至於如下定義的、有型別參數的串列，

data List a = [ ] | a :List a ,

則可理解為 List a = µ (ListF a) — List a是 ListF a的最小前定點，其中 ListF定
義如下：

ListF A X= { [ ]}∪{x : xs | xs ∈ X,x ∈ A} .

給定某型別 A,當我們要證明某性質 P對所有 List A都成立，實質上是想要證
明 List A ⊆ P（同樣地，此處 P代表所有使述語 P成立的值之集合）。我們推
論如下：

List A⊆ P

⇐ { List a = µ (ListF a) }
ListF A P⊆ P

≡ { ListF之定義 }
{ [ ]}∪{x : xs | xs ∈ P,x ∈ A} ⊆ P

≡ { (∪)的泛性質 }
{ [ ]} ⊆ P ∧ {x : xs | xs ∈ P,x ∈ A} ⊆ P .

其中 { [ ]} ⊆ P翻成口語即是「P [ ]成立」；{x : xs | xs ∈ P,x ∈ A} ⊆ P則是「若
P xs成立，對任何 x ::A, P (x : xs)成立」。
我們之所以能做自然數與串列的歸納證明，因為它們都是歸納定義出的型

別 —它們都被定義成某函數的最小前定點。若非如此，歸納證明就不適用了。
那麼，有不是歸納定義出的型別嗎？
讀者可能注意到，本節起初同時談到最小前定點與最大後定點，但後來

只討論了前者。事實上，我們也可以把一個資料型別定義為某函數的最大後
定點。這時我們說該資料型別是個餘歸納 (coinductive)定義，如此訂出的型別
稱為餘資料 (codata)。 歸納定義出的資料結構是有限的，而餘歸納定義的型
別可能包括無限長的資料結構。寫餘資料相關的證明，另有一套稱作餘歸納
(coinduction)的方法，而餘歸納也影響到我們如何寫與餘資料相關的程式。餘
歸納和歸納的相關理論剛好是漂亮的對偶。我們將在 [todo: where]介紹餘歸
納。

習題 2.35 — 回顧第 1.10節中介紹的兩種樹狀結構：

data ITree a = Null | Node a (ITree a) (ITree a) ,

data ETree a = Tip a | Bin (ETree a) (ETree a) .

說說看它們分別是什麼函數的最小前定點，並找出它們的歸納原則。

2.10 歸納定義的簡單變化

目前為止，我們所認定「良好」的函數定義是這種形式：

f 0 = ...

f (1+ n) = ...f n ...

我們知道這樣定義出的函數是個全函數、對所有輸入都會終止、和歸納法有密
切關係…。以後的幾個章節中，我們將逐步放鬆限制，允許更有彈性的函數定
義模式。我們先從歸納法的一些較簡單的變化開始。

基底狀況的變化 有些函數的值域是「正整數」或「大於 b的整數」。定義這些
函數時我們可用這樣的模式：

f1 1 = e fb b = e
f1 (1+ n) = ...f1 n... , fb (1+ n) = ...fb n ... .

我們可把 f1 理解為：另外訂了一個資料型別 data N+ = 1 | 1+ N+,以 1為基底
狀況，而 f1 是 N+ 之上的全函數。fb 的情況也類似。與使用 N的函數混用時，
我們就得在這兩個型別之間作轉換。這相當於檢查給 f1 的輸入都是大於 1的
整數。實務上為了方便，我們仍用同一個型別實作 N與 N+,就如同實務上用
Int實作 N一樣。
串列的情形也類似。有些函數若能只定義在「不是空的串列」上，其定義

會比較合理。對於這些函數，我們可想像有這麼一個「非空串列」資料型別
data List+ a = [a ] | a :List+ a，只是為了方便，我們用普通串列實作它。13

例 2.12. 假設 x↑ y傳回 x與 y中較大的值。函數 maximum+ 傳回一個非空串列

中的最大元素：

maximum+ ::List+ Int→ Int

maximum+ [x ] = x
maximum+ (x : xs) = x↑maximum+ xs .

Haskell標準函式庫中另有一個函數 maximum ::List Int→ Int, 但該函數需假設
Int中有一個相當於 −∞的值存在，以便當作 maximum [ ]的結果：

maximum [ ] =−∞
maximum (x : xs) = x↑maximum xs .

例 2.13. 回顧第 2.6.4節提及的「劃分 (partition)」：如果 concat yss = xs,且 yss中
的每個串列都不是空的，則 yss是 xs的一個劃分。如果我們限定 xss為非空串
列，計算所有劃分的函數 parts+ ::List+ a→ List (List+ (List+ a))可以寫得更簡
潔：

parts+ [x ] = [[[x ]]]
parts+ (x : xs) = concat (map (extend x) (parts+ xs)) ,

where extend x (ys : yss) = [[x ] : ys : yss,(x : ys) : yss ] .

由於每個劃分一定是非空串列，extend不需考慮輸入為 [ ]的情況。

13在一些型別系統更強大的語言中、進行更注重證明的應用時，我確實有將 N+ 與 List+ 做成
與 N和 List不同的型別的經驗與需求。

多個參數的歸納定義 函數 take/drop 的功能與 takeWhile/dropWhile 很類似，
但其定義方式卻有些不同：

take ::N→ List a→ List a
take 0 xs = [ ]

take (1+ n) [ ] = [ ]

take (1+ n) (x : xs) = x : take n xs ,

drop ::N→ List a→ List a
drop 0 xs = xs
drop (1+ n) [ ] = [ ]

drop (1+ n) (x : xs) = drop n xs .

我們可把 take/drop想成在自然數上歸納定義成的高階函數：take (1+ n)的值是
一個 List a→ List a的函數。定義 take (1+ n)時，我們假設 take n已有定義。唯
一的特殊處是我們另分出兩個子情況：串列為 [ ],或串列為 x : xs.
根據「使證明的結構符合程式的結構」的原則，如果要做關於 take/drop的

證明，一種可能做法是也依循著它們的結構：先拆解自然數，在 n :=1+ n的情
況中再分析串列的值。作為例子，我們來驗證第 42頁提到的這個性質：

定理 2.14. 對所有 n與 xs, take n xs++drop n xs = xs.

Proof. 在 n上做歸納。在 n :=[ ]的情況下，等號兩邊都化簡為 [ ]. 在 n := 1+ n
的情況中，我們再細分出兩種情形：
狀況 n :=1+ n, xs :=[ ]. 顯然等號兩邊都化簡為 [ ].
狀況 n :=1+ n, xs := x : xs:

take (1+ n) (x : xs)++drop (1+ n) (x : xs)
= { take與 drop之定義 }

x : take n xs++drop n xs
= {歸納假設 }

x : xs .

然而，本章討論的是所有資料結構都是歸納定義、所有函數也都是全函數
的世界。如果我們的世界中有不終止程式存在，上式便不見得成立了。

習題 2.36 — 請舉一個在允許不終止程式的 Haskell 中，take n xs ++
drop n xs = xs不成立的例子。
習題 2.37 — 對任何串列 xs，head xs : tail xs = xs 都成立嗎？請舉一個反
例。
習題 2.38 — 證明對自然數 m, n, take m (take (m+n) xs) = take m xs.

習題 2.39 — 證明對自然數m, n, drop m (take (m+n) xs) = take n (drop m xs).

習題 2.40 — 證明對自然數 m, n, drop (m+n) xs) = drop n (drop m xs).

函數 zip是另一個例子。我們可把 zip xs ys視為 xs之上的歸納定義：

zip ::List a→ List b→ List (a×b)
zip [ ] ys = [ ]

zip (x : xs) [ ] = [ ]

zip (x : xs) (y : yz) = (x,y) : zip xs ys .

習題 2.41 — 試定義 zipWith :: (a→ b→ c)→ List a→ List b→ List c,並證明
zipWith f xs ys = map (uncurry f ) (zip xs ys).

2.11 完全歸納

說到遞迴定義，費氏數 (Fibonacci number)是最常見的教科書例子之一。簡而
言之，第零個費氏數是 0,第一個費氏數是 1,之後的每個費氏數是之前兩個的
和。寫成遞迴定義如下：

fib ::N→ N
fib 0 = 0
fib 1 = 1
fib (2+n) = fib (1+n)+fib n .

但這和我們之前談到的歸納定義稍有不同。我們已知定義 f (1+ n)時可假設 f
在 n之上已有定義。但在 fib的定義中，fib (2+n)用到了 fib的前兩個值。這樣
的定義是可以的嗎？
函數 fib的定義可以視為完全歸納（又稱作強歸納）的例子。回顧：先前介

紹的數學歸納法中，使 P n成立的前提之一是「對所有 n,若 P n成立，P (1+ n)
亦成立」。完全歸納則把這個前提增強如下：

給定述語 P ::N→ Bool. 若

• 對所有小於 n的值 i，P i皆成立，則 P n亦成立，

則我們可得知 P對所有自然數皆成立。

以更形式化的方式可寫成：

完全歸納： (∀n ·P n) ⇐ (∀n ·P n⇐ (∀i<n ·P i)) .

請注意：前提 P n⇐ (∀i<n ·P i)隱含 P 0成立，因為當 n :=0,由於沒有自然數
i滿足 i<n,算式 (∀i<n ·P i)可化簡為 True.
在完全歸納法之中，證明 P n時，我們可假設 P對所有小於 n的值都已成

立了。對寫程式的人來說，有了完全歸納法，表示我們日後定義自然數上的函
數 f ::N→ a時，每個 f n都可以自由使用 f 在所有小於 n的輸入之上的值。因
此 fib (2+n)可以用到 fib (1+n)與 fib n,因為 n<1+n<2+n.

例 2.15. 關於完全歸納，離散數學教科書中的一個常見例子是「試證明所有自
然數都可寫成不相同的二的乘冪的和」，例如 50 = 25 +24 +2. 這可用完全歸納
證明。我們以半形式的方式論述如下：令 P n為「n可寫成一串不相同的二的乘
冪的和」. 對所有 n,我們想要證明 P n⇐ (∀i<n ·P i). 當 n為 0,這一串數字即
是空串列。當 n大於零，

• 我們可找到最接近 n但不超過 n的二之乘冪，稱之為 m（也就是 m = 2k

而且 m ⩽ n < 2×m）.
• 由於 n不是 0, m也不是 0. 也因此 n−m<n.
• 依據歸納假設 (∀i<n ·P i),以及 n−m<n, P (n−m)成立 — n−m可以寫
成不相同的二的乘冪的和。

• 也因此 n可寫成不相同的二的乘冪的和 —把 n−m加上 m即可.
上述證明僅用口語描述，因為如果形式化地寫下這個證明，就等同於寫個

將自然數轉成一串二的乘冪的程式，並證明其正確性！下述函數 binary做這樣
的轉換，例如，binary 50 = [32,16,2 ]:

binary ::N→ List N
binary 0 = [ ]

binary n = m : binary (n−m) ,

where m = last (takeWhile (⩽ n) twos)
twos = iterate (2×) 1 .

函數 binary是一個完全歸納定義，和上述的證明對應得相當密切：串列 twos
是 [1,2,4,8... ]等等所有二的乘冪，m是其中最接近而不超過 n的。遞迴呼叫
binary (n−m)是許可的，因為 n−m<n,而根據完全歸納，我們已假設對所有
i<n, binary i皆有定義。

有了完全歸納法，我們在定義自然數上的函數時可允許更靈活的函數定義：

f ::N→ a
f b = .... {一些基底情況 }
f n = ...f m ... f k ... {如果 m<n且 k<n }

f n的右手邊可以出現不只一個遞迴呼叫，只要參數都小於 n. 但我們必須確定
上述定義中的幾個子句足以包含所有狀況，沒有狀況被遺漏。例如，我們若把
fib定義中的 fib 1 = ...基底狀況去掉，計算 fib 2 = fib 1+fib 0時便會出錯。

習題 2.42 — 證明 sum (binary n) = n.

習題 2.43 — 證明當 n ⩾ 1, fib (2+n)>αn,其中 α = (1+
√

5)/2. 這個證明
可用 n :=1和 n :=2當基底狀況。

完全歸納 vs簡單歸納 完全歸納有個較早的稱呼：強歸納 (strong induction)。
原本的歸納法則相對被稱呼為簡單歸納或弱歸納。強/弱歸納的稱呼可能使人
以為完全歸納比簡單歸納更強 —意謂前者能證明出一些後者無法證明的定理。
事實上，完全歸納與簡單歸納是等價的：能用一個方法證出的定理，用另一個
方法也能證出。因此，使用哪一個純粹只是方便性的考量。

反應在程式設計上，給任一個完全歸納定義函數 f ::N→ A,我們總能做出
一個以簡單歸納定義的函數 fs ::N→ List A,滿足 fs n = map f [n,n− 1, ...0 ]. 例
如，函數 fibs n傳回 [fib n,fib (n−1) . . ,fib 0 ].

fibs ::N→ List N
fibs 0 = [0 ]
fibs 1 = [1,0 ]
fibs (1+ n) = (x1 + x0) : x1 : x0 : xs ,

where (x1 : x0 : xs) = fibs n .

由 fib 到 fibs 的轉換可能令讀者想起演算法中的動態規劃 (dynamic program-
ming)。我們將在日後談到這個話題。

串列上的完全歸納 串列與自然數是類似的資料結構。串列上的完全歸納原則
便是將 0代換為 [ ],將 1+代換為 (x:). 至於「小於」的關係，可定義為：

ys< xs≡ ys ∈ tails xs ∧ ys ≠ = xs .

也就是說 ys是 xs的一個後段，但不是 xs自己。有了如上定義，串列上的完全
歸納法是：

串列的完全歸納： (∀xs ·P xs) ⇐ (∀xs ·P xs⇐ (∀ys< xs ·P ys)) .

應用在編程上，當定義 f (xs:)時，遞迴呼叫可作用在 xs的任何後段上。
但對許多串列上的函數而言，這樣的模式還不夠靈活。我們得用下一節說

到的良基歸納。

2.12 良基歸納

良基歸納 (well-founded induction)可視為完全歸納的再推廣。如果說完全歸納
的主角是自然數，使用的是自然數上的「小於 (<)」關係，良基歸納則將其推
廣到任何型別，使用任一個良基序。

定義 2.16. 給定某型別 A之上的二元關係 (◁)。如果從任意一個 a0 ::A開始，均
不存在無限多個滿足如下關係的 a1, a2…:

. . .◁ a2 ◁ a1 ◁ a0 ,

則 (◁)可稱為一個良基序 (well-founded ordering)。

把 b ◁ a簡稱為「b小於 a」。上述定義可以這麼地直覺理解：給定任一個
a0 ::A，我們找一個滿足 a1 ◁ a0 的值 a1。這種 a1 可能已不存在，但如果存在，
我們再找一個滿足 a2 ◁ a1 的 a2. 說 (◁)是個良基序的意思便是前述過程不可
能永遠做下去：總有一天我們得停在一個「最小」的某基底 an ::A。
舉例說明：自然數上的「小於 (<)」關係是個良基序，但整數上的 (<)關係

則不是 —由於負數的存在。實數上的 (<)關係也不是。良基序並非得是個全序
(total order)。例如，我們可定義序對上的比較關係如下：

(x1,y1)◁ (x2,y2) ≡ x1 < x2 ∧ y1 < y2 ,

其中 x1, y1, x2, y2 都是自然數。這麼一來，不論 (1,4) ◁ (2,3)或 (2,3) ◁ (1,4)
都不成立，但 (◁)仍是個良基序 —任何兩個自然數形成的序對不論以什麼方
式遞減，最晚也得停在 (0,0).
如果 (◁)是個良基序，我們便可在其上做歸納。以直覺來理解的話，如果

某函數定義成如此的形式（假設這幾個子句已經包括參數的所有可能情況）：

f ::A→ B

f b = .... {一些基底情況 }
f x = ...f y ... f z ... {如果 y ◁ x且 z ◁ x }

由任何 f x開始，若 x不是基底情況之一，我們需遞迴呼叫 f y和 f z。但 y和 z
在 (◁)這個序上比 x「小」了一點。此後即使再做遞迴呼叫，每次使用的參數
又更小了一點。而由於 (◁)是良基序，f 的參數不可能永遠「小」下去 — f 非
得停在某個基底情況不可。因此 f 必須正常終止。同樣的原則也用在證明上：

給定述語 P ::A→ Bool以及 A之上的良基序 (◁)。若

• 對所有滿足 y ◁ x的值 y，P y皆成立，則 P x亦成立，

則我們可得知 P對所有 A皆成立。

或著可寫成如下形式：

良基歸納： (∀x ·P x) ⇐ (∀x ·P x⇐ (∀y ◁ x ·P y)) ,

其中 (◁)為一個良基序。

終止證明與良基歸納 我們已在許多地方強調：確定程式正常終止是很重要
的。我們也知道以簡單歸納與完全歸納定義出的程式均是會正常終止的。但這
兩種歸納定義的限制很多。雖然我們已舉了許多例子，仍有些程式難以套入它
們所要求的模板中。相較之下，良基歸納寬鬆許多。大部分我們已知、會終止
的程式都可視為良基歸納定義。
或著，上述段落應該反過來說。在函數程設中，欲證明某個遞迴定義的函

數會終止，最常見的方式是證明該函數每次遞迴呼叫時使用的參數都在某個度
量上「變小」了，而這個度量又不可能一直變小下去。因此該函數遲早得碰到
基底狀況。換句話說，該函數每次遞迴呼叫的參數符合某個良基序；當我們如
此證明一個函數會終止，其實就相當於在論證該函數是一個良基歸納定義。在
指令式編程中證明某迴圈會終止的做法也類似。最常見的方式是證明該迴圈每
多執行一次，某個量值就會變小，而該量值是不可能一直變小的。也就是說這
些量值在每趟迴圈執行時的值符合某個良基序。
以下我們將看幾個遞迴定義的例子。請讀者們想想：這些函數總會正常終

止嗎？為何？如果它們是良基歸納，使用的良基序是什麼？

例 2.17 (快速排序). 以下是大家熟悉的快速排序 (quicksort) [Hoare, 1962]:

qsort ::List Int→ List Int

qsort [ ] = [ ]

qsort (x : xs) = qsort ys++[x ]++qsort zs ,

where (ys,zs) = (filter (⩽ x) xs,filter (<x) xs) .

空串列是已經排序好的。當輸入為非空串列 x : xs，我們將 xs分為小於等於 x
的，以及大於 xs的，分別遞迴排序，再將結果接在一起。
函數 qsort會正常終止，因為每次遞迴呼叫時，作為參數的串列都會減少至

少一個元素（因為 x被取出了），而串列的長度又不可能小於 0. 若要稍微形式
地談這件事，可從良基歸納的觀點來看。如果定義：

ys ◁ xs ≡ length ys< length xs ,

在 qsort (x : xs)子句中，ys ◁ xs和 zs ◁ xs均被滿足，而 (◁)是一個良基序。因

此 qsort是一個奠立在 (◁)之上的良基歸納定義。

例 2.18 (合併排序). 在串列上，合併排序也是很常使用的排序方式。我們
在第 1.9節中示範過以全麥編程方式寫成、由下往上的合併排序。此處的寫
法則更接近大家一般的認知：拿到一個長度為 n 的串列，將之分割為長度
大致為 n / 2 的兩段，分別排序之後合併。同樣地，假設我們已有一個函數
merge ::List Int→ List Int→ List Int,如果 xs與 ys已經排序好，merge xs ys將它
們合併為一個排序好的串列。合併排序可寫成：

msort ::List Int→ List Int

msort [ ] = [ ]

msort [x ] = [x ]
msort xs = merge (msort ys) (msort zs) ,

where (ys,zs) = (take (n ‘div‘ 2) xs,drop (n ‘div‘ 2) xs) .

要論證 msort會正常終止，或著說，要將 msort視為一個良基歸納定義，我們
可用和例 2.17中一樣的良基序 (◁).
但此處請讀者小心檢查：在msort xs子句中，ys◁ xs和 zs◁ xs有被滿足嗎？當

length xs = n,串列 ys與 zs的長度分別是 n ‘div‘ 2和 n−n ‘div‘ 2. 當 length xs = 1
時，ys與 zs的長度分別是… 0和 1 — zs並沒有變短！
這是為何我們需要 msort [x ]這個子句把 length xs = 1的情況分開處理。如

果沒有這個子句，msort將有可能不終止 —讀者不妨試試看。

例 2.19 (最大公因數). 歐幾里得 (Euclid)的《幾何原本》成書於西元前三百年，
其中描述「計算最大公因數」的做法可能是世界上最古老的演算法。以下函數

計算兩個自然數 (m,n)的最大公因數。如果兩數相等，它們的最大公因數也是
自身。若兩數不相等，其最大公因數會是「大數減小數」與「小數」的最大公

因數：

gcd :: (N×N)→ N
gcd (m,n) | m = = n = n

| otherwise = gcd ((m↑n)− (m↓n),m↓n) .

這個程式總會正常終止嗎？為什麼？

事實上，若 m或 n其中之一為 0, gcd (m,n)是不會終止的 —讀者不妨也試
試看。若 m, n均為正整數呢？首先我們先確立：如果初始的 m, n均為正整數，
gcd每次遞迴呼叫拿到的參數也都是正整數 —確實如此，因為如果 m>0, n>0,
且 m ̸= = n,那麼 (m↑n)− (m↓n)與 m↓n都不會是零或負數。接下來我們可論證：
如果 m, n均為正整數，每次遞迴呼叫中，兩參數的和都變小了一些。確實：

(m↑n)− (m↓n)+(m↓n)
= m↑n
< { m, n均為正整數 }

m+n .

因此，我們可得知 gcd在 m, n均為正整數時會正常終止。如果把 gcd當作一個
良基歸納，我們用了如下的良基序：

(m1,n1)◁ (m2,n2) ≡ m1 +n1 <m2 +n2 ,

其中 m1, n1, m2, n2 均為正整數。

例 2.20 (Curried函數). 下述函數 interleave將兩個參數中的元素交錯放置。例
如 interleave [1,2,3 ] [4,5 ] = [1,4,2,5,3 ].

interleave ::List a→ List a→ List a
interleave [ ] ys = ys
interleave xs [ ] = xs
interleave (x : xs) ys = x : interleave ys xs .

這可視為一個良基歸納定義嗎？若將 interleave做為傳回函數的高階函數看待，
我們比較難看出它是定義在什麼良基序上的。但若把 interleave的兩個參數一
起考慮，我們不難看出什麼度量在遞迴呼叫後「變小」了：兩個參數長度的和！

凡是遇到像 interleave的 curried函數，我們也可考慮它的 uncurried版本：

interleave′ :: (List a×List a)→ List a
interleave′ ([ ], ys) = ys
interleave′ (xs, [ ]) = xs
interleave′ (x : xs,ys) = x : interleave′ (ys,xs) .

函數 interleave′ 是個良基定義 —參數中的兩個串列雖然交換位置，但它們長度
的總和會變小。也就是說 interleave′可視為定義在這個良基序上的函數：

(xs1,ys1)◁ (xs2,ys2) ≡ length xs1 + length ys1 < length xs2 + length ys2 .

凡是 interleave′有的性質，不難找出 interleave的相對應版本；證明 interleave
的性質時，可當成是在證明 interleave′ 的相對性質。因此我們也會比較寬鬆地
說 interleave也是 (◁)之上的良基歸納定義。

例 2.21. 下列函數被稱作「McCarthy 91函數」：

mc91 ::N→ N
mc91 n | n>100 = n−10

| otherwise = mc91 (mc91 (n+11)) .

讀者不妨先猜猜看 mc91會傳回什麼？答案是，mc和以下函數是等價的：

mc91′ | n>100 = n−10
| otherwise = 91 .

[todo: finish this.]

2.13 詞典序歸納

我們終於要定義第 1.9節與 2.12節中都提到的「合併」函數：將兩個已排序好的
串列合而為一。函數 merge最自然的寫法可能是：

merge ::List Int→ List Int→ List Int

merge [ ] ys = ys
merge (x : xs) [ ] = x : xs
merge (x : xs) (y : ys) = if x ⩽ y then x : merge xs (y : ys)

else y : merge (x : xs) ys .

如果兩個串列之中有一個為空串列，合併的結果是另一個。如果兩個都不是空
串列，我們比較其第一個元素，以便決定將哪個當作合併後的第一個元素。
但，merge最後一個子句的第一個遞迴呼叫中，y : ys沒有變短；第二個遞

迴呼叫中，x : xs沒有變短。這種程式會終止嗎？如果會，是哪種歸納定義呢？
一種看法是將 merge視作和例 2.20中的 interleave類似的歸納定義：兩個參

數的長度和在遞迴呼叫中變小了。另一個可能是將函數 merge視作詞典序歸納
(lexicographic induction)的例子 —詞典序歸納也是良基歸納的一個特例。
先介紹詞典序。我們怎麼決定兩個英文單字在詞典中的先後順序呢？通常

是先比較其第一個字母，如果第一個字母便分出了大小，就以此大小為準，不
論剩下的字母為何。如果第一個字母一樣，便從第二個字母開始比起。若要
以形式化的方式寫下詞典序的定義，我們考慮一個較簡單的狀況：如何比較
x1 y1 和 x2 y2 兩個長度均為二的字串？如果 (<)是比較單一字元大小的順序，
我們把兩個字元的詞典序寫成 (<;<)，定義如下：

x1y1(<;<)x2y2 ≡ x1 < x2 ∨ (x1 = x2∧ y1 < y2) .

如前所述，先比較 x1 與 x2,如果相等，再比較 y1 與 y2.
我們可以再稍微擴充一些，考慮 xi 與 yi 型別不同的情況：

定義 2.22. 給定義在型別 A之上的序 (◁)和型別 B之上的序 (≺),它們的詞典
序 (lexicographic ordering)，寫做 (◁;≺)，是 (A×B)上的一個序，定義為：

(x1,y1)(◁;≺)(x2,y2) ≡ x1 ◁ x2 ∨ (x1 = x2∧ y1 ≺ y2) .

上述定義也可擴充到三個、四個… 元素的序對上。此處便不把他們寫出來了。
關於詞典序的有趣性質相當多，此處僅用到下述性質

定理 2.23. 如果 (◁)與 (≺)均為良基序，(◁;≺)也是良基序。

因此，(◁;≺)也可用來做歸納定義。我們把使用詞典序的良基序歸納稱作「詞
典序歸納」。
回頭看 merge的定義。先考慮下述、在第 1.9節中出現的 uncurried版本：

merge′ :: (List Int×List Int)→ List Int

merge′ ([ ], ys) = ys
merge′ (x : xs, [ ]) = x : xs
merge′ (x : xs,y : ys) = if x ⩽ y then x : merge′ (xs,y : ys)

else y : merge′ (x : xs,ys) .

如果 (◁)是比較串列長度的良基序，我們可說 merge′ 是在 (◁;◁)之上的歸納
定義。確實，

• (xs,y : ys) (◁;◁) (x : xs,y : ys),因為 xs ◁ x : xs;
• (x : xs,ys) (◁;◁) (x : xs,y : ys),因為 x : xs = x : xs且 ys ◁ y : ys.

至於 merge則是 merge′的 curried版本，因此也是定義良好的。
如前所述，函數 merge 的定義不一定得看成辭典序歸納 — 它也可和

interleave 一樣看成另一種較簡單的良基歸納 — 比較兩參數的長度之和。接
下來的例子就得倚靠辭典序歸納了。

例 2.24. 知名的 Ackermann函數 (Ackermann’s function)是一個遞增得相當快的
函數。

ack ::N→ N→ N
ack 0 n = 1+ n
ack (1+ m) 0 = ack m 1
ack (1+ m) (1+ n) = ack m (ack (1+ m) n) .

該函數定義上的特殊處之一是 ack (1+ m) n的結果又被當作 ack m的參數，因
此較難以用理解 interleave的方式理解。但它可以視為詞典序 (<;<)上的歸納：

• (m,1) (<;<) (1+ m,0)，因為 m<1+ m;
• (1+ m,n) (<;<) (1+ m,1+ n)，因為 n<1+ n;
• (m,ack (1+ m) n) (<;<) (1+ m,1+ n)，因為 m<1+ m.

2.14 交互歸納

許多工作無法由一個函數獨立完成，而需要許多函數彼此呼叫。本章最後談談
這類的交互歸納 (mutual induction)定義。下列函數定義中，even判斷其輸入
是否為偶數。第二個子句告訴我們：如果 n是奇數，1+ n便是偶數。但如何判
斷一個數字是否為奇數？如果 n是偶數，1+ n便是奇數：

even ::N→ Bool

even 0 = True

even (1+ n) = odd n ,

odd ::N→ Bool

odd 0 = False

odd (1+ n) = even n .

這類彼此呼叫的定義可以視為一整個大定義。為讓讀者習慣，我們先把 even
與 odd的定義改寫為 λ 算式與 case:

even = λn→ case n of
0 → True

1+ n→ odd n ,

odd = λn→ case n of
0 → False

1+ n→ even n .

上述的定義可以合併成一個：evenOdd 是一個序對，其中有兩個函數 N→
Bool，其中是 fst evenOdd就是 even，snd evenOdd就是 odd:

evenOdd :: ((N→ Bool)× (N→ Bool))

evenOdd = (λn→ case n of 0 → True

1+ n→ snd evenOdd n,
λn→ case n of 0 → False

1+ n→ fst evenOdd n) .

習題 2.44 — 證明 all (xs = =) (zipWith (++) (inits xs) (tails xs)). (unfinished)

2.15 參考資料

快速排序最初由 Hoare 在 Communications of the ACM 的演算法專欄中發表
為兩個獨立的演算法：將陣列分割為大、小兩塊的「演算法 63: PARTITION」
[Hoare, 1961b]，以及用前述演算法將陣列排序的「演算法 64: QUICKSORT」
[Hoare, 1961c]。至於「演算法 65: FIND」[Hoare, 1961a]的功能則是：給定 k，
尋找一個陣列中第 k小的元素。該演算法也使用了 PARTITION,我們現在常把
它稱為「快速選擇 (quickselect)」。該專欄要求作者使用 Algol語言。Algol支援
遞迴，Hoare也大方地用了遞迴。當時遞迴仍是新觀念，Hoare在另一篇論文
[Hoare, 1962]中（以大量文字）描述不用遞迴的作法。
例 2.17中的快速排序常被當作「函數編程（或 Haskell）漂亮又簡潔」的例

證：快速排序用其他語言得寫得落落長，用 Haskell 可在兩三行內清楚地寫
完。但這樣的比較並不很公平：例 2.17排序的是串列，而拿來比較的對象通常
是將陣列做排序的指令式語言程式。快速排序的主要挑戰之一，也是「演算法
63: PARTITION」的重點，是在 O(n)的時間、O(1)的額外空間之內完成陣列的
分塊。這點在串列版本中並沒有（或不須）表達出來。

Manna and McCarthy [1969] Manna and Pnueli [1970]

搜尋樹

本節以搜尋樹為例，示範樹狀結構上的歸納證明。本節於初次閱讀時可跳過。
許多應用需要將資料收集在某種表達「集合」的資料結構中。我們希望能

做的操作包括：尋找並判斷某筆資料是否在集合中、加入某筆新資料、刪除某
筆資料，等等。許多演算法都仰賴這些操作能有效率地被實作。
我們可以用串列表達一個集合，但在一般的簡單實作中，在串列中尋找與

刪除都需要與串列長度成正比的時間。若集合中的資料是能比大小的，使用一
個陣列並將元素排序好，可用二元搜尋法在對數時間內尋找某筆資料。但在陣
列中加入與刪除元素也都需要與陣列長度成正比的時間挪出或消除空位。
因此，許多針對此類應用的樹狀結構被發展出來。

3.1 二元搜尋樹

二元搜尋樹 (binary search tree)是一類樹狀資料結構的統稱。簡單的二元搜尋樹
是一種類似 ITree的結構（為說明方便，我們讓樹中的元素皆為 Int）：

data BTree= Nul | Nod BTree Int BTree ,

另有個附加限制：在所有內部節點 Nod t x u中，t之中的所有元素均小於 x, u
之中的所有元素均大於 x. 以下我們將此限制簡稱為「有序性」。給定一個有序
的二元搜尋樹 t，下述函數 search k t判斷 k是否出現在 t之中：

search :: Int→ BTree→ Bool

search k Nul = False

search k (Nod t x u) | k< x = search k t
| k = = x = True

| x< k = search k u .

93



DR
AF
T

94 CHAPTER 3. 搜尋樹

這是一個歸納定義。由於有序性，每遇到一個 Nod t x u節點，我們可比較 k與
x的大小，藉以決定往哪一個子樹做搜尋。
如果二元樹是平衡的 —意指在樹中的每個 Nod t x u之中，t和 u的元素數

目大致相等，或著至少其比例為某個常數，search k (Nod t x u)每次選擇往 t或
u搜尋時都會排除一定比例的元素。如此一來，搜尋一個含 n個元素的樹可在
O(logn)的時間內完成。然而，維持「平衡」並不容易，也是各種二元搜尋樹
各顯神通之處。
以下我們多給一些定義，以便更形式化地描述有序性。我們假設有個抽象

的「集合」型別 Set a. 含單一元素 x的集合寫成 {x};聯集寫作 (∪),並具有結合
律、交換律等性質。下述函數 elems傳回一個樹中的所有元素（類似練習 2.32
中的 tags）：

elems ::BTree→ Set Int

elems Nul = [ ]

elems (Nod t x u) = elems t∪{x}∪ elems u .

有序性可表示為下列述語 sorted:

sorted ::BTree→ Bool

sorted Nul = True

sorted (Nod t x u) = all (<x) (elems t) ∧ all (x<) (elems u) ∧
sorted t ∧ sorted u .

其中 all :: (a→ Bool)→ Set a→ Bool 檢查一個集合中的元素是否均滿足某述
語。函數 all滿足以下性質：

all p /0 = True

all p {x} = p x
all p (s∪ t) = all p s ∧ all p t .

在二元搜尋樹中插入新元素的函數可定義如下：

insert :: Int→ BTree→ BTree

insert k Nul = Nod Nul k Nul
insert k (Nod t x u) | k< x = Nod (insert k t) x u

| k = = x = Nod t x u
| x< k = Nod t x (insert k u) .

函數 insert和 search的結構很類似：都藉由比較 k與 x決定該往哪兒插入元素；
函數 search只在碰到 Nul時才能傳回 False,函數 insert k也只將 Nod Nul k Nul

放在原本 Nul出現之處 —新插入的節點永遠是最邊緣的葉節點。
我們自然會希望 insert保持有序性：如果 t是有序的，insert k t也應該是。

表示為定理如下：

定理 3.1. 對所有 t與 k, sorted t⇒ sorted (insert k t).

Proof. 在 t之上做歸納證明。基底情況 t :=Nul之中，sorted (Nod Nul k Nul)顯
然成立。歸納情況有 k< x, k = = x, x< k三種。以下我們只列出第一種。
狀況 t :=Nod t x u, k< x:

sorted (insert k (Nod t x u))
≡ { insert之定義, k< x }

sorted (Nod (insert k t) x u)
≡ { sorted之定義 }

all (<x) (elems (insert k t)) ∧ all (x<) (elems u) ∧
sorted (insert k t) ∧ sorted u
≡ {練習 3.1: elems (insert k t) = {k}∪ elems t }

all (<x) ({k}∪ elems t) ∧ all (x<) (elems u) ∧
sorted (insert k t) ∧ sorted u
≡ { all之性質 }

k< x ∧ all (<x) (elems t) ∧ all (x<) (elems u) ∧
sorted (insert k t) ∧ sorted u
⇐ { k< x,歸納假設 }

all (<x) (elems t) ∧ all (x<) (elems u) ∧
sorted t ∧ sorted u
≡ { sorted之定義 }

sorted (Nod t x u) .

函數 insert並不保證做出的樹能平衡 —恰恰相反，我們一不小心便會做出
偏一邊的樹。例如，insert 5 (insert 4 (insert 3 (insert 2 (insert 1 Nul))))得到的結
果會是：

Nod Nul 1 (Nod Nul 2 (Nod Nul 3 (Nod Nul 4 (Nod Nul 5 Nul)))) .

在這個樹中搜尋 5,和在串列 [1,2,3,4,5 ]之中尋找 5基本上是一樣的。
許多建立在二元搜尋樹的進階資料結構會在插入元素時多做些操作，想辦

法維持樹的平衡。我們將在第 3.2節介紹的紅黑樹就是一個例子。

習題 3.1 — 證明對所有 t與 k, elems (insert k t) = {k}∪ elems t.

3.2 紅黑樹

在二元搜尋樹的基礎上，紅黑樹 (red-black tree)多加了個屬性：每個節點都是
紅色或黑色之一。表示成 Haskell資料結構如下：

data RBTree= E | R RBTree Int RBTree

| B RBTree Int RBTree ,

其中 E為沒有資料的葉節點，R為紅色內部節點，B為黑色內部節點 —葉節點
E被視為黑色的。定義 data Color = Red | Blk,下述定義的 color t傳回 t的根部
節點的顏色：

color ::RBTree→ Color

color E = Blk

color (R ) = Red

color (B ) = Blk .

我們要求紅黑樹滿足下列性質：
1. 紅黑樹也是二元搜尋樹，意即它得是有序的。
2. 從根部開始到每個葉節點 E的路徑上的黑節點數目均相同。我們說這樣
的一棵樹是平衡的。

3. 紅節點的兩個子代都必須是黑色的，黑節點則無此限制。為方便說明，
我們把滿足此條件的樹稱為準紅黑樹。

4. 根節點為黑色的。
其中，關於有序性的討論和前一節原則上相同，此節將之省略。我們假設存在
某函數 sorted ::RBTree→ Bool判斷一棵紅黑樹是否有序。我們看看其他性質
如何形式化。
首先，函數 bheight定義一棵樹的「黑高度」—所有路徑上黑色節點的最多

數目：

bheight ::RBTree→ N
bheight E = 0
bheight (R t x u) = bheight t ↑bheight u
bheight (B t x u) = 1+(bheight t ↑bheight u) .

說一棵樹「平衡」意指每個內節點中，兩個子樹的黑高度均相等。

balanced ::RBTree→ Bool

balanced E = True

balanced (R t x u) = bheight t = = bheight u ∧ balanced t ∧ balanced u
balanced (B t x u) = bheight t = = bheight u ∧ balanced t ∧ balanced u .

函數 semiRB檢查一棵樹是否為準紅黑 —紅節點的兩棵子樹均為黑色：

semiRB ::RBTree→ Bool

semiRB E = True

semiRB (B t x u) = semiRB t ∧ semiRB u
semiRB (R t x u) = color t = = Blk ∧ color u = = Blk ∧ semiRB t ∧ semiRB u

最後，如前所述，紅黑樹需滿足 sorted, balanced, semiRB, 並且根節點須為黑
色：

redBlack ::RBTree→ Bool

redBlack t = sorted t ∧ balanced t ∧ semiRB t ∧ color t = = Blk .

在這樣一棵二元樹中，由於 balanced被滿足，每條路徑上的黑節點數目均
相同；由於 semiRB被滿足，紅節點不會連續出現。因此最長路徑之長度不超
過最短路徑的兩倍。在這樣的樹中做二元搜尋，總會在 O(logn)的時間內找到
資料或走到葉節點。

Figure 3.1: 在紅黑樹中插入新元素後做的四種旋轉。

3.2.1 紅黑樹插入

在紅黑樹中插入元素的方式最初和二元搜尋樹相同：一邊搜尋一邊往下走，如
果我們碰到 E,便是插入新元素之處。新元素總在邊緣被插入，而樹的有序性
仍保持著。
新加入的節點該是什麼顏色呢？讓新節點為紅色可能是個合理選擇。如此

一來，插入新元素不會改變一棵樹的黑高度，也因此如果該樹原本是平衡的，
加入新節點後的新樹仍是平衡的。
但這麼做可能破壞準紅黑性質：路徑上有可能出現兩個連續的紅節點。因

此我們在插入後的回程途中適時做旋轉，如圖 3.1所示。每當我們看到一個黑
節點之下有兩個連續的紅節點，必定是圖中四個角落的四種情形之一（我們只
插入了一個元素，最多只有一個多出的紅節點。圖中 s, t, u,與 v仍是平衡的準
紅黑樹）。我們將每個情形都旋轉成圖中央的情況。如此一來圖中央以 x, z為
根部的兩顆子樹都是黑根部的紅黑樹，y是紅節點。於是我們再往上檢查，如
果又出現四種情況之一就再旋轉，否則往上重建路徑，直到回到根部。如此做
出一棵滿足準紅黑性質的樹。最後，如果根部節點為紅色，便直接改成黑色。
我們看看前述的插入演算法可如何表示為 Haskell程式。我們將主要執行插

入與旋轉的函數稱為 ins :: Int→ RBTree→ RBTree. 函數 insert則在呼叫 ins之
後將樹根改為黑色的：

insert :: Int→ RBTree→ RBTree

insert k t = blacken (ins k t) ,

where blacken (R t x u) = B t x u
blacken t = t .

函數 ins在遇上紅節點（R t x u）時和第 3.1中的 insert很類似：比較 k與 x以
決定該往哪邊插入，並在歸納呼叫後以 R重做節點。但遇到黑節點 B t x u時，
我們不使用 B,而是額外呼叫 rotate函數：

紅黑樹旋轉只需四種情況

本節的的紅黑樹旋轉方式來自 Okasaki [1999]。熟悉資料結構的讀者可能發現它
們比一般教科書或網路資源中的處理方式簡單許多：一般資料中常會分出八種以
上的狀況，除了主要路徑上的三個節點，也會考慮其兄弟節點的顏色。

Okasaki發現只需本節的這四種情況就足夠了。讀者也應可發現，採用較簡單
的版本，對於證明以及暸解紅黑樹的性質幫助很大。
那麼，為何一般教科書會用上那麼多情況呢？Okasaki認為可能是效率考量。一

般書中的版本中，有些情況可不需旋轉，只直接改變節點顏色。如此一來需要更
動的欄位數目較少。有些情況中轉出的樹根部已是黑色的。在指令式語言中，此
時該程序就可以直接結束。
然而，在函數語言中，我們無論如何都需重建整個路徑上的節點。因此上述優

點均不明顯。另一方面，Okasaki也認為此種情況較少的版本是比較適合用於教
學中的。

ins :: Int→ RBTree→ RBTree

ins k E= R E k E
ins k (R t x u) | k< x = R (ins k t) x u

| k = = x = R t x u
| k> x = R t x (ins k u)

ins k (B t x u) | k< x = rotate (ins k t) x u
| k = = x = B t x u
| k> x = rotate t x (ins k u) .

回顧：圖 3.1的四種旋轉都只在根節點為黑色時啟動，因此我們也只在碰上黑
節點時呼叫 rotate。在 rotate s x t之中，s為目前的左子樹，x為目前的（黑色）
節點中的標籤，t則為目前的右子樹。函數 rotate的定義如下：

rotate ::RBTree a→ a→ RBTree a→ RBTree a
rotate (R (R s x t) y u) z v = R (B s x t) y (B u z v)
rotate (R s x (R t y u)) z v = R (B s x t) y (B u z v)
rotate s x (R (R t y u) z v) = R (B s x t) y (B u z v)
rotate s x (R t y (R u z v)) = R (B s x t) y (B u z v)
rotate s x t = B s x t .

其中，前四個情況的左邊分別對應到圖 3.1的四個情況；他們的右手邊都是一
樣的，對應到圖 3.1正中央的樹。最後的 rotate s x t則是四種情況之外、不用旋
轉的情形。

3.2.2 紅黑樹之性質：高度

許多討論紅黑樹的教材在將插入、刪除等等操作的實作呈現讀者看過之後就結
束了，對於其性質的討論意外地不完整。然而，這類資料結構之所以有效，正
因為它們需有的性質一直被保持著。談資料結構上的操作卻不證明它們怎麼維
護資料結構的性質，可說是缺了最重要的一塊。本章剩下的篇幅中，我們將描
述並證明紅黑樹的一些主要性質。

首先我們談談黑高度。讀者稍加嘗試之後會發現，insert k t有時會增加 t的
黑高度，有時不會。我們怎知道紅黑樹何時會長高呢？
原來，函數 ins其實是不會讓樹長高的！我們有如下的定理 — ins k前後樹

的黑高度不變：

定理 3.2. 對所有 k與 t, bheight (ins k t) = bheight t.

函數 insert k t呼叫 ins k t,得到的樹仍有原來的黑高度。如果 blacken把樹由紅
轉黑，新樹的黑高度才因此加一。否則樹的黑高度仍不變。

系理 3.3. 對所有 k與 t，如果 ins k t為黑色，bheight (insert k t) = bheight t. 否則
bheight (insert k t) = 1+bheight t.

我們將嘗試證明定理 3.2. 回顧我們的原則：證明的結構依循程式的結構。
由於 insert呼叫 ins,欲證明關於 insert的系理 3.3，我們需要關於 ins的定理 3.2.
同樣地，由於 ins呼叫 rotate,欲證明定理 3.2,我們需要一個關於 rotate的引裡：

引理 3.4. 對所有 t, u與 z, bheight (rotate t z u) = 1+(bheight t ↑bheight u).

這也不意外：rotate的兩個參數 t與 u原本在黑節點之下，旋轉後的黑高度
不變，仍是 1+(bheight t ↑bheight u).
以下我們證明定理 3.2.

Proof. 在 t之上做歸納。以下只列出幾個代表性狀況。
狀況 t :=E.

bheight (ins k E)
= bheight (R E k E)
= 0
= bheight E .

狀況 t :=R t x u, k< x:

bheight (ins k (R t x u))
= { ins之定義；k< x }

bheight (R (ins k t) x u)
= { bheight之定義 }

bheight (ins k t)↑bheight u
= {歸納假設 }

bheight t ↑bheight u
= { bheight之定義 }

bheight (R t x u) .

狀況 t :=B t x u, k< x:

bheight (ins k (B t x u))
= { ins之定義；k< x }

bheight (balance (ins k t) x u)

= {引理 3.4 }
1+(bheight (ins k t)↑bheight u)

= {歸納假設 }
1+(bheight t ↑bheight u)

= { bheight之定義 }
bheight (B t x u) .

為求完整，我們也簡述引理 3.4之證明：

Proof. 由於 rotate沒有遞迴、不呼叫其他函數，但有許多狀況，關於 rotate的
證明也大都僅是檢查每個情況，雖不難但可能很繁瑣。以下只舉一種狀況為
例。
狀況 (t,z,u) :=(R (R t x u) y v,z,w):

bheight (rotate (R (R t x u) y v) z w)
= { rotate之定義 }

bheight (R (B t x u) y (B v z w))
= { bheight之定義 }
(1+(bheight t ↑bheight u))↑ (1+(bheight v↑bheight w))

= {由於 (k+ x)↑ (k+ y) = k+(x↑ y), (↑)有結合律 }
1+(((bheight t ↑bheight u)↑bheight v)↑bheight w)

= { bheight之定義 }
1+(bheight ((R (R t x u) y v))↑bheight w)

3.2.3 紅黑樹之性質：平衡

之所以討論黑高度，目的之一當然是要討論平衡。我們可證明函數 ins k維持
平衡性：

定理 3.5. 對所有 k與 t, balanced t⇒ balanced (ins k t).

而由於 balanced t⇒ balanced (blacken t),定理 3.5蘊含 insert k也維持輸入樹的
平衡：balanced t⇒ balanced (insert k t).
同樣地，要證明定理 3.5，我們也需要一個關於 rotate的引理：

引理 3.6. 對所有 t與 u,

balanced t ∧ balanced u ∧
bheight t = bheight u⇒ balanced (rotate t x u) .

由於這些證明與前面提到的證明類似，我們將它們留給讀者做練習。

習題 3.2 — 證明定理 3.5. 你將用得上引理 3.6與定理 3.2.
習題 3.3 — 證明引理 3.6.

3.2.4 紅黑樹之性質：顏色

最後，我們談談紅黑樹插入之後的顏色。我們也許希望函數 ins k能保持準紅
黑性，意即 semiRB t⇒ semiRB (ins k t). 但這顯然不成立：我們已經知道 ins k
可能在一個路徑上產生連續兩個紅節點。
為描述此時的特殊狀況，我們另外定義一個性質：滿足下列述語的樹被稱

作紅外 (infrared)樹 —取自比紅色還紅一點之意：

infrared ::RBTree→ Bool

infrared (R t x u) = (color t = = Blk ∨ color u = = Blk) ∧
semiRB t ∧ semiRB u

infrared t = False .

紅外樹幾乎是一棵根部為紅色的準紅黑樹，但兩個子樹 t與 u之中最多有一個
可以是紅色！我們將其表示成 color t = = Blk | color u = = Blk. 此外，t與 u仍必須是
準紅黑樹。其他的情形（E或是 B ）是黑色的，都不是紅外樹。
那麼，我們是否能證明：給定 ins k準紅黑樹 t，ins k t總是一個紅外樹，意

即 semiRB t⇒ infrared (ins k t)？對歸納證明熟悉的讀者可能立刻覺得事有蹊
蹺：有二就有三，證明歸納狀況時，如果歸納假設中的子樹有兩個連續的紅節
點，在歸納狀況中可能看到三個連續紅節點。如此一來似乎沒完沒了。
這是一個必須嘗試證明一個更強的性質才能使歸納證明成立的例子。函數

ins真正滿足的是一個更強的性質：給定準紅黑樹 t,如果 t是紅色，ins k t則是
一棵紅外樹；如果 t是黑色，ins k t也將是一棵準紅黑樹：

定理 3.7. 對所有 t:
1. semiRB t ∧ color t = Red⇒ infrared (ins k t),
2. semiRB t ∧ color t = Blk⇒ semiRB (ins k t).

為證明定理 3.7，我們也需要一個與 rotate相關的引理：只要 t與 u之中有一個
是準紅黑樹，另一個是準紅黑樹或紅外樹，rotate t x u就會是準紅黑樹：

引理 3.8. 對所有 t與 u,
1. (infrared t ∨ semiRB t) ∧ semiRB u⇒ semiRB (rotate t x u);
2. semiRB t ∧ (infrared u ∨ semiRB u)⇒ semiRB (rotate t x u).

有了定理 3.7，由於 infrared t⇒ semiRB (blacken t)，我們立刻得知 insert k保持
準紅黑性：

系理 3.9. semiRB t⇒ semiRB (insert k t).

以下證明定理 3.7:

Proof. 定理 3.7的 1, 2兩個小性質需在同一個歸納中證明。注意：1與 2的合取
蘊含了

semiRB t⇒ (infrared (ins k t) ∨ semiRB (ins k t)) . (3.1)

我們將用到此性質。
在 t之上做歸納。我們只舉出兩個具代表性的例子：

狀況: t :=B t x u, k< x:

semiRB (ins k (B t x u))
= { ins之定義, k< x }

semiRB (rotate (ins k t) x u)
⇐ {引理 3.8 }
(infrared (ins k t) ∨ semiRB (ins k t)) ∧ semiRB u
⇐ {歸納假設，(3.1) }

semiRB t ∧ semiRB u
= { semiRB之定義 }

semiRB (B t x u) .

狀況: t :=R t x u, k< x:

infrared (ins k (R t x u))
= { ins之定義 }

infrared (R (ins k t) x u)
= { infrared之定義 }
(color (ins k t) = Blk ∨ color u = Blk) ∧ semiRB (ins k t) ∧ semiRB u

= {歸納假設 }
(color (ins k t) = Blk ∨ color u = Blk) ∧ semiRB t ∧ color t = Blk ∧ semiRB u
⇐ {命題邏輯: ((P ∨ Q) ∧ R)⇐ (Q ∧ R) }

color t = color u = Blk ∧ semiRB t ∧ semiRB u
= { semiRB之定義 }

semiRB (R t x u) .

習題 3.4 — 證明引理 3.8.

關於語意的基本概念

第 1章介紹了函數語言的基礎概念，第 2章談了歸納定義與證明。我們將運用
這些知識在第 5章之中推導一些程式、解一些編程問題。但在那之前，我們得
暫緩一下，釐清一些和語意相關的概念。
語意 (semantics)一詞由語言學中借用而來。語意學是關於字句、詞語的意

思的研究。一段程式的「意思」為何？在電腦技術發展的早期這似乎不是個問
題。當時，「程式語言」只是屬於某台特定型號電腦的指令集，指令的意思就
如同其操作手冊所說。若有疑義，用電腦執行看看就知道了。隨著技術發展，
同一個程式語言可在不同電腦上執行，各廠商、研究機構都可以實作同一語言
的編譯器。同時，實用性質的程式語言也發展得更加複雜，一個程式語言中常
有不太能憑直覺理解的細微處。我們因而需要有個不依賴特定硬體、特定編譯
器，也可討論程式語言的「意思」的方法。不僅可作為各家實作程式語言的依
據，也方便大家溝通：每當我們設計新的語言、符號，我們也常需要釐清它的
語意是什麼。
描述語意的方式有許多種。本書目前為止其實在不知不覺中混用了兩種語

意：指稱語意談一個程式是什麼，操作語意談一個程式做什麼。以下我們以非
常粗略的方式介紹它們。

4.1 指稱語意

指稱語意 (denotational semantics)談一個程式是什麼。在我們的討論範圍中，
有堅實基礎、有不含糊的定義的東西只有數學物件。因此，指稱語意試圖把程
式語言對應到數學物件。最簡單的指稱語意可能是基於集合論的：把型別視
為集合，把程式語言中的函數視為集合論中的函數。例如，N便是自然數的集

合；(A×B)是 A與 B的笛卡兒積：1

(A×B) = {(a,b) | a ∈ A,b ∈ B} .

但函數又是什麼呢？集合論中，一個型別為 A→ B的函數可視為 A×B的一個
子集；若 f x = y，該子集中便有 (x,y)這個元素。例如，double ::N→N可表示
成如下的集合：

{(0,0),(1,1),(2,4),(3,6),(4,8) . . .} .

能稱為函數的集合還需滿足兩個額外條件：
• 簡單性 (simplicity): 對所有 x ∈ A,該集合中僅存在一對唯一的 (x,y)。意即
每個輸入只對應到一個輸出。

• 完整性 (totality): 對所有 x ∈ A,在該集合中都存在某對 (x,y)。意即值域中
的每個元素都被涵蓋到。

至於遞迴函數呢？以階層 fact為例，其定義為：

fact ::N→ N
fact 0 = 1
fact (1+ n) = (1+ n)× fact n .

我們可想成：有一個從集合到集合的函數 factF，

factF X = {(0,1)}∪{(1+ n,(1+ n)×m) | (n,m) ∈ X} .

給任何集合 X，factF傳回這樣的集合：
• 新集合中有 (0,1) —這對應到 fact 0 = 1.
• 對 X 之中的每一個 (n,m), 新集合中有 (1+ n,(1+ n)×m) — 這對應到

fact (1+ n) = (1+ n)× fact n.
而函數 fact 的語意就是 factF 唯一的定點 (fixed point). 意即 fact 是唯一滿足
fact = factF fact的集合。關於定點的較完整理論可參照第 2.9節。確實，fact可
寫成集合如下：

fact = {(0,1),(1,1),(2,2),(3,6),(4,24)...} .

若將 fact餵給 factF,

factF fact
= {(0,1)}∪{(1+ n,(1+ n)×m) | (n,m) ∈ {(0,1),(1,1),(2,2),(3,6)...}}
= {(0,1)}∪{(1,1),(2,2),(3,6),(4,24)...} ,

我們又得到了 fact. 因此 fact確實是 factF的定點。
但，要把「fact便是 factF的定點」當作 fact的定義，我們還得確定：確實

只有這麼一個集合滿足 fact = factF fact。讀者稍加檢查一下，即可發現確實如
此 —例如，把 fact添一項或刪一項，都會使得 factF fact不等於 fact.

1在更嚴謹的說法中，我們通常用一個語意函數 J_K將語法對應到其意義。因此我們會說JA×BK = JAK× JBK. 此處採用較不正式的說法。

並不是所有從集合到集合的函數都有唯一的定點。考慮如下兩個函數：

g 0 = 0
g x = g (1+ x) ,

h 0 = 0
h x = (−1)×h x .

若以類似 fact的方式試圖將它們寫成某函數的定點，我們可能寫出如下的 gF
和 hF:

gF X= {(0,0)}∪{(n,m) | (1+ n,m) ∈ X} ,

hF X= {(n,(−1×m)) | (n,m) ∈ X} .

但 gF的定點有

{(0,0),(1,1),(2,1),(3,1)...} ,

{(0,0),(1,2),(2,2),(3,2)...} ,

{(0,0),(1,10),(2,10),(3,10)...} ...

等無限多個。而 hF 有一個唯一定點：空集合，但空集合違反了前述的完整性
要求。
如同 g和 f 這樣的函數，在我們目前介紹的簡單指稱語意中是沒有定義的。

它們沒有語意，在我們的語言中是不該有的存在。它們剛好也是執行起來不會
終止的函數。因此，我們似乎可以猜測「有唯一定點」和「執行會終止」似乎
有些關聯。但，如前所述，我們的指稱語意中沒有「執行」的觀念。那是操作
語意擅長描述的。
值得一提地，指稱語意採取了一個靜態的世界觀。一個函數單純地是輸入

與輸出的對應，沒有「執行」的觀念，沒有執行快慢的考量，也沒有預設「終
止」的概念。
函數 double1 x = x+ x與 double2 x = 2× x使用不同的演算法，但它們的語

意都是同一個集合，因此被視為同一個函數 —在語意上我們無法區分 double1

與 double2。
在指稱語意中，當我們說兩個函數相等，意指它們的語意是同一個集合。

例如，map 1+ · concat = concat ·map (map (1+)),因為兩者的語意都是下述的集
合：

{([ ], [ ]), ...([[1 ], [2 ]], [2,3 ]), ...([[1 ], [2,3 ]], [2,3,4 ]), ...} .

4.2 操作語意

操作語意 (operational semantics)談一個程式做什麼。在操作語意中，我們通常
不談符號的「意思」是什麼。符號 0只代表它本身，1+ 0也只代表它本身。操
作語意著重的是一串符號如何變成另一段符號。在這個意義下，前述的 fact定
義：

fact ::N→ N
fact 0 = 1
fact (1+ n) = (1+ n)× fact n ,

從操作語意的觀點可被讀解成覆寫規則：看到 fact 0, 均可改寫為 1; 看到
fact (1+ n),均可改寫為 (1+ n)× fact n. 歸約 fact (1+ (1+ 0))這個式子可被視為
是不斷使用這兩條覆寫規則：

fact (1+ (1+ 0))
= (1+ (1+ 0))× fact (1+ 0)
= (1+ (1+ 0))× (1+ 0)× fact 0
= (1+ (1+ 0))× (1+ 0)×1 ,

其中的每個等號都可讀解為「改寫成」。
操作語意中較容易談「執行」與「終止」的概念。改寫一個式子相當於執行

它，如果已經沒有可改寫之處，也就是碰到了範式，執行便終止了。在操作語
意中，兩個數值（例如 4+ 4與 2× 4）相等意謂它們可被歸約成同一個範式；
兩個函數 f 和 g相等則意謂對於任何 x, f x與 g x都相等。
有了兩種談語意的方式，我們自然希望它們有些一致性。確實，我們有如

下的定理：

定理 4.1. 操作語意保持指稱語意。意即，給定一個有指稱語意的算式 e,若算
式 e能在零步或多步之內依據操作語意改寫成 e′,則 e與 e′的指稱語意相同。

定理 4.2. 給定一個遞迴的函數定義 f = F f。若在操作語意中，該函數對所有輸
入都正常終止，則在指稱語意中，f 是 F的唯一定點。

我們在下一節會

一般程式推導

程式推導 (program derivation)是本書的重要主題。給定一個待解決的問題，
並假設該問題能描述成邏輯、數學、或其他形式語言（稱作一個形式規格
(specification)），程式推導泛指以嚴謹方式將該規格轉換成一個解決該問題的
程式的方法。指令式語言與函數語言都能做程式推導，而且有許多道理可相
通。本書討論函數語言的程式推導。一個簡單、典型的函數程式語言推導可能
有如下的形式：

spec
= {性質 1 }

e1

= {性質 2 }
....

= {性質 n }
en .

其中 spec是問題的規格，通常也是一個函數語言程式。根據性質 1, spec等於
e1；根據性質 2，e1等於 e2… 長此以往，直到我們導出一個符合我們要求的 en.
光這麼看來，這和我們前幾章作的數學證明似乎沒什麼不同。差別在於，做證
明時我們已有 spec和 en，要做的是把 spec到 en 之間的步驟補足。但做程式推
導時，我們是從 spec出發，希望透過種種跡象找出一個滿意的（可能是夠快
的、不佔空間的、或滿足某些其他性質的）en. 當然，找出 en 之後，從 spec推
出 en 的過程就成了 spec = en 的證明。當程式推導的進行方式類似數學演算，
有人可能稱之為程式演算 (program calculation)。本書混用這兩個詞彙，並不區
分。
為何做程式推導？第一個理由是我們希望程式正確。此處「正確」指的是

en確實滿足了最初的規格 spec。但如果只為了正確性，我們為何不能先無論如

107



DR
AF
T

108 CHAPTER 5. 一般程式推導

何把 en 寫出，再試著證明 spec = en？原因之一是通常程式寫好了，大家便不
想證它了。更重要的是：程式開發時，若沒有把「怎麼證明它」列入考量，寫
出的程式常常是很難證明的。為了確保有證明，最好讓「產生證明」這件事成
為程式開發過程的一部分，甚至讓證明引導程式的開發。Dijkstra說道：

先給一個程式再證明它，在某個意義上是把馬車放在馬前面。更有

希望的做法是讓正確性證明與程式一起長出來：如此一來，我們能

選擇證明的結構，然後設計一個能用這種方法證明出來的程式。這

還有個額外的好處：正確性考量可以成為程式該怎麼寫的引導與啟

發。[Dijkstra, 1974]

Dijkstra的最後一句話帶到了程式推導的第三個理由：提倡者認為，程式推
導的種種方法與技巧可以引導我們去分析、思考、解決問題；這是解問題、甚
至發現新演算法的方法（Backhouse [2003], Backhouse [2011]）。藉由良好的符
號設計，我們可以發現常見的程式推導模式，並將其推廣到其他有類似性質的
問題上。
第 0章的結尾說道「函數語言的價值便是：它是個適於演算的語言。」便於

做程式推導，是我覺得函數語言最突出的特質。我們可以由規格出發、在符號
上操作，將程式如同求代數的解一樣地算出來。只要每個小步驟都正確，最後
的程式就是正確的。由第 1章至今，我們做了許多準備工作，備齊需要的基本
知識。現在我們終於可以開始做一些演算了。

5.1 展開-收回轉換

考慮這麼一個例子：給定一個整數形成的串列，我們想計算其每個數的平方的
和。例如當輸入是 [2,6,5,3 ],我們希望算出 22 + 62 + 52 + 32 = 74. 這項工作可
以簡短地描述如下：

sumsq = sum ·map square . (5.1)

函數 sumsq的型別為 List Int→ Int. 其中 map square將輸入串列的每個元素
都平方，然後由 sum計算其總和。第 2.4節中給過一個 sum的歸納定義，重複
如下：

sum ::List Int→ Int

sum [ ] = 0
sum (x : xs) = x+ sum xs .

消除中間串列 對大部分的應用而言，如上定義的 sumsq已經很堪用了。但作
為一個例子，我們來挑惕些仍不滿意之處。執行 sumsq xs時，map square會產
生另一個（存放 xs每個元素的平方的）串列，該串列隨即由 sum消掉 —感覺
上似乎很浪費空間與時間。1 有不產生這個中間串列的方法嗎？

1在惰性求值的情況下，該中間串列的每個節點被產生後立刻被 sum消去，因此不會真的佔用
和 xs同樣長度的空間。這也是大家覺得惰性求值有助於模組化、使小函數變得易於重用的例子之
一。但「產生一個新節點、立刻消去」仍耗了一些不必要的時間。

我們試著做些計算。當 sumsq的參數是 [ ]時：

sumsq [ ]

= { sumsq之定義 }
sum (map square [ ])

= { map之定義 }
sum [ ]

= { sum之定義 }
0 .

由此我們得知 sumsq [ ]會被計算成 0. 當輸入不是空串列時呢？試計算：

sumsq (x : xs)
= { sumsq之定義 }

sum (map square (x : xs))
= { map之定義 }

sum (square x : map square xs)
= { sum之定義 }

square x+ sum (map square xs)
= { sumsq之定義 }

square x+ sumsq xs .

可得知對任何 x與 xs, sum (x : xs)歸約的結果和 square x+ sumsq xs會是相同的。
總結說來，藉由計算，我們發現 sumsq滿足以下兩條性質：

sumsq [ ] = 0

sumsq (x : xs) = square x+ sumsq xs .
(5.2)

但如果我們翻轉過來，把這兩條性質當作 sumsq的新定義呢？這是一個依
據歸納法定義的良好程式，而且不會產生中間串列！
但我們怎知道 sumsq的新定義滿足我們最初的要求 (5.1)，即 sumsq = sum ·

map square呢？回顧起來，我們的演算只證明了當 sumsq滿足 (5.1)，它也滿足
(5.2)，卻還不能據此宣稱另一個方向：若 (5.2)成立，(5.1)也成立。我們將在第
4章詳細討論這個問題。目前可暫時這麼說：如果我們從某個問題規格 f = e起
始，發現 f 滿足某一組等式，而這些等式剛好可湊成一個會正常終止的歸納定
義，則 e確實是唯一滿足這些等式的解。由於如此的 e是唯一的，我們也可倒
過來以這組等式為 f 的定義，並同時宣稱 f = e這個性質成立。
回顧起來，在 sumsq的計算中，我們僅是把 sumsq的定義展開，接著展開

map, sum等等元件的定義，直到我們又看到 sumsq的定義出現在式子中、剛好
可以收回為止。這是一種單純而歷史悠久的程式推導方法，稱作展開-收回轉
換 (fold-unfold transformation)。雖然簡單，有時這個方法意外地有用。
在這個例子中，我們真正做到的是將一個單行、使用全麥編程的 sumsq定

義轉換成了一個歸納定義。新定義的 sumsq比起原版稍有效率些，但這只是恰
巧發生的 —歸納定義的程式不見得總會比較有效率。程式推導確保程式的正
確性 —意即導出的程式與原本的規格是同一個函數。但新程式的效率仍須單
獨分析。這是下一節的主題。

習題 5.1 — 下述函數 descend n傳回 [n,n−1,n−2 ...0 ]:

descend ::N→ List N
descend 0 = [ ]

descend (1+ n) = 1+ n : descend n .

定義 sumseries = sum ·descend. 請找出 sumseries的歸納定義。

習題 5.2 — 承接習題 5.1。函數 repeatN :: (N×a)→ List a的定義為

repeatN (n,x) = map (const x) (descend n) .

因此，repeatN (n,x)會傳回一個含 n個 x的串列。例如 repeatN (3,'a') =
"aaa". 請算出一個歸納定義的 repeatN.

習題 5.3 — 承接習題 5.2。遊程編碼 (run-length encoding)是一種簡單的壓
縮方式：將字串中重複的字元表達成其出現的數字。例如 "aaabbbbcdd"
可以表達為 "3a4b1c2d". 下列函數 rld ::List (N×a)→ List a則是抽象過的
「遊程解碼」，將已經表示成（次數 ×字元）的壓縮文展開：

rld = concat ·map repeatN .

例如，rld [(2,'a'),(3,'b'),(1,'c')] = "aabbbc". 請導出 rld的歸納定義。

習題 5.4 — 下列函數 delete將輸入串列中的每個元素輪流刪除：

delete :: List a→ List (List a)
delete [ ] = [ ]

delete (x : xs) = xs : map (x:) (delete xs) ,

例如，delete [1,2,3,4 ] = [[2,3,4 ], [1,3,4 ], [1,2,4 ], [1,2,3 ]]. 函數 select ::
List a → List (a × List a) 則將一個串列中的元素依次選出。例如，
select [1,2,3,4 ] = [(1, [2,3,4 ]),(2, [1,3,4 ]),(3, [1,2,4 ]),(4, [1,2,3 ])]. 函數
select恰巧可用 delete定義出來：

select xs = zip xs (delete xs) .

請 推 導 出 select 的 歸 納 定 義。 提 示： 下 述 性 質 可 能 有 用 —
zip xs (map f ys) = map (id× f ) (zip xs ys).

習題 5.5 — 函數 delete有另一個可能定義：

delete xs = map (del xs) [0 . . length xs−1 ]
where del xs i = take i xs++drop (1+ i) xs ,

（此處我們利用了當 n為負數時，[0 . .n ]化簡成 [ ]的特性。）請用此定義推
導出 delete的歸納定義。提示：你可能用得上下述性質：

[0 . .n ] = 0 : map (1+) [0 . .n−1 ], if n ⩾ 0, (5.3)

5.2 關於執行效率

有點意外地，本書直到現在才較正式地談執行效率。在本書中，我們假設數
字四則運算、邏輯運算元等等都可在常數時間內完成。資料結構方面，使用
資料建構元或對其做樣式配對都只需要常數時間。例如，將 x與 y做成 (x,y)
是常數時間內可完成的動作；一個型別是序對的值如果已經是弱首範式，將
其配對成 (x,y)以取出其中的 x和 y也只需要常數時間。如果有資料結構定義
data T= A | B,給一個已經是弱首範式的 x ::T,只需常數時間便可判斷它究竟是
A還是 B.
回顧 List的定義：

data List a = [ ] | a :List a .

我們可看出這是一個偏一邊的表示法。對於串列的左邊的操作都可以在常數時
間內完成。因此，我們可在常數時間內判斷一個串列 xs究竟是 [ ]還是可分解
成頭和尾；產生 [ ]只需要常數時間；在 xs的左邊添加一個元素，傳回 x : xs，
也是常數時間內可完成的動作。但如果我們要拿出某串列最右邊的元素，或著
在串列的右邊加東西呢？回顧 (++)的定義：

(++) ::List a→ List a→ List a
[ ] ++ ys = ys
(x : xs)++ ys = x : (xs++ ys) .

若我們試著看看 [1,2,3 ]++[4,5 ]是怎麼被算出來的

(1 : 2 : 3 : [ ])++(4 : 5 : [ ])
= 1 : ((2 : 3 : [ ])++(4 : 5 : [ ]))
= 1 : 2 : ((3 : [ ])++(4 : 5 : [ ]))
= 1 : 2 : 3 : ([ ]++(4 : 5 : [ ]))
= 1 : 2 : 3 : 4 : 5 : [ ] ,

可發現 (++)需把第一個參數從頭到尾走過一遍。因此，若第一個參數的長度
是 n，(++)是一個時間複雜度為 O(n)的函數。函數 last的情況也類似。

習題 5.6 — 回顧 last的定義，試著展開 last [1,2,3,4 ]並確認 last xs需要的
時間是否為 O(length xs)。

諸如 sum, length, maximum之類的函數將串列從頭到尾走過一次。當輸入串列
長度為 n,它們均需時 O(n)。如果函數 f 需時 O(t)，map f 需時 O(t× n). 函數
filter p, takeWhile p等等在最壞情況下需將串列走完，因此它們也是需要線性時
間的函數。函數 zip需要的時間與兩個串列中較短者成正比。
在習題 2.5中，我們曾請讀者定義一個函數 reverse ::List a→ List a，將輸入

的串列反轉，例如 reverse [1,2,3,4,5 ] = [5,4,3,2,1 ]. 一個可能的答案如下：

reverse [ ] = [ ]

reverse (x : xs) = reverse xs++[x ] .

這個程式的效率如何呢？我們看看 reverse [1,2,3,4 ]如何被歸約：

reverse [1,2,3,4 ]
= reverse [2,3,4 ]++[1 ]
= (reverse [3,4 ]++[2 ])++[1 ]
= ((reverse [4 ]++[3 ])++[2 ])++[1 ]
= (((reverse [ ]++[4 ])++[3 ])++[2 ])++[1 ]
= ((([ ]++[4 ])++[3 ])++[2 ])++[1 ] .

為了把 [1 ] 接在左邊，(++[1 ]) 需要走過一個長度為 3 的串列。而在那之前，
(++[2 ])需要走過一個長度為 2的串列。推廣說來，要反轉一個長度為 n的串
列，(++)會被使用O(n)次。每個 (++)左邊的串列長度也是O(n)，因此 reverse
是一個需時 O(n2)的演算法！「反轉串列」這個看來很基本的操作竟需要 O(n2)

的時間，似乎令人難以接受。是否有更快的做法呢？我們將在第 5.6 節討論
到。

習題 5.7 — 回顧第 1.10與 2.8節中提及的外標籤二元樹：

data ETree a = Tip a | Bin (ETree a) (ETree a) .

以下函數傳回樹中所有的標籤：

tips ::ETree a→ List a
tips (Tip x) = [x ]
tips (Bin t u) = tips t++ tips u .

函數 tips最壞情況的時間複雜度為何？請做出一個含有 n個標籤的樹 t，
使得 tips t僅需要 O(n)的時間算完；也請做出一個含有 n個標籤的樹 u，
使得 tips u需要 O(n2)的時間。

5.3 用展開-收回轉換增進效率

在前面的例子中，我們手動推導出的 sumsq只比原來的版本快了一點點，並沒
有複雜度上的改進。本節我們來看一些使用程式推導改進複雜度的例子。

5.3.1 計算多項式 – Horner法則

給定整數串列 as = [a0,a1,a2 . . .an]以及 x :: Int,我們想計算如下的多項式：

a0 +a1x+a2x2 + ...+anxn .

這問題的規格能清楚寫成：

poly x as = sum (zipWith (×) as (iterate (×x) 1)) ,

其中 iterate (×x) 1產生無限串列 [1,x,x2,x3...], zipWith計算 [a0,a1x, . . .anxn], sum
計算總和。
讀者應已對 sum 和 zipWith 很熟悉了。函數 iterate 在第 1.8.2 節中使用過，

iterate f x 會展開為無限長的串列 [x, f x, f (f x)... ]，每個元素分別是把 f 使用
0次、1次、2次… 的結果。函數 iterate可定義為

iterate :: (a→ a)→ a→ List a
iterate f x = x : map f (iterate f x) .

我們可將之理解成：iterate f x 的第一個元素是 x；剩下的元素呢？是把
iterate f x本身拿來，對每個元素多做一次 f !
讀到此的讀者可能有些疑問：這是一個合法的歸納定義嗎？以及，我們原

已說定不談無限的資料結構，何以在此卻出現了呢？
上述 iterate的定義方式確實不是歸納，而是「餘歸納」(coinduction)的一個

例子。第 2.9節中曾提及，餘歸納與歸納互為對偶，以餘歸納定義出的資料結
構稱作「餘資料」，可以是能無限地展開的。「餘串列」和歸納定義的串列應
該視為不同的型別，但它們可共存於同一個程式中，只要我們確定不在餘資料
上做歸納定義或證明。
為培養一些對 iterate的直覺，我們試著展開它：

iterate f x
= { iterate之定義 }

x : map f (iterate f x)
= { map之定義 }

x : map f (x : map f (iterate f x))
= { map融合 }

x : f x : map (f · f ) (iterate f x)
= { iterate與 map之定義 }

x : f x : f (f x) : map (f · f ) (map f (iterate f x))
= { map融合 }

x : f x : f (f x) : map (f · f · f ) (iterate f x)

可發現越展開，式子中便累積越多個 map f .
在 poly之中，iterate雖產生無限長的餘資料，但立刻被 zipWith截短了。2

若我們試著展開 poly x [a,b,c,d ],會得到：

poly x [a,b,c,d ]
= sum (zipWith (×) [a,b,c,d ] (iterate (×x) 1))
= {展開 iterate,將「f自我組合四次」記為 f 4 }

sum (zipWith (×) [a,b,c,d ]
(1 : (1× x) : (1× x× x) : (1× x× x× x) : map (×x)4 (iterate (×x) 1)))

= a + b× x + c× x× x + d× x× x× x .

可看到式子越長，便累積越多個 (×x)。當 as長度為 n,需要的乘法數目為
O(n2). 我們有可能降低做乘法的次數嗎？
我們試著找出 poly在 as上的歸納定義。當 as :=[ ]時，poly x [ ]可歸約為 0.

考慮 as := a : as的情況，和做證明時一樣，我們先將 poly x (a : as)展開，然後
試著整理出 sum (zipWith (×) as (iterate (×x) 1)，以便收回成為 poly x as。計算
中的每一步都以此為目的，試著將 sum與 zipWith移動至 iterate旁邊:

poly x (a : as)
= { poly的定義 }

sum (zipWith (×) (a : as) (iterate (×x) 1))
= { iterate的定義 }

sum (zipWith (×) (a : as) (1 : map (×x) (iterate (×x) 1)))
= { zipWith與 sum的定義 }

a+ sum (zipWith (×) as (map (×x) (iterate (×x) 1)))
= { zipWith (×) as ·map (×x) = map (×x) · zipWith (×) as，見習題 }

a+ sum (map (×x) (zipWith (×) as (iterate (×x) 1)))
= { sum ·map (×x) = (×x) · sum }

a+(sum (zipWith (×) as (iterate (×x) 1)))× x
= { poly的定義 }

a+(poly x as)× x .

第 4步中關於 zipWith與 map的性質幫助我們將 map (×x)往外提、將 zipWith
往裡推。事實上，該性質不限於乘法，而可適用於任何滿足結合律的運算子
(⊗)。我們可非正式地理解如下：

zipWith (⊗) [a,b,c ] (map (⊗x) [d,e, f ])
= [a⊗ (d⊗ x),b⊗ (e⊗ x),c⊗ (f ⊗ x)]
= {結合律: m⊗ (n⊗ k) = (m⊗n)⊗ k }
[(a⊗d)⊗ x,(b⊗ e)⊗ x,(c⊗ f )⊗ x ]

= map (⊗x) (zipWith (⊗) [a,b,c ] [d,e, f ]) .

第 5步之中的 sum ·map (×x) = (×x) · sum在習題 2.7中證明過，需要乘法與加
法的分配律。在本推導中，它的功能是將 sum往右推。它也是使 poly可以加

2此處我們假設 as為有限長的串列，並把 zipWith視為在其第一個參數之上的歸納定義。

速的關鍵性質：共同的 (×x)可以提出來 —左手邊可能做了的許多次 (×x)其實
只需做一次。追根究底，poly之所以能算得更快，都歸功於乘法與加法的分配
律。經過上述計算，我們可得：

poly x [ ] = 0
poly x (a : as) = a+(poly as)× x .

在這個定義中，函數 poly遞迴多少次，便做多少個乘法。因此本演算法所需
的乘法數目為 O(n).
快速版本的函數 poly相當於把 a0 +a1x+a2x2 + ...+anxn 轉換成

a0 + x× (a1 + x× (a2 + ...+(an−1 + x×an))) .

這條規則在William George Horner 1819年的一篇論文中出現並證明，因此
通常被稱作 Horner法則，雖然 Horner本人和許多歷史學家們都相信該規則可
被追溯得更早。

習題 5.8 — 試證明：如果 (⊗) 滿足結合律，zipWith (⊗) as ·map (⊗x) =
map (⊗x) · zipWith (⊗) as。

5.3.2 二進位表示法

回顧第 2.2節中的函數 exp。該函數計算乘冪 — exp b n = bn，其定義如下:

exp ::N→ N→ N
exp b 0 = 1
exp b (1+ n) = b× exp b n .

以這個演算法計算 bn 時會需要 O(n)個乘法。是否有更快的做法呢？
我們先定義函數 binary ::N→ List Bool，可將一個自然數轉換成反轉的二進

位表示法（即最低有效位在左邊，最高有效位在右邊）。以下我們將 False簡寫
為 0, True簡寫為 1：

binary 0 = [ ]

binary n | even n = 0 : binary (n ‘div‘ 2)
| odd n = 1 : binary (n ‘div‘ 2) .

例如，map binary [1,2,3,4 ] = [[1 ], [0,1 ], [1,1 ], [0,0,1 ]]. 函數 binary在每次遞迴
呼叫時將參數減半，因此 binary n只需要 O(logn)的時間。我們讓 binary傳回
反轉的二進位數是為了方便定義 decimal ::List Bool→N — binary的反函數，將
binary的結果轉回成原有的數字：

decimal [ ] = 0
decimal (c : cs) = if c then 1+2×decimal cs else 2×decimal cs .

我們可證明 decimal ·binary = id.
回到 exp,試計算如下

exp b
= { id為 (·)的單位元 }

exp b · id
= { decimal ·binary = id }

exp b ·decimal ·binary .

由於 binary n只需 O(logn)的時間，如果我們能把 exp b ·decimal的計算時間也
縮減到 O(logn),我們就有個只需對數時間的演算法了！
令 roll b = exp b ·decimal。顯然 roll b [ ] = 1. 考慮輸入為 c : cs的情況，在以

下的推導中，我們假設 exp b n = bn 擁有乘冪該有的各種算術性質：

roll b (c : cs)
= { exp2與 decimal之定義 }

exp b (if c then 1+2×decimal cs else 2×decimal cs)
= {函數分配進 if }

if c then exp b (1+2×decimal cs) else exp b (2×decimal cs)
= {算術：bi+2×x = bi× (bx)2 }

if c then b× square (exp b (decimal cs)) else square (exp b (decimal cs))
= { roll之定義 }

if c then b× square (exp2 b cs) else square (roll b cs) .

因此，我們推導出了在 O(logn)時間內計算 bn 的程式如下：

exp b = roll b ·binary

roll b [ ] = 1
roll b (c : cs) = if c then b× square (exp2 b cs) else square (exp2 b cs) .

習題 5.9 — 如何得知 binary會終止？它的定義用的是什麼歸納方式？
習題 5.10 — 證明 decimal ·binary = id.

習題 5.11 — 展開 exp b = roll b · binary以導出一個不產生中間串列的 exp
定義。這個定義用的是什麼歸納方式呢？

5.3.3 小結與提醒

如果回顧本節發生了什麼，該說：我們為 poly 和 roll · decimal 找出了歸納定
義。它們的效率因此提升了，但這只能說是湊巧：兩個演算中，都有些代數性
質可運用，使得推導出的歸納定義在每一步需要做的工作不多，剛好是有效率
的。
一般說來，歸納定義和效率提升不見得能畫上等號。導出了某函數的歸納

定義後，仍需針對它做分析，才能知道這個推導是否值得。有些情況下，推導
出的程式有較好的時間複雜度，這樣的程式通常能表現得比原來的程式好。有
些情況下，找出歸納定義能消除中間串列，或著減少走訪資料結構的次數。這
時，導出的程式仍有同樣的時間複雜度，但可能有較小的常數。此時需注意：

本章談的僅是時間複雜度，而且只考慮一些特定運算元被使用的次數。實際上
的執行效率受到許多因素的影響，例如：不同運算元花費不同的時間；空間的
使用量也影響效率（例如，使用大量記憶體的程式可能需要較多次垃圾收集）；
某些演算法適合快取，等等。我們在之後的章節中將看到一些歸納定義程式走
訪資料結構的次數雖較少，但反而執行得慢的例子。

5.4 變數換常數

接下來的幾節中，我們將介紹幾種常見的程式推導技巧。[todo: generalize]
本節先以一個簡單但時常用上的技巧作為開頭。回顧在例 1.26中提及的函

數 positions:

positions z = map fst ·filter ((= = z) · snd) · zip [0.· ] .

positions z xs傳回 z在 xs中出現的所有位置。上述的定義方式會產生許多中間
串列。我們能用第 5.1節中的方式，利用展開-收回轉換為 positions推導出一個
歸納定義，並藉此將中間串列消除嗎？
我們針對 positions z的輸入做分析，在歸納狀況中，先將 positions z (x : xs)

展開，希望最後能收回 positions z xs. 由於讀者對相關計算應已熟悉，以下的演
算以較快的步調進行：

positions z (x : xs)
= map fst ·filter ((= = z) · snd) · zip [0 . . ] $ (x : xs)
= { zip之定義，[0 . . ] = 0 : [1 . . ] }

map fst ·filter ((= = z) · snd) $ (0,x) : zip [1 . . ] (x : xs)
= { map與 filter之定義 }

if x = = z then 0 : map fst (filter ((= = z) · snd) (zip [1 . . ] (x : xs)))
else map fst (filter ((= = z) · snd) (zip [1 . . ] (x : xs))) .

結果我們卡在這兒了：最後的式子中，zip的參數是 [1 . . ],但 positions的定義要
求這個參數得是 [0 . . ]. 我們無法將式子收回得到 positions z xs.
看來，問題是 positions的定義把 0給寫死了。如果我們索性把該位置變成

一個變數呢？我們定義：

posFrom z i = map fst ·filter ((z = =) · snd) · zip [i . . ]

函數 positions是 posFrom的特例：positions z = posFrom z 0. 而 posFrom的歸納
定義可用展開-收回轉換找出：

posFrom z i (x : xs)
= map fst ·filter ((= = z) · snd) · zip [i . . ] $ (x : xs)
= { zip之定義，[i . . ] = i : [1+ i . . ] }

map fst ·filter ((= = z) · snd) $ (i,x) : zip [1+ i . . ] (x : xs)
= { map與 filter之定義 }

if x = = z then i : map fst (filter ((= = z) · snd) (zip [1+ i . . ] (x : xs)))
else map fst (filter ((= = z) · snd) (zip [1+ i . . ] (x : xs)))

= { posFrom之定義 }
if x = = z then i : posFrom z (1+ i) xs

else posFrom z (1+ i) xs .

由此我們可得：

posFrom z i [ ] = [ ]

posFrom z i (x : xs) = if x = = z then i : posFrom z (1+ i) xs
else posFrom z (1+ i) xs .

由於將一個常數換成變數，posFrom比 positions多了些彈性，因此較容易
收回。這個技巧在許多場合用得上，往往是許多程式推導的第一步。但我並不
建議大家看到任一個定義，便一股腦地把所有常數都換成變數。如同第 2.7節
中提及的原則，該把哪些常數換掉仍應由計算中發現，只用在必要之處。

習題 5.12 — 函數 index為串列中的每個元素標上位置。

index ::List a→ List (N×a)
index = zip [0 . . ] .

請試著為 index導出一個歸納定義。如果不成功，找出一個更通用的輔助
函數，並導出其歸納定義。

5.5 組對

接下來我們介紹另一個重要的技巧：「組對 (tupling)」。

5.5.1 陡串列

給定一個整數串列。當我們說它很「陡」，意思是它由左到右下降得極快，快
到每一個元素都大於其右邊所有元素的和。形式化的定義如下：

steep ::List Int→ Bool

steep [ ] = True

steep (x : xs) = x> sum xs ∧ steep xs .

如果當作一個程式，當輸入串列長度為 n,由於反覆呼叫 sum,上述的程式需要
O(n2)的時間。但每次算出的 sum xs和 x比較後便立刻被丟棄，似乎很浪費。
我們能否把 sum 的結果存下來呢？下述函數 steepsum 把 steep 與 sum 都算出
來，放在一個序對中：

steepsum ::List Int→ (Bool× Int)

steepsum xs = (steep xs,sum xs) .

我們看看 steepsum能否算得快一點？
根據定義，steepsum [ ] = (True,0). 我們看看 xs = x : xs的例子：

steepsum (x : xs)
= { steepsum之定義 }
(steep (x : xs),sum (x : xs))

= { steep與 sum之定義 }
(x> sum xs ∧ steep xs,x+ sum xs)

= {將子算式提取至 let中 }
let (b,s) = (steep xs,sum xs)
in (x> s ∧ b,x+ s)

= { steepsum之定義 }
let (b,s) = steepsum xs
in (x> s ∧ b,x+ s) .

我們已推導出：

steepsum [ ] = (True,0)
steepsum (x : xs) = let (b,s) = steepsum xs

in (x> s ∧ b,x+ s) .

這是一個只用 O(n) 時間的程式。有了 steepsum，我們可重新定義 steep 為
steep = fst · steepsum。
讓一個函數多傳回一些值的動作稱作組對 (tupling) —因為多傳回的值被放

在一個序對中。3 我們常用此技巧來存下可重複使用的中間值，並減少走訪資
料結構的次數。
最後一提：若使用第 1.6.3節，36頁介紹的「分裂」運算元：

⟨·, ·⟩ :: (a→ b)→ (a→ c)→ a→ (b× c)
⟨f ,g⟩ x = (f x,g x) .

函數 steepsum的定義可較簡潔地寫成：

steepsum = ⟨steep,sum⟩ .

本書將在適當時採用這種寫法。

責任越大，能力越強？ 某個意義上，函數 steepsum做的事比 steep多：後者
只判斷輸入是否為陡串列，前者不只如此，還多附送了串列的和。然而，傳回
較多東西、似乎做了更多事的程式，反倒可以執行得比較快。竟出現「責任越
大，能力越強」這種違反直覺的現象，這是怎麼回事呢？4

其實，在歸納定義與歸納證明中，這都是常見的。有些比較通用的程式反
倒比較容易定義；有些定理本身不好證明，為了證明它，我們把它變得更廣
泛、更強些，反倒好證了。箇中原因說穿了便不難理解，函數 steepsum便是

3在一些函數語言中，‘pair”指含兩個成員的序對，‘tuple”則不限定為兩個成員。Tuple也可當
作動詞，指做出一個 tuple. 本書將動詞的 tuple譯為「組對」–組出一個對。有些語言中 pair與
tuple有更根本的差異，但本書中不做區分。

4「能力越強，責任越大 (with great power comes great responsibility)」是 2002年版《蜘蛛人》電
影的名句。根據考證 [O’Toole, 2017],法國國民公會 1793年的政令中即出現過類似的想法，包括邱
吉爾和羅斯福等人也都說過類似話語的不同版本。

一個容易明白的好例子：一個函數若傳回較多資訊，在遞迴呼叫它時，我們便
有更多資訊可直接取用。同樣地，一個定理若保證更強的性質，表示在使用歸
納假設的步驟中，我們有了更強的性質可用。
如果我們把一個函數或待證的定理擴充得太強，確實也有可能使它們強到

寫不出來、證不出來。做歸納定義或證明的重要技巧之一，便是找到這麼一個
平衡點：將一個待定義或證明的物件擴充到足以提供歸納步驟需要的所有資
訊，又不至於強到無法寫出、證出。
在這一節以及下一節中，我們都會看到許多如此的例子。

5.5.2 以串列標記樹狀結構

我們再舉一個組對的好例子。回顧第 1.10與 2.8節中提及的外標籤二元樹：

data ETree a = Tip a | Bin (ETree a) (ETree a) .

下述函數 size ::ETree a→ Int計算一棵樹中標籤的數目（對 ETree而言是 Tip出
現的次數）；repl t xs則將 t原有的標籤丟棄，改用串列 xs由右至左依序重新為
t上標籤。觀察：在遞迴呼叫中，我們用 take和 drop將串列 xs截成適當的長
度：

size (Tip ) = 1
size (Bin t u) = size t+ size u ,

repl ::ETree a→ List b→ ETree b
repl (Tip ) xs = Tip (head xs)
repl (Bin t u) xs = Bin (repl t (take n xs)) (repl u (drop n xs))

where n = size t .

如果 t是一個向左傾斜的二元樹，repl t xs不僅會反覆計算 size，也會反覆地將
take與 drop用在 xs上，使得上述的 repl成為一個 O(n2)的演算法。利用組對
的技巧，我們能讓 repl的時間複雜度小一些嗎？
我們試著把 repl作用在一個稍微左斜的樹，(Bin (Bin t u) v)之上，看看有

什麼能做的。令 t, u的 size分別為 n1 與 n2. 如果我們希望導出一個線性時間
的程式，其中一個提示是：我們希望在 repl (Bin (Bin t u) v) xs 中，xs 的每個
元素最多都只被 take和 drop各碰過一次。依此原則，以下的推導基嘗試做到
兩點：首先，把連續的 take消去；其次，若已做了 take n1 xs，就避免再出現
take (n1 +n2) xs，因為後者的存在會讓 take重複處理 xs中的元素。我們會用到
習題 2.38 – 2.40中提到的三個性質：

take m (take (m+n) xs) = take m xs ,

drop m (take (m+n) xs) = take n (drop m xs) ,

drop (m+n) xs = drop n (drop m xs) .

其中 m, n均為自然數。試演算如下：

repl (Bin (Bin t u) v) xs
= { repl之定義，兩次 }

Bin (Bin (repl t (take n1 (take (n1 +n2) xs)))
(repl u (drop n1 (take (n1 +n2) xs))))

(repl v (drop (n1 +n2) xs))
= {習題 2.38 – 2.40 }
Bin (Bin (repl t (take n1 xs))

(repl u (take n2 (drop n1 xs))))
(repl v (drop n2 (drop n1 xs))) .

演算到此，與 t有關的是 repl t, take (size t),與 drop (size t)三項；與 u有關
的是 repl u, take (size u),與 drop (size u)三項。如果我們把 repl t, take (size t),與
drop (size t)取出，當作一個函數之內完成的動作：

repTail ::ETree a→ List b→ (ETree b×List b)
repTail s xs = (repl s (take n xs),drop n xs) ,

where n = size s .

那麼 Bin (repl t (take n1 xs)) (repl u (take n2 (drop n1 xs)))似乎有可能收回成為這
樣的式子：xs先被丟給 repTail t，將 t 標記好，並得到剩下的串列 drop n1 xs。
這個剩下的串列又可以丟給 repTail u,兩者都只把 xs走過一次。我們試著導出
repTail的歸納定義。基底狀況 s :=Tip y比較容易，我們考慮 s :=Bin t u的情
況，並演算如下（令 n1 = size t，n2 = size u,因此 size (Bin t u) = n1 +n2）：

repTail (Bin t u) xs
= { repTail之定義 }
(repl (Bin t u) (take (n1 +n2) xs),drop (n1 +n2) xs)

= { repl之定義，令 n1 = size t }
(Bin (repl t (take n1 (take (n1 +n2) xs)))

(repl u (drop n1 (take (n1 +n2) xs))),drop (n1 +n2) xs)
= {習題 2.38 – 2.40 }
(Bin (repl t (take n1 xs))

(repl u (take n2 (drop n1 xs))),drop n2 (drop n1 xs))
= {提出共同項 }

let (t′,xs′) = (repl t (take n1 xs),drop n1 xs)
(u′,xs′′) = (repl u (take n2 xs′),drop n2 xs′)

in (Bin t′ u′,xs′′)
= { repTail之定義 }

let (t′,xs′) = repTail t xs
(u′,xs′′) = repTail u xs′

in (Bin t′ u′,xs′′) .

因此我們得到：

repTail (Tip ) xs = (Tip (head xs), tail xs)
repTail (Bin t u) xs = let (t′,xs′) = repTail t xs

(u′,xs′′) = repTail u xs′

in (Bin t′ u′,xs′′) .

確實如同所預期的，串列 xs被 repTail t使用，得到標籤過的新樹 t′,和剩下的
串列 xs′. 後者再被 repTail u用來給 u上標籤。最後我們得傳回剩下的串列 xs′′.
實際上把串列變短的動作發生在基底狀況 repTail (Tip )中。串列中的每個元
素只會在每次遇見 Tip時被取出一次，因此這是一個線性時間的演算法。

[todo: repsort t = rep t (sort (leaves t [ ])).]
Burstall and Darlington [1977]

習題 5.13 — 回顧第 1.10節中談到的 ITree:

data ITree a = Null | Node a (ITree a) (ITree a) .

猴麵包樹 (baobab)，又稱猢猻木，是一種樹幹相當粗的樹。a 如果一個
ITree Int的每個標籤都大於其兩個子樹的標籤總和，我們便說它是一棵猴
麵包樹。以下的函數判定一棵樹是否為猴麵包樹（其中 sumT :: ITree Int→
Int計算一個樹中所有標籤的總和）：

baobab :: ITree Int→ Bool

baobab Null = True

baobab (Node x t u) = baobab t ∧ baobab u ∧
x> (sumT t+ sumT u) .

因反覆呼叫 sumT ,當樹的大小為 n時，baobab的執行時間為 O(n2). 請使
用組對的技巧，在 O(n)的時間內算出 baobab.

a猴麵包樹原產於馬達加斯加、非洲等地，也被寫進了《小王子》之中。

習題 5.14 — 本題出自Hu et al. [1997]。函數 depth定義一棵 ETree的深度。
Tip的深度為零，Bin的深度則為兩子樹中較深者的深度加一：

depth ::ETree a→ N
depth (Tip ) = 0
depth (Bin t u) = 1+ (depth t ↑depth u) .

下列函數 deepest則傳回一棵樹中最深的標籤：

deepest ::ETree a→ List a
deepest (Tip x) = [x ]
deepest (Bin t u) | m<n = deepest u

| m = = n = deepest t++deepest u
| m>n = deepest t

where (m,n) = (depth t,depth u) .

請用組對的技巧，避免重複計算 depth. 注意: 完成的程式中，(++)仍可能
需要 O(n2)的時間。我們將在習題 5.27處理這個問題。



DR
AF
T

5.5. 組對 123

習題 5.15 — 第 3.2節中的函數 balanced ::RBTree→ Bool檢查一棵紅黑樹
是否平衡。由於重複呼叫 bheight,這是一個需要 O(n2)時間的函數。請用
組對的技巧推導出一個可在線性時間內判斷平衡的版本。

5.5.3 代換為最小標籤 —循環程式

下列函數 minE曾出現在第 2.8節中，找出一棵 ETree中最小的值。函數 rep則
可視為 repl的簡單版，將樹中的每個標籤都代換成同一個值 y。

minE ::ETree Int→ Int

minE (Tip x) = x
minE (Bin t u) = minE t ↓minE u ,

rep :: b→ ETree a→ ETree b
rep y (Tip ) = Tip y
rep y (Bin t u) = Bin (rep y t) (rep y u) .

給一棵樹 t,我們要把 t之中的每個標籤都代換成 t的最小標籤。直觀的做法是
寫成 let m = minE t in rep t m. 如此一來，t會被走訪兩次，第一次被 minE走訪
以計算 m,第二次由 rep進行代換。Bird [1984]提出挑戰：有可能在只把 t走訪
一次的情況下，完成上述工作嗎？
以下函數 repmin同時傳回代換過後的樹，以及原樹中的最小標籤：

repmin ::ETree Int→ a→ (ETree a× Int)

repmin y = ⟨rep y,minE⟩ .

使用本節的技巧，讀者們應該已經可以為 repmin導出如下的歸納定義：

repmin y (Tip x) = (Tip y,x)
repmin y (Bin t u) = let (t′,m) = repmin y t

(u′,n) = repmin y u
in (Bin t′ u′,m↓n) .

該定義只將 t走訪一次。然後我們定義：

transform ::ETree Int→ ETree Int

transform t = let (t′,m) = repmin m t in t′ .

函數 transform用 repmin算出 t的最小標籤 m,同時又用 m來標記 t… 於是，似
乎確實用一次走訪就完成了兩件事！這是怎麼做到的呢？
函數 transform有個特殊之處：變數 m既是 repmin的傳回值，又是其參數。

這是一個循環程式 (circular program)。這樣的程式之所以能正常終止，有賴於
Haskell的範式順序/惰性求值（見第 19頁）。實際上發生的事情如此：假設輸入
為

t = Bin (Bin (Tip 4) (Tip 2))
(Bin (Bin (Tip 3) (Tip 1)) (Tip 5))

repmin將輸入 t走訪一遍，邊走邊建立了一個未算出的算式 (4↓2)↓ ((3↓1)↓5).
該算式就是 m的值，其實也可視作一棵樹，其結構和 t一樣，只是把每個 Bin

代換成 (↓). 函數 repmin的另一項工作是新建一棵樹 t′，該樹之中每個 Tip的標
籤都指到這個算式。根據範式順序求值，這個算式還不用立刻被算出來。直到
我們終於不得不算出它，例如當 t′ 被傳回，我們要求電腦把 t′印出來，或著有
別的函數需檢查 t′ 中的標籤時，該算式才被歸約成一個數字 —這需要把該算
式走訪一遍。（但，根據惰性求值，一但 m被算成一個數字，下次使用 m時就
不再需要反覆計算了。）
需注意，雖然「t只被走訪一次」確實成立，這並不表示 transform必然比老

實將樹走訪兩次的程式快 —效率是個複雜的問題，本章將陸續談到更多。我
的建議是：循環程式應視為有趣、優雅的謎題，而不是以效率為目的的程式設
計技巧。

5.5.4 小結與提醒

「組對」的技巧讓函數多傳回一些值。我們可能藉此省下一些重複的計算，增
進效率。
我們自然想問：給一個有重複計算、待改進的函數 f ,怎知道該讓它多傳回

什麼值？在本節的例子中，我們可由符號演算看出一些端倪：將 f 的定義展
開，辨識出被重複計算的子算式，這些子算式就可能用來與原函數組對。但廣
泛說來，「將一個函數或性質通用化」是編程與證明中最困難、最需要經驗、
智慧的一步。只要找到正確的通用化，例如 steepsum或 repTail,剩下的推導都
可相當機械化地進行。唯有「通用化」這一步，我們無法保證有任何機械化、
公式化的方法可作為解決所有問題的萬靈丹 —否則編程就是一件可完全自動
化的事情了。
雖然如此，我們仍希望基於符號演算的形式方法能給我們一定程度的幫助

與指引。程式語言研究的目標之一便是分辨出編程的過程中，哪些部分是瑣
碎、機械化的，哪些部分是真正需要靈感與智慧的，並且盡量使用符號幫助，
使我們在進行思考時有更多工具。
組對可用於減少走訪資料結構的次數，但這麼做並不見得有效率上的好

處。例如，以下函數計算一個串列中元素的平均值：

average xs = sum xs/ length xs .

利用組對，我們可以另定義一個函數 sumlen = ⟨sum, length⟩, 並推導其歸納定
義，在一次走訪中同時計算串列的和與長度：

sumlen [ ] = (0,0)
sumlen (x : xs) = let (s, l) = sumlen xs

in (x+ s,1+ l) .

然後平均便可定義成 average′ xs = let (s, l) = sumlen xs in s/ l.
然而，根據 Hu et al. [1997]，average′ 通常比 average慢。兩者都是 O(n)的

演算法。函數 sumlen在傳回值時會產生一個序對，該序對立刻被上層拆掉。
因此 average′ 每處理一個元素耗費的時間較多，往往不如乾脆將 xs走訪兩次。
Hu et al. [1997]認為需有更有效率的序對實作法，組對才是值得做的轉換。

習題 5.16 — 函 數 allpairs 傳 回 輸 入 串 列 中 任 兩 個 元 素 （依
其 原 本 順 序） 形 成 的 序 對。 例 如 allpairs [1,2,3,4 ] 可 得 到
[(1,2),(1,3),(1,4),(2,3),(2,4),(3,4)]。

allpairs ::List a→ List (a×a)
allpairs [ ] = [ ]

allpairs (x : xs) = map (λy→ (x,y)) xs++allpairs xs .

而 maxdiff 則計算一個串列中任兩元素前者與後者的最大差：

maxdiff ::List Int→ Int

maxdiff = maximum ·map (λ (x,y)→ x− y) ·allpairs .

當輸入串列長度為 n,如此定義的 maxdiff 是一個需時 O(n2)的演算法。定
義：

mdm xs = (maxdiff xs,???) .

找出 ???可能的值，使得 mdm能在 O(n)時間之內完成計算。你可假設以
下性質：

maximum (map (x−) xs) = x−minimum xs . (5.4)

並假設 maximum與 minimum在空串列上的值分別為 −∞與 ∞.

5.6 累積參數

在第 5.2節中，我們曾提及該處定義的 reverse ::List a→ List a需要 O(n2)的時
間。是否有比較快的作法呢？

5.6.1 串列反轉與連接

下列函數 revcat 似乎比 reverse更通用一些：它拿兩個參數 xs與 ys,不僅將 xs
反轉，還把 ys接到反轉後結果的右邊：5

revcat ::List a→ List a→ List a
revcat xs ys = reverse xs++ ys .

原有的 reverse可以視為 revcat的特例 — reverse xs = revcat xs [ ]. 這裡是否也會
出現責任越大，能力越強的現象：看似做了較多事情的 revcat其實反倒有比較
有效率的實作呢？我們試著演算看看！當 xs := [ ], 顯然 revcat [ ] ys = ys. 考慮
xs := x : xs的情況：

revcat (x : xs) ys
= { revcat之定義 }

5revcat為 ‘reverse”與 ‘concat”的簡寫。

reverse (x : xs)++ ys
= { reverse之定義 }
(reverse xs++[x ])++ ys

= { (++)之結合律 }
reverse xs++([x ]++ ys)

= { (++)與 revcat之定義 }
revcat xs (x : ys) .

因此我們有了：

revcat [ ] ys = ys
revcat (x : xs) ys = revcat xs (x : ys) .

我們看看 revcat [1,2,3,4 ] [ ]如何歸約：

revcat (1 : 2 : 3 : 4 : [ ]) [ ]
= revcat (2 : 3 : 4 : [ ]) (1 : [ ])
= revcat (3 : 4 : [ ]) (2 : 1 : [ ])
= revcat (4 : [ ]) (3 : 2 : 1 : [ ])
= revcat [ ] (4 : 3 : 2 : 1 : [ ])
= [4,3,2,1 ] .

串列 [1,2,3,4 ]中的元素一個個被搬到 revcat的第二個參數中。這是一個常數
時間內可完成的動作，而每個元素只會被搬動一次。因此當 xs 的長度是 n，
reverse xs ys可在 O(n)的時間內執行完畢！
我們剛看到的技巧和第 5.5節中的組對互為對偶。在第 5.5節中，我們為了

導出一個函數的較快版本，讓它多傳回一些資訊。而在本節，我們則讓一個函
數多吃些參數，多接受些資訊。通常這些新介紹的參數用於「累積」計算到目
前為止的結果，例如在 revcat中，參數 ys存放被反轉到一半的串列。因此這
個技巧被稱作累積參數 (accumulating parameters)。
很多情況下，累積參數的運用仰賴某些運算元的結合律。觀察 revcat的推

導，關鍵的一步便是 (++)的結合律。它使我們能夠把 (reverse xs++ [x ])++ ys
轉換為 reverse xs++([x ]++ ys)，將 [x ]往右搬，才能收回、累積到 revcat 的第
二個參數 ys中。
回顧起來，我們最初如何發明出 revcat xs ys = reverse xs++ ys 這樣的定義

呢？一個解釋便是：reverse之所以慢，是因為 (++)的括號都括錯了方向，往
左邊括了。因此我們在右邊補一個 (++ys),希望用 (++)的結合律，將括號往右
挪。
我們多看一個例子。以下函數 tags傳回一棵 ITree中所有的標籤：

tags :: ITree a→ List a
tags Null = [ ]

tags (Node x t u) = tags t++[x ]++ tags u .

和習題 5.7的情況類似，當 t是一棵向左傾斜的樹，tags會需要 O(n2)的時間。
例如，考慮這棵樹 t（以下將 Node x Null Null簡寫為 lv x）：

t = Node 7 (Node 6 (Node 4 (Node 2 (lv 1) (lv 3))
(lv 5)))

(lv 8) .

將 tags t展開（並為說明方便，將一些 [ ]++[x ]++[ ]化簡為 [x ]），我們會得到：

((([1 ]++[2 ]++[3 ])++[4 ]
++[5 ])++[6 ])++[7 ]++[8 ] .

這個式子的結構和 t 一樣，只是將 Node x...都變為 . .++[x ]++ . .. 我們可看到
(++[6 ])需要走過一個長度為 5的串列，(++[7 ])需要走過一個長度為 6的串列。
為了避免 (++)的重複走訪，我們在 tags t的左邊補一個 (++ys),希望透過結

合律改變括號的順序。我們定義：

tagsAcc :: ITree a→ List a→ List a
tagsAcc t ys = tags t++ ys .

利用 (++)的結合律，讀者應不難導出如下的歸納定義：

tagsAcc Null ys = ys
tagsAcc (Node x t u) ys = tagsAcc t (x : tagsAcc u ys) .

之後我們便可重新定義 tags t = tagsAcc t [ ]. 至於 tagsAcc的效率如何呢？如果
將 tagsAcc t ys展開，我們順利地得到：

1 : (2 : (3 : (4 : (5 : (6 : (7 : (8 : ys))))))) .

習題 5.17 — 由 tagsAcc t ys = tags t++ ys推導出上述的歸納定義。
習題 5.18 — 確認 tagsAcc t ys確實是 1 : (2 : (3 : (4 : (5 : (6 : (7 : (8 : ys))))))).

習題 5.19 — 接續習題 5.7，利用累積參數，推導出一個只需線性時間的
tips ::ETree a→ List a.

5.6.2 尾遞迴

關於第 5.6.1 節的 revcat 有許多面向可談。本節先用它帶出一個重要的觀念：
尾遞迴。
從第 2章起，我們常見的歸納定義形式如下：

f [ ] = ...

f (x : xs) = ...f xs ... .

sum [ ] = 0
sum (x : xs) = x+ sum xs .

右側的 sum函數便是一個典型的例子。我們通常必須對遞迴呼叫的結果做些加
工。例如此處 sum xs的結果需要經過 (x+),才成為 sum (x : xs)的值。實作上，
電腦在執行 sum xs的程式碼前必須記下「sum xs回傳後，必須回來執行 (x+)」
這件事。這可能用堆疊或著其他方式達成。無論如何，當 xs長度為 n,便有大

約 n個這種「尚待完成的計算」被暫存著。當 xs被走訪到了基底狀況，這些暫
存的計算才一個個被收回。
但 revcat的情況卻有點不同。觀察其程式碼，會發現遞迴呼叫 revcat xs (x :

ys)的結果本身就是其左手邊 revcat (x : xs) ys的值，不需其他的加工：

revcat [ ] ys = ys
revcat (x : xs) ys = revcat xs (x : ys) .

因此，revcat的實作中，遞迴呼叫完成後，電腦不需回到原呼叫之處再執行什
麼東西：最後的結果 ys可直接傳回到最上層、第一個呼叫 revcat的地方！6

當一個函數 f 呼叫函數 g時，如果該呼叫本身就是函數 f 最後的結果，並
沒有針對傳回值的額外計算，我們將它稱之為一個尾呼叫 (tail call)。此名稱的
由來可能是因為該呼叫是一連串計算後「最尾端」的動作。如果這是一個遞迴
呼叫，則稱之為尾遞迴 (tail recursion)。

尾遞迴與迴圈 函數語言中的尾遞迴程式和指令式語言中的迴圈有相當密切的
關係。回顧 revcat [1,2,3,4 ] [ ]的歸約過程：

revcat (1 : 2 : 3 : 4 : [ ]) [ ]
= revcat (2 : 3 : 4 : [ ]) (1 : [ ])
= revcat (3 : 4 : [ ]) (2 : 1 : [ ])
= revcat (4 : [ ]) (3 : 2 : 1 : [ ])
= revcat [ ] (4 : 3 : 2 : 1 : [ ])
= [4,3,2,1 ] ,

其實看來就像是一個有兩個變數的迴圈，其中一個由 1 : 2 : 3 : 4 : [ ] 逐漸縮短
為 [ ]，另一個由 [ ] 逐步增長為 4 : 3 : 2 : 1 : [ ]. 函數 reverse 的定義 reverse xs =
revcat xs [ ]就是為這兩個變數設定初始值：如果要計算 xs的反轉，兩變數應該
分別初始化為 xs與 [ ]. 在 revcat的定義中，revcat [ ] ys表示該迴圈在第一個變
數為 [ ]時終止，此時程式傳回 ys；而 revcat (x : xs) ys = revcat xs (x : ys)則指定
了在迴圈的每一步中兩個變數的值如何改變。函數 reverse與 revcat的組合相
當於是這樣的一個指令式語言迴圈（假設 xs與 ys為變數，XS為欲反轉的串
列）：7

xs,ys :=XS, [ ];
do xs ≠ = [ ]→

xs,ys := tail xs,head xs : ys
od;
return ys

而該迴圈的恆式 (loop invariant)是什麼呢？正是 reverse XS= reverse xs++ys
—右手邊就是 revcat的定義。

6如果該語言的實作確實有做到這點，我們說它實作了尾呼叫消除 (tail call elimination)或尾呼
叫最佳化 (tail call optimisation)。有些語言到了蠻晚的版本才支援這個最佳化。

7xs,ys := e1,e2 將 e1 與 e2 兩個值同時給予 xs與 ys; do B→ S od表示一個迴圈，當 B還成立時
便反覆執行 S.

一般說來，一個尾遞迴函數可視為一個迴圈，其參數就是迴圈中的變數。
遞迴呼叫時參數改變，相當於更新這些變數的值，其基底狀況則相當於迴圈的
終止條件。當我們為了計算某個函數 f 而設計了另一個較通用、可寫成尾遞迴
的函數 fAcc, fAcc的定義 (例如 reverse xs++ ys)往往就是這個迴圈的恆式 (loop
invariant)。在指令式程式推導的方法學中，設計一個迴圈最難之處便是決定其
恆式。如第 5.5.4節所言，一般來說，並沒有一個固定的方法讓我們由 f 找出
fAcc。但我們仍可讓符號推演幫我們一些忙，並歸納出一些常見的模式。

串列總和 累積參數的技巧常被用來推導尾遞迴程式（雖然累積參數的應用並
不僅止於此）。我們多看一個例子。如前所述，sum函數每次被遞迴呼叫時會
需要用堆疊或其他等價的方式記住每個 (x+)。是否能避免這麼做呢？我們定
義：

sumAcc ::List Int→ Int→ Int

sumAcc xs y = y+ sum xs .

如此定義 sumAcc的直覺理由是：我們想用參數 y存放累積的結果，希望利用
(+)的結合律逐步把 sum xs的結果搬入 y中。同樣地，一但找到了這個定義，
之後的演算便相當日常、例行了。我們先記下：sum xs = sumAcc xs 0. 接下來
試圖導出 sumAcc的歸納定義。在基底情況中，我們會需要 y+0 = y。為了目
睹結合律如何運作，我們看看 xs := x : xs的狀況：

sumAcc (x : xs) y
= y+ sum (x : xs)
= y+(x+ sum xs)
= { (+)之結合律 }
(y+ x)+ sum xs

= sumAcc xs (y+ x) .

讀者可能發現我們之所以定義 sumAcc xs y為 y+ sum xs,而不是 sum xs+ y,是為
了在使用結合律的那一步之中讓 x可以就近與 y結合。當然，(+)也滿足交換
律，因此兩個定義其實是一樣的。當我們處理沒有交換律的運算元時，就得對
這類細節更小心了。總之，我們可導出以下的程式：

sumAcc [ ] y = y
sumAcc (x : xs) y = sumAcc xs (y+ x) .

這可翻譯成下述的兩種迴圈之一。左手邊的程式將串列 XS加總，使用的
兩個變數 xs與 y對應到 sumAcc的兩個參數。在右邊的程式中，我們假設輸入
資料 [x0,x1 . .xN−1 ]依序被存放在一個陣列 X中。陣列有 N個元素，其索引範
圍為 [0 . .N−1 ]. 我們把陣列視為索引到內容的函數，因此 X i存放著 xi.

xs,y :=XS,0;
do xs ̸= = [ ]→ y := y+head xs;

xs := tail xs
od;
return y

i,y :=0,0;
do i ̸= = N→ y := y+X i;

i := i+1
od;
return y

確實，這就是我們一般用指令式語言加總一個串列/陣列時的常見迴圈寫
法：由左到右走訪，用一個變數存放目前為止的和。以往許多人可能都沒注
意到：同樣是「將 xs由左到右走一遍」，這個程式和 sum xs是不同的演算法！
sum [1,2,3,4 ]算出的是 1+(2+(3+(4+0))),而上述的、我們常用的那個迴圈
算出的是 (((0+1)+2)+3)+4. 多虧 (+)的結合律，它們剛好是一樣的。

在迴圈中處理串列 在本節的結尾，我們更完整地討論一下歸納式的串列處理
與迴圈的關係。初次閱讀的讀者可跳過本段。假設某函數 f ::List A→ B能寫
成如下的形式：

f [ ] = e
f (x : xs) = x⊕ f xs ,

其中 e ::B, (⊕) ::A→ B→ B,此處不假設 (⊕)滿足結合律。我們能用一個尾遞
迴函數（或著說用一個迴圈）計算 f 嗎？直覺上，我們可用一個迴圈將串列從
右往左走一遍，並用變數紀錄目前為止算出的值。確實，如果我們要求如下的
規格：

loop xs (f ys) = f (xs++ ys) , (5.5)

（所以 f xs = loop xs (f [ ]) = loop xs e）並分析 xs :=[ ]和 xs := xs++[x ]的情況如
下：

loop [ ] (f ys)
= { loop之規格 }

f ([ ]++ ys)
= { (++)之定義 }

f ys ,

loop (xs++[x ]) (f ys)
= { loop之規格 }

f (xs++[x ]++ ys)
= { (++)之結合律，loop之規格 }

loop xs (f (x : ys))
= loop xs (x⊕ f ys) .

可知如下定義的 loop能滿足 (5.5)：

loop [ ] z = z
loop (xs++[x ]) z = loop xs (x⊕ z) .

但從串列尾端取出元素較不方便。如果我們改用下式作為規格：

loop xs (f ys) = f (reverse xs++ ys) , (5.6)

可推導出

loop [ ] z = z
loop (x : xs) z = loop xs (x⊕ z) .

但此時要用 loop計算出 f xs，得先將 xs反轉：f xs = loop (reverse xs) e. 這具體
印證了我們的觀察：在串列上歸納定義出的函數，處理元素的順序和通常寫法
的迴圈是相反的。



DR
AF
T

5.6. 累積參數 131

假設有 N個元素的串列 xs = [x0,x1 ...xN−1 ]被反過來存放在陣列 X之中，意
即 X 0 = xN−1, X 1 = xN−2, … X (n−1) = x0. 前述的兩種 loop函數都可在不同的
詮釋下理解為如下的指令式程式。

i,z :=0,e;
do i ̸= = N→ z :=X i⊕ z;

i := i+1
od;
return z .

5.6.3 更多尾遞迴範例

我們多看些使用結合律推導出尾遞迴程式，並增進效率的例子。

快速乘冪 說到結合律的應用，似乎不得不提作為經典例子的乘冪。我們在第
5.3.2節中曾用二進位表示法導出一個用 O(logn)個乘法計算 bn 的程式。此處
我們使用結合律推導出尾遞迴的版本。回顧函數 exp在第 2.2節中的定義，乘
冪便是連續的乘法。我們定義以下的函數 expAcc，把 exp b n乘上一個累積參
數 x，希望將一些中間結果搬移到 x中：

expAcc ::N→ N→ N→ N
expAcc b n x = x× exp b n .

如果 expAcc 有快速的定義，我們可以令 exp b n = expAcc b n 1. 然後我們
針對 n 為零、n 為非零的偶數，以及 n 為奇數三種情況作分析。當 n := 0,
expAcc b 0 x = x. 當 n 為偶數時，可以被改寫成 2× n. 以下的推導中我們將
exp b n寫成 bn,並假設它已有乘冪該有的各種性質：

expAcc b (2×n) x
= x×b2×n

= {因 bm×n = (bm)n }
x× (b2)n

= { expAcc之定義，b2 = b×b }
expAcc (b×b) n x .

當 n是奇數，我們將它改寫成 1+n:

expAcc b (1+n) x
= x×b1+n

= { exp之定義 }
x× (b×bn)

= { (×)之結合律 }
(x×b)×bn

= { expAcc之定義 }
expAcc b n (x×b) .

將語法改寫成 Haskell能接受的形式（例如將 2×n與 n改寫成 n與 n ‘div‘ 2）
之後，我們得到這樣的程式：

expAcc b 0 x = x
expAcc b n x | even n = expAcc (b×b) (n ‘div‘ 2) x

| odd n = expAcc b (n−1) (x×b) .

確實，這是一般在指令式語言中快速計算乘冪的方式。一個操作性的理解法
是：expAcc b n x開始執行後，第一個參數中總是存放著 b的「2的某個次方」
的乘冪（b, b2, b4…）。只在 n 是奇數時，當時的 b 才會被乘入累積參數 x 之
中。

習題 5.20 — 本節的 expAcc函數相當於怎樣的指令式語言迴圈？其迴圈恆
式為何？

習題 5.21 — 請推導出一個尾遞迴版本的 length函數。

習題 5.22 — 第 2.3節中介紹了經典的階層函數：

fact ::N→ N
fact 0 = 1
fact (1+ n) = (1+ n)× fact n .

請推導出一個尾遞迴版本。您利用了關於 (×)的什麼性質？最後的程式是
否像是一個計算 n階層的指令式語言迴圈呢？

習題 5.23 — 第 2.2節中把乘法定義為連續的加法。確實，一些早期、簡
單的微電腦中沒有專做乘法的電路，只能以加法實作乘法。但如果有乘以
二、除以二、以及判斷一個數字是奇數或偶數的指令（都是簡單的位元運
算），我們只需 O(logm)個加法即可計算 m×n. 定義：

mulAcc m n k = k+m×n .

請推導出一個只使用加減法、乘以二、除以二、奇偶判斷，並在 O(logm)

的時間內算出 mulAcc m n k 的歸納定義。這個定義能被改寫成一個迴圈
嗎？其恆式為何？

習題 5.24 — 下述函數 dtoN 將一個以串列表達的十進位數字轉成自然數。
例如 dtoN [4,1,6,0 ] = 4160:

dtoN ::List N→ N
dtoN [ ] = 0
dtoN (d : ds) = d×10length ds +dtoN ds .

其中 length ds的反覆計算使得 dtoN成為一個需時 O(n2)的演算法（n為輸
入串列的長度）。

1.使用組對的技巧，導出一個能在 O(n)時間內計算 dtoN 以及一些其
他輔助結果的函數。

2.使用累積參數，導出一個能在 O(n)時間內計算 dtoN 的尾遞迴函數。
提示：可試試看用這樣的定義 dtoNAcc ds n = . .n . .+dtoN ds.

多個累積參數 我們以一個稍微複雜的例子結束本節。給定以下函數：8

masc ::N→ N
masc 0 = 1
masc (2×n) = 2×masc n
masc (1+2×n) = n+masc n .

我們想用累積參數的技巧，推導出一個以尾遞迴計算 masc n的演算法。最
難的一步總是尋找一個合適的通用化。我們試著展開 masc,看看是否能找到什
麼規律。以 masc 43為例：

masc 43
= 21+masc 21
= 21+10+masc 10
= 31+masc 10
= 31+2×masc 5
= 31+2× (2+masc 2)
= 35+2×masc 2
= 35+4×masc 1
= 35+4×1
= 39 .

我們發現式子總能展開成為 a+b×masc n的形式。因此我們定義

mascAcc ::N→ N→ N→ N
mascAcc n a b = a+b×masc n .

但在開始推導 mascAcc前，我們得先確定 masc能由 mascAcc算得出來。幸好，
我們可以讓 masc n = mascAcc n 0 1.

8masc為 “mostly ascending”的縮寫。這個函數大致上遞增，但偶爾會掉下來一點點。

現在我們試著尋找 mascAcc的尾遞迴定義。當 n :=0, mascAcc 0 a b = a+b.
當 n為非零的偶數，可以將它代換為 2×n。我們推導：

mascAcc (2×n) a b
= a+b×masc (2×n)
= { masc之定義 }

a+b×2×masc n
= mascAcc n a (2×b) .

而當 n為奇數，可以寫成 1+2×n的形式：

mascAcc (1+2×n) a b
= a+b×masc (1+2×n)
= { masc之定義 }

a+b× (n+masc n)
= { (×)分配入 (+), (+)之結合律 }
(a+b×n)+b×masc n

= mascAcc n (a+b×n) b .

因此我們有了如下的尾遞迴定義：

mascAcc 0 a b = a+b
mascAcc n a b | even n = mascAcc (n ‘div‘ 2) a (2×b)

| odd n = mascAcc (n ‘div‘ 2) (a+b× (n ‘div‘ 2)) b .

在推導 mascAcc的過程中使用了各種四則運算的性質，但總之目標是將式子整
理回 a+b×masc n的形式，以便收回成為 mascAcc。

習題 5.25 — 本題來自 Kaldewaij [1990]. 給定以下函數：

fusc ::N→ N
fusc 0 = 0
fusc 1 = 1
fusc (2×n) = fusc n
fusc (1+2×n) = fusc n+ fusc (1+n)

請推導出一個計算 fusc的尾遞迴程式。提示: 以 fusc 78為例展開，看是否
能找到什麼規律。



DR
AF
T

5.6. 累積參數 135

習題 5.26 — 我們能以尾遞迴的方式做 map f 嗎？如果以 mapAcc f xs ys
= ys++map f xs為規格,我們會導出如下的歸納定義：

mapAcc f [ ] ys = ys
mapAcc f (x : xs) ys = mapAcc f xs (ys++[ f x ]) .

但重複的 (++[ f x ])需要 O(n2)的時間。請問用什麼樣的規格才能導出下面
的歸納定義呢？

mapAcc f [ ] ys = reverse ys
mapAcc f (x : xs) ys = mapAcc f xs (f x : ys) .

5.6.4 尾遞迴的效率考量

如前一節所述，歸納定義的 sum需用堆疊（或其他功能相當的機制）記下每個
遞迴呼叫的結果該怎麼加工。這會佔用與輸入串列長度成正比的額外空間，相
當不理想 —直覺上，「將一個串列加總」應該是只需定量的額外空間即可完成
的計算。尾遞迴的 sumAcc則只需用一個變數存放目前為止的總和，當串列走
訪到底，可直接將這個總和傳到最上層，似乎合理多了。因此，在以效率為考
量的函數語言程式庫中，諸如加總、算最小值等等的函數幾乎都是以尾遞迴方
式寫成的。大部分函數語言使用及早求值 —在呼叫一個函數之前，總是把其
參數先算成範式（見第 1.1節，第 19頁）。對這些語言來說，在上述場合使用尾
遞迴函數確實只使用了定額的額外空間，效果相當好。
但對使用惰性求值的 Haskell 來說，情況又更複雜一些 — 若以分析工具

實測，我們會發現純以本章的方式寫出的 sumAcc 仍用了和串列長度成正
比的記憶體空間！這是怎麼回事呢？如前所述，sumAcc [1,2,3,4... ] 0 的值是
((((0+1)+2)+3)+4)+ .... 串列越長，這個式子越長。根據惰性求值的原則，
Haskell不拖到被強迫求值的最後一刻是不會將算式歸約的。因此在走訪串列
的過程中，0+1不會歸約成 1, (0+1)+2不會歸約成 3… 這個大算式就這麼存
放在記憶體中。直到整個串列被走訪完，sumAcc的呼叫者要檢查其傳回值了
（可能是要將它印出來，或著做樣式配對），這個大算式才又一步步被化簡成一
個單一數值。
為改善這類情況下的效率，Haskell提供了一些方法讓我們早點把一些數值

強迫算成範式。例如在資料型別上標注某些欄位為「嚴格 (strict)」的、使用
內建函數 seq將數值歸約、或甚至使用更低階、屬於特定編譯器的「無盒型別
(unboxed type)」等等。對實務導向的 Haskell編程員來說這些都是實用的技巧，
只是已超出本書的範疇。
如果函數傳回的不是數值，而是結構化的資料，而程式語言支援惰性求值，

情形又有所不同。如習題 5.26中所見，我們可用尾遞迴的方式做 map f :

mapAcc f [ ] ys = reverse ys
mapAcc f (x : xs) ys = mapAcc f xs (f x : ys) .

乍看之下，這個程式的問題似乎是需要多做一次 reverse。但這可能並不很嚴
重：reverse也可用 revcat 實作，在線性時間內完成。該定義和原歸納定義的
map f 的最大差別是：mapAcc f 需等到輸入串列整個被走訪完畢後才會開始傳
回第一個結果。而回顧 map f 的歸納定義：

map f [ ] = [ ]

map f (x : xs) = f x : map f xs ,

歸納狀況中，f x : ... 可在走訪到 x 時便先產生。考慮這樣的程式：length ·
filter p ·map f。根據惰性求值，f x : ...會立刻被 filter與 length接收，然後才開
始計算 map f xs. 因此 map f 回傳的中間串列其實並不會被完整地產生。在大部
分情況下，歸納定義的 map f 是比尾遞迴的 mapAcc f 更有效率的函數。在許
多情形中，早點產生部分的結果、早點讓它被使用掉，會是在空間與時間上都
更有效率的做法。

5.6.5 函數作為串列

既然說到串列反轉，本節延伸介紹一個相關且相當有用的技巧。回顧 revcat的
尾遞迴定義，若將最後一個參數省去，其定義可改寫成：

revcat ::List a→ (List a→ List a)
revcat [ ] = id
revcat (x : xs) = revcat xs · (x:) .

revcat是一個傳回 List a→ List a的函數。在基底狀況，revcat [ ]傳回 id. 歸納
狀況中，revcat xs傳回的函數和 (x:)組合在一起。這和 reverse其實很像：在基
底狀況，reverse [ ]傳回空串列 [ ],歸納狀況中，reverse (x : xs)傳回 reverse xs++
(x : [ ])。好像把空串列代換成 id,把 (++)變成 (·),我們就得到 revcat了。
這令我們聯想：有沒有可能把 List a→ List a視為串列的另一種表示法，其

中 id就是空串列，而串列連接就是 (·)呢？定義：

type DList a = List a→ List a ,

一個型別為 DList a的函數 f 表示一個「尾段尚未確定」的串列—餵給它一個尾
巴 ys, f ys便會傳回一個真正的串列。若 xs是一個型別為 List a的串列，(xs++)

便是如此的一個 DList a. 要將一個 DList a轉成 List a,則只需將 [ ]傳進去即可。
下列兩個函數幫我們在這兩種表示法之間作轉換：

toDList ::List a→ DList a
toDList xs = (xs++) ,

toList ::DList a→ List a
toList xs = xs [ ] .

空串列被轉換成 ([ ]++),化簡一下之後確實得到了 id. 「含一個單一元素 x」的
DList可寫成 (x:) —給任何串列，(x:)把 x接到其最左邊。
兩個 DList a如何連接在一起呢？使用函數組合 (·)。確實，(xs++) · (ys++)

是一個接收任一個尾段 ws,傳回 xs++(ys++ws)的函數。DList版本的 ‘cons”建
構元則可定義成

nil ::DList a
nil = id ,

cons :: a→ DList a→ DList a
cons x xs = (x:) · xs .

List上的 (++)若往左結合，效率會較不好。例如 (xs++ ys)++ zs會需要把 xs走
訪兩次。DList的情況呢？考慮 ((xs++) · (ys++)) · (zs++)，其中左邊的兩個 DList

被括在一起。我們演算看看給了一個尾端 ws後的情況：

((xs++) · (ys++)) · (zs++) $ ws
= { (·)之定義 }
(xs++) · (ys++) $ zs++ws

= { (·)之定義 }
(xs++) $ ys++(zs++ws)

= xs++(ys++(zs++ws)) .

由於 (·)的定義，即使 (xs++) · (ys++)先被組合在一起，我們仍得到括號往右括
的 xs++(ys++(zs++ws))！
我們多研究一個例子。在習題 5.7中，給定一個 ETree a,我們想傳回其所有

的標記。如果寫成 tips (Bin t u) = tips t++ tips u，當樹往左邊傾斜時，程式會需
要 O(n2)的時間。但如果我們改用 DList:

tipsD ::ETree a→ DList a
tipsD (Tip x) = (x:)
tipsD (Bin t u) = tipsD t · tipsD u .

上述函數會邊走訪輸入的樹，邊產生一個結構與輸入樹相同的 DList。例如，
考慮如下的樹：

t = Bin (Bin (Bin (Tip 5) (Tip 4))
(Bin (Tip 3) (Tip 2)))

(Tip 1) ,

tipsD t 將會是 (((5:) · (4:)) · ((3:) · (2:))) · (1:) –結構和 t 相同，只是將 t 的每個
Tip x 代換成 (x:), 每個 Bin 代換成 (·). 但當我們傳一個空串列進去，這個由
(x:)和 (·)形成的「樹」將被走訪一遍，並在線性時間內得到 5 : 4 : 3 : 2 : 1 : [ ]。
事實上，如果我們為 tipsD 補一個參數並展開，我們將得到和習題 5.19 中一
樣的結果。有了 tipsD,我們可將原有的 tips改定義為 tips = toList · tipsD,或著
tips t = tipsD t [ ].

Hughes [1986]

習題 5.27 — 在習題 5.14中，我們使用組對可能得到一個型別為 ETree a→
(List a,N)的函數。該函數仍需要 O(n2)的時間連接串列（其中 n為輸入樹
的大小）。請設計一個 O(n)的演算法。

摺

第 2章中的許多歸納函數定義都循著同一個固定模式。以 sum, length,與 map f
為例：

sum ::List Int→ Int

sum [ ] = 0
sum (x : xs) = x+sum xs ,

length ::List a→ N
length [ ] = 0
length (x : xs) = 1+ (length xs) ,

map :: (a→ b)→ List a→ List b
map f [ ] = [ ]

map f (x : xs) = f x :map f xs .

它們都在輸入為 [ ]時傳回某個基底值，在輸入為 x : xs時在 xs上遞迴呼叫，並
將呼叫結果稍作加工。三者的不同之處只在橘色的部分，即基底值以及用於加
工的函數：sum使用 0與 (+), length使用 0與 1+, map f 則使用 [ ]與 (f x:). 如
果說「抽象化」是一個高階程式語言給我們的最重要能力，我們能否將這個模
式抽象出來呢？
我們把上述三個定義中橘色的部分抽出變成參數，將餘下的函數稱為 foldr:

foldr :: (a→ b→ b)→ b→ List a→ b
foldr f e [ ] = e
foldr f e (x : xs) = f x (foldr f e xs) .

如此一來，sum, length,與 map f 都是 foldr的特例：

139

sum = foldr (+) 0 ,

length = foldr (λx n→ 1+ n) 0 ,

map f = foldr (λx ys→ f x : ys) [ ] .

函數 foldr是串列上的「摺 (fold)」— foldr一詞是 fold與「右邊 (right)」的縮寫，
意謂該函數是一個往右結合的摺。我們將在下一節解釋。

6.1 串列的摺

「摺」有許多方式可理解。我們可說 foldr 捕捉了最常見的一種歸納定義模式，
並將它形式化地表達出來。在第 2章中，許多函數定義都遵循這樣的模式：

h ::List a→ b
h [ ] = e
h (x : xs) = ...x ...h xs ...

在 h [ ]的情況傳回某個基底值；在 h (x : xs)的情況中可使用 x與 h xs的值。如
果上述定義中 ...之處沒有出現 xs,則 h的定義就能寫成一個 foldr.
我們也可將 foldr視為組件 (combinator)函數之一。第 1.8.3節之中介紹了組

件函數的觀念：如同 map, take, drop, zip等等的組件函數捕捉了常見的程式設
計模式。每個組件負責一項單一、通用、易重用的功能。函數 foldr也可視為
一個組件，只是它比一些其他組件更抽象、更通用 —我們稍後將發現許多我
們見過的組件函數都是 foldr的特例。
還有一個理解 foldr 的方式：foldr 替換了串列中的建構元。回顧：任何

有限長度的串列都是由 [ ] 開始，有限次地套用 (:) 而來。例如 [x0,x1,x2 ] 是
x0 : (x1 : (x2 : [ ]))的簡寫。考慮 foldr (⊕) e [x0,x1,x2 ]:

foldr (⊕) e (x0 : (x1 : (x2 : [ ])))
= x0⊕ foldr (⊕) e (x1 : (x2 : [ ]))
= x0⊕ (x1⊕ foldr (⊕) e (x2 : [ ]))
= x0⊕ (x1⊕ (x2⊕ foldr (⊕) e [ ]))
= x0⊕ (x1⊕ (x2⊕ e)) .

我們可看到 foldr將串列走訪一次，將每個 (:)替換成 (⊕),將 [ ]替換成 e. 式子
中的括號往右邊結合，這是 foldr 的名字中字母 r 的由來。這種理解也便於解
釋 foldr的型別。回想串列的兩個建構元，

• [ ]的型別是 List a,
• (:)的型別是 a→ List a→ List a.

函數 foldr (⊕) e接收一個 List a，把其中的建構元分別替換為 e與 (⊕)，藉此算
出一個型別為 b的值。因此，

• e是輸入為 [ ]時立刻傳回的值，其型別必須是 b.
• 至於 (⊕)的型別，考慮 x0⊕ (x1⊕ (x2⊕ e))這個式子。其中 x0 的型別為 a,

x1⊕ (x2⊕e)是建構元已被替換過的串列，型別應該為 b. 而 (⊕)拿到這兩
個輸入後，得算出一個型別為 b的值. 因此 (⊕)的型別為 a→ b→ b.

注意：e與 (⊕)的型別分別是將 [ ]與 (:)的型別中的 List a代換成 b而來。綜合
言之，foldr的型別是 (a→ b→ b)→ b→ List a→ b.
為方便說明，此後我們將 foldr (⊕) e之中的 e稱作基底值 (base value)，將

(⊕)稱作步驟函數 (step function). 函數 foldr的型別可以理解為：給一個型別為
a→ b→ b的步驟函數，和一個型別為 b的基底值，foldr就能將一個 List a轉
換為 b.
串列的摺 (foldr)只是一個常用的特例—「將資料結構中的建構元代換掉」的

動作也可推廣到其他資料結構上。我們在之後的章節中將看到一些其他資料結
構上的摺。
下一節將舉更多使用 foldr 的例子。在那之前我們再次提醒讀者：在

foldr (⊕) e (x : xs) 的狀況中，(⊕) 可以使用 x 與 foldr (⊕) e xs 的結果，但不
能直接使用 xs.

6.1.1 更多串列上的摺

回顧起來，我們可發現第 2章介紹的許多函數都是 foldr.

例 6.1. 以下函數都可寫成 foldr:
• concat = foldr (++) [ ].
• filter p = foldr (λx xs→ if p x then x : xs else xs) [ ],
• takeWhile p = foldr (λx xs→ if p x then x : xs else [ ]) [ ],
• elem x = foldr (λy b→ x = = y ∨ b) False,
• all p = foldr (λx b→ p x ∧ b) True.

串列連接 (++) :: List a→ List a→ List a 雖是個二元運算，若將 (++ys) ::
List a→ List a視為一個函數，它可寫成一個 foldr:

(++ys) = foldr (:) ys .

一個重要的特例是當 ys = [ ]時。對任何 xs, xs++[ ] = xs. 因此 (++[ ])是串列上
的 id:

id ::List a→ List a
id = foldr (:) [ ] .

確實，將一個串列中的 (:)代換成 (:), [ ]代換成 [ ],我們還是得到原來的串列。
我們日後還會用到「串列上的 id是一個 foldr」的性質。
計算所有前段的函數 inits ::List a→ List (List a)可寫成 foldr:

inits = foldr (λx xss→ [ ] : map (x:) xss) [[ ]] .

計算所有後段的 tails ::List a→ List (List a)也可以寫成 foldr，但需要用一個小
性質。回顧其定義：

tails [ ] = [[ ]]

tails (x : xs) = (x : xs) : tails xs .

乍看之下這不符合 foldr的模式：參數 xs出現在 ... : tails xs的左邊，但在 foldr
的模式中，xs不能出現在遞迴呼叫之外。幸好 tails有一個剛好在此有用的小
特性：tails xs傳回的所有後段中，第一個就是 xs本身：head (tails xs) = xs. 因
此我們可將 tails寫成：

tails = foldr (λx xss→ (x : head xss) : xss) [[ ]] .

由於 tails永遠傳回非空串列，使用 head xss是安全的。

習題 6.1 — 請將以下函數寫成 foldr:
1.perms ::List a→ List (List a) (見第 2.6.4節),
2.sublists ::List a→ List (List a) (見第 2.6.4節),
3.splits ::List a→ List (List a×List a)（見習題 2.27）。

6.1.2 不是 foldr的函數

並非所有輸入為串列的函數都是 foldr. 最明顯的例子是 tail: 我們無法由 x和
tail xs算出 tail (x :xs)。例如，tail [1,2,3 ] = [2,3 ]，但 tail [2,3 ] = [3 ],而 [2,3 ]無
法由 1和 [3 ]組出來。
另一個例子是 dropWhile p. 回顧其定義：

dropWhile p [ ] = [ ]

dropWhile p (x : xs) = if p x then dropWhile p xs else x : xs ,

在歸納情況中，else的分支需傳回 x : xs — xs出現在遞迴呼叫以外的地方。這
樣的程式不是 foldr。當然，這只表示 dropWhile p的這個定義不符合 foldr的模
式. 是否有其他的方式能將 dropWhile p寫成 foldr呢？不論 dropWhile p是怎麼
定義的，考慮 dropWhile even [4,3,6,2 ] = [3,6,2 ],但 dropWhile even [3,6,2 ] = [ ]

—看來，dropWhile p丟掉了太多資訊，使得我們無法保證能從 dropWhile p xs
重組出 dropWhile p (x : xs)。因此，dropWhile p和 tail一樣，是先天上無法寫成
foldr的。

習題 6.2 — 考慮第 2.6.4節的函數 fan:

fan :: a→ List a→ List (List a)
fan y [ ] = [[y ]]
fan y (x : xs) = (y : x : xs) : map (x:) (fan y xs) .

為何這個定義目前的形式不是一個 foldr? 有沒有可能將 fan y 寫成一個
foldr呢？

6.2 摺融合定理

第 0章中提及，「抽象化」意指抽取出我們認為重要的概念、成分，給予一個名
字或符號。如此一來，這個概念正式地「存在」了，我們可以談論它、研究其

性質，並將研究結果應用在所有符合這個抽象概念的事物上。一個程式語言最
重要的功能之一是提供良好的抽象化機制。由於高階函數等等性質，函數語言
讓我們能較容易地對程式結構作抽象。
「摺」是我們找到的一個抽象，許多程式可以表達為摺。而一旦辨識出了摺
這個結構，我們可開始討論所有摺都滿足的性質，這些性質則將可適用於所有
是摺的程式上。
關於摺的性質中，最重要的也許是本節的摺融合定理 (fold-fusion theorem)。
摺融合定理告訴我們一個摺如何能與串接於其後的函數融合起來，成為單

獨的一個摺：

定理 6.2 (摺融合定理 (串列版)). 給定 f ::a→ b→ b, e ::b, h ::b→ c. 如果 h (f x y) =
g x (h y)對所有 x :: a與在 foldr f e的值域中的 y :: b成立，則

h · foldr f e = foldr g (h e) .

性質 h (f x y) = g x (h y) 是該融合能成立的充分條件，我們日後將稱之為
「融合條件 (fusion condition)」。如果定理本身看來太抽象，下述例子也許可給讀
者一些直覺。考慮 [x0,x1,x2 ]:

h (foldr f e [x0,x1,x2 ])

= { foldr之定義 }
h (f x0 (f x1 (f x2 e)))

= {融合條件: h (f x y) = g x (h y) }
g x0 (h (f x1 (f x2 e)))

= {融合條件: h (f x y) = g x (h y) }
g x0 (g x1 (h (f x2 e)))

= {融合條件: h (f x y) = g x (h y) }
g x0 (g x1 (g x2 (h e)))

= { foldr之定義 }
foldr g (h e) [x0,x1,x2 ] .

由此例可看出融合條件 h (f x y) = g x (h y)的作用 —將 h往右推，並將途中經
過的 f 都變成 g,直到碰到 e為止。
定理 6.2可用例行的歸納證明證成：

Proof. 假設融合條件成立，我們需證明對所有 xs, h (foldr f e xs) = foldr g (h e) xs.
情況 xs :=[ ]:

h (foldr f e [ ])
= h e
= foldr g (h e) [ ] .

情況 xs := x : xs:

h (foldr f e (x : xs))
= { foldr之定義 }

h (f x (foldr f e xs))
= {融合條件: h (f x y) = g x (h y) }

g x (h (foldr f e xs))
= {歸納假設 }

g x (foldr g (h e) xs)
= { foldr之定義 }

foldr g (h e) (x : xs) .

註記 我們在歸納情況的第二步使用了融合條件 h (f x y) = g x (h y). 欲使該步
成立，融合條件不須對所有 y都成立 —我們只需要它在 y是 foldr f e的可能結
果時成立即可。這是定理 6.2中「在 foldr f e的值域中的 y」這句話的由來。
在本章接下來大部分的例子中，我們其實可以證明融合條件對所有 y均成

立。但只要我們處理的演算法問題稍微複雜些，我們便會常遇到融合條件只對
foldr f e的值域中的 y成立的情況。我們將在第 6.2.6節中看到一些例子。

6.2.1 將摺融合用於定理證明

定理 6.2有幾種用法：
• 一種可能是用於證明性質：我們希望證明 h · foldr f e與 foldr g (h e)相等，
此時我們已知 h, f , g,與 e.

• 另一種可能是用於生成程式。此時我們通常已知 h, f ,與 e,但不知道 g. 我
們希望找到一個讓融合條件成立的 g,使得 h · foldr f e能在一個 foldr之中
完成。

我們先討論第一種情況。

例 6.3. 回顧 map融合定理 (2.2): map f ·map g = map (f ·g). 第 63頁提供了一個
歸納證明。由於 map g是一個摺，我們也可用摺融合定理證明如下。

map f ·map g
= { map的摺定義 }

map f · foldr (λx ys→ g x : ys) [ ]
= {摺融合 }

foldr (λx ys→ f (g x) : ys) [ ]
= { map的摺定義 }

map (f ·g) .

第二步需要的融合條件只需簡單展開定義即可滿足：

map f (g x : ys)
= { map之定義 }

f (g x) : map f ys .

將上述例子與第 2.4節的歸納證明比較。歸納證明中，最關鍵的是「使用歸
納步驟」的一步，而使用摺融合定理的證明卻沒有這步 —歸納步驟的使用被包
裝、隱藏在摺融合定理中了。而上述例子中關於融合條件的證明，恰巧是原歸
納證明中和問題本身最相關的部分。
我們可說：摺融合定理之於證明，就如同摺之於程式。摺是抽象出的常見

程式骨架，將拆解輸入串列、做遞迴呼叫等動作包裝起來。有了摺，我們不需
自己做遞迴呼叫，只需填入針對特定問題的 f , e等參數的值。摺融合定理則是
抽象出的常見證明骨架，將狀況分析、使用歸納假設等動作包裝起來。有了摺
融合定理，我們不需自己做狀況分析、引用歸納假設，只需填入針對這個問題
的融合條件的證明。
事實上，map融合定理是下述定理的特例：

定理 6.4 (foldr-map融合定理). foldr f e ·map g = foldr (f ·g) e.

我們時常看到 foldr與 map一起出現，此時定理 6.4相當好用。

例 6.5. 我們嘗試證明 sum ·map (2×) = (2×) · sum. 首先考慮等號左手邊的
sum ·map (2×). 由於 sum是一個 foldr,我們可用定理 6.4將該式合併為一個 foldr:

sum ·map (2×)
= { sum之摺定義 }

foldr (+) 0 ·map (2×)
= {定理 6.4: foldr-map融合 }

foldr ((+) · (2×)) 0 .

另一方面，(2×) · sum可以融合成同一個 foldr:

(2×) · sum
= (2×) · foldr (+) 0
= {摺融合 }

foldr ((+) · (2×)) 0 .

其中的融合條件證明如下：

2× (x+ y)
= {乘法與加法之分配率 }

2× x+2× y
= { (·)之定義 }
((+) · (2×)) x (2× y) .

由此我們證明了 sum ·map (2×) = (2×) · sum.

許多等式可用類似的模式證明：為證明 e1 = e2,我們對兩邊都做融合，看是
否能製造出同一個 foldr.

例 6.6. 回顧練習 2.11：證明對所有 f , xs, 與 ys, map f (xs++ ys) = map f xs++
map f ys. 若把 xs提出，這相當於證明：

map f · (++ys) = (++map f ys) ·map f .

其中 (++ys)是 foldr. 因此我們可使用摺融合與 foldr-map融合：

(++map f ys) ·map f
= { foldr-map融合 }

foldr ((:) · f ) (map f ys)
= {摺融合 }

map f · (++ys) .

其中，最後一步的融合條件為

map f (x : zs)
= { map之定義 }

f x : map f zs
= { (·)之定義 }
((:) · f ) x (map f zs) .

雖然看來複雜，其實是運用符號、展開定義即可證成的性質。

在本節的許多例子中，使用摺融合定理大大簡化了證明 —幾乎到了只要把
式子寫下就快要證完了，「沒什麼可說」的地步。我們再看最後一個例子。

例 6.7. 考慮證明第 2.5節中提及的性質: sum · concat = sum ·map sum. 我們可使
用 foldr-map融合定理與摺融合定理：

sum ·map sum
= foldr (+) 0 ·map sum
= { foldr-map融合 }

foldr (λxs n→ sum xs+n) 0
= {摺融合 }

sum · foldr (++) [ ]

= sum · concat .

第二步的 foldr-map融合能成立的原因是 (+) · sum展開之後確實成為 (λxs n→
sum xs+n). 這一步也可改用摺融合定理證明，其融合條件為 sum (sum xs : ys) =
sum xs+ sum ys,只需展開定義即可證成。
倒數第二步的摺融合的條件為：sum (xs++ys) = sum xs+sum ys. 這是第 2.5節

的證明中必須發明的關鍵性質。我們再一次看到：使用摺融合定理讓我們只需

提供一個證明中最與問題相關的關鍵部分。

習題 6.3 — 請用摺融合證明 foldr-map融合定理 (6.4).

習題 6.4 — 使用 foldr-map 融合和摺融合證明 sum ·map length = length ·
concat.
習題 6.5 — 使用摺融合定理證明對所有 f , map f · concat = concat ·
map (map f ).

習題 6.6 — 給定 f ,g ::a→ List a與 p ::a→Bool,試證明：如果 filter p (f x) =
if p x then g x else [ ],則 concat ·map (filter p · f ) = concat ·map g ·filter p.

6.2.2 以摺融合生成程式

如前所述，另一種使用摺融合的理由是生成程式：我們希望 h · foldr f e能在一
個 foldr之中完成。此時我們已知 h, f ,與 e,希望用融合條件找出適合的步驟函
數。

例 6.8. 回顧第 5.1節的例子：給定 sumsq = sum ·map square，我們希望找出一個
不產生中間串列的版本。由於 map square是一個 foldr,我們嘗試將 sum融合進
map square中，希望找出能滿足 sumsq= foldr g e的 g與 e . 顯然 e= sumsq [ ] = 0.
為了找出滿足融合條件的步驟函數 g,我們推算：

sum (square x : xs)
= { sum之定義 }

square x+ sum xs
= {提出 x與 sum xs }
(λx y→ square x+ y) x (sum xs) .

因此，根據定理 6.2, sumsq = foldr (λx y→ square x+ y) 0.

例 6.9. 給定一個整數串列，其中由左到右的數字表示對一個帳戶存款或提款的
金額：正數為存款、負數為提款。我們想確定在任何一個時刻帳戶金額不至於

變成負數。一個可能做法是：對該串列的每一個前段算總和，我們可得到每個

時刻的帳戶金額。接著看看其中最小值是否為負數即可。定義函數 noOverdraft
如下：

noOverdraft :: Int→ Bool

noOverdraft = (⩾ 0) ·minimum ·map sum · inits .

我們試著導出一個比較快速的版本。

為增進效率，我們試試看能否把 minimum ·map sum · inits合併為一個 foldr.
回顧：inits = foldr (λx xss→ [ ] : map (x:) xss) [ ]. 我們可以一口氣把 minimum ·
map sum融合進 inits,也可分兩次進行，先將 map sum · inits融合成一個 foldr,再
與 minimum融合。
此處我們嘗試後者，先將map sum · inits融合。基底值為map sum [[ ]] = [0 ],而

步驟函數 step1須滿足的融合條件為map sum ([ ] :map (x:) xss)= step1 x (map sum xss).
試計算如下：

map sum ([ ] : map (x:) xss)
= 0 : map sum (map (x:) xss)
= { map融合 }

0 : map (sum · (x:)) xss
= { sum (x : xs) = x+ sum xs, map融合 }

0 : map (x+) (map sum xss) .

因此我們得到

map sum · inits = foldr (λx ys→ 0 : map (x+) ys) [0 ] .

下一步是將 minimum融合進 map sum · inits. 基底值為 minimum [0 ] = 0,而步
驟函數 step2 須滿足 minimum (0 : map (x+) ys) = step2 x (minimum ys). 計算如下：

minimum (0 : map (x+) ys)
= 0↓minimum (map (x+) ys)
= { minimum (x+) ys = x+minimum ys，後述 }

0↓ (x+minimum ys) .

最後一步使用的性質 minimum (x+) ys = x+minimum ys尚待證明，其關鍵性質
是 (x+)可分配進 (↓)之中：x+(y↓ z) = (x+ y)↓ (x+ z). 總之，我們得到

noOverdraft
= (⩾ 0) ·minimum ·map sum · inits
= (⩾ 0) · foldr (λx y→ 0↓ (x+ y)) 0 .

這是一個只需線性時間的演算法。

我們能否把 (⩾ 0)也融入 foldr之中呢？要使這個融合成立，我們得找到滿
足 0↓ (x+ y)⩾ 0≡ step3 x (y ⩾ 0)的 step3. 演算如下：

0↓ (x+ y)⩾ 0
≡ { a↓b ⩾ c≡ a ⩾ c ∧ b ⩾ c }

0 ⩾ 0 ∧ x+ y ⩾ 0
≡ x+ y ⩾ 0
≡ {希望找到這樣的 step3 }

step3 x (y ⩾ 0) .

然而我們無法找到這樣的 step3 —僅由 y ⩾ 0我們無法得知 x+ y ⩾ 0是否成立。
我們可說 (⩾ 0)丟失了太多資訊，使得融合無法成立。
也由於同樣的理由，如果我們最初把問題定義為：

noOverdraft = and ·map (⩾ 0) ·map sum · inits ,

函數 map (⩾ 0)將無法融合進 map sum · inits之中。

使用摺融合論證兩個式子相等的證明常有如下的形式：

h1 · foldr f1 e1

= {摺融合定理 }

foldr g (h1 e1)

= {摺融合定理 }
h2 · foldr f2 e2 .

此時，我們也常需要藉由兩個融合條件之一來發現步驟函數 g是什麼。

例 6.10. 習題 2.28曾證明 length · sublists = exp 2 · length. 此處我們用摺融合定理
試試看。

考慮等式的左手邊，我們嘗試將 length · sublists 融合為一個 foldr. 由於
sublists = foldr (λx xss→ xss++map (x:) xss) [[ ]]（習題 6.1(2)），融合後的 foldr
之基底值為 length [[ ]] = 1. 為找出步驟函數，我們推算：

length (xss++map (x:) xss)
= length xss+ length (map (x:) xss)
= { length (map f ) = length }

2× length xss ,

由此得到步驟函數 (λx n→ 2×n).
因此該等式可證明如下：

length · sublists
= length · foldr (λx xss→ xss++map (x:) xss) [[ ]]
= {摺融合定理，如上 }

foldr (λx n→ 2×n) 1
= {摺融合定理，如下 }

exp 2 · foldr 1+ 0
= exp 2 · length .

在第二次摺融合中，基底值 exp 2 0 確實是 1. 融合條件為 exp 2 (1+ n) = 2×
exp 2 n.

習題 6.7 — 回顧例 6.9. 試著將 map (⩾ 0)融入 map sum · inits中，說說看為
何該融合會失敗。
習題 6.8 — 使用摺融合定理證明 sum (xs++ ys) = sum xs+ sum ys. 提示：這
相當於證明 sum · (++ys) = (+(sum ys)) · sum.

習題 6.9 — 參考習題 6.2, 使用摺融合定理證明 length (fan y xs) =

1+ (length xs).

習題 6.10 — 回顧第 5.3.2節中將反轉表示的二進位數字轉為自然數的函數
decimal ::List Bool→ N. 該函數可寫成一個摺：

decimal = foldr (λc n→ if c then 1+2×n else 2×n) 0

請使用摺融合將 exp b ·decimal表示成單一的摺。

摺融合與尋找歸納定義 回顧：id ::List a→ List a可以寫成一個 foldr — id =

foldr (:) [ ]. 而任何函數 f ::List a→ b都等於 f · id. 如果我們將 f 與 id融合，會
發生什麼事呢？首先我們需要找出基底值 f [ ]是什麼。接著我們要找到滿足
f (x : xs) = step x (f xs)的步驟函數 step. 但這其實就是使用展開-收回轉換尋找 f
的歸納定義！只是此處要求的歸納定義比較嚴格：在 f xs之外不能使用 xs.
確實，第 5.1與 5.3節中許多尋找歸納定義的演算都可以視為使用摺融合生

成程式的例子。以第 5.3.1節的 poly為例。找出其歸納定義的過程可以視為摺
融合：

poly x
= poly x · id
= { id = foldr (:) [ ] }

poly x · foldr (:) [ ]
= {摺融合定理 }

foldr step (poly x [ ]) .

其中基底值 poly x [ ] = 0. 而函數 step須滿足融合條件 poly x (a:as)= step a (poly x as)。
尋找 step的過程和第 114頁的計算完全相同。我們會得到

poly x (a : as) = a+(poly x as)× x .

到此為止我們便找到了 poly x 的歸納定義。也可以說，我們已得知 poly x =

foldr (λa b→ a+b× x) 0.

6.2.3 掃描

本節將介紹一個本書首次提及，初見時較難理解，但在許多演算法中扮演重要
角色的組件函數：掃描（scan）。
如我們所知，函數 sum :: List Int→ Int 計算一個串列的總和。如果我們想

計算一個串列由右到左的累計和，例如當給定串列 [3,7,2,4 ]，我們希望得到
[16,13,6,4,0 ]（其中 6 = 2+4, 13 = 7+2+4，16 = 3+7+2+4,而 0是空串列
的和），該怎麼做呢？
在第 2.6.3節中，我們曾提及計算一個串列所有後段 (suffixes)的函數 tails ::

List a→ List (List a)。例如，tails [3,7,2,4 ]將得到 [[3,7,2,4 ], [7,2,4 ], [2,4 ], [4 ],
[ ]]。對串列的每一個後段算總和，我們便得到累計和 [16,13,6,4,0 ]了：

runsum ::List Int→ List Int

runsum = map sum · tails .

由於使用多個 sum函數走訪每個後段，如此定義出的 runsum將是一個執行時
間為 O(n2)的函數。但讀者想必已覺得可不用如此費事：我們應該可以在由右
到左走訪串列的過程中記住目前為止的和，避免重算 sum。這該怎麼做呢？
回想：sum可寫成一個摺。因此我們可稍微推廣一下，定義函數 scanr 如

下：

scanr :: (a→ b→ b)→ b→ List a→ List b
scanr f e = map (foldr f e) · tails .

給定一個串列 xs, scanr f e 先算出 xs 的所有後段，然後對每一個後段都做
foldr f e. 前述的 runsum其實是 scanr的特例：runsum = scanr (+) 0.
如果把上述的 scanr定義當作演算法，處理長度為 n的串列時呼叫 f 的次數

為 O(n2). 我們找找看是否有比較快的演算法。
第 6.1.1節中提及 tails是一個 foldr:

tails = foldr (λx xss→ (x : head xss) : xss) [[ ]] ,

也許我們可試著把 map (foldr f e)融合入 tails中，看看是否能找出一個較有效
率的 scanr定義。其融合條件如下：

map (foldr f e) ((x : head xss) : xss)
= { map之定義 }

foldr f e (x : head xss) : map (foldr f e) xss
= { foldr之定義 }

f x (foldr f e (head xss)) : map (foldr f e) xss
= { g (head ys) = head (map g ys);令 g := foldr f e, ys := xss }

f x (head (map (foldr f e) xss)) : map (foldr f e) xss
= {取出 map (foldr f e) xss }

let ys = map (foldr f e) xss
in f x (head ys) : ys .

於是我們推導出了 scanr的另一個定義：

引理 6.11 (掃描引理). 對所有 f , e,

scanr f e = foldr (λx ys→ f x (head ys) : ys) [e ] .

若將 foldr的定義展開，函數 scanr可寫成下列的歸納形式，也許比較容易
理解：

scanr f e [ ] = [e ]
scanr f e (x : xs) = f x (head ys) : ys ,

where ys = scanr f e xs .

在第 6.1.1節中，使得 tails能寫成一個 foldr 的重要性質是 head (tails xs) = xs
– tails xs 的第一個元素就是 xs 本身。使用 tails 定義的 scanr 自然繼承了相
關的性質：scanr f e xs 的第一個元素就是 foldr f e xs, 因此可直接用 head 取
出，不需每次重新計算。用本節開頭的例子說明，以下我們令 scr = (λx ys→
x+head ys : ys):

scanr (+) 0 [3,7,2,4 ]
= scr 3 (scr 7 (scr 2 (scr 4 [0 ])))
= scr 3 (scr 7 (scr 2 [4+0,0 ]))
= scr 3 (scr 7 [2+4+0,4+0,0 ])
= scr 3 [7+2+4+0,2+4+0,4+0,0 ]
= [3+7+2+4+0,7+2+4+0,2+4+0,4+0,0 ]
= [16,13,6,4,0 ] ,

其中每個 scr都可直接使用之前累積計算的結果，不用從頭加起。

6.2.4 香蕉船定理

第 5.5.4節簡短地提到一個例子：令 sumlen= ⟨sum, length⟩, 1直接執行的話，sum
與 length將各自走訪輸入串列一次，但我們可推導出 sumlen的歸納定義，得
到一個只走訪串列一次的版本。
上述例子還可以更通用一些。考慮 ⟨foldr f1 e1, foldr f2 e2⟩ —這個算式拿一個

串列當輸入，兩個 foldr 分別將串列走訪一次，兩個結果分別存放在序對中。
我們有可能將它變成一個摺（因此只走訪串列一次）嗎？下述的「香蕉船定理
(banana-split theorem)」告訴我們：含兩個摺的分裂，可以寫成一個摺。

定理 6.12 (香蕉船定理). 給定 f1 :: a→ b→ b, e1 :: b , f2 :: a→ c→ c, e2 :: c,下述等
式成立：

⟨foldr f1 e1, foldr f2 e2⟩= foldr g (e1,e2) ,

其中 g x (y,z) = (f1 x y, f2 x z).

分裂運算元 ⟨·, ·⟩ 的英文稱呼是 ‘split”, 在程式推導圈內有時會用一套稱作
「香蕉括號 (banana brackets)」的符號表示摺，兩者合起來便是 ‘banana-split” —
甜點「香蕉船」的英文名稱。
定理 6.12相當於把兩個處理同一份資料的迴圈合併成一個。但如同第

5.5.4節提及，這並不保證效率會比較好。我們會使用定理 6.12的原因可能是如
果確定某函數是摺，我們能做更多後續處理（例如，使用掃描定理或其他只對
摺成立的性質）。

習題 6.11 — 證明香蕉船定理。你可以在輸入串列上做歸納證
明，也可以利用 ⟨foldr f1 e1, foldr f2 e2⟩ = ⟨foldr f1 e1, foldr f2 e2⟩ · id =

⟨foldr f1 e1, foldr f2 e2⟩ · foldr (:) [ ]的特性，使用摺融合定理。

「組對」常可視為分裂與 id的融合。例如，在第 5.5.1節的陡串列問題中，我們
定義 steepsum xs = (steep xs,sum xs)，並試著推導其歸納定義。該過程也可視為
將 ⟨steep,sum⟩與 id融合：

⟨steep,sum⟩
= { f · id = f }
⟨steep,sum⟩ · id

= { id = foldr (:) id }
⟨steep,sum⟩ · foldr (:) [ ]

= {摺融合 }
foldr (λx (b,s)→ (x> s ∧ b,x+ s)) (True,0) .

其融合條件的證明與第 5.5.1節中幾乎相同。

1分裂運算元 ⟨·, ·⟩的定義請參照第 1.6.3節，36頁。

6.2.5 累積參數與摺融合

第 5.6節介紹的「累積參數」技巧也常可視為高階函數與摺的融合。以第 5.6.1節
的經典例子 —串列反轉為例。函數 reverse ::List a→ List a是一個 foldr:

reverse = foldr (λx xs→ xs++[x ]) [ ] .

為增進其效率，我們創造了函數 revcat，其定義為：

revcat ::List a→ List a→ List a
revcat xs ys = reverse xs++ ys .

但如果把參數都移除，上述定義其實等同於：

revcat = (++) · reverse .

推導 revcat的歸納定義就是計算 (++)與 reverse的融合！
為了導出一個較快的 revcat 實作，我們嘗試把 (++) · reverse 融合為一個

foldr. 其推導大綱如下：

(++) · reverse
= (++) · foldr (λx xs→ xs++[x ]) [ ]
= {摺融合，試著計算出 base與 step }

foldr step base .

我們可暫停一下，看看這個式子的型別。函數 (++)的型別為 List a→ (List a→
List a), (++) · reverse與 foldr step base的型別也相同。如果摺融合成功，我們會
得到的是一個傳回函數的 foldr —輸入為 List a,輸出為 List a→ List a. 其中 base
的型別為 List a→ List a, 而 step 的型別將是 a→ (List a→ List a)→ (List a→
List a) — step x將一個函數轉成另一個函數。
根據摺融合定理，基底值 base是

(++) [ ] = (λxs→ (++) [ ] xs)
= (λxs→ [ ]++ xs)
= (λxs→ xs)
= id .

步驟函數 step須滿足的融合條件如下

(++) ((λx xs→ xs++[x ]) x xs) = step x ((++) xs) .

簡單地化簡等號左手邊，我們得到：

(++) (xs++[x ]) = step x ((++) xs) .

這個式子無法再規約，因為 (++)還需要一個參數。因此我們根據外延相等（定
義 1.8），在等號兩邊各補一個參數 ys:

(++) (xs++[x ]) ys = step x ((++) xs) ys .

為找出 step,演算如下：

(++) (xs++[x ]) ys
= (xs++[x ])++ ys
= { (++)之結合律 }

xs++([x ]++ ys)
= { (·)之定義 }
(((++) xs) · (x:)) ys

= {將 x, ((++) xs),與 ys提出 }
(λx f → f · (x:)) x ((++) xs) ys .

根據外延相等，我們已證明

(++) (xs++[x ]) = (λx f → f · (x:)) x ((++) xs) .

因此 step = (λx f → f · (x:)),而 revcat可寫成如下的摺：

revcat = foldr (λx f → f · (x:)) id .

例如 revcat "abc" = id · ('c':) · ('b':) · ('a':), 而 revcat "abc" ys = 'c' : ('b' :
('a' : ys)).

6.2.6 引入脈絡

第 6.2節開頭（第 144頁）曾提及：使用摺融合定理將 h · foldr f e融合成 foldr g (h e)
時，融合條件 h (f x y) = g x (h y)並不需對所有 y都成立，而只需對在 foldr f e
的值域內的 y成立即可。截至目前為止我們看了不少摺融合的例子，但我們所
證明的融合條件，均是較寬鬆、對所有 y都成立的。我們還沒看過只對特定 y
成立（因此可能較難證明的）融合條件。
知道「y在 foldr f e的值域內」，意味著證明融合條件時，我們可以假設 y滿

足所有 foldr f e的傳回值該滿足的性質。習慣上，「使用這些性質」被稱作「引
入脈絡」(bringing in the context) —意指做證明時知道 y不是任意的值，而是由
foldr f e產生（有特定脈絡）的。此處我們看一個引入脈絡的簡單例子。

例 6.13. 回顧 sumsq= sum ·map square. 我們在例 6.8中曾導出 sumsq= foldr (λx y→
square x+ y) 0. 現在考慮下述函數：

psuc n = if n ⩾ 0 then n+1 else 0 .

如果參數是非負整數，psuc將它加一，否則傳回 0. 我們能將 psuc · sumsq融合
為一個 foldr嗎？
直覺看來，由於 sumsq一定傳回非負整數，psuc只是將其結果加一。但如

果使用摺融合，需要的融合條件是 psuc (square x+ y) = step x (psuc y). 我們找
不到一個對任意 x與 y都能使該條件成立的 step.
此時我們需要引入脈絡：由於 y是 sumsq的結果，必定是個非負整數。我們

可證明對於非負整數 m與 n,

psuc (m+n) = m+psuc n .

由於 square x與 y都是非負整數，我們有

psuc (square x+ y) = square x+psuc y .

因此我們可選 step x y = square x+ y —與 sumsq的步驟函數相同。至於基底值
則是 psuc 0 = 1. 因此，

psuc · sumsq = foldr (λx y→ square x+ y) 1 .

例 6.13是個刻意設計的、簡單的例子。在第 TODO節中，我們會看到其他
需要引入脈絡的演算法。

6.3 左摺、串列同構、與 Paramorphism

為了完整性，本節介紹與串列的摺相關的另外幾個組件函數。初次閱讀時可跳
過本節。

6.3.1 左摺

除了 foldr, Haskell的標準函式庫有另一個函數 foldl,名字中的字母 l為「左」之
意：foldr將串列中的元素往右括，foldl則往左括，例如，

foldl (▷) e [x0,x1,x2,x3 ] = (((e ▷ x0)▷ x1)▷ x2)▷ x3 .

有了 foldl,串列反轉可直接定義為 reverse = foldl (λxs x→ x : xs) [ ].
函數 foldl可在輸入串列的長度上歸納定義如下：

foldl :: (b→ a→ b)→ b→ List a→ b
foldl (▷) e [ ] = e
foldl (▷) e (xs++[z ]) = foldl (▷) e xs ▷ z .

由於運算元 (▷)右邊的參數才是目前的元素，其型別為 b→ a→ b. 由此出發，
我們不難導出下述將串列從左邊開始拆解的定義：

foldl (▷) e [ ] = e
foldl (▷) e (x : xs) = foldl (▷) (e ▷ x) xs .

我們可注意到這是一個尾遞迴定義，因此，如果將 sum, prod 等函數定義為
foldl (+) 0, foldl (×) 1等等，執行時可不佔用堆疊的空間。由於這個特性，對
一些將 Haskell 用於著重效率的應用的人們來說，foldl 才是他們較常使用的
「摺」。2

關於 foldl與 foldr的關係，Bird [1998]提及了三個定理。首先，
2為了效率因素更常被使用的可能是另一個嚴格 (strict)版的函數 foldl′,該函數在遞迴呼叫前會

先將 e ▷ x規約成正規式，避免尚待計算的算式被累積著。

定理 6.14 (第一對偶定理 (The First Duality Theorem)). 如果 (⊕) :: a→ a→ a滿
足遞移律，以 e :: a為單位元素，則 foldr (⊕) e xs = foldl (⊕) e xs.

提醒一下：該定理及本節的其他定理中，xs 需為歸納定義、有限長度的
串列。遞移律意指 x⊕ (y⊕ z) = (x⊕ y)⊕ z — 由於此項要求，(⊕) 的型別得是
a→ a→ a. 由於 (+), (×), (↓), (↑)等運算元都有遞移律，sum, prod, maixmum,
minimum等函數都可用 foldl定義。事實上，他們在標準 Haskell Report中的定
義便是如此。
定理 6.14其實是下述定理的特例。函數 foldl有個尾遞迴定義，其中參數 e

的角色就像是個累積參數。第 5.6.2節中曾提及，尾遞迴函數時常和遞移律有
關。函數 foldl是否也牽涉了某種遞移律呢？

定理 6.15 (第二對偶定理 (The Second Duality Theorem)). 如果二元運算元 (◁) ::
a→ b→ b及 (▷) :: b→ a→ b滿足 x ◁ (y ▷ z) = (x ◁ y) ▷ z以及 x ◁ e = e ▷ x，
則 foldr (◁) e xs = foldl (▷) e xs.

定理 6.15中的 x ◁ (y ▷ z) = (x ◁ y) ▷ z可視為一種擴充到兩個運算元的遞
移律。定理 6.14則是 (◁) = (▷)的特殊情況。
定理 6.15的證明頗值得研究：

Proof. 我們在 xs 上做歸納。基底狀況 xs := [ ] 中，等號兩邊都歸約為 e. 至於
xs := x : xs的情況，我們把兩邊都化簡並使用歸納假設，推論如下：

foldr (◁) e (x : xs)
= { foldr之定義 }

x ◁ foldr (◁) e xs
= {歸納假設 }

x ◁ foldl (▷) e xs
= {論證如後述 }

foldl (▷) (x ◁ e) xs
= {假設：x ◁ e = e ▷ x }

foldl (▷) (e ▷ x) xs
= { foldl之定義 }

foldl (▷) e (x : xs) .

在等式推論中段，我們希望 x ◁ foldl (▷) e xs = foldl (▷) (e ▷ x) xs. 這會需要另
一個歸納證明。但我們若直接證明此等式，到中途便會無法進行，且會發現我
們需要稍做一下推廣，改證明一個較強的性質：

x ◁ foldl (▷) y xs = foldl (▷) (x ◁ y) xs . (6.1)

兩者的差別是後者不針對特定的值 e,而是任何的 y. 這是一個「較強的性質
反而比較好證明」的例子。
等式(6.1)的證明如此進行：在 xs :=[ ]的情況中，等號兩側都規約為 x ◁ y.

歸納情況 xs := z : xs證明如下：

x ◁ foldl (▷) y (z : xs)
= { foldl之定義 }

x ◁ foldl (▷) (y ▷ z) xs
= {歸納假設 }

foldl (▷) (x ◁ (y ▷ z)) xs
= {假設：x ◁ (y ▷ z) = (x ◁ y)▷ z }

foldl (▷) ((x ◁ y)▷ z) xs
= { foldl之定義 }

foldl (▷) (x ◁ y) (z : xs) .

最後一個對偶定理則確認了我們的一個直覺理解：將一個串列反轉做 foldl,
便相當於 foldr:

定理 6.16 (第三對偶定理 (The Third Duality Theorem)). 對所有 f :: a→ b→ b,
e :: b,以及 xs ::List a, foldr f e xs = foldl (flip f ) e (reverse xs). 其中 flip f x y = f y x.

6.3.2 串列同構

考慮函數 h ::List a→ b. 如果存在 e :: b, f :: a→ b,和 (⊙) :: b→ b→ b使得 h滿足
下列等式：

h [ ] = e
h [x ] = f x
h (xs++ ys) = h xs⊙h ys ,

我們便說 h是一個串列同構 (list homomorphism),記為 h = hom (⊙) f e. 注意第
三個等式蘊含 (⊙)（至少在 h的值域內）須滿足結合律：h xs⊙ (h ys⊙ h zs) =
h (xs++(ys++ zs)) = h ((xs++ ys)++ zs) = (h xs⊙h ys)⊙h zs.
串列同構有平行計算的可能性：計算 hom (⊙) f e xs時，如果 xs有不只一個

元素，我們可將 xs由中間任意截成兩段，分別給不同的處理器計算，再將結
果用 (⊙)結合。
顯然，hom dot f e可以寫成 foldr,也能寫成 foldl:

hom (⊙) f e = foldr (λx y→ f x⊙ y) e
= foldl (λy x→ y⊙ f x) e .

下述定理則告訴我們反過來也成立：如果 h同時可寫成 foldr及 foldl,則 h是一
個串列同構：

定理 6.17 (第三同構定理 (The Third Homomorphism Theorem)). 考慮 h ::List a→
b。如果存在 e :: b, (◁) :: a→ b→ b, 及 (▷) :: b→ a→ b 使得 h = foldr (◁) e =

foldl (▷) e,則存在 (⊙) ::b→ b→ b使得 h = hom (⊙) f e（其中 f x = x ◁ e = e ▷ x）
.

6.3.3 Paramorphism與本原遞迴

如果我們把 foldr的限制放寬，允許 xs出現在遞迴呼叫之外，得到的模式稱為
paramorphism. 串列版的 paramorphism可定義如下：

para :: (a→ List a→ b→ b)→ b→ List a→ b
para f e [ ] = e
para f e (x : xs) = f x xs (para f e xs) .

para f e = fst · foldr (λx (y,xs)→ (f x xs y,x : xs)) (e, [ ]) .

6.4 自然數的摺

自然數可以視為有兩個建構元 0和 1+ 的歸納資料結構。我們也可為自然數定
義一個摺：

foldN :: (a→ a)→ a→ N→ a
foldN f e 0 = e
foldN f e (1+ n) = f (foldN f e n) .

和串列的情況類似，函數 foldN f e拿一個自然數，將其中的 0代換成 e, 1+ 代
換成 f . 留意型別：0與 1+ 的型別分別是 N與 N→ N, foldN 要將一個型別為 N
的值轉變為型別為 a的值,因此 e的型別為 a,而 f 的型別為 a→ a. 為方便稱呼，
我們也將 e與 f 分別稱為基底值與步驟函數。
許多我們定義過的、自然數上的函數都可寫成自然數的摺：
• exp b = foldN (b×) 1,
• (+n) = foldN (1+) n,
• (×n) = foldN (n+) 0.

自然數上的 id ::N→ N也可寫成摺：id = foldN (1+) 0.
自然數的摺也有一個融合定理：

定理 6.18 (摺融合定理 (自然數版)). 給定 f :: a→ a, e :: a, h :: a→ b. 如果對所有
在 foldN f e值域中的 x :: a,融合條件 h (f x) = g (h x)成立,則

h · foldN f e = foldN g (h e) .

例 6.19. 判斷一個自然數是否為偶數的述語 even ::N→ Bool可寫成一個摺:

even = foldN not True .

函數 even ·(+n)判斷一個數值加上 n之後是否為偶數。由於 (+n) = foldN (1+) n,
我們可以嘗試把 even融入 (+n),變成一個摺。根據定理 6.18,基底值為 even n;
而由於 even (1+ n) = not (even n),步驟函數為 not. 因此：

even · (+n) = foldN not (even n) .

習題 6.12 — 回顧第 2.11節中提到的費氏數定義：

fib 0 = 0
fib 1 = 1
fib (2+n) = fib (1+n)+fib n .

若直接將上述定義當作演算法，我們得做許多重複的計算。定義 fib2 n =

(fib (1+n),fib n). 請將 fib2融合進 id ::N→ N,以便得到一個遞迴呼叫次數
為 O(n)的演算法。

6.5 其他資料結構

既然串列與自然數都有摺，其他的資料結構也可以有。

二元樹 回顧我們提及的兩種常見二元樹：

data ITree a = Null | Node a (ITree a) (ITree a) ,

data ETree a = Tip a | Bin (ETree a) (ETree a) .

其中 ITree的摺可定義如下：

foldIT :: (a→ b→ b→ b)→ b→ ITree a→ b
foldIT f e Null = e
foldIT f e (Node x t u) = f x (foldIT f e t) (foldIT f e u) .

內標記二元樹 ITree 的兩個建構元之型別分別為 Null :: ITree a 與 Node :: a→
ITree a→ ITree a→ ITree a. 和串列的摺一樣，內標記二元樹的摺將一個 ITree a
轉成一個型別為 b的值 —藉由將 Null代換為基底值 e :: b,以及將 Node代換為
步驟函數 f :: a→ b→ b→ b.
外標記二元樹的摺則可定義如下：

foldET :: (b→ b→ b)→ (a→ b)→ ETree a→ b
foldET f k (Tip x) = k x
foldET f k (Bin t u) = f (foldET f k t) (foldET f k u) .

型別 ETree 的建構元分別為 Tip :: a→ ETree a 和 Bin :: ETree a→ ETree a→
ETree a. 由於 Tip 是一個由 a 到 ETree a 的函數，取代它的得是一個型別為
a→ b的基底函數. 取代 Bin的步驟函數之型別則為 b→ b→ b. 有了這兩者，
我們便可將 ETree a轉換為 b.
例如，第 2.8節中曾提到幾個定義在樹之上的函數：tags傳回一個 ITree的

所有標記；size傳回其大小；minE 傳回一個 ETree的最小元素，mapE f 對樹
中的每個標記做 f . 它們都可用摺定義：

tags = foldIT (λx xs ys→ xs++[x ]++ ys) [ ] ,

size = foldIT (λx m n→ 1+m+n) 0 ,

minE = foldET (↓) id ,

mapE f = foldET Bin f .

兩個二元樹的摺也有它們的融合定理：

定理 6.20 (摺融合定理 (ITree版)). 給定 f :: a→ b→ b→ b, e :: b, h :: b→ c, 與
g :: a→ c→ c→ c. 如果融合條件 h (f x y z) = g x (h y) (h z) 對任何 x :: a 與在
foldIT f e值域中的 y,z :: b成立，我們有 h · foldIT f e = foldIT g (h e).

定理 6.21 (摺融合定理 (ETree版)). 給定 f :: b→ b→ b, k :: a→ b, h :: b→ c, 與
g :: c→ c→ c. 如果融合條件 h (f x y) = g (h x) (h y)對任何在 foldET f k值域中
的 x,y :: b成立，我們有 h · foldET f k = foldET g (h · k).

兩個定理的融合條件都依循著與串列版相同的原則：當 h與步驟函數碰在一
起，融合條件讓我們將 h往裡推。兩個定理都可用單純的歸納法證明。

習題 6.13 — 以摺融合定理證明 length (tags t) = size t.

習題 6.14 — 串列有 foldr-map 融合定理 (6.4)，ETree 也有類似的 foldET-
mapE融合定理。請寫出該定理並證明之。
習題 6.15 — 函數 mapI :: (a→ b)→ ITree a→ ITree b將一個 a→ b的函數
作用在 ITree 的每一個標記上。請用 foldIT 定義 mapI, 並寫下 foldIT 與
mapI 的融合定理並證明之。
習題 6.16 — 以摺融合定理證明 minE (mapE (x+) t) = x+minE t.

習題 6.17 — 將 (++) · tags融合，以便推導出一個在線性時間內收集 ITree

內所有標籤的演算法。

非空串列 第 2.10節中曾提及我們可把至少有一個元素的串列想像成一個資料
結構：data List+ a = [a ] | a :List+ a. 此種串列的摺可定義為：

foldrn :: (a→ b→ b)→ (a→ b)→ List+ a→ b
foldrn f k [x ] = k x
foldrn f k (x : xs) = f x (foldrn f k xs) .

例 6.22. 例 2.13中的 parts+可以寫成

parts+ ::List+ a→ List (List+ (List+ a))
parts+ = foldrn (λx→ concat ·map (extend x)) wrap3 ,

where extend x (ys : yss) = [[x ] : ys : yss,(x : ys) : yss ]
wrap3 x = [[[x ]]] .

我們考慮一個定義在非空串列上的簡單演算法推導練習。其次，這也將是
一個使用 foldrn與「引入脈絡」的例子，
下述函數 ascending ::List+ Int→ Bool判斷一個串列是否為遞增：

ascending [x ] = True

ascending (x : y : xs) = x ⩽ y ∧ ascending (y : xs) .

給定一個整數串列，如何將它切成一個個區段，使得每個區段都是遞增的？如
果我們讓每個元素都自己成一段，似乎是滿足需求，但這沒什麼意思。我們希
望讓遞增區段盡量連續，也就是說，我們要區段數目最少的分割法。下述函數
upHills將輸入串列以最精簡的方式切成段：

upHills ::List Int→ List (List Int)

upHills = shortest ·filter (all ascending) ·parts+ .

其中 parts+ 把串列任意切段，filter (all ascending)挑出每個區段都是遞增的分
割法，而 shortest挑選元素數目最少的串列。我們能由此導出一個比較快的演
算法嗎？
我們先將 filter (all ascending)融入 parts+ 之中。經過一些稍繁瑣但原則上

並不困難的計算，我們可得：

filter (all ascending) ·parts+ =

foldrn (λx→ concat ·map (extendAsc x)) wrap3 .

這和 parts+的差別只在 extend變成了 extendAsc. 後者的定義為：

extendAsc x (ys : yss) = if x ⩾ head ys then [[x ] : ys : yss,(x : ys) : yss ]
else [[x ] : ys : yss ] .

函數 extendAsc比 extend多做了一個檢查，只在 x ⩾ head ys時將 (x : ys) : yss列
為一個可能選項。注意：由於 ys的型別是 List+ Int, head一定可成功。如果我
們使用 List Int,在這裡就得多做些條件判斷。雖然每個非空串列 List+ 都可用
List表達,有些問題使用 List+描述時會比較便於證明與推論。
接著我們試圖融合 shortest 與 filter (all ascending) · parts+. 基底函數為

(shortest ·wrap3) x = shortest [[[x ]]] = [[x ]]. 至於步驟函數，我們希望找到滿
足下述融合條件的 step:

shortest (concat (map (extendAsc x) ysss)) = step x (shortest ysss) .

由左手邊開始，由於 shortest可分配進 concat（意即 shortest ·concat = shortest ·
map shortest），我們可推論：

shortest (concat (map (extendAsc x) ysss))
= { shortest分配進 concat; map融合 }

shortest (map (shortest · extendAsc x) ysss) .

為了有些進展，我們看看 shortest ·extendAsc x能如何化簡。將輸入（非空串列）
寫成 ys : yss:

shortest (extendAsc x (ys : yss)) =
= { extend′之定義;提出 if }

if x ⩾ head ys then shortest [[x ] : ys : yss,(x : ys) : yss ]
else shortest [[x ] : ys : yss ]

= { shortest挑選較短之串列 }
if x ⩾ head ys then (x : ys) : yss else [x ] : ys : yss .

因此，融合條件的左手邊可歸約如下：

shortest (map (shortest · extendAsc x) ysss)
= {前述推導 }

shortest (map (λ (ys : yss)→ if x ⩾ head ys then (x : ys) : yss
else [x ] : ys : yss) ysss) .

我們希望繼續將 shortest往裡推，但此時似乎卡住了。

shortest (map (λ (ys : yss)→ if x ⩾ head (head (head ysss)) then (x : ys) : yss
else [x ] : ys : yss) ysss) .

= shortest (if x ⩾ head (head (head ysss))
then map (λ (ys : yss)→ (x : ys) : yss) ysss
else map (λ (ys : yss)→ [x ] : ys : yss) ysss)

= if x ⩾ head (head (head ysss))
then shortest (map (λ (ys : yss)→ (x : ys) : yss) ysss)
else shortest (map (λ (ys : yss)→ [x ] : ys : yss) ysss)

= if x ⩾ head (head (head ysss))
then (λ (ys : yss)→ (x : ys) : yss) (shortest ysss)
else (λ (ys : yss)→ [x ] : ys : yss) (shortest ysss)

= let (ys : yss) = shortest ysss
in if x ⩾ head ys

then (x : ys) : yss else [x ] : ys : yss .

upHills = foldrn step (λx→ [[x ]]) ,

where step x (ys : yss) = if x ⩾ head ys then (x : ys) : yss
else [x ] : ys : yss .

習題 6.18 — 將 filter (all ascending) · parts+ 融合為 foldrn (λx → concat ·
map (extendAsc x)) wrap3, 並在過程中推導 extendAsc 的定義。您可能用
得上習題 6.6提及的性質：如果 filter p (f x) = if p x then g x else [ ], 則
concat ·map (filter p · f ) = concat ·map g ·filter p.

6.6 參考資料

Gibbons et al. [2001]

區段問題

一個串列中連續的任意一截被稱作一個「區段」(segment)。例如，[1,2,3 ]的
區段包括 [ ],[1 ],[2 ],[3 ],[1,2 ],[2,3 ],以及 [1,2,3 ]本身（注意：空串列也是一個區
段）。也許因為歷史因素，許多有趣的演算法問題有這樣的形式：給定一個串
列 xs ::List A，一個述語 p ::List A→ Bool,和一個函數 f ::List A→ B，我們想在
xs的所有區段中，找出滿足述語 p並使得 f 值最大的那個。我們把這類問題統
稱為「區段問題」。
本節討論一些簡單的區段問題。我們先回顧一下區段的形式定義。第

2.6.3節曾提及，下述函數 segments算出一個串列的所有區段：

segments ::List a→ List (List a)
segments = concat ·map inits · tails .

其中 tails :: List a → List (List a) 計算一個串列所有的後段 (suffixes)，例如
tails [1,2,3 ] = [[1,2,3 ], [2,3 ], [3 ], [ ]]；函數 inits ::List a→ List (List a)則計算一
個串列的所有前段 (prefixes)，例如 inits [1,2,3 ] = [[ ], [1 ], [1,2 ], [1,2,3 ]]. 對每一
個後段，計算所有的前段，便得到所有的區段了。1 事實上，如此的定義之下
segments [1,2,3 ]會將空串列傳回很多次。不過以本章的目的而言，我們對這些
重複的 [ ]並不在意。
為方便讀者，我們將 inits與 tails的歸納定義重複如下。

inits ::List a→ List (List a)
inits [ ] = [[ ]]

inits (x : xs) = [ ] : map (x:) (inits xs) ,

tails ::List a→ List (List a)
tails [ ] = [[ ]]

tails (x : xs) = (x : xs) : tails xs .

1我們也可以反過來，定義 segments = concat ·map tails · inits. 本節中的所有推導與證明可調整
成相對應的版本。

163

7.1 最大區段和

給定一個串列。請問它的眾多區段中，總和最大的和是多少？這個最大區段和
(maximum segment sum)問題可說是最經典的區段問題。我們可將其寫成如下的
規格：

mss ::List Int→ Int

mss = maximum ·map sum · segments ,

這幾乎便是問題的字面翻譯：算出所有的區段，對每個區段算其和，然後挑出
最大的一個。2 當輸入串列長度為 n,這個定義本身是一個時間複雜度為 O(n3)

的演算法 —該串列的區段有 O(n2)個，每個都需分別算總和。我們能導出一個
更快的演算法嗎？

前段-後段分解 許多區段問題的推導都以如下方式開頭：將 segments展開成
inits與 tails,並將 maximum往右推，與 inits放在一起：

maximum ·map sum · segments
= maximum ·map sum · concat ·map inits · tails
= { map f · concat = concat ·map (map f ) (習題 2.13) }

maximum · concat ·map (map sum) ·map inits · tails
= { maximum · concat = maximum ·map maximum }

maximum ·map maximum ·map (map sum) ·map inits · tails
= { map融合（定理 2.2）}

maximum ·map (maximum ·map sum · inits) · tails .

細看 maximum ·map sum · inits這個子算式，其意思是「給定一個串列，計算其
所有前段的和的最大值」。我們為這個子算式取個名字，令 mps = maximum ·
map sum · inits，其中 mps為「最大前段和 maximum prefix sum」的縮寫。經由
上述演算，我們得知

mss = maximum ·map mps · tails ,

意思是：要找出所有區段的最大和，我們可以對每一個後段，找出其最大前段
和。
這是解許多區段問題的常見模式：要解決最佳區段問題，先試著解最佳前

段問題。要算出最佳區段，可對每一個後段，算出其最佳前段。3

使用掃描引理 接下來我們注意到 map mps · tails這個子算式。回顧掃描引理
(6.11),重複如下：

map (foldr f e) · tails = scanr f e = foldr (λx ys→ f x (head ys) : ys) [e ] .

2函數 maximum ::List Int→ Int的定義見例 2.12. 我們得假設整數中有個 −∞作為 maximum [ ]
的結果。若要避免 −∞，可注意到 inits, tails,和 segments都不會傳回空串列 —它們的型別都可寫成
更嚴格的 List+ (List a)。因此我們可改用例 2.12中的 maximum+. 本節的推導稍加修改後即可適用。

3反過來當然也可以。如果我們定義 segments = concat ·map tails · inits,此處的策略就變成「對
每個前段，算出其最佳後段」。

如果我們能把 mps變成一個 foldr，map mps · tails可改寫為 scanr. 如果該摺
的步驟函數只花常數時間，我們就有了一個線性時間的演算法了！
如何把 mps 變成摺呢？由於 inits 是摺，我們可使用摺融合。此處的計

算和例 6.9很類似，我們可以把 map sum 與 maximum 分兩次融合進 inits, 也
可以一次把 maximum ·map sum 融合進 inits. 這次我們試試看後者。基底值
maximum (map sum [[ ]]) = 0. 我們想要尋找滿足融合條件maxmium (map sum ([ ] :
map (x:) xss)) = step x (maximum (map sum xss))的函數 step. 計算如下：

maximum (map sum ([ ] : map (x:) xss))
= { map與 sum之定義 }

maximum (0 : map sum (map (x:) xss))
= { map融合 }

maximum (0 : map (sum · (x:)) xss)
= { sum之定義 }

maximum (0 : map ((x+) · sum) xss)
= { maximum之定義 }

0↑maximum (map ((x+) · sum) xss)
= { maximum ·map (x+) = (x+) ·maximum }

0↑ (x+maximum (map sum xss)) .

因此我們推導出了：

maximum ·map sum · inits
= maximum ·map sum · foldr (λx xss→ [ ] : map (x:) xss) [[ ]]
= {摺融合定理，融合條件如上 }

foldr (λx s→ 0↑ (x+ s)) 0 .

注意：步驟函數推導的最後一步中，為了將 maximum往右推，使用了如下
的性質

maximum ·map (x+) = (x+) ·maximum . (7.1)

在 (7.1)的左手邊，我們將一個串列的每個元素都加上 x,然後取最大值。性質
(7.1)告訴我們，其實我們可以先取最大值，然後做一個加法即可。這個步驟允
許我們在每一步省下了 O(n)個加法，是使得整個演算法之所以能加速的關鍵
一步。而 (7.1)的證明只需使用例行的歸納法，但其中的關鍵一步需要如下的
分配律：

x+(y↑ z) = (x+ y)↑ (x+ z) . (7.2)

加法與 (↑)的分配率是使我們能有一個線性時間演算法的關鍵性質。
既然 mps已經是一個摺，我們可以使用掃描引理：

maximum ·map sum · segments
= {上述計算 }

maximum ·map (foldr (λx s→ 0↑ (x+ s)) 0) · tails

= {掃描引理 6.11 }
maximum · scanr (λx s→ 0↑ (x+ s)) 0 .

我們得到

mss = maximum · scanr (λx s→ 0↑ (x+ s)) 0 .

這是一個使用線性時間、線性空間的演算法。
藉由程式推導，我們不僅找到了一個較快的演算法，也找出了使得該演算

法之所以成立的根本性質。這使我們能輕易將該演算法推廣：不僅是加法與
(↑)，該演算法能用在任何滿足 (7.2)的一組運算元之上。

習題 7.1 — 證明性質 (7.1)。

常數空間 使用掃描引理導出的 mss能在線性時間內對輸入串列的每個後段算
出其 mps（即「最大前段和」）並存放在一個串列中。方法是使用一個 scanr將
串列由右到左走訪一遍，在每一步將 x與 mps xs的值（存放在串列中）相加，
並和 0比大小。每個 mps值之中最大的那個，便是我們要的答案。在函數語言
圈內，關於最大區段和的討論大多到此為止：我們已經有了一個漂亮的線性時
間演算法了。
若要再挑惕，這個演算法的不盡滿意之處是需要線性的空間 — scanr 會產

生一個中間串列，由 maximum 消掉。我們有可能把 maximum 與 scanr 融合，
得到一個不產生中間串列的演算法嗎？根據摺融合定理，我們將需要找到滿足
以下融合條件的函數 step：

maximum (0↑ (x+head ys) : ys) = step x (maximum ys) .

這顯然做不到：從 maximum ys是無法取出 head ys的。
這時就用得上組對的技巧了 —既然需要 head，就把它一併歸納地算出來

吧！定義：

msps = ⟨maximum,head⟩ · scanr (λx s→ 0↑ (x+ s)) 0 .

要把 ⟨maximum,head⟩融入 scanr,我們依融合條件推導：

⟨maximum,head⟩ (0↑ (x+head ys) : ys)
= { ⟨·, ·⟩之定義 }
(maximum (0↑ (x+head ys) : ys),head (0↑ (x+head ys) : ys))

= { maximum與 head之定義 }
((0↑ (x+head ys))↑maximum ys,0↑ (x+head ys))

= {取出 (maximum ys,head ys) }
(λm s→ ((0↑ (x+ s))↑m,0↑ (x+ s))) (maximum ys,head ys)

= {取出重複項 0↑ (x+ s) }
(λm s→ let s′ = 0↑ (x+ s) in (s′ ↑m,s′)) (maximum ys,head ys) .

我們得到



DR
AF
T

7.2. 最長高原問題 167

指令式版本的最大區段和

根據第 5.6.2節的討論，當輸入串列 [x0 . .xN−1 ]被逆向存放在陣列 X之中，函數
msps相當於以下的指令式程式：

i,m,s :=0,0,0;
do i ≠ = N→ s :=0↑ (X i+ s);

m :=m↑ s;
i := i+1

od;
return m

這在指令式程式推導的領域中也是一個經典範例。

mss = fst ·msps ,

msps ::List Int→ (Int× Int)

msps = foldr (λm s→ let s′ = 0↑ (x+ s) in (s′ ↑m,s′)) (0,0) .

或著，我們把 msps的 foldr定義展開，也許比較容易理解：

msps [ ] = (0,0)
msps (x : xs) = let (m,s) = msps xs

s′ = 0↑ (x+ s)
in (s′ ↑m,s′) .

這是一個使用線性時間、常數空間計算最大區段和的演算法。當 (m,s) =
msps xs, m是 xs的最大區段和，s則是 xs的最大前段和。每考慮一個新元素 x,
最大前段和被更新為 0↑ (x+ s) —如果把 x接上後仍是正數，x+ s就是最好的
前段和，否則最大前段和是空串列的和 0. 新的最大前段和再與 m比較，得到
新的最大區段和。

7.2 最長高原問題

如果一個區段的每個元素都相等，我們稱之為一個「高原 (plateau)」。本節
考慮這個問題：給一個串列，找出其中最長的高原的長度。例如當輸入為
[2,3,3,2,2,2,1,6,6 ]時，輸出應為 3 —即最長的高原 [2,2,2 ]的長度。以下是本
問題的一種可能的規格寫法：

lp ::List Int→ Int

lp = maximum ·map length ·filter plateau · segments+ ,

其中 segments+ 算出所有的區段，plateau判斷一個區段是否為高原。過濾出所
有的高原後，我們計算每個高原的長度，然後找出最大值。4

函數 segments+和 segments類似，其定義為：
4函數 lp與 plateau可以有更通用的型別 Eq a⇒ List a→ Int, Eq a⇒ List a→ Bool.

區段問題不傳回區段？

為何我們在最大區段和問題中只傳回最大區段的和、在最長高原問題中只傳回
長度，而不傳回該區段本身呢？

這是許多演算法題目常見的簡化：當真正的問題是尋找「使得某值最大的聚
合資料結構」時，我們常只要求解題者傳回該值，而不是整個聚合資料結構。如
此做的好處之一是讓我們能暫時忽略「如何組出需傳回的資料結構」的細節，更
專注在問題本身。這使得問題的規格與推導過程都簡單許多。

如果我們真的需要整個區段，我們總能將前述的簡單版程式擴充為傳回資
料結構的程式。讀者不妨試試能否將第 ⁇ 頁的 mpsAll 改寫為型別為 List Int→
List (Int×List Int)的程式 —對每個後段，計算其最佳前段和，以及該前段。您會
發現程式結構並沒有改變，只是多了些繁瑣的細節。
另一個理由是：具有最大和的區段可能不只一個。在函數程式推導之中，如

果我們要傳回區段，由於規格也是函數，我們似乎必須在規格中就決定傳回哪一
個。若希望規格是「傳回任一個具有最大和的區段」，我們會需要更多機制，例如
使用關係 (relation)或使用單子。

segments+ ::List a→ List (List+ a)
segments+ = concat ·map inits+ · tails+ ,

函數 inits+與 tails+則與 inits與 tails類似，但只傳回非空的前後段，定義如下：

inits+ ::List a→ List (List+ a)
inits+ [ ] = [ ]

inits+ (x : xs) = [x ] : map (x:) (inits+ xs) ,

tails+ ::List a→ List (List+ a)
tails+ [ ] = [ ]

tails+ (x : xs) = (x : xs) : tails+ xs .

以上三個函數的回傳型別都是 List (List+ a) —每個傳回的前/後/區段都是非空
的，但並非每個串列都有非空的前/後/區段。函數 lp仍可接受空串列作為輸
入，但（我們等下將看到）在規格中使用 segments+ 將帶來不少方便。我們也
可使用 segments定最長平原問題的規格，只是會增加一些不利於講解的細節。
函數 plateau可定義如下：

plateau ::List+ Int→ Bool

plateau [x ] = True

plateau (x : y : xs) = x = = y ∧ plateau (y : xs) .

由於輸入是非空串列，plateau的定義中考慮的是「有一個元素」和「有兩個以
上元素」的兩種情況。

前段-後段分解 和最大區段和問題一樣，我們先嘗試把這個區段問題分解為
「對每個後段，計算最佳前段」：

maximum ·map length ·filter plateau · segments+

= maximum ·map length ·filter plateau · concat ·map inits+ · tails+

= {因 filter p · concat = concat ·map (filter p) }
maximum ·map length · concat ·map (filter plateau · inits+) · tails+

= {因 map f · concat = concat ·map (map f ), map融合 }

maximum · concat ·map (map length ·filter plateau · inits+) · tails+

= {因 maximum · concat = maximum ·map maximum, map融合 }
maximum ·map (maximum ·map length ·filter plateau · inits+) · tails+ .

嘗試使用掃描引理 下一步：我們把 maximum ·map length ·filter plateau · inits+

簡寫為 lpp，並嘗試將它寫成 (⁇)的形式，以便使用掃描引理。演算如下：

(maximum ·map length ·filter plateau · inits+ $ (x : xs))
= (maximum ·map length ·filter plateau $ [x ] : map (x:) (inits+ xs))
= { filter之定義，plateau [x ] = True }

maximum (1 : map length (filter plateau (map (x:) (inits+ xs))))
= 1↑maximum (map length (filter plateau (map (x:) (inits+ xs)))) .

演算至此的問題是：如何把 map (x:)提出來呢？定理 2.4允許我們把 map搬
到 filter的左邊：

filter p ·map f = map f ·filter (p · f ) .

至於 plateau · (x:)能否再化簡？觀察 plateau定義的第二個子句：

plateau (x : y : xs) = x = = y ∧ plateau (y : xs) ,

寫成函數組合的形式便是：

plateau · (x:) = ((x = =) ·head) ∧̇plateau ,

其中 (∧̇)為「提升成函數版」的 (∧),定義為

(∧̇) :: (a→ Bool)→ (a→ Bool)→ a→ Bool

(f ∧̇g) x = f x ∧ g x .

函數 (∧) ::Bool→ Bool→ Bool拿兩個布林值、算出一個新布林值，(∧̇)則拿兩
個函數 f ,g :: a→ Bool，組合成另一個型別為 a→ Bool的函數，其結果是 f 與
g之傳回值的合取。函數 filter與 (∧̇)有如下的性質：

filter (p ∧̇q) = filter p ·filter q . (7.3)

有了以上眾多性質，我們演算如下：

filter plateau ·map (x:)
= {定理 2.4 }

map (x:) ·filter (plateau · (x:))
= { plateau之定義 }

map (x:) ·filter (((x = =) ·head) ∧̇plateau)
= {因 (7.3): filter (p ∧̇q) = filter p ·filter q }

map (x:) ·filter plateau ·filter ((x = =) ·head) .

現在整個式子成為 1 ↑ (maximum ·map (length · (x:)) · filter plateau · filter ((x = =) ·
head) · inits+) $ xs.
把 map (x:)往左搬之後，式子的右邊出現了 filter ((x = =) · head) (inits+ xs) —

產生所有 xs 的非空前段，取出第一個元素為 x 的。但讀者們可能立刻發現：
inits+傳回的每個前段的第一個元素都是一樣的！也就是說我們有以下性質。

filter ((x = =) ·head) (inits+ xs) =

if x = = head xs then inits+ xs else [ ] .
(7.4)

考慮非空前段的好處在這兒可看出：head對非空串列才有值。
回到 lpp，我們可繼續推導如下：

lpp (x : xs)
= {前述演算, map融合 }

1↑ (maximum ·map (length · (x:)) ·filter plateau ·
filter ((x = =) ·head) · inits+ $ (x : xs))

= {因 (7.4) }
1↑ (maximum ·map (length · (x:)) ·filter plateau $

if x = = head xs then inits+ xs else [ ])
= { (2.3): 函數分配進 if }

if x = = head xs
then 1↑ (maximum ·map (length · (x:)) ·filter plateau · inits+ $ xs)
else 1↑ (maximum ·map (length · (x:)) ·filter plateau $ [ ])

= { length · (x:) = (1+) · length,及其他化簡 }
if x = = head xs

then 1+(maximum ·map length ·filter plateau · inits+ $ xs)
else 1

= { lpp之定義 }
if x = = head xs then 1+ lpp xs else 1 .

由此我們得到

lpp [x ] = 1
lpp (x : xs) = if x = = head xs then 1+ lpp xs else 1 .

然而，我們雖為 lpp推導出了一個歸納定義，該定義並不符合 (⁇)！後者要求
lpp的右手邊必須是 x⊕ lpp xs的形式 —在 lpp xs之外不能有其他的 xs,而上述
的 lpp右手邊多了一個 head xs.
這時組對的技巧又派上用場了。定義：

lpph xs = (lpp xs,head xs) ,

我們可推導出

lpph [x ] = (1,x)
lpph (x : xs) = (if x = = y then 1+n else 1,x) ,

where (n,y) = lpph xs .

該函數符合 (⁇)的形式，其中 e = (1,x), x⊕ (y,n) = (if x = = y then 1+n else 1,x).

總結 綜合目前為止的推導，函數 lpp的推導大架構如下：

lpp
= {前段-後段分解 }

maximum ·map (maximum ·map length ·filter plateau · inits+) · tails+

= {前述演算：尋找歸納定義 }
maximum ·map lpp · tails+

= {因 lpp = fst · lpph }
maximum ·map (fst · lpph) · tails+

= {掃描引理 }
maximum ·map fst · lpphAll ,

其中 lpphAll的定義如下：

lpphAll [x ] = [(1,x)]
lpphAll (x : xs) = (if x = = y then 1+n else 1,x) : (n,y) : ys ,

where ((n,y) : ys) = lpphAll xs .

這是一個使用線性時間、線性空間的演算法。

7.3 參考資料

最大區段和 對試圖推銷程式推導的人來說，最大區段和問題幾乎有模範問題
該有的一切特質：目標不難理解，但又不容易一眼看出怎麼解得快；大部分的
推導過程都能用很形式化、依符號推導的方式進行；推出的程式有顯著的效率
提升。程式僅需短短幾行，乍看之下卻不容易理解為何會正確。因此最大區段
和問題是程式推導圈內常常提及的經典例子。
根據 Bentley的 Programming Pearls[Bentley, 1986]一書，最大區段和問題最

初由 Brown大學的 Ulf Grenander所提出。他當時正設計一個圖形配對的函式。
其中，具有最大區段和的二維子陣列是圖形中最有可能含有指定樣式的區域。
二維問題比較難解，因此他先考慮一維的情況：

1977年，[Grenander]把該問題講給UNILOGIC公司的Michael Shamos
聽，後者一夜之間就設計出了演算法 3. Shamos不久後告訴我這個
問題時，我們都認為那大概就是最好的解法了；… 又幾天後，Shamos
在 Carnegie-Mellon大學的專題討論會中講了這個問題和它的來龍去
脈。當時在場的統計學家 Jay Kadane一聽，幾分鐘內就設計出了演
算法 4.

— Jon Bentley, Programming Pearls (第一版),第 76頁.

Kadane的演算法 4就是現在廣為人所知的（指令語言版）線性時間解（見
第 167頁）。

Shamos的演算法 3則是一個採取分而治之 (divide and conquer)法的演算法：
將陣列分成長度略等的兩半 xs++ ys,分別計算 xs與 ys的最大區段和。但除此
之外，還得考慮跨越 xs與 ys的區段。因此該演算法在遞迴的每一層又多用了

Shamos的最大區段和演算法

以下是我根據個人理解將 Shamos的演算法 3寫成的 Python程式。

def mss (l,u) :

if l>u :
return 0 #空陣列的情況

else if l = = u :
return (0↑a [l ]) #單一元素陣列的情況

else :
m = (l+u)/2

#計算 a[l..m]的最大後段和
sum,maxToLeft = 0,0
for i in range (m, l−1,−1) :

sum = sum+a [i ]
maxToLeft = maxToLeft ↑ sum

#計算 a[m+1..u]的最大前段和
sum,maxToRight = 0,0
for i in range (m+1,u+1) :

sum = sum+a [i ]
maxToLeft = maxToRight ↑ sum

maxCrossing = maxToLeft+maxToRight

#遞迴計算 a[l..m]與 a[m+1..u]的最大區段和
maxInL = mss (l,m)
maxInR = mss (m+1,u)

return (maxInL↑maxCrossing↑maxInR)

兩個迴圈，分別計算 xs的最大後段和與 ys的最大前段和，兩者之和就是橫跨
xs與 ys的最大區段和。（見第 172頁。）該演算法使用 O(n logn)的時間。
事後回顧，Shamos其實不需在每層都把最大前段與後段和重頭算起。我們

可用類似組對的想法，對每個子陣列都由下至上地算出以下四個值：最大前段
和、最大區段和、最大後段和，以及總和。如此一來我們可得到一個線性時間
演算法。這可看作給我們的一個暗示：「分而治之」在此也許是不必要的，我
們不需要把陣列從中切半。事實上，把陣列分成頭與尾反而讓事情簡化不少。
Kadane立刻想出了演算法 4，不知是否用了同樣的思路？
我最初是在 Gibbons [1997]中見到此問題，當時便覺得印象深刻。Gibbons

的這篇文章是很好的函數程式推導簡介。Bird et al. [1996]改用關係 —函數的
一種擴充 —解最大區段和問題，並推廣到其他資料結構。Mu [2008]則以最大
區段和問題為開端，討論一些有趣的變形：例如有長度限制的最大區段和，和
最大平均區段。

區段問題 Zantema [1992]

單子與副作用

從純函數語言的觀點而言，程式是函數，而函數的「本分」是把一個輸入值
對應到一個輸出值，除此之外發生的都是副作用 (side effects)。一個程式執行
時若可能改變某個可變變數 (mutable variable)的值，是一種副作用 —這可能
是大家最耳熟能詳的例子。1 此外，讀寫檔案、抓滑鼠的位置、往螢幕寫東
西、和系統問現在的日期時間、取亂數等等，也是副作用。程式若拋出例外
(exception)、不傳回值了，是一種副作用。一個程式若有不止一個可能的傳回
值（程式員無法預測是哪一個），也是一種副作用。在更嚴格的定義中，一個
程式如果進入無窮迴圈而不終止（因此不傳回值），也可視為副作用。
許多程式設計典範都大量依靠副作用完成各種工作：更新變數、輸出入、

用例外處理執行時間遇到的錯誤…。在大家的一般印象中，純函數語言反其道
而行，其特徵就是「不能有副作用」。但這其實是個過度簡化的說法。實情正
好相反：純函數語言不僅允許副作用，甚至允許多種副作用並存，並且可把一
個程式所被允許的副作用標示在其型別中。只是，副作用也必須被納入嚴謹的
數學架構中。在純函數語言中，我們需好好地談每個副作用到底「是什麼」（即
其語意為何），滿足哪些性質，而這些性質便可用來推論含副作用的程式的正
確性。
例如，「拋出例外 (exception)」算是一種副作用。如果我們定義如下的型別：

data Except b = Exception String | Return b ,

那麼一個正常終止時傳回型別為 b的值、但也有可能拋一個例外（並附帶一個
字串錯誤訊息）的程式可視為一個值域為 Except b的全函數 —正常終止時傳回

1在數學式或函數語言程式中，例如 x+ 2× y+ 1, 其中的 x和 y被稱為 “variables”，譯為「變
數」。這名稱的由來是：相對於 1, 2等等常數，「變數」的值被其所屬環境而決定。但在同一個環
境與範圍 (scope)中，一個變數仍代表同一個值。命令式語言中的「變數」則是存放值的容器，我
們可使用賦值 (assignment)指令改變變數中的值。為了區分，我們把後者這種能被賦值而改變內
容的變數稱作「可變變數 (mutable variables)」。

173

Return x,「拋出例外」則可用傳回 Exception msg來模擬。有多個可能傳回值
（型別均為 b）的程式可視為值域為 Set b的全函數 —所有可能的回傳值都被收
集在集合之中。假設有一個型別為 s的可變變數，那麼一個執行完畢後會傳回
b、但可能改變該可變變數的程式可以視為一個型別為 s→ (b,s)的函數 —輸入
的 s為該可變變數的初始值，輸出的序對中，b是程式的結果，s是可變變數
的新值。例如，當 s為 Int,一個傳回該可變變數的現有值、並同時將變數加一
的「指令」可表示成一個函數 λ s→ (s,1+ s).
如此一來，我們可在純函數語言中實作出許多種副作用。這可能和讀者心

目中的「實作」相當不同 —此處的「將可變變數加一」並沒有真正去修改原
記憶體位置中的值，而是產生了一個新值；「拋例外」並未採用高效率的實作
（例如到系統堆疊中把回傳位置一個個 pop出來），而和一般函數一樣，只是傳
回一個資料結構。但我們可把前述這些較有效率的做法當作交給編譯器做的最
佳化 —當編譯器看到傳回 Except b的函數，或著看到 s→ (b,s)這樣的函數，
可自己想辦法去找有效率的編譯方式。2 身為程式員，我們在乎的是將「會拋
例外的程式其實是什麼？會修改可變變數的程式其實是什麼？」給描述出來。
另一個可能令讀者困擾的是：這樣寫程式實在不方便。一個使用 Except的

程式可能得有許多 case將子程式的結果配對拆開。一個程式若由許多型別為
s→ (b,s)的組成，得手動將序對配對，將輸入與輸出的 s傳來傳去。這些做法
都太低階，我們希望能在抽象一點的層次上工作，方便我們寫程式、讓我們能
做抽象些的思考。這個抽象概念必須能適用於種種副作用，抓出這些副作用的
共同模式。
在 Haskell中，描述副作用最常用的抽象概念是單子 (monad)。大家常認為

單子是學習函數語言碰上的第一個難關。希望讀者讀完本章後，會發現它其實
沒那麼難。一般談單子的方式是由下往上的：由例外、可變變數等等特例的實
作開始，歸納出這些特例的共同模式，使讀者可發現引出單子這個概念的動
機。從第 8.1節開始，我們也將由這種觀點出發解說單子。但也有另一種由上
而下的看法：先討論單子與每個副作用的運算子該滿足什麼性質，才談他們可
怎麼實作。我們認為這種看法一樣重要。

例：簡單算式求值 以下的資料結構 Expr表達一個只有整數、負號、和加法
的數學式：

data Expr = Num Int | Neg Expr | Add Expr Expr .

例如 −3+ 4可表達為 Add (Neg (Num 3)) (Num 4). 我們可輕易寫一個歸納定
義，將一個 Expr計算成一個整數：

eval ::Expr→ Int

eval (Num n) = nw
eval (Neg e) =−(eval e)
eval (Add e0 e1) = eval e0 + eval e1 .

看來這是個再單純也不過的純函數。但只要我們稍加擴充這個數學式型別，
情況就變複雜了。如果數學式有除法，分母為零時 eval就得拋出例外；我們

2實務上這相當不容易…TODO

可以為數學式加入變數，那麼 eval碰到變數時得查閱該變數的值；我們也許
想要用一個可變變數幫我們記住 eval做了多少次加法。程式只要稍微變複雜，
我們就會開始需要許多種副作用。
我們將用這個例子介紹單子。在接下來的幾個章節中，我們將以不同的方

式擴充 Expr型別以及 eval函。每個擴充需要不同的副作用。我們將由如何模
擬出副作用為動機出發，介紹單子的概念。引入單子這種抽象化後，我們會發
現我們只需改用不同的單子，就可以在不太改變 eval的主要程式的情況下，為
eval加入新的副作用。因此單子可以視為模組化地加入副作用的方法

8.1 例外處理

假設我們添了一個整數除法運算元：

data Expr = Num Int | Neg Expr | Add Expr Expr | Div Expr Expr ,

我們便碰到了一個小難題：要怎麼處理分母是零的狀況呢？
Haskell的 div函數可做整數除法。碰到分母為零時，div和大部分語言的除

法函數一樣讓整個程式當掉。我們不希望程式當掉，因此得在呼叫 div前先檢
查一下分母是否為零。但當分母真為零，eval應該傳回什麼呢？

eval (Div e0 e1) = let v1 = eval e1

in if v1 ̸= = 0 then eval e0 ‘div‘ v1 else ???

用Maybe表達部份函數 一種看法是，有了除法後，eval對於某些輸入沒有對
應的值，因此 eval成了一個部份 (partial)函數。
操作上，我們希望 Haskell 提供拋出例外的功能，eval 無值可傳的時候便

拋出例外。在程式的另外某處可能接住這些例外，另行處理。本章開頭提及，
一個傳回 Int、但可能拋出例外的函數可用傳回 Except Int 的全函數來模擬。
Except再給一次如下：

data Except b = Expt String | Return b .

我們把 eval的型別改為 Expr→ Except Int,正常運算結束便傳回 Return,碰
到分母為零的情形則傳回 Expt.
這個版本的 eval該怎麼寫呢？碰到 Num n自然是傳回 Return n:

eval (Num n) = Return n .

處理 Neg e 的條款則比以前稍微複雜了。我們得檢查 eval e 的結果，若是
Return v，應傳回 −v，但也需包上一個 Return. 如果遇到 Expt msg,表示 e的運
算失敗了，我們只好把 Expt msg再往上傳。

eval (Neg e) = case eval e of
Return v → Return (−v)
Expt msg→ Expt msg .

Add的情況下，程式碼看來更冗長了些：

eval (Add e0 e1) = case eval e0 of
Return v0→ case eval e1 of

Return v1→ Return (v0 + v1)

Expt msg → Expt msg
Expt msg → Expt msg .

我們得分別檢查 eval e0 與 eval e1 的結果是 Return或是 Expt，因此用兩層 case
分出了四個狀況 —雖然它們是重複性很高的程式碼。最後是 Div e0 e1 的情況，
我們先計算 e1,如果是零的話傳回 Expt,含錯誤訊息 "division by zero",否
則做除法運算。在這過程中我們也需要許多 case：

eval (Div e0 e1) = case eval e1 of
Return 0 → Expt "division by zero"
Return v1→ case eval e0 of

Return v0→ Return (v0 ‘div‘ v1)

Expt msg → Expt msg
Expt msg → Expt msg .

很自然地，我們會希望把這些例行公事抽象掉。

Except作為單子 再看一次處理 Neg的條款：

eval (Neg e) = case eval e of
Return v → Return (−v)
Expt msg→ Expt msg .

我們真正想表達的僅是「把 eval e的結果送給 λv→ Return (−v) —如果有結果
的話」。這只是多加了一點手續的函數應用 (application)。回想 Haskell有個函
數應用運算元 ( $) :: (a→ b)→ a→ b. 給定 f :: a→ b和 x :: a,那麼 f $ x就是把 x
餵給 f（也就是 f x）。也許我們可以自己定義一個函數應用運算元 (=<<). 概念
上，f =<< x仍是把 x餵給 f，但 x的型別是 Except a, f 的型別是 a→ Except b,
而「檢查 x是 Return或 Expt」的動作就可藏在 (=<<)之中（讀者可把 (=<<)的
型別和 ( $)做比較）：

(=<<) :: (a→ Except b)→ Except a→ Except b
f =<<Return x = f x
f =<<Expt msg = Expt msg .

如此一來，eval (Neg e)的條款可簡潔地寫成：

eval (Neg e) = (λv→ Return (−v))=<< eval e ,

如果定義 negate x = −x, 上式的右手邊可以更簡潔地寫成 Return · negate=<<
eval e.

但，目前較常見的習慣是把這個運算子的左右顛倒：參數寫前面，函數寫
後面。也就是定義:

(>>=) :Except a→ (a→ Except b)→ Except b
Just x >>= f = f x
Nothing>>= f = Nothing .

運算子 (>>=)唸作 “bind”,是我們將會細談的重要元素。有了它，函數 eval的
前兩個條款可以改寫如下：

eval (Num n) = return n
eval (Neg e) = eval e>>=λv→ return (−v) .

我們把 Return 寫成函數：return = Return。等下會解釋為什麼。注意：依照
Haskell的運算元優先順序，m>>=λv→ e會被理解成 m>>=(λv→ e). Add和
Div的情況也可以用 (>>=)改寫:

eval (Add e0 e1) = eval e0 >>=λv0→
eval e1 >>=λv1→
return (v0 + v1)

eval (Div e0 e1) = eval e1 >>=λv1→
if v1 = = 0 then throw "division by zero"
else eval e0 >>=λv0→

return (v0 ‘div‘ v1) .

此處我們把 Expt寫成 throw.
如此一來，我們的程式變得簡短了一些。例如，eval (Add e0 e1)直覺上彷

彿在說「把 eval e0 的結果叫做 v0, eval e1 的結果叫做 v1,然後傳回 v0 + v1」，至
於判斷 eval e0 與 eval v1 到底是 Return還是 Expt的動作則藏在 (>>=)之中了。
但引入 (>>=)這個抽象的用意並不只是把程式變短，而是可推廣到其他的副作
用之上。這將是接下來幾小節的主題。
提醒一下：有些讀者可能認為 eval 的後兩個條款看來像命令式語言：

eval e1 >>= λv1 → . . 看來就像是 v1 := eval e1 — 把 eval e1 的值寫到 v1 中。但
其實 (>>=)並沒有賦值（改變變數現有的值）之意，要做比較的話，它更像是
增強版的 let. 算式 p>>=λx→ e就像是 let x = p in e. 兩者都是「把 p的值算出，
給予一個名字 x,然後計算 e」。差別則是：

• let處理純的值：在 let x = p in e之中，如果 p的型別是 a, x的型別也是 a;
算式 e的型別若是 b,整個式子的型別也是 b; e的型別相同。

• (>>=)處理含有 Except的值：在 p>>=λx→ e之中，
– p的型別得有 Except. 如果 p的型別是 Except a,則 x的型別是 a;
– e的型別得有 Except. 如果 e的型別是 Except b,整個式子的型別也是

Except b.
– 因此 λx→ e是一個型別為 a→ Except b的函數。

捕捉例外 最後，我們可以定義一個捕捉例外的運算子。算式 catch p hdl嘗試
配對 p的值，如果是 Return x,就傳回結果 x；如果是 Expt msg,就把錯誤訊息

msg交給例外處理函數 hdl。當 p的型別是 Except a, catch p hdl的型別也是 a,
而 hdl的型別必須是 String→ a:

catch ::Except a→ (String→ a)→ a
catch (Return x) hdl = x
catch (Expt msg) hdl = hdl msg .

例如，下式嘗試計算 e並把結果轉成字串。如果 e能正常計算，字串會是
"Result:"開頭，否則是 "Error:"開頭：

catch (eval e) (λv→ "Result: "++ show v)
(λmsg→ "Error: "++msg) .

下式則無論如何都傳回 Int, 如果 e之中出現了分母為零的除法，我們便傳回
65536:

catch (eval e) id (const 65536) .

8.2 單子與單子律

第 0章提及，「抽象化」意指抽取出一個問題中最關鍵的元素。一個成功的抽
象化能抓到同類問題的共通本質，因此可以推廣到其他相關的情境、問題，用
同一套方法表達這些情境、解決這些問題。我們在第 8.1節發現的便是一個關
於計算與副作用的成功抽象化。
我們可以把第 8.1節中的 Except a理解成一個尚待完成的計算，或是一個

尚待執行的程式。用 case去配對它便是把這個計算算出來、把程式執行出來。
如果順利執行，計算結果的型別為 a,但也可能產生副作用。而由其型別建構
子 Except可看出此處可能的副作用是「可能拋出例外」。運算子 (>>=)的功用
是將小計算/程式串接起來，組成較大的計算/程式。
推廣出去，Except其實可以是任意一個型別建構元 m: 一個型別為 m a的值

表示一個尚待完成的計算，或是一個尚待執行的程式。如果順利執行，計算結
果的型別為 a,但執行過程中可能產生副作用。至於究竟是什麼副作用，則看
m是什麼而定。此外，每個 m附帶兩個運算子：

• return :: a→ m a;
• (>>=) :: m a→ (a→ m b)→ m b.

它們三者合稱為一個單子 (monad).
給任一個值 x :: a,運算子 return x代表一個純粹傳回 x,沒有副作用的計算。

以 m = Except為例，自然的選擇是令 return = Return. 我們之後會看到其他例
子。
運算子 (>>=)將小計算串接在一起。在 p>>= f 之中，p :: m a是一個回傳型

別將為 a,但可能有副作用的計算。如果 p被算出來，其結果會被傳給 f . 函數
f :: a→ m b是一個拿到 p的結果後算出一個計算的計算. 整個式子 p>>= f 的型
別也是 m b. 這是理解單子程式的一個重點：函數本身沒有副作用，但函數可
以算出含有副作用的程式。如果我們把 f 寫成 λ 表示式，在 p>>=λx→ e之

中，p的結果被取名為 x，而 e :: m b之中可以使用 x. 此處 (>>=)可理解成由兩
個小計算 p與 e組出一個大計算的接著劑，兩個小計算用 x來溝通。
運算子 return與 (>>=)是用來組合出含有副作用的程式的元件，因此得滿

足一些該有的性質。正式說來，

定義 8.1 (單子). 一個單子 (monad)包含一個型別建構子 m, 以及兩個運算子
return :: a→ m a與 (>>=) :: m a→ (a→ m b)→ m b. 它們須滿足以下三條單子律：

右單位律 (right identity): p>>= return = p , (8.1)

左單位律 (left identity): return x>>= f = f x , (8.2)

結合律 (associativity): (p>>= f )>>=g = p>>=(λx→ f x>>=g) . (8.3)

右單位律若寫成 p>>=λx→ return x也許更好理解。計算 p的結果 x被傳到
(>>=)的右邊，但後者只是單純地把 x再傳出。這和只寫 p得是一樣的。
在左單位律的左手邊 return x>>= f 之中，函數 f 接收 return x的結果 —而

return x的結果就是 x! 因此這和 f x得是一樣的。
最後一條是 (>>=)的結合律：「用 p與 f 串接出一個程式，再把 g串接到右

邊」與「把 f 與 g串接在一起，並把 p串接在它們左邊」得到的必須是一樣的
程式。但由於型別的問題，在後者的情況我們需要一個 x表示 p的結果，前者
則不用。
每定義一個新的單子，我們需選定一個型別 m，定義 return與 (>>=)兩個

運算子，然後證明它們滿足單子律。需要推論含單子的程式的性質與正確性
時，這三條單子律也是我們會用到的性質。
最後一提：定義 8.1 並不是單子的唯一定義方式。其他等價的定義詳見

[todo: where].

習題 8.1 — 證明第 8.1 節中的定義滿足單子律。意即當 m = Except,
return = Return,而 (>>=)的定義如第 8.1節所示時，三條單子律都成立。

單子 class 既然 Haskell有 type class機制，我們可以把單子定義為一個 class:

class Monad m where
return :: a→ m a
(>>=) :: m a→ (a→ m b)→ m b .

如此明確規定一個屬於Monad的型別建構子必須有 return與 (>>=). Except
型別則是其中的一個特例：

instance Monad Except where
return = Return

Expt msg>>= f = Expt msg
Return x >>= f = f x .

之後我們討論其他的 m時，也將把它們宣告為Monad的特例。但這麼做只
是為了兩個方便性：1. 讓不同的 m可以共用 return與 (>>=)兩個符號，不用重

新取名；2. 看到使用 return與 (>>=)的程式時，可以暫時不指定它們到底使用
那一個特例。若非為了這些小方便，單子與 type class不一定得有關聯。
此外，Haskell中的 type class宣告也只保證 return與 (>>=)的型別是正確

的。目前 Haskell並無法保證使用者定義的 return與 (>>=)滿足前述的單子律。

副作用特定的運算子 不知讀者是否想起：那麼 throw與 catch呢？他們也應
該滿足一些性質，例如，throw msg>>= f = throw msg（拋出例外後，後面的 f
不會執行）和 catch (throw msg) hdl = hdl msg（throw拋出的例外由 hdl處理）
應該都是合理以及必要的性質。另一方面，它們是屬於「例外」這個副作用的
運算子，其他的副作用並不見得會有它們。反之，別的副作用可能會有該副作
用的特定運算子。例如，要談可變變數，可能會有個賦值運算子，這是 Except

沒有的。
由此得知，用於表達副作用時，單子的運算子可分為兩大類：1. return與

(>>=)是各個副作用共通的，每個單子都必須定義 return與 (>>=),並確保它們
滿足單子律。2. 此外，每個副作用還可能擁有特有的運算子，它們可能也有些
該滿足的性質。我們將在介紹各個副作用的小節中看到更多例子。

8.3 讀取單子

為了進入下一個例子，我們為算式型別加上變數：

data Expr = Num Int | Neg Expr | Add Expr Expr |
Var Name | Let Var Expr Expr ,

type Name= String .

Name是變數名稱，Var v是出現在算式中的變數，Let則用來宣告新的區域變
數。例如，以下式子表達的是 let x = 3+1 in (x+4)+ x:

Let "x" (Add (Num 3) (Num 1))
(Add (Add (Var "x") (Num 4)) (Var "x")) ,

變數 x的值為 3+1,因此整個算式的值是 ((3+1)+4)+(3+1) = 12.

8.3.1 變數與環境

有了變數後，我們不能只問「Add (Var "x") (Var "y")的（整數）值是什麼」了，
因為我們不知道 x和 y的值。有自由變數的算式得放在一個環境下才能談它的
值。所謂「環境」是一個變數到值的函數，告訴我們每個變數的值。我們可以
用一個串列表示：3

type Env = [(Name, Int)] .

例如 Add (Var "x") (Num 4)在環境 [("x",3),("y",6)]下的值是 7. 我們也假設
有一個「查表」函數：

3或著，我們也可以定義 type Env =Name→Maybe Int,而 lookup env x = env x.

lookup ::Env→Maybe Int .

例如，當 env = [("x",3),("y",6)]，lookup env "x"是 Just 3，lookup env "z"則
是 Nothing.
提醒一下讀者：Let 是在宣告區域變數，而不是賦值。例如，下式表達

let x = 3 in x+(let x = 4 in x)+(−x)：

Let "x" (Num 3)
(Add (Add "x" (Let "x" (Num 4) (Var "x")))
(Neg (Var "x")))

它的值是 3+4+(−3)，而不是 3+4+(−4) —內層宣告的 x僅是暫時遮蓋到外
面的 x。
有了這些鋪陳，我們看看新的 eval函數該怎麼寫。既然算式要在環境之下

才有值，eval得把環境也納為參數之一：

eval ::Expr→ Env→ Int

函數 eval拿一個算式和一個環境，計算該算式的值。新的 eval中，最初的三
個條款基本上是一樣的，只是多了一個參數 env得往下傳：

eval (Num n) env = n
eval (Neg e) env =−(eval e env)
eval (Add e0 e1) env = eval e0 env+ eval e1 env .

碰到變數時，我們到環境中查變數的值：

eval (Var x) env = case lookup env x of Just v→ return v .

這裡的 case算式只處理了 Just的情形。如果 lookup傳回的是 Nothing，也就是
變數 x並不在環境 env中，該怎麼辦呢？我們等下再談。最後，碰到 Let x e0 e1

時，我們先把 e0 的值在 env這個環境之下算出，然後算 e1:

eval (Let x e0 e1) env =
let v = eval e0 env
in eval e1 ((x,v) : env)

但計算 e1 時須使用新的環境 (x,v) : env，這讓 e1 可以用到 x。變數 x在新環境
中的值是 v.

算式的語意是函數 第 ⁇節開頭提及，算式沒有變數、沒有除法時，eval的
型別是 Expr→ Int —此時一個算式的「意思」就是一個整數。有了除法後，為
了表示可能拋出例外的函數，eval的型別變成 Expr→ Except Int. 此時 Except a
代表一個尚待完成、可能有副作用（拋出例外）的計算。
加上變數、考慮環境後，eval的型別變成了 Expr→ (Env→ Int)：eval拿一

個算式，傳回一個函數；該函數又拿一個環境，才算出一個整數值。也就是
說，一個算式的語意是「拿一個環境，傳回一個整數的函數」。的確，既然算

式算成的那個整數必須由環境決定，算式其實不能看作一個數值，而應該是從
環境到整數的函數才對。算式 eval (Add (Var "x") (Var "y"))是一個函數，如
果給它 [("x",3),("y",2)],我們得到 5;如果給 [("x",4),("y",−3)],我們得到 1.
如果把「給一個環境，傳回型別為 a的結果的函數」叫做 Reader a,也就是

說定義 type Reader a = Env→ a，函數 eval的型別成了 Expr→ Reader a. 此處，
Reader a也可視為一個尚待完成、可能發生副作用的計算。此處可能的副作用
包括「和環境詢問變數的值」。習慣上，我們把這種副作用稱作「讀者 (reader)」
或「讀取」。「給環境」的動作，就是執行運算，把值算出來。我們把其中兩個
條款改寫成 λ 算式：

eval (Num n) = λenv→ n
eval (Add e1 e2) = λenv→ eval e1 env+ eval e2 env

可以理解成：eval (Num n) 傳回一個計算，無論環境為何，該計算都傳回 n;
eval (Add e1 e2) 也傳回一個計算，該計算拿到環境後，在同一個環境之下把
eval e1 和 eval e2 算成值，然後傳回他們的和。
然而，手動把 env傳來傳去是個重複性高、容易出錯、也嫌累贅的動作。

在第 8.1節中，我們把重複性地產生與拆開 Except型別的動作抽象成 return和
(>>=)，使得 Except成為一個單子。對於本節的 Reader,我們也能設計出一組
return和 (>>=),把重複的動作抽象掉，使得 Reader成為一個單子嗎？

8.3.2 「讀取」副作用是單子

讀取單子 如第 8.2節所述，return製作一個沒有副作用的計算，(>>=)則用於
把兩個計算接起來。把 m代換成 Reader,它們的型別分別是：

return :: a→ Reader a
(>>=) ::Reader a→ (a→ Reader b)→ Reader b ,

其中 Reader a = Env→ a.
函數 return比較單純：return x是一個無論 env是什麼，都傳回 x的計算：

return x env = x .

至於 m>>= f 則可定義如下：

(p>>= f ) env = f (p env) env .

p>>= f 的型別為 Reader b, 意謂它可以收一個環境 env 當參數，而等號右手
邊必須算出一個型別為 b的結果。變數 p ::Reader a是一個尚待完成的計算，
p env把它算出來，得到型別為 a的結果。函數 f 的型別為 a→ Reader b,因此
f (p env)意謂 f 得到 p env :: a的結果，並算出一個新的計算. 這個計算又在得到
env之後才真正被算出來，最後型別為 b.
有了 return與 (>>=), eval的前三個條款可改寫如下：

eval ::Expr→ Reader Int

eval (Num n) = return n

eval (Neg e) = eval e>>=λv→ return (−v) .

eval (Add e0 e1) = eval e0 >>=λv0→
eval e1 >>=λv1→
return (v0 + v1) ,

除了 Except變成 Reader之外和第 8.1節完全相同！由於使用了單子這個適當
的抽象，只要把型別從 Except變成 Reader，主程式在不需大幅更動的情況下
就可以沿用。使用單子，我們能模組化地挑選我們想要的副作用。

讀取單子的特定運算 本節新添加的 Var 與 Let兩個情況是第 8.1節沒有的。
要處理它們，需要讀取副作用特有的運算子。
處理 Var x時，eval得到環境中查變數 x的值。我們姑且先為此專門定義一

個函數：

lookupVar ::Var→ Reader Int

lookupVar x env = case lookup env x of Just v→ return v ,

之後再考慮更通用的情況。有了它，eval遇上 Var x的條款可寫成：

eval (Var x) = lookupVar x .

計算 Let x e1 e2 時，我們需要更動環境，把「x的值是 e1 的值」這個資訊
加到環境中，並在新環境內執行 eval e2,但執行完之後就回到原有的環境 —新
環境只是局部的。由於這是常見的模式，我們可定義一個更通用的運算子。給
定一個函數 f ::Env→ Env用於製作局部的環境，如果目前的環境是 env,算式
local f p在新環境 f env之下執行 p:

local :: (Env→ Env)→ Reader a→ Reader a
local f p env = p (f env) .

有了 local的幫忙，eval遇上 Let時可寫成：

eval (Let x e0 e1) = eval e0 >>=λv→
local ((x,v):) (eval e1) .

我們用 ((x,v):)幫環境增加一筆資料，在這之下計算 eval e2.

習題 8.2 — 使用本節的定義，將 eval (Add e0 e1) env與 eval (Let x e0 e1) env
展開，確認它們和第 8.3.1節的定義一樣。意即：

eval (Add e0 e1) env = eval e0 env+ eval e1 env ,

eval (Let x e0 e1) env = eval e1 ((x,eval e0 env) : env) .

更通用的環境單子 為了說明方便，目前為止我們假設「環境」是某個固定的
型別: Env. 我們當然可以把這部份也抽象掉：

type Reader e a = e→ a .

現在Reader多吃一個參數 e,代表環境的型別。對任意的 e, Reader e都是一個單
子，函數 local更通用的型別是 (e→ e)→ Reader e a→ Reader e a. 本節之前的
每個 Reader都需代換成 Reader Env. 例如 eval的型別是 Expr→ Reader Env Int.
但改變的都只有型別，程式不需變動。
函數 lookupVar ::Var→ Reader Env Int只能在 e為 Env的情況下運作。更一

般來說，讀取單子通常會有一個運算子 ask,把整個環境傳出來供我們使用：

ask ::Reader e e
ask env = env .

函數 lookupVar則可改用 ask定義為：

lookupVar ::Var→ Reader Int

lookupVar x = ask>>=λenv→
case lookup env x of Just v→ return v .

最後，本節之中讓 Reader e與前一節的 Except共用 return, (>>=)等符號。
若需在 Haskell中如此，我們得將 Reader e宣告為 type class Monad的一個特
例。但由於一些型別檢查的技術問題，Haskell不允許用 type宣告的別名成為
type class特例。我們得把 Reader用 data宣告成一個資料型別：4

data Reader e a = Reader (e→ a) ,

instance Monad (Reader e) where
return a = Reader (λe→ a)
Reader r>>= f = Reader (λe→ f (r e) e) .

8.3.3 推論讀取單子程式的性質

給一個單子程式，如何討論它的性質？比如說，對某一個 e, 如何得知 eval
(Let "x" (Num 3) (Let "y" (Num 4) e))的值是什麼？它和 eval (Let "y" (Num 4)
(Let "x" (Num 3) e))的值是否總是相等？
由於 eval是用 return, (>>=), local … 等等運算子定義出來的，而這幾個運算

子的定義也都已經有了，我們總是可以把他們的定義都展開，回到最基礎的層
次證明任何我們想確認的性質。但一來如此的證明可能非常瑣碎，二來 return,
(>>=) 等運算子的定義可能還會改變。我們是否能在稍微抽象一點的層次運
作，假裝我們不知道這些單子運算子的定義，只討論它們具有什麼性質，並用
這些性質來做證明？

4更普遍的做法是用 newtype宣告：newtype Reader e a = Reader {runReader :: (e→ a)}. 但本
書不談 newtype.

我們應可以合理要求一個「正確」的讀取單子實作應該要滿足下列的性質。
首先，假設 e是一個不含變數 v的算式：

ask>>=λv→ return e = return e ,如果 v不出現在 e之中。 (8.4)

等號兩邊都只是傳回 e的值，而 e的值不受 v影響，因此 ask是可以省略掉的。
其次，連續使用 ask兩次可以縮減為一次就好：

ask>>=λv0→ ask>>=λv1→ f v0 v1 = ask>>=λv→ f v v . (8.5)

在等號左手邊，我們把問了環境兩次之後的計算抽象為一個函數呼叫
f v0 v1. 在右手邊我們則讓 f 的兩個參數都是 v —詢問環境一次的結果。
下面兩個式子討論當 local遇上 return與 (>>=)時會如何：

local g (return e) = return e (8.6)

local g (p>>= f ) = local g p>>=(local g · f ) (8.7)

在 (8.6)的左手邊，改變環境之後立刻 return e,其實和單純做 return e一樣。性
質 (8.7)則告訴我們 local g可以分配到 (>>=)的兩側。由於型別之故，(>>=)的
左邊是 local g p,右邊則得用函數合成 (·).
最後，下列性質將 local與 ask關聯在一起：

local g ask = ask>>=(return ·g) . (8.8)

在 local g的環境之下做 ask, 相當於先做 ask, 然後把得到的環境交給 g加工。
我們可說算式 (8.8)藉由這兩個運算子的互動定義出了 local的「意思」。
有了這些性質，我們不需引用 local, ask, (>>=) … 等等的定義，也可論證讀

取單子程式的性質了。例如，我們來看看 let x = 4 in x+ x的值會是什麼：

eval (Let "x" (Num 4) (Add (Var "x") (Var "x")))
= { eval之定義 }

eval (Num 4)>>=λv→ local (("x",v):) (eval (Add (Var "x") (Var "x")))
= { eval之定義 }

return 4>>=λv→ local (("x",v):) (eval (Add (Var "x") (Var "x")))
= {單子律：左單位律 (8.2) }

local (("x",4):) (eval (Add (Var "x") (Var "x")))
= { eval之定義 }

local (("x",4):) (eval (Var "x")>>=λv0→
eval (Var "x")>>=λv1→ return (v0 + v1))

= { (8.7) }
local (("x",4):) (eval (Var "x"))>>=λv0→
local (("x",4):) (eval (Var "x"))>>=λv1→
local (("x",4):) (return (v0 + v1))

我們將 local (("x",4):) (eval (Var "x"))抽出來化簡：

local (("x",4):) (eval (Var "x"))
= { eval與 lookupVar之定義 }

local (("x",4):) (ask>>=λenv→ case lookup env "x" of Just v→ return v)
= { (8.7) }

local (("x",4):) ask>>=λenv→
local (("x",4):) (case lookup env "x" of Just v→ return v)

= { (8.8),左單位律 (8.2) }
local (("x",4):) (case lookup (("x",4) : env) "x" of Just v→ return v)

= { lookup (("x",4) : env) "x"= Just 4 }
local (("x",4):) (return 4)

= { (8.6) }
return 4 .

由此我們得知 local (("x",4):) (eval (Var "x"))就是 return 4. 將它放回原式中：

local (("x",4):) (eval (Var "x"))>>=λv0→
local (("x",4):) (eval (Var "x"))>>=λv1→
local (("x",4):) (return (v0 + v1))

= {前述計算 }
return 4>>=λv0→ return 4>>=λv1→ return (v0 + v1)

= {單子律：左單位律 (8.2) }
return (4+4) .

因此，eval (Let "x" (Num 4) (Add (Var "x") (Var "x")))就是 return 8.
提醒讀者注意一點：eval (Let "x" (Num 4) (Add (Var "x") (Var "x"))) 和

return 8都不是基礎型別，而是「尚待完成的計算」。論證單子程式時我們常常
不是在基礎型別的層次上運作，而是證明一個計算與另一個計算是等價的。這
意味著它們傳回同樣的值，也發生同樣的副作用。我們日後會看到更多此類的
例子。

習題 8.3 — 證明對所有 e ::Expr,

eval (Let "x" (Num 3) (Let "y" (Num 4) e)) =
eval (Let "y" (Num 4) (Let "x" (Num 3) e)) ,

如果 (("x",3):) · (("y",4):) = (("y",4):) · (("x",3):)成立。
習題 8.4 — 然而，(("x",3):) ·(("y",4):)= (("y",4):) ·(("x",3):)並不成立。

我們稍早曾遇到這個問題：如果給這樣的式子 eval (Var "x") [("y",0)],變數 x
並不在環境中，lookup將傳回 Nothing，這時該怎麼辦？
我們可以再擴充 Reader的型別，讓 eval也可以傳回一個 Except結果：

type ReaderExcept e a = e→ Except a

而 return和 (>>=)也得隨之擴充：

return a = λenv→ Just a
rm>>= f = λenv→ case rm env of

Just v→ f v env
Nothing→ Nothing

這個 ReaderExcept 型別綜合了讀取單子與例外單子的功能，其 return 與
(>>=) 定義也像是兩個單子定義的混合。但這並不是令人相當滿意的做法。
ReaderExcept比起 Reader又更複雜了一點。日後我們也許會想要有更多功能，
例如狀態、輸出入等。ReaderExcept已經夠抽象難解了，我們並不希望設計、
維護越來越龐大的單子。
既然Maybe單子讓一個程式加上「例外」的副作用，讀取單子讓一個程式

加上可存取環境的功能，我們能否把這兩項功能分別模組化地加入呢？
也就是說，給了兩個單子 M1和 M2,能否把他們的功能加在一起，產生另

一個新單子呢？

8.4 狀態單子

沿用第 8.1節之中的型別：

data Expr = Num Int | Neg Expr | Add Expr Expr | Div Expr Expr .

這次我們暫且忽略分母為零的可能，而考慮另一個應用：我們想知道計算過程
中做了多少次除法。指令式語言中，一個常見的作法是用一個可變變數來計
數。在函數語言中我們怎麼模擬這種行為呢？
在指令式語言中，可變變數的值常用來表示整個系統目前的「狀態 (state)」：

諸如已處理過的資料個數、處理中的元素號碼、棋盤上每個棋子的位置… 等
等。因此我們把「存取可變變數」的能力稱作狀態 (state)副作用。
經過前幾節的熟悉，我們現在應可以更抽象地、由上而下地想像單子了：

對於一個我們需要的副作用，先假設其單子存在，考慮它該有哪些運算子、這
些運算子該滿足什麼性質、用它們如何寫程式。然後才考慮這個單子的實作。
對於狀態這個副作用，我們該怎麼設計其運算子呢？
在命令式語言中，x := x+ 1這行指令把可變變數 x的值遞增。這一行看似

單純的指令其實包含幾項特性：
• 變數是有名字的；
• 當變數名字出現在 :=的右手邊，表示讀取變數的值；
• 當變數名字出現在 :=的左手邊，表示將值寫入該變數。

函數語言中討論狀態副作用時偏好用單純些、使用時比較繁瑣、但比較好討論
性質的設計：變數沒有名字，並且把「讀取」與「寫入」兩個動作分為單獨的
運算子。
具體說來，令 State s a表示一個結果型別為 a,但在執行時可讀寫一個型別

為 s的可變變數的計算。我們要求 State s是一個單子 —意謂存在著滿足單子
律的 return :: a→ State s a和 (>>=) ::State s a→ (a→ State s b)→ State s b。一
個型別為 State s a的運算中隱藏的可變變數沒有名字，只能用下述兩個運算子
存取：

get ::State s s ,

put :: s→ State s () .

運算子 get和第 8.3節中的 ask類似，讀出該可變變數的值；put e則把 e的值
寫到可變變數中，並傳回 ()。例如，當 s = Int,我們可以定義如下的操作 inc，
把可變變數加一：

inc ::State Int ()
inc = get>>=λv→ put (1+ v) .

關於 put 有幾件事情可提醒。首先，當我們呼叫 put e, 其中的 e已經是一
個型別為 s的純數值，不含副作用。例如在 inc之中，我們必須先將變數的值
get 出來（這是一個有副作用的動作），才能計算新的值並寫回去。這讓程式
寫起來很繁瑣，但也使得推論程式的性質容易許多。其次，關於 put e的型別
State s (). 回想：()是一個只有一個值的型別，該值在 Haskell中也寫作 (). 給
定 e :: s後，put e ::State s ()是一個會存取型別為 s的可變變數，並傳回 ()的計
算。在此，「傳回 ()」可理解為傳回一個不帶資訊的值，僅表達「我做完了」。
這個傳回值既然沒有資訊，通常不會被用上。因此當 put 不是函數的最後

一個動作時，程式常有如下的模樣：

...put e>>=\_→ ...

由於「執行一段程式，但只需要它的副作用，不需要它的結果」這件事在本章
還會常常發生，我們另外訂一個運算元：

(>>) :: m a→ m b→ m b
p>>q = p>>=\_→ q .

如此一來 put e>>=\_→ q可以簡寫成 put e>>q. （注意：p>>q並不要求 p的
傳回值型別為 ().）
回到 get與 put。如果它們的某個實作確實表達了我們前面口述的意圖，該

實作應該會滿足以下的規則：

get-put: get>>=put = return () , (8.9)

put-get: put e>>get = put e>> return e , (8.10)

put-put: put e0 >>put e1 = put e1 , (8.11)

以及一個和第 8.3節中的 ask類似的性質：

get-get: get>>=λv0→ get v1 >>=λv1→ f v0 v1 = get>>=λv→ f v v . (8.12)

其中，get-put意謂：將可變變數的值讀出後立刻寫入相當於什麼都不做。規
則 put-get意謂：剛做完 put e之後立刻讀取可變變數的值，必定得到 e. 規則
put-put意謂：連做兩個 put 之後，只有第二個 put 寫入的值會被留下。規則
get-get則意謂：連做兩次 get所得到的值必定相同，可以只 get一次就好。
如前所述，用 get與 put可定義出 inc,而我們只需在 eval遇到 Div的情況中

呼叫 inc，就可記錄做了多少次除法了：

eval (Div e0 e1) = eval e1 >>=λv1→
eval e0 >>=λv0→

inc>>
return (v0 ‘div‘ v1) .

函數 eval的其他條款可完全不變，只需把所用單子的型別改成 State Int. 由於
單子捕捉了含副作用的程式的共通模式，我們只需選用不同的副作用運算子，
就可在不更動大部分程式的情況下使用我們需要的副作用。

8.4.1 河內塔問題

本節以河內塔問題為例子，示範狀態單子的推論。

hanoi 0 = return ()

hanoi (1+ n) = hanoi n>> inc>>hanoi n .

hanoi n = put (2↑n−1).

ix A
習題解答

Answer (習題 1.1) —

myeven :: Int→ Bool

myeven x = x ‘mod‘ 2 = = 0 .

Answer (習題 1.2) —

area ::Float→ Float

area r = (22/7)× r× r .

Answer (習題 1.6) —根據外延相等，我們需證明 (∀x · id · f = f = f · id). 推
論如下：

(id · f ) x
= { (·)之定義 }

id (f x)
= { id之定義 }

f x
= { id之定義 }

f (id x)
= { (·)之定義 }
(f · id) x .

Answer (習題 1.10) —欲證明 curry ·uncurry = id：

curry ·uncurry = id
≡ {外延相等及 id之定義 }
(∀f : curry (uncurry f ) = f )
≡ {外延相等 }
(∀f x y : curry (uncurry f ) x y = f x y) .

因此我們證明 curry (uncurry f ) x y = f x y如下：

curry (uncurry f ) x y
= { curry之定義 }

uncurry f (x,y)
= { uncurry之定義 }

f x y .

與此相似，欲證明 uncurry · curry = id我們須證明 uncurry (curry f ) (x,y) =
f (x,y)，其證明也與上面的證明類似。

Answer (習題 1.11) — map (uncurry ( $)) ::List ((a→ b)× a)→ List b. 它拿
一個內容均為「(函數 ×參數)」序對的串列，將每個函數作用在其參數
上。例如：

map (uncurry ( $)) [((1+),3),(square,4),(smaller 5,3)]

會得到 [1+3,4×4,3 ].

Answer (習題 1.12) —

tails ::List a→ List (List a)
tails xs = map (λn→ drop n xs) [0 . . length xs ] .

Answer (習題 1.13) —

squaresUpTo :: Int→ List Int

squaresUpTo n = takeWhile (⩽ n) (map square [0 . . ]) .

Answer (習題 1.14) — Answer (習題 1.14) — zip = zipWith (,), 或 zip =

zipWith (λx y→ (x,y)).

Answer (習題 2.5) —

reverse ::List a→ List a
reverse [ ] = [ ]

reverse (x : xs) = reverse xs++[x ] .

另，關於 reverse效率的討論詳見第 5.2節。

Answer (習題 2.7) —欲證明 sum (map (x×) ys) = x× sum ys,在 ys上歸納。
情況 ys :=[ ],兩邊都歸約為 0.
情況 ys := y : ys:

sum (map (x×) (y : ys))
= { map之定義 }

sum (x× y : map (x×) ys)
= { sum之定義 }

x× y+ sum (map (x×) ys)
= {歸納假設 }

x× y+ x× sum ys
= {乘法與加法之分配律 }

x× (y+ sum ys)
= { sum之定義 }

x× sum (y : ys) .

Answer (習題 2.10) —檢視 length, (++), 與 (+) 的定義，會發現等號兩邊
都須對 xs做分析才能化簡。因此我們對 xs做歸納。
狀況 xs :=[ ].

length ([ ]++ ys)
= { (++)之定義 }

length ys
= { (+)之定義 }

0+ length ys
= { length之定義 }

length [ ]+ length ys .

狀況 xs := x : xs.

length ((x : xs)++ ys)
= { (++)之定義 }

length (x : (xs++ ys))
= { length之定義 }

1+ (length (xs++ ys))
= {歸納假設 }

1+ (length xs+ length ys)
= { (+)之定義 }
(1+ (length xs))+ length ys

= { length之定義 }
length (x : xs)+ length ys .

Answer (習題 2.12) —考慮 xs : xss的情況：

length (concat (xs : xss))
= length (xs++ concat xss)

= {因 length (xs++ ys) = length xs+ length ys }
length xs+ length (concat xss)

= {歸納假設 }
length xs+ sum (map length xss)

= sum (length xs : map length xss)
= sum (map length (xs : xss)) .

我們需要的性質是 length (xs++ ys) = length xs+ length ys .

Answer (習題 2.13) —考慮 xs : xss的情況：

map f (concat (xs : xss))
= map f (xs++ concat xss)
= {因 map f (xs++ ys) = map f xs++map f ys }

map f xs++map f (concat xss)
= {歸納假設 }

map f xs++ concat (map (map f xss))
= concat (map f xs : map (map f xss))
= concat (map (map f (xs : xss))) .

我們需要的性質是 map f (xs++ ys) = map f xs++map f ys .

Answer (習題 2.19) —在 xs上做歸納。

all (∈ (x : xs)) (filter p (x : xs))
≡ all (∈ (x : xs)) (if p x then x : filter p xs else filter p xs)
≡ if p x then all (∈ (x : xs)) (x : filter p xs)

else all (∈ (x : xs)) (filter p xs)
≡ if p x then x ∈ (x : xs) ∧ all (∈ (x : xs)) (filter p xs)

else all (∈ (x : xs)) (filter p xs)
⇐ if p x then all (∈ (x : xs)) (filter p xs)

else all (∈ (x : xs)) (filter p xs)
≡ { (2.4) }

all (∈ (x : xs)) (filter p xs)
⇐ {因習題 2.17, z ∈ (x : xs)⇐ z ∈ xs }

all (∈ xs) (filter p xs)
⇐ True .

Answer (習題 2.20) —

inits+ ::List a→ List (List a)
inits+ [ ] = [ ]

inits+ (x : xs) = [x ] : map (x:) (inits+ xs) .

Answer (習題 2.21) —在 xs上做歸納。
情況 xs :=[ ]. 此時等號兩邊都是 [[ ]].
情況 xs := x : xs.

map (λn→ take n (x : xs)) (upto (length (x : xs)))
= { length之定義 }

map (λn→ take n (x : xs)) (upto (1+ (length xs)))
= { upto之定義 }

map (λn→ take n (x : xs)) (0 : map (1+) (upto (length xs)))
= { length之定義 }

map (λn→ take n (x : xs)) (0 : map (1+) (upto (length xs)))
= { map之定義，take 0 (x : xs) = [ ] }
[ ] : map (λn→ take n (x : xs)) (map (1+) (upto (length xs)))

= {定理 2.2：map f ·map g = map (f ·g) }
[ ] : map (λn→ take (1+ n) (x : xs)) (upto (length xs))

= { take之定義 }
[ ] : map (λn→ x : take n xs) (upto (length xs))

= {定理 2.2：map f ·map g = map (f ·g) }
[ ] : map (x:) (map (λn→ take n xs) (upto (length xs)))

= {歸納假設 }
[ ] : map (x:) (inits xs)

= { inits之定義 }
inits (x : xs) .

Answer (習題 2.22) —考慮如何從 segments [1,2,3 ]湊出 segments [0,1,2,3 ].
前者應當有的結果可能是：

[[ ], [1 ], [1,2 ], [1,2,3 ], [2 ], [2,3 ], [3 ]] .

而 segments [0,1,2,3 ]的結果可能是：

[[ ], [0 ], [0,1 ], [0,1,2 ], [0,1,2,3 ], [1 ], [1,2 ], [1,2,3 ], [2 ], [2,3 ], [3 ]] .

乍看之下，要多做的一步是：在空串列或所有 1開頭的串列前面補上 0. 之
所以只擴充 1開頭的串列，因為只有這些是原本便靠在左邊，和 0相鄰的
（也就是說它們是 [1,2,3 ]的前段）。
然而，當輸入串列型別為 List a（而不是特定的 List Int, List Char時），我
們無法用比較的方式找出這些前段。函數 segments 傳回的資訊不夠多。
如果要歸納定義，我們必須把「前段」們和其他的區段分開。

Answer (習題 2.25) —在 xs上做歸納。基底狀況很容易成立。考慮 xs :=
y : xs:

length (fan x (y : xs))
= length ((x : y : xs) : map (y:) (fan x xs))

= 1+ (length (map (y:) (fan x xs)))
= {因 length ·map f = length (練習 2.6) }

1+ (length (fan x xs))
= {歸納假設 }

1+ (1+ (length xs))
= 1+ (length (y : xs)) .

Answer (習題 2.26) —欲 證 明 map length (perms xs) =

map (const (length xs)) (perms xs),在 xs之上做歸納。基底狀況 xs :=[ ]中，
等號兩邊都化約為 [0 ]. 考慮歸納步驟 xs := x : xs:

map length (perms (x : xs))
= { perms之定義 }

map length · concat ·map (fan x) ·perms $ xs
= { map f · concat = concat ·map (map f ) (習題 2.13), map融合 }

concat ·map (map length · fan x) ·perms $ xs
= { map length · fan x = (λn→ repeatN n n) · length }

concat ·map ((λn→ repeatN n n) · (1+) · length) ·perms $ xs
= { map融合 }

concat ·map ((λn→ repeatN n n) · (1+)) ·map length ·perms $ xs
= {歸納假設 }

concat ·map ((λn→ repeatN n n) · (1+)) ·map (const (length xs)) ·perms $ xs
= { map融合 }

concat ·map ((λn→ repeatN n n) · (1+) · const (length xs)) ·perms $ xs
= {歸約 λ 算式 }

concat ·map (repeatN (1+ (length xs)) · (1+) · length) ·perms $ xs
= { length · fan x = (1+) · length (習題 2.25) }

concat ·map (repeatN (1+ (length xs)) · length · fan x) ·perms $ xs
= { [todo: ⁇?] map (const y) = repeatN y · length }

concat ·map (map (const (1+ (length xs))) · fan x) ·perms $ xs
= { map f · concat = concat ·map (map f ) (習題 2.13) }

map (const (length (x : xs))) · concat ·map (fan x) ·perms $ xs
= { perms之定義 }

map (const (length (x : xs))) (perms (x : xs)) .

Answer (習題 2.27) —

splits ::List a→ List (List a×List a)
splits [ ] = [([ ], [ ])]

splits (x : xs) = ([ ],x : xs) : map ((x:)× id) (splits xs) .

其中 (f ×g) (x,y) = (f x,g y).

Answer (習題 2.28) —將式子改寫為 length (sublists xs) = exp 2 (length xs),
在 xs上做歸納。歸納步驟 xs := x : xs為：

length (sublists (x : xs))
= length (sublists xs++map (x:) (sublists xs))
= {因 (2.1) length (xs++ ys) = length xs+ length ys }

length (sublists xs)+ length (map (x:) (sublists xs))
= {因 length ·map = length (練習 2.6) }

length (sublists xs)+ length (sublists xs)
= {因 x+ x = 2× x }

2× length (sublists xs)
= {歸納步驟 }

2× exp 2 (length xs)
= exp 2 (1+ (length xs))
= exp 2 (length (x : xs)) .

Answer (習題 2.29) —

((y:) · (ys++)) xs
= { definition of (·) }

y : (ys++ xs)
= { definition of (++) }
(y : ys)++ xs .

Answer (習題 2.30) —

((++zs) · (y:)) xs
= { definition of (·) }
(++zs) (y : xs)

= (y : xs)++ zs
= { definition of (++) }

y : (xs++ zs)
= (y:) ((++zs) xs)
= { definition of (·) }
((y:) · (++zs)) xs .

Answer (習題 2.31) —

ITree上之歸納法： (∀t :: ITree a ·P t) ⇐
(P Null ∧ (∀x t u ·P (Node x t u)⇐ P t ∧ P u) .

Answer (習題 2.32) —

tags :: ITree a→ List a
tags Null = [ ]

tags (Node x t u) = tags t++[x ]++ tags u .

Answer (習題 2.33) —

size :: ITree a→ N
size Null = 0
size (Node x t u) = 1+ (size t+ size u) .

Answer (習題 2.34) —在 t之上做歸納。當 t :=Null，顯然等號兩邊都歸約
為 0. 當 t :=Node x t u:

length (tags (Node x t u))
= { length與 leaves之定義 }

1+ (length (tags t)+ length (tags u))
= {歸納假設 }

1+ (size t+ size u)
= { size之定義 }

size (Node x t u) .

Answer (習題 2.37) — head [ ] : tail [ ] ≠ = [ ].

Answer (習題 2.38) —針對 m做歸納。當 m :=0,等號兩邊都歸約成 [ ]。考
慮 m :=1+ m. 此時針對 xs做分析。當 xs :=[ ],等號兩邊亦都歸約成 [ ]。考
慮 xs := x : xs的情況：

take (1+ m) (take (1+ m+n) (x : xs))
= { (+)與 take之定義 }

take (1+ m) (x : take (m+n) xs)
= { take之定義 }

x : take m (take (m+n) xs)
= {歸納假設 }

x : take m xs
= { take之定義 }

take (1+ m) xs .

Answer (習題 2.39) —針對 m 做歸納。當 m := 0, 等號兩邊都歸約成
take n xs。考慮 m := 1+ m. 此時針對 xs 做分析。當 xs := [ ], 等號兩邊亦
都歸約成 [ ]。考慮 xs := x : xs的情況：

drop (1+ m) (take (1+ m+n) (x : xs))
= { (+)與 take之定義 }

drop (1+ m) (x : take (m+n) xs)
= { take之定義 }

drop n (take (m+n) xs)
= {歸納假設 }

take n (drop m xs)
= { take之定義 }

take n (drop (1+ m) (x : xs)) .

Answer (習題 2.40) —針對 m 做歸納。當 m := 0, 等號兩邊都歸約成
drop n xs。考慮 m := 1+ m. 此時針對 xs 做分析。當 xs := [ ], 等號兩邊亦
都歸約成 [ ]。考慮 xs := x : xs的情況：

drop (1+ m+n) (x : xs)
= { (+)與 take之定義 }

drop (m+n) xs
= {歸納假設 }

drop n (drop m xs)
= { drop之定義 }

drop n (drop (1+ m) (x : xs)) .

Answer (習題 2.42) —使用完全歸納。
狀況 n :=0: 等號兩邊都歸約為 [ ].
狀況 n>0。假設對於任何 0 ⩽ i<n, sum (binary i) = i.

sum (binary n) =
= {令 m = last (takeWhile (⩽ n) twos), sum之定義 }

m+ sum (binary (n−m))

= {因 n−m<n,歸納假設 }
m+(n−m)

= n .

Answer (習題 2.43) —本證明的關鍵性質是 α2 = (3+
√

5)/2 = α +1. 使用
完全歸納證明 fib (2+n)>αn，以 n :=1和 n :=2當基底狀況。
狀況 n :=1: fib 3 = 2>α .
狀況 n :=2: fib 4 = 3> (3+

√
5)/2 = α2.

狀況 n :=2+n且 2+n>4。假設對於任何 3 ⩽ i<2+n, fib i>α i−2. 論證：

fib (2+n)
= fib (1+n)+fib n
> {歸納假設 }

αn−1 +αn−2

= {提出 αn−2 }
(α +1)×αn−2

= { α2 = α +1 }
α2×αn−2

= αn .

Answer (習題 2.44) —在 xs上做歸納。

all ((x : xs) = =) (zipWith (++) (inits (x : xs)) (tails (x : xs)))
≡ all ((x : xs) = =) (zipWith (++) ([ ] : map (x:) (inits xs)) ((x : xs) : tails xs))
≡ all ((x : xs) = =) (([ ]++(x : xs)) : zipWith (++) (map (x:) (inits xs)) (tails xs))
≡ ((x : xs) = = (x : xs)) ∧

all ((x : xs) = =) (zipWith (++) (map (x:) (inits xs)) (tails xs))

Answer (習題 3.2) —在 t之上做歸納。基底情況 t :=E很容易建立。以下
只示範一種歸納情況。
狀況 t :=B t x u, k< x:

balanced (ins k (B t x u))
= { ins之定義；k< x }

balanced (rotate (ins k t) x u)
⇐ {引理 3.6 }

balanced (ins k t) ∧ balanced u ∧ bheight (ins k t) = bheight u
= {定理 3.2 }

balanced (ins k t) ∧ balanced u ∧ bheight t = bheight u
⇐ {歸納假設 }

balanced t ∧ balanced u ∧ bheight t = bheight u
= { balanced之定義 }

balanced (B t x u) .

Answer (習題 3.3) —以下只示範其中一種狀況。
狀況: (t,x,u) :=(R (R t x u) y v,z,w).

balanced (rotate (R (R t x u) y v) z w)
= { rotate之定義 }

balanced (R (B t x u) y (B v z w))
= { balanced之定義 }

bheight (B t x u) = bheight (B v z w) ∧
balanced (B t x u) ∧ balanced (B v z w)

= { bheight之定義 }
1+(bheight t ↑bheight u) = 1+(bheight v↑beight w) ∧
bheight t = bheight u ∧ balanced t ∧ balanced u ∧
bheight v = bheight w ∧ balanced v ∧ balanced w
⇐ { balanced與 bheight之定義；算數性質 }

bheight (R t x u) = bheight v ∧
balanced (R t x u) ∧ balanced v ∧ balanced w ∧
bheight t ↑bheight u↑bheight v = bheight w

= { balanced與 bheight之定義 }
balanced (R (R t x u) y v) ∧ balanced w ∧

bheight R (R t x u) y v = bheight w .

Answer (習題 3.4) —本證明只是一一檢查每個狀況。以 1. (infrared t ∨
semiRB t) ∧ semiRB u⇒ semiRB (rotate t x u)為例，由於有 semiRB u, 我們
只需檢查 rotate的第一、第二、與最後一個狀況。以第一個狀況為例：
狀況: (t,x,u) :=(R (R t x u) y v,z,w):

semiRB (rotate (R (R t x u) y v) z w)
= { rotate之定義 }

semiRB (R (B t x u) y (B v z w))
= { semiRB與 color之定義 }

color t = color u = color v = color w = Blk ∧
semiRB t ∧ semiRB u ∧ semiRB v ∧ semiRB w

= { semiRB之定義 }
color v = color w = Blk ∧
semiRB (R t x u) ∧ semiRB v ∧ semiRB w

= { infrared之定義 }
infrared (R (R t x u) y v) ∧ semiRB w

= {命題邏輯，semiRB (R (R t x u) y v) = False }
(infrared (R (R t x u) y v) ∨ semiRB (R (R t x u) y v)) ∧ semiRB w .

Answer (習題 5.1) —顯然 sum (descend 0) = 0. 考慮歸納情況：

sum (descend (1+ n))
= { descend之定義 }

sum (1+ n : descend n)
= { sum之定義 }
(1+ n)+ sum (descend n)

= { sumseries之定義 }
(1+ n)+ sumseries n .

因此

sumseries 0 = 0
sumseries (1+ n) = (1+ n)+ sumseries n .

Answer (習題 5.2) —顯然 repeatN (0,x) = [ ]. 至於歸納情況，演算如下：

repeatN (1+ n,x)
= { repeatN 之定義 }

map (const x) (descend (1+ n))
= { descend之定義 }

map (const x) (1+ n : descend n)
= { map與 const之定義 }

x : map (const x) (descend n)
= { repeatN 之定義 }

x : repeatN (n,x) .

因此，

repeatN (0, x) = [ ]

repeatN (1+ n,x) = x : repeatN (n,x) .

Answer (習題 5.3) —基底狀況：

rld [ ]

= { rld之定義 }
concat (map repeatN [ ])

= { map與 concat之定義 }
[ ] .

歸納狀況：

rld ((n,x) : xs)
= { rld之定義 }

concat (map repeatN ((n,x) : xs))
= { map之定義 }

concat (repeatN (n,x) : map repeatN xs)
= { concat之定義 }

repeatN (n,x)++ concat (map repeatN xs)
= { rld之定義 }

repeatN (n,x)++ rld xs .

因此我們已推導出：

rld [ ] = [ ]

rld ((n,x) : xs) = repeatN (n,x)++ rld xs .

Answer (習題 5.4) —顯然 select [ ] = [ ]。考慮 xs := x : xs的情況：

select (x : xs)
= { select與 delete之定義 }

zip (x : xs) (xs : map (x:) (delete xs))
= { zip之定義 }
(x,xs) : zip xs (map (x:) (delete xs))

= { zip xs (map f ys) = map (id× f ) (zip xs ys) }
(x,xs) : map (id× (x:)) (zip xs (delete xs))

= { select之定義 }
(x,xs) : map (id× (x:)) (select xs) .

因此，

select [ ] = [ ]

select (x : xs) = (x,xs) : map (id× (x:)) (select xs) .

Answer (習題 5.5) —以下只列出歸納狀況：

delete (x : xs)
= { delete之定義 }

map (del (x : xs)) [0 . . length (x : xs)−1 ]
= { length之定義，簡單運算 }

map (del (x : xs)) [0 . . length xs ]
= {由於 length xs ⩾ 0,使用 (5.3) }

map (del (x : xs)) (0 : map (1+) [0 . . length xs−1 ])
= { map之定義 }

del (x : xs) 0 : map (del (x : xs)) (map (1+) [0 . . length xs−1 ])
= { map融合 }

del (x : xs) 0 : map (del (x : xs) · (1+)) [0 . . length xs−1 ] .

在此暫停一下，觀察 del (x : xs). 顯然, del (x : xs) 0 = xs. 至於 del (x : xs) ·
(1+)，我們演算看看:

(del (x : xs) · (1+)) i
= { (·)與 del之定義 }

take (1+ i) (x : xs)++drop (1+(1+ i)) (x : xs)
= { take與 drop之定義 }

x : take i xs++drop (1+ i) xs
= { del之定義 }

x : del xs i
= { (·)之定義 }
((x:) ·del xs) i .

繼續之前的演算：

del (x : xs) 0 : map (del (x : xs) · (1+)) [0 . . length xs−1 ]
= {前述之演算 }

xs : map ((x:) ·del xs) [0 . . length xs−1 ]
= { map融合 }

xs : map (x:) (map (del xs) [0 . . length xs−1 ])
= { delete之定義 }

xs : map (x:) (delete xs) .

由此，我們導出了正文中 delete的歸納定義.

Answer (習題 5.7) —令 t為一個向右傾斜的樹：

t = Bin (Tip 1) (Bin (Tip 2) . .(Bin (Tip (n−1)) (Tip n))) ,

tip t 展開成為 [1 ]++([2 ] . .([n− 1 ]++ [n ])),可在 O(n)時間內歸約成範式。
這是最好的情況。令 u為一個向左傾斜的樹：

u = Bin (Bin (...(Bin (Bin (Tip 1) (Tip 2)) (Tip 3))) ...
(Tip (n−1))) (Tip n) ,

tip u展開成為 ((. .(([1 ]++[2 ])++[3 ]) . .) . .++[n−1 ])++[n ],需要 O(n2)的時
間，也是 tips最壞情況的時間複雜度.

Answer (習題 5.8) —欲 證 明 zipWith (⊗) as (map (⊗x) bs) =

map (⊗x) (zipWith (⊗) as bs), 在 as 上做歸納。當 as := [ ], 等號兩邊
都歸約成 [ ]. 考慮 as := a : as 的情況。若 bs := [ ], 等號兩邊仍均為 [ ]. 當
bs :=b : bs:

zipWith (⊗) (a : as) (map (⊗x) (b : bs))
= { zipWith與 map之定義 }
(a⊗ (b⊗ x)) : zipWith (⊗) as (map (⊗x) bs)

= {歸納假設 }
(a⊗ (b⊗ x)) : map (⊗x) (zipWith (⊗) as bs)

= { (⊗)滿足結合律 }
((a⊗b)⊗ x) : map (⊗x) (zipWith (⊗) as bs)

= { zipWith與 map之定義 }
map (⊗x) (zipWith (⊗) (a : as) (b : bs)) .

Answer (習題 5.11) —

exp b 0 = 1
exp b n | even n = square (exp b (n ‘div‘ 2))

| odd n = b× square (exp b (n ‘div‘ 2)) .

由於遞迴呼叫中的 n總是變小，本定義可視為 n之上的良基歸納。

Answer (習題 5.12) —試著展開-收回 index (x : xs)：

index (x : xs)
= zip [0 . . ] (x : xs)

= { [0 . . ] = 0 : [1 . . ], zip之定義 }
(0,x) : zip [1 . . ] xs .

我們又發現 zip [1..] 無法收回成 index xs. 因此，我們定義：

indexFrom ::N→ List a→ List (N×a)
indexFrom i = zip [i . . ] .

試著推導 indexFrom的歸納定義如下：

indexFrom i (x : xs)
= zip [i . . ] (x : xs)
= { [i . . ] = i : [1+ i . . ], zip之定義 }
(i,x) : zip [1+ i . . ] xs

= (i,x) : indexFrom (1+ i) xs .

因此我們已得到：

indexFrom i [ ] = [ ]

indexFrom i (x : xs) = i,x) : indexFrom (1+ i) xs .

Answer (習題 5.13) —考慮如下的定義：

baosum ::Tree Int→ (Bool, Int)

baosum = ⟨baobab,sumT⟩ .

如果 baosum 有時間效率為 O(n) 的定義，我們可重定義 baobab = fst ·
baosum. 當 t :=Null,我們有 baosum Null= (True,0). 考慮 t :=Node x t u:

baosum (Node x t u)
= { baosum之定義 }
(baobab (Node x t u),sumT (Node x t u))

= { baobab與 sumT 之定義 }
(baobab t ∧ baobab u ∧ x> (sumT t+ sumT u),
x+ sumT t+ sumT u)

= {引入區域變數 }
let (b,y) = (baobab t,sumT t)

(c,z) = (baobab u,sumT u)
in (b ∧ c ∧ x> (y+ z),x+ y+ z)

= { baosum之定義 }
let (b,y) = baosum t

(c,z) = baosum u
in (b ∧ c ∧ x> (y+ z),x+ y+ z) .

如此，我們已經導出：

baosum Null = (True,0)
baosum (Node x t u) =

let (b,y) = baosum t
(c,z) = baosum u

in (b ∧ c ∧ x> (y+ z),x+ y+ z) .

Answer (習題 5.14) —定義：

dd ::ETree a→ (List a,N)
dd = ⟨deepest,depth⟩ .

顯然 dd (Tip x) = ([x ],0). 以下我們考慮歸納情況時，用一個非標準的語法
同時處理守衛算式的三個式子：

dd (Bin t u)
= { dd, deepest,與 depth之定義 }
((m< n→ deepest u
| m = = n → deepest t++deepest u
| m> n→ deepest t),1+(m↑n))

where (m,n) = (depth t,depth u)
= {取出 deepest t與 deepest u }
((m< n→ ys
| m = = n → xs++ ys
| m> n→ xs),1+(m↑n))

where ((xs,m),(ys,n)) = ((deepest t,depth t),(deepest u,depth u))
= { dd之定義，函數分配進條件判斷 }
(m< n→ (ys,1+n)
| m = = n → (xs++ ys,1+n)
| m>n→ (xs,1+m))

where ((xs,m),(ys,n)) = (dd t,dd u) .

因此我們得到：

dd (Tip x) = ([x ],0)
dd (Bin t u) | m<n = (ys,1+n)

| m = = n = (xs++ ys,1+n)
| m>n = (xs,1+m) ,

where ((xs,m),(ys,n)) = (dd t,dd u) .

Answer (習題 5.15) —定義：

balHeight ::RBTree→ (Bool,N)
balHeight = ⟨balanced,bheight⟩ .

我們有 balanced = fst · balHeight. 顯然 balHeight E = (True,0). 考慮 B t x u
的狀況：

balHeight (B t x u)
= { balHeight之定義 }
(balanced (B t x u),bheight (B t x u))

= { balanced與 bheight之定義 }
(bheight t = = bheight u ∧ balanced t ∧ balanced u,

1+(bheight t ↑bheight u))
= {將 balanced與 bheight的呼叫取出 }

let (bt, ht) = (balanced t,bheight t)
(bu,hu) = (balanced u,bheight u)

in (ht = = hu ∧ bt ∧ bu,1+(ht ↑hu))
= { balHeight之定義 }

let (bt, ht) = balHeight t
(bu,hu) = balHeight u

in (ht = = hu ∧ bt ∧ bu,1+(ht ↑hu)) .

當輸入為 R t x u的情況亦雷同。我們可推導出：

balHeight ::RBTree→ (Bool,N)
balHeight E = (True,0)
balHeight (R t x u) = let (bt, ht) = balHeight t

(bu,hu) = balHeight u
in (ht = = hu ∧ bt ∧ bu,(ht ↑hu))

balHeight (B t x u) = let (bt, ht) = balHeight t
(bu,hu) = balHeight u

in (ht = = hu ∧ bt ∧ bu,1+(ht ↑hu)) .

Answer (習題 5.16) —由 (5.4)我們推測：

mdm ::List Int→ (Int× Int)

mdm xs = (maxdiff xs,minimum xs) .

我們試圖找出 mdm 的歸納定義。顯然 mdm [ ] = (−∞,∞). 考慮歸納情況
（令 minus (x,y) = x− y）：

mdm (x : xs)
= { mdm, maxdiff ,與 minimum之定義 }
(maximum (map minus (allpairs (x : xs))),x↓minimum xs)

= (maximum (map minus (map (λy→ (x,y)) xs++allpairs xs)),
x↓minimum xs) .

集中焦點在序對的第一個元素：

maximum (map minus (map (λy→ (x,y)) xs++allpairs xs))
= {因 map f (xs++ ys) = map f xs++map f ys (習題 2.11) }

maximum (map minus (map (λy→ (x,y)) xs)++
map minus (allpairs xs))

= { map-fusion }
maximum (map (x−) xs++map minus (allpairs xs))

= {因 maximum (xs++ ys) = maximum xs↑maximum ys }
maximum (map (x−) xs)↑maximum (map minus (allpairs xs))

= {因 (5.4) }
(x−minimum xs)↑maximum (map minus (allpairs xs))

= { maxdiff 之定義 }
(x−minimum xs)↑maxdiff xs .

回到推導主體：

mdm (x : xs)
= {上述計算 }
((x−minimum xs)↑maxdiff xs,x↓minimum xs)

= {使用 let }
let (y,z) = (maxdiff xs,minimum xs)
in ((x− z)↑ y,x↓ z)

= { mdm之定義 }
let (y,z) = mdm xs
in ((x− z)↑ y,x↓ z) .

於是我們得知：

mdm [ ] = (−∞,∞)

mdm (x : xs) = let (y,z) = mdm xs
in ((x− z)↑ y,x↓ z) .

Answer (習題 5.17) —僅看 t :=Node x t u的狀況：

tagsAcc (Node x t u) ys
= tags (Node x t u)++ ys
= (tags t++[x ]++ tags u)++ ys
= { (++)的結合律 }

tags t++(x : tags u++ ys)
= tagsAcc t (x : tagsAcc u ys) .

我們可導出：

tagsAcc Null ys = ys
tagsAcc (Node x t u) ys = tagsAcc t (x : tagsAcc u ys) .

Answer (習題 5.19) —定義 tipsAcc t ys = tips t++ ys. 如果 tipsAcc有個有效
率的定義，我們可改定義 tips t = tipsAcc t [ ]. 當 t :=Tip x, tipsAcc (Tip x) ys
可直接展開為 x : ys. 當 t :=Bin t u,

tipsAcc (Bin t u) ys
= (tips t++ tips u)++ ys
= { (++)之結合律 }

tips t++(tips u++ ys)
= tipsAcc t (tipsAcc u ys) .

因此，

tipsAcc (Tip x) ys = x : ys
tipsAcc (Bin t u) ys = tipsAcc t (tipsAcc u ys) .

Answer (習題 5.20) —假設我們要計算 BN:

b,n,x :=B,N,1;
do n ≠ = 0→ if even n→ b,n :=b×b,n ‘div‘ 2

| odd n→ n,x :=n−1,x×b
fi

od;
return x

其迴圈恆式為 BN = x×bn.

Answer (習題 5.22) —我們預期將利用 (×)的結合律。定義 factAcc n y =

y× fact n。顯然 factAcc 0 y = y. 至於 n :=1+ n的狀況：

factAcc (1+ n) y
= y× fact (1+ n)
= y× ((1+ n)× fact n)
= { (×)之結合律 }
(y× (1+ n))× fact n

= factAcc n (y× (1+ n)) .

因此我們有了下述定義。由於 fact n = factAcc n 1. 這相當於右邊的迴圈：

factAcc 0 y = y
factAcc (1+ n) y =

factAcc n (y× (1+ n)) .

n,y :=N,0;
do n ̸= = 0→ n,y :=n−1,y×n od;
return y .

在推導 factAcc 的歸納定義時，我們不僅用到 (×) 的結合律，也用到了
y×1 = y.

Answer (習題 5.23) —針對 m做分析。基底情況中，mulAcc 0 n k = k+0×
n = k. 當 m為非零的偶數，可被改寫為 2×m並演算如下：

mulAcc (2×m) n k
= k+(2×m)×n
= { (×)之交換律與結合律 }

k+m× (2×n)
= mulAcc m (2×n) k .

當 m為奇數，可被改寫為 1+m,並演算如下：

mulAcc (1+m) n k
= k+(1+m)×n
= { (×)分配進入 (+) }

k+n+m×n
= { (+)之結合律，mulAcc之定義 }

mulAcc m n (k+n) .

因此我們得到：

mulAcc 0 n k = k
mulAcc m n k | even n = mulAcc (m ‘div‘ 2) (2×n) k

| odd n = mulAcc (m−1) n (k+n) .

如果改寫成計算M×N的迴圈，其恆式為M×N= k+m×n.

Answer (習題 5.24) —要使用組對避免重複計算，我們可以定義一個函數
同時傳回 dtoN ds以及 10的 length ds次方。

dtoNT ::List N→ (N×N)
dtoNT ds = (dtoN ds,10length ds) .

我們可輕易導出：

dtoNT [ ] = (0,1)
dtoNT (d : ds) = (d× t+n,10× t) ,

where (n, t) = dtoNT ds .

若使用累積參數，可定義

dtoNAcc ds n = n×10length ds +dtoN ds .

若有歸納定義，我們可令 dtoN ds = dtoNAcc ds 0. 基底狀況為
dtoNAcc [ ] n = n. 至於 ds :=d : ds的情況可演算如下：

dtoNAcc (d : ds) n
= n×10length (d:ds)+dtoN (d : ds)
= n×10length (d:ds)+d×10length ds +dtoN ds

= { length與乘冪之定義 }
n×10×10length ds +d×10length ds +dtoN ds

= {算數運算 }
(n×10+d)×10length ds +dtoN ds

= { dtoNAcc之定義 }
dtoNAcc ds (n×10+d) .

可推導出：

dtoNAcc [ ] n = n
dtoNAcc (d : ds) n = dtoNAcc ds (n×10+d) .

Answer (習題 5.25) —由於我們展開的式子都有 a× fusc n+b× fusc (1+n)
的形式，定義

fuscAcc n a b = a× fusc n+b× fusc (1+n) .

如果 fuscAcc有歸納定義，我們可令 fusc n = fuscAcc n 1 0.
為推導 fuscAcc,針對 n做分析。當 n :=0, fuscAcc 0 a b = b. 當 n為非零偶
數，可被改寫為 2×n:

fuscAcc (2×n) a b
= a× fusc (2×n)+b× fusc (1+2×n)
= { fusc之定義 }

a× fusc n+b× (fusc n+ fusc (1+n))
= {四則運算 }
(a+b)× fusc n+b× fusc (1+n)

= fuscAcc n (a+b) b .

當 n為奇數，也就是可改寫為 1+2×n的形式時：

fuscAcc (1+2×n) a b
= a× fusc (1+2×n)+b× fusc (2+2×n)
= a× fusc (1+2×n)+b× fusc (2× (1+n))
= { fusc之定義 }

a× (fusc n+ fusc (1+n))+b× fusc (1+n)
= {四則運算 }

a× fusc n+(a+b)× fusc (1+n)
= fuscAcc n a (a+b) .

因此我們有了如下的程式：

fuscAcc 0 a b = b
fuscAcc n a b | even n = fuscAcc (n ‘div‘ 2) (a+b) b

| odd n = fuscAcc (n ‘div‘ 2) a (a+b) .

Answer (習題 5.26) —使用如下的規格：

mapAcc :: (a→ b)→ List a→ List b→ List b
mapAcc f xs ys = reverse ys++map f xs .

因此 mapAcc f [ ] ys = reverse ys. 而 xs := x : xs的情況推導如下：

mapAcc f xs ys
= reverse ys++map f (x : xs)
= reverse ys++(f x : map f xs)
= (reverse ys++[ f x ])++map f xs
= reverse (f x : ys)++map f xs
= mapAcc f xs (f x : ys) .

Answer (習題 5.27) —假設習題 5.14 中推導出的函數為 dd :: ETree a →
(List a,N). 我們定義：

ddD ::ETree a→ (DList a,N)
ddD t = let (xs,n) = dd t in ((λ zs→ xs++ zs),n) .

有了 ddD後，我們可改定義 dd t = let (f ,n) = ddD t in (f [ ],n).
試著推導 ddD 的歸納定義，不難得到 ddD (Tip x) = ((x:),0). 考慮
ddD (Bin t u),為計算方便將 dd改寫為 if-then-else之形式：

ddD (Bin t u)
= { ddD之定義 }

let (xs,n) = dd (Bin t u) in ((xs++),n)
= { dd之定義 }

let (xs,m) = dd t
(ys,n) = dd u

in (((if m<n then xs else if m = = n then xs++ ys else ys)++),

1+(m↑n))
= {函數應用分配入 if }

let (xs,m) = dd t
(ys,n) = dd u

in (if m<n then (xs++) else if m = = n then (xs++ ys++) else (ys++),

1+(m↑n))
= {抽出 (xs++)與 (ys++) }

let (f ,m) = (let (xs,m) = dd t in ((xs++),m))

(g,n) = (let (ys,n) = dd u in ((ys++),n))
in (if m<n then f else if m = = n then f ·g else g,

1+(m↑n))
= { ddD之定義 }

let (f ,m) = ddD t

(g,n) = ddD u
in (if m<n then f else if m = = n then f ·g else g,

1+(m↑n)) .

因此我們得到：

ddD (Tip x) = ((x:),0)
ddD (Bin t u) | m<n = (f ,1+n)

| m = = n = (f ·g,1+n)
| m>n = (g,1+m) ,

where ((f ,m),(g,n)) = (ddD t,ddD u) .

Answer (習題 6.1) —
1.perms = foldr (λx xss→ concat (map (fan x) xss)) [[ ]]
2.sublists = foldr (λx xss→ xss++map (x:) xss) [[ ]]
3.splits可定義如下：

splits = foldr spl [([ ], [ ])] ,

where spl x ((xs,ys) : zss) =
([ ],x : xs++ ys) : map ((x:)× id) ((xs,ys) : zss) .

Answer (習題 6.2) —由於 xs 出現在遞迴呼叫以外的地方 — ((y : x : xs):),
此處的 fan 定義並不是一個 foldr. 但由於 tail (head (fan y xs)) =

xs （例如，fan 5 [1,2,3 ] = [[5,1,2,3 ], [1,5,2,3 ], [1,2,5,3 ], [1,2,3,5 ]], 因此
tail (head (fan 5 [1,2,3 ])) = [1,2,3 ]），我們可將 fan寫成：

fan y = foldr (λx xss→ (y : x : tail (head xss)) : map (x:) xss) [[y ]] .

Answer (習題 6.4) —

sum ·map length
= { sum = foldr (+) 0, foldr-map融合 }

foldr ((+) · length) 0
= {摺融合 }

length · foldr (++) [ ]

= length · concat .

其融合條件證明如下：

length (xs++ ys)
= { length與 (++)的同態性 }

length xs+ length ys
= { (·)之定義 }
((+) · length) xs (length ys) .

Answer (習題 6.6) —回顧：filter p= foldr (λx xs→ if p x then x :xs else xs) [ ],
concat = foldr (++) [ ]. 我們演算如下：

concat ·map g ·filter p
= {摺融合，如下述 }

foldr (λx ys→ filter p (f x)++ ys) [ ]
= { foldr-map融合，如下述 }

concat ·map (filter p · f ) .

摺融合的條件為：

concat (map g (if p x then x : xs else xs))
= { concat ·map g分配進 if之中 }

if p x then g x++ concat (map g xs) else concat (map g xs)
= {提出 concat (map g xs) }
(if p x then g x else [ ])++ concat (map g xs)

= {假設 }
filter p (f x)++ concat (map g xs) .

至於 foldr-map融合,只需驗證 ((++) ·filter p · f ) x ys確實等於 filter p (f x)++
ys.

Answer (習題 6.7) —
我們需要滿足融合條件 map (⩾ 0) (0 : map (x+) ys) = step x (map ⩾ 0 ys)的
step. 演算如下：

map (⩾ 0) (0 : map (x+) ys)
= True : map ((⩾ 0) · (x+)) ys
= step x (map (⩾ 0) ys) .

然而我們無法由 map (⩾ 0) ys算出 map ((⩾ 0) · (x+)) ys.

Answer (習題 6.8) —

sum · (++ys)
= sum · foldr (:) ys
= {摺融合 }

foldr (+) (sum ys)
= {摺融合 }
(+(sum ys)) · foldr (+) 0

= (+(sum ys)) · sum .

其中第一個摺融合的條件為 sum (x : xs) = x+ sum xs —我們由此發現融合
後的步驟函數為 (+). 第二個摺融合的條件證明如下：

(+(sum ys)) (x+ y)
= (x+ y)+ sum ys

= x+(y+ sum ys)
= x+((+(sum ys)) y) .

Answer (習題 6.9) —相當於證明 length · fan y = 1+ · length. 推論如下：

length · fan y
= length · foldr (λx xss→ (y : x : tail (head xss)) : map (x:) xss) [[y ]] .

= {摺融合定理 }
foldr 1+ 1

= {摺融合定理 }
1+ · foldr 1+ 0

= 1+ · length .

其中第一次融合的融合條件可證明如下：

length ((y : x : tail (head xss)) : map (x:) xss)
= { length之定義 }

1+ (length (map (x:) xss))
= { length ·map f = length }

1+ (length xss) .

由此發現步驟函數為 1+. 第二次融合的融合條件則只需展開定義即可證
成。

Answer (習題 6.10) —基底值為 base = exp b 0 = 1. 為找出步驟函數，我們
推論：

exp b (if c then 1+2×n else 2×n)
= {函數分配進 if }

if c then exp b (1+2×n) else exp b (2×n)
= {因 mx+y = mx×my }

if c then b× exp b (2×n) else exp b (2×n)
= {因 m2n = (mn)2,回顧：square x = x× x }

if c then b× square (exp b n) else square (exp b n) .

因此可得

exp b ·decimal = foldr (λd x→ if c then b× square x
else square x) 1 .

Answer (習題 6.12) —回顧 id = foldN (1+) 0. 欲將 fib2 融入 id, 基底值為
fib2 0 = (fib 1,fib 0) = (1,0). 為得到步驟函數，我們演算如下：

fib2 (1+ n)
= (fib (1+ (1+ n)),fib (1+ n))

= { fib之定義 }
(fib (1+ n)+fib n,fib (1+ n))

= (λ (x,y)→ (x+ y,x)) (fib2 n) .

因此我們得到

fib2 = foldN (λ (x,y)→ (x+ y,x)) (1,0) .

Answer (習題 6.13) —即 證 明 length · tags = size. 回 顧 tags =

foldIT (λx xs ys → xs ++ [x ] ++ ys) [ ] 使用摺融合，由於 length [ ] = 0
以 及 length (xs ++ [x ] ++ ys) = 1 + length xs + length ys, 我 們 得 到
length · tags = foldIT (λx m n→ 1+m+n) 0 = size.

Answer (習題 6.14) — ETree上的 foldET-mapE融合定理為：

foldET f k ·mapE g = foldET f (k ·g) .

由於 mapE g = foldET Bin f , 欲證明上式可用摺融合定理。其融合條件
foldET f k (Bin t u) = f (foldET f k t) (foldET f k u)恰巧是 foldET 之定義。

Answer (習題 6.15) —函數 mapI 可定義如下：

mapI f = foldIT (λx t u→ Node (f x) t u) Null .

考慮 foldIT f e ·mapI g之融合。其基底值為 foldIT f e Null= e. 步驟函數的
推導如下：

foldIT f e (Node (g x) t u)
= { foldIT 之定義 }

f (g x) (foldIT f e t) (foldIT f e u) .

因此 foldIT f e ·mapI g = foldIT (λx y z→ f (g x) y z) e.

Answer (習題 6.16) —即證明 minE ·mapE (x+) = (x+) ·minE. 推論如下：

minE ·mapE (x+)

= { foldET-mapE融合，見習題 6.14 }
foldET (↓) (id · (x+))

= { foldET 融合，如下述 }
(x+) ·minE .

融合的基底函數為 id · (x+) = (x+) · id,融合條件則為 x+(y ↓ z) = (x+ y) ↓
(x+ z)。

Answer (習題 6.17) —回想 tags = foldIT (λx xs ys→ xs++[x ]++ ys) [ ]. 融合
後之基底值為 (++) [ ] = id. 融合後之步驟函數 step須滿足 (++) (xs++[x ]++
ys) = step x (xs++) (ys++). 但由於左手邊的 (++)還需一個參數才能化簡，
我們在左右兩邊各補上一個參數 zs. 演算如下：

(++) (xs++[x ]++ ys) zs
= (xs++[x ]++ ys)++ zs
= { (++)之遞移律 }

xs++(x : (ys++ zs))
= { (·)之定義 }
((xs++) · (x:) · (ys++)) zs .

因此我們得到 (++) · tags = foldIT (λx f g→ f · (x:) ·g) id.

Answer (習題 6.18) —基底函數為 filter (all ascending) ·wrap3 = wrap3. 為
求出步驟函數，我們推論：

filter (all ascending) (concat (map (extend x) ysss))
= { filter p · concat = concat ·map (filter p), map融合 }

concat (map (filter (all ascending) · extend x) ysss)
= {推導 extendAsc,如下述 }

concat (map (extendAsc x) (filter (all ascending) ysss)) ,

欲使得最後一步成立，我們使用習題 6.6中的性質，試圖找到滿足下述條
件的 extendAsc：

filter (all ascending) (extend x yss) =
if all ascending yss then extendAsc x yss else [ ] .

我們演算如下：

filter (all ascending) (extend x (ys : yss))
= filter (all ascending) [[x ] : ys : yss,(x : ys) : yss ]
= { filter之定義;因 ascending [x ] = True }

if all ascending (ys : yss) then
([x ] : ys : yss) : filter (all ascending) [(x : ys) : yss ]
else filter (all ascending) [(x : ys) : yss ]

= { filter與 ascending之定義；重安排 if的幾個分支 }
if all ascending (ys : yss) then

if x ⩾ head ys then [[x ] : ys : yss,(x : ys) : yss ]
else [[x ] : ys : yss ]

else [ ]
= {抽取出 extendAsc如下 }

if all ascending (ys : yss) then extend′ x (ys : yss) else [ ]

其中 extendAsc的定義如下：

extendAsc x (ys : yss) = if x ⩾ head ys then [[x ] : ys : yss,(x : ys) : yss ]
else [[x ] : ys : yss ] .



DR
AF
T

222 APPENDIX A. 習題解答

Answer (習題 8.2) —考慮 eval (Add e0 e1):

eval (Add e0 e1) env
= { eval之定義 }
(eval e0 >>=λv0→ eval e1 >>=λv1→ return (v0 + v1)) env

= { (>>=)之定義 }
(λv0→ eval e1 >>=λv1→ return (v0 + v1)) (eval e0 env) env

= {函數應用 }
(eval e1 >>=λv1→ return (eval e0 env+ v1)) env

= { (>>=)之定義,函數應用 }
return (eval e0 env+ eval e1 env) env

= { return之定義 }
eval e0 env+ eval e1 env .

考慮 eval (Let x e0 e1) env:

eval (Let x e0 e1) env
= { eval之定義 }
(eval e0 >>=λv→ local ((x,v):) (eval e1)) env

= { (>>=)之定義,函數應用 }
local ((x,eval e0 env):) (eval e1) env

= { local之定義 }
eval e1 ((x,eval e0 env) : env) .

